

# NAG Library Routine Document

## C09DAF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

C09DAF computes the one-dimensional maximal overlap discrete wavelet transform (MODWT) at a single level. The initialization routine C09AAF must be called first to set up the MODWT options.

### 2 Specification

```
SUBROUTINE C09DAF (N, X, LENC, CA, CD, ICOMM, IFAIL)
  INTEGER          N, LENC, ICOMM(100), IFAIL
  REAL (KIND=nag_wp) X(N), CA(LENC), CD(LENC)
```

### 3 Description

C09DAF computes the one-dimensional MODWT of a given input data array,  $x_i$ , for  $i = 1, 2, \dots, n$ , at a single level. For a chosen wavelet filter pair, the output coefficients are obtained by applying convolution to the input,  $x$ . The approximation (or smooth) coefficients,  $C_a$ , are produced by the low pass filter and the detail coefficients,  $C_d$ , by the high pass filter. Periodic (circular) convolution is available as an end extension method for application to finite data sets. The number  $n_c$  of coefficients  $C_a$  or  $C_d$  is returned by the initialization routine C09AAF.

### 4 References

Percival D B and Walden A T (2000) *Wavelet Methods for Time Series Analysis* Cambridge University Press

### 5 Arguments

- 1: N – INTEGER *Input*  
*On entry:* the number of elements,  $n$ , in the data array  $x$ .  
*Constraint:* this must be the same as the value N passed to the initialization routine C09AAF.
- 2: X(N) – REAL (KIND=nag\_wp) array *Input*  
*On entry:* X contains the input dataset  $x_i$ , for  $i = 1, 2, \dots, n$ .
- 3: LENC – INTEGER *Input*  
*On entry:* the dimension of the arrays CA and CD as declared in the (sub)program from which C09DAF is called. This must be at least the number,  $n_c$ , of approximation coefficients,  $C_a$ , and detail coefficients,  $C_d$ , of the discrete wavelet transform as returned in NWC by the call to the initialization routine C09AAF. Note that  $n_c = n$  for periodic end extension, but this is not the case for other end extension methods which will be available in future releases.  
*Constraint:*  $LENC \geq n_c$ , where  $n_c$  is the value returned in NWC by the call to the initialization routine C09AAF.
- 4: CA(LENC) – REAL (KIND=nag\_wp) array *Output*  
*On exit:* CA( $i$ ) contains the  $i$ th approximation coefficient,  $C_a(i)$ , for  $i = 1, 2, \dots, n_c$ .

- 5: CD(LENC) – REAL (KIND=nag\_wp) array *Output*  
*On exit:* CD( $i$ ) contains the  $i$ th detail coefficient,  $C_d(i)$ , for  $i = 1, 2, \dots, n_c$ .
- 6: ICOMM(100) – INTEGER array *Communication Array*  
*On entry:* contains details of the discrete wavelet transform and the problem dimension as setup in the call to the initialization routine C09AAF.  
*On exit:* contains additional information on the computed transform.
- 7: IFAIL – INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.  
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**  
*On exit:* IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, N is inconsistent with the value passed to the initialization routine:  $N = \langle value \rangle$ , N should be  $\langle value \rangle$ .

IFAIL = 3

On entry, array dimension LENC not large enough:  $LENC = \langle value \rangle$  but must be at least  $\langle value \rangle$ .

IFAIL = 6

On entry, the initialization routine C09AAF has not been called first or it has not been called with WTRANS = 'T', or the communication array ICOMM has become corrupted.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

## 7 Accuracy

The accuracy of the wavelet transform depends only on the floating-point operations used in the convolution and downsampling and should thus be close to *machine precision*.

## 8 Parallelism and Performance

C09DAF is not threaded in any implementation.

## 9 Further Comments

None.

## 10 Example

This example computes the one-dimensional maximal overlap discrete wavelet decomposition for 8 values using the Daubechies wavelet, WAVNAM = 'DB4'.

### 10.1 Program Text

```

Program c09dafa

!      C09DAF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: c09aaf, c09daf, c09dbf, nag_wp
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Integer                     :: ifail, n, nf, nwc, nwl, ny
      Character (14)              :: mode, wavnam, wtrans
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: ca(:), cd(:), x(:), y(:)
      Integer                     :: icomm(100)
!      .. Executable Statements ..

!      .. Executable Statements ..
      Write (nout,*) 'C09DAF Example Program Results'
!      Skip heading in data file
      Read (nin,*)
!      Read problem parameters.
      Read (nin,*) n
      Read (nin,*) wavnam, mode
      Allocate (x(n),y(n))

      Write (nout,99999) wavnam, mode
!      Read array
      Read (nin,*) x(1:n)
      Write (nout,*) 'Input Data      X : '
      Write (nout,99997) x(1:n)
!      Query wavelet filter dimensions
      wtrans = 'Time invariant'

!      ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
      Call c09aaf(wavnam,wtrans,mode,n,nwl,nf,nwc,icomm,ifail)

      Allocate (ca(nwc),cd(nwc))

      ifail = 0
      Call c09daf(n,x,nwc,ca,cd,icomm,ifail)

```

```

Write (nout,99998)
Write (nout,99997) ca(1:nwc)
Write (nout,99996)
Write (nout,99997) cd(1:nwc)

ny = n

ifail = 0
Call c09dbf(nwc,ca,cd,ny,y,icomm,ifail)

Write (nout,99995)
Write (nout,99997) y(1:ny)

99999 Format (1X,'MODWT :: Wavelet: ',A,', End mode: ',A)
99998 Format (1X,'Approximation coefficients CA : ')
99997 Format (1X,8(F8.4,1X),:)
99996 Format (1X,'Detail coefficients          CD : ')
99995 Format (1X,'Reconstruction          Y : ')
      End Program c09dafa

```

## 10.2 Program Data

```

C09DAF Example Program Data
8          : n
DB4  Periodic      : wavnam, mode
1.0
3.0
5.0
7.0
6.0
4.0
5.0
2.0          : x(1:n)

```

## 10.3 Program Results

```

C09DAF Example Program Results
MODWT :: Wavelet: DB4          , End mode: Periodic
Input Data      X :
 1.0000  3.0000  5.0000  7.0000  6.0000  4.0000  5.0000  2.0000
Approximation coefficients CA :
 2.7781  1.5146  2.2505  4.8788  6.6845  6.3423  4.7869  3.7644
Detail coefficients          CD :
-0.6187  0.6272  0.1883 -1.1966  1.2618  0.3354 -0.3314 -0.2660
Reconstruction          Y :
 1.0000  3.0000  5.0000  7.0000  6.0000  4.0000  5.0000  2.0000

```

---