

NAG Library Routine Document

D01ESF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

D01ESF approximates a vector of definite integrals \mathbf{F} over the unit hypercube $\Omega = [0, 1]^d$, given the vector of integrands $\mathbf{f}(\mathbf{x})$.

$$\mathbf{F} = \int_{\Omega} \mathbf{f}(\mathbf{x}) d\mathbf{x} = \int_0^1 \int_0^1 \dots \int_0^1 \mathbf{f}(x_1, x_2, \dots, x_d) dx_1 dx_2 \dots dx_d.$$

The routine uses a sparse grid discretisation, allowing for computationally feasible estimations of integrals of high dimension ($d \sim O(100)$).

2 Specification

```
SUBROUTINE D01ESF (NI, NDIM, F, MAXDLV, DINEST, ERREST, IVALID, IOPTS,      &
                  OPTS, IUSER, RUSER, IFAIL)
INTEGER           NI, NDIM, MAXDLV(NDIM), IVALID(NI), IOPTS(100),      &
                  IUSER(*), IFAIL
REAL (KIND=nag_wp) DINEST(NI), ERREST(NI), OPTS(100), RUSER(*)
EXTERNAL          F
```

3 Description

D01ESF uses a sparse grid to generate a vector of approximations $\hat{\mathbf{F}}$ to a vector of integrals \mathbf{F} over the unit hypercube $\Omega = [0, 1]^d$, that is,

$$\hat{\mathbf{F}} \approx \mathbf{F} = \int_{[0,1]^d} \mathbf{f}(\mathbf{x}) d\mathbf{x}.$$

3.1 Comparing Quadrature Over Full and Sparse Grids

Before illustrating the sparse grid construction, it is worth comparing integration over a sparse grid to integration over a full grid.

Given a one-dimensional quadrature rule with N abscissae, which accurately evaluates a polynomial of order P_N , a full tensor product over d dimensions, a full grid, may be constructed with N^d multidimensional abscissae. Such a product will accurately integrate a polynomial where the maximum power of any dimension is P_N . For example if $d = 2$ and $P_N = 3$, such a rule will accurately integrate any polynomial whose highest order term is $x_1^3 x_2^3$. Such a polynomial may be said to have a maximum combined order of P_N^d , provided no individual dimension contributes a power greater than P_N . However, the number of multidimensional abscissae, or points, required increases exponentially with the dimension, rapidly making such a construction unusable.

The sparse grid technique was developed by Smolyak (Smolyak (1963)). In this, multiple one-dimensional quadrature rules of increasing accuracy are combined in such a way as to provide a multidimensional quadrature rule which will accurately evaluate the integral of a polynomial whose maximum order appears as a monomial. Hence a sparse grid construction whose highest level quadrature rule has polynomial order P_N will accurately integrate a polynomial whose maximum combined order is also P_N . Again taking $P_N = 3$, one may, theoretically, accurately integrate a polynomial such as $x^3 + x^2 y + y^3$, but not a polynomial such as $x^3 y^3 + xy$. Whilst this has a lower maximum combined order than the full tensor product, the number of abscissae required increases significantly slower than the equivalent full grid, making some classes of integrals of dimension

$d \sim O(100)$ tractable. Specifically, if a one-dimensional quadrature rule of level ℓ has $N \sim O(2^\ell)$ abscissae, the corresponding full grid will have $O((2^\ell)^d)$ multidimensional abscissae, whereas the sparse grid will have $O(2^\ell d^{\ell-1})$. Figure 1 demonstrates this using a Gauss–Patterson rule with 15 points in 3 dimensions. The full grid requires 3375 points, whereas the sparse grid only requires 111.




Figure 1

Three-dimensional full (left) and sparse (right) grids, constructed from the 15 point Gauss–Patterson rule

3.2 Sparse Grid Quadrature

We now include a brief description of the sparse grid construction, sufficient for the understanding of the use of this routine. For a more detailed analysis, see Gerstner and Griebel (1998).

Consider a one-dimensional n_ℓ -point quadrature rule of level ℓ , Q_ℓ . The action of this rule on a integrand f is to approximate its definite one-dimensional integral ${}_1F$ as,

$${}_1F = \int_0^1 f(x)dx \approx Q_\ell(f) = \sum_{i=1}^{n_\ell} w_{\ell,i} \times f(x_{\ell,i}),$$

using weights $w_{\ell,i}$ and abscissae $x_{\ell,i}$, for $i = 1, 2, \dots, n_\ell$.

Now construct a set of one-dimensional quadrature rules, $\{Q_\ell \mid \ell = 1, \dots, L\}$, such that the accuracy of the quadrature rule increases with the level number. In this routine we exclusively use quadrature rules which are completely nested, implying that if an abscissae $x_{\ell,k}$ is in level ℓ , it is also in level $\ell + 1$. The quantity L denotes some maximum level appropriate to the rules that have been selected.

Now define the action of the tensor product of d rules as,

$$(Q_{\ell_1} \otimes \dots \otimes Q_{\ell_d})(f) = \sum_{i_1=1}^{n_{\ell_1}} \dots \sum_{i_d=1}^{n_{\ell_d}} w_{\ell_1,i_1} \dots w_{\ell_d,i_d} f(x_{\ell_1,i_1}, \dots, x_{\ell_d,i_d}),$$

where the individual level indices ℓ_j are not necessarily ordered or unique. Each tensor product of d rules defines an action of the quadrature rules Q_ℓ , $\ell = (\ell_1, \ell_2, \dots, \ell_d)$ over a subspace, which is given a level $|\ell| = \sum_{j=1}^d \ell_j$. If all rule levels are equal, this is the full tensor product of that level.

The sparse grid construction of level ℓ can then be declared as the sum of all actions of the quadrature differences $\Delta_k = (Q_k - Q_{k-1})$, over all subspaces having a level at most $\ell - d + 1$,

$${}_dF \approx Q_\ell^d(f) = \sum_{\text{level at most } \ell-d+1} (\Delta_{k_1} \otimes \cdots \otimes \Delta_{k_d})(f). \quad (1)$$

By definition, all subspaces used for level $\ell - 1$ must also be used for level ℓ , and as such the difference between the result of all actions over subsequent sparse grid constructions may be used as an error estimate.

Let L be the maximum level allowable in a sparse grid construction. The classical sparse grid construction of $\ell = L$ allows each dimension to support a one-dimensional quadrature rule of level at most L , with such a quadrature rule being used in every dimension at least once. Such a construction lends equal weight to each dimension of the integration, and is termed here ‘isotropic’.

Define the set $\mathbf{m} = (m_j, j = 1, 2, \dots, d)$, where m_j is the maximum quadrature rule allowed in the j th dimension, and m_q to be the maximum quadrature rule used by any dimension. Let a subspace be identified by its quadrature difference levels, $\mathbf{k} = (k_1, k_2, \dots, k_d)$.

The classical construction may be extended by allowing different dimensions to have different values m_j , and by allowing $m_q \leq L$. This creates non-isotropic constructions. These are especially useful in higher dimensions, where some dimensions contribute more than others to the result, as they can drastically reduce the number of function evaluations required.

For example, consider the two-dimensional construction with $L = 4$. The classical isotropic construction would have the following subspaces.

Subspaces generated by a classical sparse grid with $L = 4$.

Level	Subspaces
1	(1, 1)
2	(2, 1), (1, 2)
3	(3, 1), (2, 2), (1, 3)
4	(4, 1), (3, 2), (2, 3), (1, 4)

If the variation in the second dimension is sufficiently accurately described by a quadrature rule of level 2, the contributions of the subspaces (1, 3) and (1, 4) are probably negligible. Similarly, if the variation in the first dimension is sufficiently accurately described by a quadrature rule of level 3, the subspace (4, 1) is probably negligible. Furthermore the subspace (2, 3) would also probably have negligible impact, whereas the subspaces (2, 2) and (3, 2) would not. Hence restricting the first dimension to a maximum level of 3, and the second dimension to a maximum level of 2 would probably give a sufficiently acceptable estimate, and would generate the following subspaces.

Subspaces generated by a non-isotropic sparse grid with $L = 4$, $m_q = 3$ and $\mathbf{m} = (3, 2)$.

Level	Subspaces
1	(1, 1)
2	(2, 1), (1, 2)
3	(3, 1), (2, 2)
4	(4, 1), (3, 2)

Taking this to the extreme, if the variation in the first and second dimensions are sufficiently accurately described by a level 2 quadrature rule, restricting the maximum level of both dimensions to 2 would generate the following subspaces.

Subspaces generated by a sparse grid construction with $L = 4$, $m_q = 2$ and $\mathbf{m} = (2, 2)$.

Level	Subspaces
1	(1, 1)
2	(2, 1), (1, 2)
3	(2, 2)
4	None

Hence one subspace is generated at level 3, and no subspaces are generated at level 4. The level 3 subspace (2,2) actually indicates that this is the full grid of level 2.

3.3 Using D01ESF

D01ESF uses optional parameters, supplied in the option arrays IOPTS and OPTS. Before calling D01ESF, these option arrays must be initialized using D01ZKF. Once initialized, the required options may be set and queried using D01ZKF and D01ZLF respectively. A complete list of the options available may be found in Section 11.

You may control the maximum level required, L , using the optional parameter **Maximum Level**. Furthermore, you may control the first level at which the error comparison will take place using the optional parameter **Minimum Level**, allowing for the forced evaluation of a predetermined number of levels before the routine attempts to complete. Completion is flagged when the error estimate is sufficiently small:

$$|\hat{F}_d^k - \hat{F}_d^{k-1}| \leq \max(\epsilon_a, \epsilon_r \times \hat{F}_d^k),$$

where ϵ_a and ϵ_r are the absolute and relative error tolerances, respectively, and $k \leq L$ is the highest level at which computation was performed. The tolerances ϵ_a and ϵ_r can be controlled by setting the optional parameters **Absolute Tolerance** and **Relative Tolerance**.

Owing to the interlacing nature of the quadrature rules used herein, abscissae \mathbf{x} required in lower level subspaces will also appear in higher-level subspaces. This allows for calculations which will be repeated later to be stored and re-used. However, this is naturally at the expense of memory. It may also be at the expense of computational time, depending on the complexity of the integrands, as the lookup time for a given value is (at worst) $O(d)$. Furthermore, as the sparse grid level increases, fewer subsequent levels will require values from the current level. You may control the number of levels for which values are stored by setting the optional parameter **Index Level**.

Two different sets of interlacing quadrature rules are selectable using the optional parameter **Quadrature Rule**: Gauss–Patterson and Clenshaw–Curtis. Gauss–Patterson rules offer greater polynomial accuracy, whereas Clenshaw–Curtis rules are often effective for oscillatory integrands. Clenshaw–Curtis rules require function values to be evaluated on the boundary of the hypercube, whereas Gauss–Patterson rules do not. Both of these rules use precomputed weights, and as such there is an effective limit on m_q ; see the description of the optional parameter **Quadrature Rule**. The value of m_q is returned by the queriable optional parameter **Maximum Quadrature Level**.

D01ESF also allows for non-isotropic sparse grids to be constructed. This is done by appropriately setting the array MAXDLV. It should be emphasised that a non-isometric construction should only be used if the integrands behave in a suitable way. For example, they may decay toward zero as the lesser dimensions approach their bounds of Ω . It should also be noted that setting $\text{MAXDLV}(k) = 1$ will not reduce the dimension of the integrals, it will simply indicate that only one point in dimension k should be used. It is also advisable to approximate the integrals several times, once with an isometric construction of some level, and then with a non-isometric construction with higher levels in various dimensions. If the difference between the solutions is significantly more than the returned error estimates, the assumptions of dimensional importance are probably incorrect.

The abscissae in each subspace are generally expressible in a sparse manner, because many of the elements of each abscissa will in fact be the centre point of the dimension, which is termed here the ‘trivial’ element. In this routine the trivial element is always returned as 0.5 owing to the restriction to the $[0, 1]$ hypercube. As such, the subroutine F returns the abscissae in Compressed Column Storage (CCS) format (see the F11 Chapter Introduction). This has particular advantages when using accelerator hardware to evaluate the required functions, as much less data must be forwarded. It also, potentially, allows for calculations to be computed faster, as any sub-calculations dependent upon the trivial value may be potentially re-used. See the example in Section 10.

4 References

Caflish R E, Morokoff W and Owen A B (1997) Valuation of mortgage backed securities using Brownian bridges to reduce effective dimension *Journal of Computational Finance* **1** 27–46

Gerstner T and Griebel M (1998) Numerical integration using sparse grids *Numerical Algorithms* **18** 209–232

Smolyak S A (1963) Quadrature and interpolation formulas for tensor products of certain classes of functions *Dokl. Akad. Nauk SSSR* **4** 240–243

5 Arguments

1: NI – INTEGER *Input*

On entry: n_i , the number of integrands.

Constraint: $NI \geq 1$.

2: NDIM – INTEGER *Input*

On entry: d , the number of dimensions.

Constraint: $NDIM \geq 1$.

3: F – SUBROUTINE, supplied by the user. *External Procedure*

F must return the value of the integrands f_j at a set of n_x d -dimensional points \mathbf{x}_i , implicitly supplied as columns of a matrix $X(d, n_x)$. If X was supplied explicitly you would find that most of the elements attain the same value, x_{tr} ; the larger the number of dimensions, the greater the proportion of elements of X would be equal to x_{tr} . So, X is effectively a sparse matrix, except that the ‘zero’ elements are replaced by elements that are all equal to the value x_{tr} . For this reason X is supplied, as though it were a sparse matrix, in compressed column storage (CCS) format (see the F11 Chapter Introduction).

Individual entries $x_{k,i}$ of X , for $k = 1, 2, \dots, d$, are either trivially valued, in which case $x_{k,i} = x_{tr}$, or are non-trivially valued. For point i , the non-trivial row indices and corresponding abscissae values are supplied in elements $c(i) = \text{ICOLZP}(i), \dots, \text{ICOLZP}(i+1) - 1$, for $i = 1, 2, \dots, n_x$, of the arrays IROWIX and XS, respectively. Hence the i th column of the matrix X is retrievable as

$$X(\text{IROWIX}(c(i)), i) = \text{XS}(c(i)),$$

$$X(k \notin \text{IROWIX}(c(i)), i) = x_{tr}.$$

An equivalent integer valued matrix Q is also implicitly provided. This contains the unique indices $q_{k,i}$ of the underlying one-dimensional quadrature rule corresponding to the individual abscissae $x_{k,i}$. For trivial abscissae, the implicit index $q_{k,i} = 1$. Q is supplied in the same CCS format as X , with the non-trivial values supplied in QS.

The specification of F is:

```
SUBROUTINE F (NI, NDIM, NX, XTR, NNTR, ICOLZP, IROWIX, XS, QS,      &
              FM, IFLAG, IUSER, RUSER)
```

```
INTEGER          NI, NDIM, NX, NNTR, ICOLZP(NX+1),                &
                  IROWIX(NNTR), QS(NNTR), IFLAG, IUSER(*)
REAL (KIND=nag_wp) XTR, XS(NNTR), FM(NI, NX), RUSER(*)
```

1: NI – INTEGER *Input*

On entry: n_i , the number of integrands.

2: NDIM – INTEGER *Input*

On entry: d , the number of dimensions.

3:	NX – INTEGER	<i>Input</i>
	<i>On entry:</i> n_x , the number of points x_i , corresponding to the number of columns of X , at which the set of integrands must be evaluated.	
4:	XTR – REAL (KIND=nag_wp)	<i>Input</i>
	<i>On entry:</i> x_{tr} , the value of the trivial elements of X .	
5:	NNTR – INTEGER	<i>Input</i>
	<i>On entry:</i> if IFLAG > 0, the number of non-trivial elements of X . If IFLAG = 0, the total number of abscissae from the underlying one-dimensional quadrature.	
6:	ICOLZP(NX + 1) – INTEGER array	<i>Input</i>
	<i>On entry:</i> the set $\{ICOLZP(i), \dots, ICOLZP(i+1) - 1\}$ contains the indices of IROWIX and XS corresponding to the non-trivial elements of column i of X and hence of the point \mathbf{x}_i , for $i = 1, 2, \dots, n_x$.	
7:	IROWIX(NNTR) – INTEGER array	<i>Input</i>
	<i>On entry:</i> the row indices corresponding to the non-trivial elements of X .	
8:	XS(NNTR) – REAL (KIND=nag_wp) array	<i>Input</i>
	<i>On entry:</i> $x_{k,i} \neq x_{tr}$, the non-trivial entries of X .	
9:	QS(NNTR) – INTEGER array	<i>Input</i>
	<i>On entry:</i> $q_{k,i} \neq 1$, the indices of the underlying quadrature rules corresponding to $x_{k,i} \neq x_{tr}$.	
10:	FM(NI, NX) – REAL (KIND=nag_wp) array	<i>Output</i>
	<i>On exit:</i> $FM(p, i) = f_p(\mathbf{x}_i)$, for $i = 1, 2, \dots, n_x$ and $p = 1, 2, \dots, n_i$.	
11:	IFLAG – INTEGER	<i>Input/Output</i>
	<i>On entry:</i> if IFLAG = 0, this is the first call to F. $n_x = 1$, and the entire point \mathbf{x}_1 will satisfy $x_{k,1} = x_{tr}$, for $k = 1, 2, \dots, d$. In addition, NNTR contains the total number of abscissae from the underlying one-dimensional quadrature; XS contains the complete set of abscissae and QS contains the corresponding quadrature indices, with $XS(1) = x_{tr}$ and $QS(1) = 1$. This will always be called in serial. In subsequent calls to F, IFLAG = 1. Subsequent calls may be made from within an OpenMP parallel region. See Section 8 for details. <i>On exit:</i> set IFLAG < 0 if you wish to force an immediate exit from D01ESF with IFAIL = -1.	
12:	IUSER(*) – INTEGER array	<i>User Workspace</i>
13:	RUSER(*) – REAL (KIND=nag_wp) array	<i>User Workspace</i>
14:	PUSER – TYPE (C_PTR)	<i>Communication Array</i>
	F is called with the arguments IUSER, RUSER and PUSER as supplied to D01ESF. You should use the arrays IUSER, RUSER and PUSER to supply information to F.	

F must either be a module subprogram USED by, or declared as EXTERNAL in, the (sub) program from which D01ESF is called. Arguments denoted as *Input* must **not** be changed by this procedure.

- 4: MAXDLV(NDIM) – INTEGER array *Input*
On entry: **m**, the array of maximum levels for each dimension. MAXDLV(j), for $j = 1, 2, \dots, d$, contains m_j , the maximum level of quadrature rule dimension j will support.
 The default value, $\min(m_q, L)$ will be used if either $\text{MAXDLV}(j) \leq 0$ or $\text{MAXDLV}(j) \geq \min(m_q, L)$ (for details on the default values for m_q and L and on how to change these values see the options **Maximum Level**, **Maximum Quadrature Level** and **Quadrature Rule**).
 If $\text{MAXDLV}(j) = 1$ for all j , only one evaluation will be performed, and as such no error estimation will be possible.
Suggested value: $\text{MAXDLV}(j) = 0$ for all $j = 1, 2, \dots, d$.
Note: setting non-default values for some dimensions makes the assumption that the contribution from the omitted subspaces is 0. The integral and error estimates will only be based on included subspaces, which if the 0 contribution assumption is not valid will be erroneous.
- 5: DINEST(NI) – REAL (KIND=nag_wp) array *Output*
On exit: DINEST(p) contains the final estimate \hat{F}_p of the definite integral F_p , for $p = 1, 2, \dots, n_i$.
- 6: ERREST(NI) – REAL (KIND=nag_wp) array *Output*
On exit: ERREST(p) contains the final error estimate E_p of the definite integral F_p , for $p = 1, 2, \dots, n_i$.
- 7: IVALID(NI) – INTEGER array *Output*
On exit: IVALID(p) indicates the final state of integral p , for $p = 1, 2, \dots, n_i$.
 IVALID(p) = 0
 The error estimate for integral p was below the requested tolerance.
 IVALID(p) = 1
 The error estimate for integral p was below the requested tolerance. The final level used was non-isotropic.
 IVALID(p) = 2
 The error estimate for integral p was above the requested tolerance.
 IVALID(p) = 3
 The error estimate for integral p was above $\max(0.1|\hat{F}_p|, 0.01)$.
 IVALID(p) < 0
 You aborted the evaluation before an error estimate could be made.
- 8: IOPTS(100) – INTEGER array *Communication Array*
 9: OPTS(100) – REAL (KIND=nag_wp) array *Communication Array*
 The arrays IOPTS and OPTS **must not** be altered between calls to any of the routines D01ESF, D01ZKF and D01ZLF.
- 10: IUSER(*) – INTEGER array *User Workspace*
 11: RUSER(*) – REAL (KIND=nag_wp) array *User Workspace*
 12: PUSER – TYPE (C_PTR) *Communication Array*
 IUSER, RUSER and PUSER are not used by D01ESF, but are passed directly to F and should be used to pass information to this routine.
- 13: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output arguments may be useful even if $IFAIL \neq 0$ on exit, the recommended value is -1 . **When the value -1 or 1 is used it is essential to test the value of $IFAIL$ on exit.**

On exit: $IFAIL = 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry $IFAIL = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Note: D01ESF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

$IFAIL = 1$

The requested accuracy was not achieved for at least one integral.

$IFAIL = 2$

No accuracy was achieved for at least one integral.

$IFAIL = 11$

On entry, $NI = \langle value \rangle$.

Constraint: $NI \geq 1$.

$IFAIL = 21$

On entry, $NDIM = \langle value \rangle$.

Constraint: $NDIM \geq 1$.

$IFAIL = 1001$

Either the option arrays IOPTS and OPTS have not been initialized for D01ESF, or they have become corrupted.

$IFAIL = -1$

Exit requested from F with $IFLAG = \langle value \rangle$.

$IFAIL = -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

$IFAIL = -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

$IFAIL = -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

For each integral p , an error estimate E_p is returned, where,

$$E_p = \left| \hat{F}_p^k - \hat{F}_p^{k-1} \right| \approx \left| \hat{F}_p - F_p \right|,$$

where $k \leq L$ is the highest level at which computation was performed.

8 Parallelism and Performance

8.1 Direct Threading

D01ESF is directly threaded for parallel execution. For each level, at most n_t threads will evaluate the integrands over independent subspaces of the construction, and will construct a partial sum of the level's contribution. Once all subspaces from a given level have been processed, the partial sums are combined to give the total contribution of the level, which is in turn added to the total solution. For a given number of threads, the division of subspaces between the threads, and the order in which a single thread operates over its assigned subspaces, is fixed. However, the order in which all subspaces are combined will necessarily be different to the single threaded case, which may result in some discrepancy in the results between parallel and serial execution.

To mitigate this discrepancy, it is recommended that D01ESF be instructed to use higher-than-working precision to accumulate the actions over the subspaces. This is done by setting the option **Summation Precision** = HIGHER, which is the default behaviour. This has some computational cost, although this is typically negligible in comparison to the overall runtime, particularly for non-trivial integrands.

If **Summation Precision** = WORKING, then the accumulation will be performed using working precision, which may provide some increase in performance. Again, this is probably negligible in comparison to the overall runtime.

For some problems, typically of lower dimension, there may be insufficient work to warrant direct threading at lower levels. For example, a three-dimensional problem will require at most 3 subspaces to be evaluated at level 2, and at most 6 subspaces at level 3. Furthermore, level 2 subspaces typically contain only 2 new multidimensional abscissae, while level 3 subspaces typically contain 2 or 4 new multidimensional abscissae depending on the **Quadrature Rule**. If there are more threads than the number of available subspaces at a given level, or the amount of work in each subspace is outweighed by the amount of work required to generate the parallel region, parallel efficiency will be decreased. This may be mitigated to some extent by evaluating the first s_l levels in serial. The value of s_l may be altered using the optional parameter **Serial Levels**. If $s_l \geq L$, then all levels will be evaluated in serial and no direct threading will be utilized.

If you use direct threading in the manner just described, you must ensure any access to the user arrays IUSER and RUSER is done in a thread-safe manner. These are classed as OpenMP SHARED, and are passed directly to the subroutine F for every thread.

8.2 Parallelization of F

The vectorized interface also allows for parallelization inside the subroutine F by evaluating the required integrands in parallel. Provided the values returned by F match those that would be returned without parallelizing F, the final result should match the serial result, barring any discrepancies in accumulation. If you wish to parallelize F, it is advisable to set a large value for **Maximum Nx**, although be aware that increasing **Maximum Nx** will increase the memory requirement of the routine. In general, parallelization of F should not be necessary, as the higher-level parallelism over different subspaces scales well for many problems.

9 Further Comments

Not applicable.

10 Example

The example program evaluates an estimate to the set of integrals

$$\mathbf{F} = \int_{\Omega} \begin{pmatrix} \sin(1 + |\mathbf{x}|) \\ \vdots \\ \sin(n_i + |\mathbf{x}|) \end{pmatrix} \log |\mathbf{x}| d\mathbf{x}$$

where $|\mathbf{x}| = \sum_{j=1}^d jx_j$. It also demonstrates a simple method to safely use IUSER and RUSER as workspace for sub-calculations when running in parallel.

10.1 Program Text

```
! D01ESF Example Program Text
! Mark 26 Release. NAG Copyright 2016.

Module d01esfe_mod

! D01ESF Example Program Module:
! User-defined Routines

! .. Use Statements ..
Use nag_library, Only: nag_wp
! .. Implicit None Statement ..
Implicit None
! .. Accessibility Statements ..
Private
Public                                :: f
Contains
Subroutine f(ni,ndim,nx,xtr,nntr,icolzp,irowix,qs,qs,fm,iflag,iuser,      &
  ruser)

! .. Use Statements ..
Use nag_library, Only: x06adf
! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: xtr
Integer, Intent (Inout)         :: iflag
Integer, Intent (In)            :: ndim, ni, nntr, nx
! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Out) :: fm(ni,nx)
Real (Kind=nag_wp), Intent (Inout) :: ruser(*)
Real (Kind=nag_wp), Intent (In) :: xs(nntr)
Integer, Intent (In)            :: icolzp(nx+1), irowix(nntr), qs(nntr)
Integer, Intent (Inout)         :: iuser(*)
! .. Local Scalars ..
Real (Kind=nag_wp)              :: s_ntr, s_tr
Integer                          :: i, j, logs_hi, logs_lo, s_hi, s_lo, &
  tid
! .. Intrinsic Procedures ..
Intrinsic                        :: log, real, sin, sum
! .. Executable Statements ..

! For each evaluation point x_i, i = 1, ..., nx, return in fm the
! computed values of the ni integrands f_j, j = 1, ..., ni defined by

! fm(j,i) = f_j(x_i)
!
! = sin(j + S(i))*log(S(i)), where S(i) = Sum_{k=1}^{ndim} k*x_i(k).
!
! Split the S expression into two components, one involving only the
! 'trivial' value xtr:

! S(i) = Sum_{k=1}^{ndim} (k*xtr) + Sum_{k=1}^{ndim} (k*(x_i(k)-xtr))
!
```

```

!           ndim*(ndim+1)   ndim
!           = xtr * ----- + Sum (k*(x_i(k)-xtr))
!                   2       k=1

!           := s_tr          + s_ntr(i)

!       By definition the summands in the s_ntr(i) term on the right-hand side
!       are zero for those k outside the range of indices defined in irowix.

!       As a demonstration of safely operating with the user arrays iuser and
!       ruser when running in parallel, 'partition' these based on the current
!       thread number. Store some of the s_tr and s_ntr computations in these
!       array sections.

!       The thread number, converted to 1-based numbering.
tid = x06adf() + 1

s_lo = iuser(tid)
s_hi = s_lo + nx - 1
logs_lo = s_hi + 1
logs_hi = logs_lo + nx - 1

If (iflag==0) Then
!       First call: nx=1, no non-trivial dimensions.
!       The constant s_tr can be reused by all subsequent calculations.
s_tr = 0.5E0_nag_wp*xtr*real(ndim*(ndim+1),kind=nag_wp)
ruser(1) = s_tr
ruser(s_lo) = s_tr
ruser(logs_lo) = log(s_tr)
Else
!       Calculate S(i) = s_tr + s_ntr(i).
s_tr = ruser(1)
Do i = 1, nx
    s_ntr = sum(real(irowix(icolzp(i):icolzp(i+1)-1),kind=nag_wp)*(xs(icolzp(i):icolzp(i+1)-1)-xtr)) &
    ruser(s_lo+i-1) = s_ntr + s_tr
    ruser(logs_lo+i-1) = log(s_ntr+s_tr)
End Do
End If

!       Finally we obtain fm(j,:) = sin(j+S(:))*log(S(:))
Do j = 1, ni
    fm(j,:) = sin(real(j,kind=nag_wp)+ruser(s_lo:s_hi))* &
    ruser(logs_lo:logs_hi)
End Do

Return
End Subroutine f
End Module d01esfe_mod
Program d01esfe

!       .. Use Statements ..
Use nag_library, Only: d01esf, d01zkf, d01zlf, nag_wp, x06acf
Use d01esfe_mod, Only: f
!       .. Implicit None Statement ..
Implicit None
!       .. Parameters ..
Integer, Parameter :: nout = 6
!       .. Local Scalars ..
Real (Kind=nag_wp) :: rvalue
Integer :: ifail, j, liuser, lruser, maxnx, &
ndim, ni, optype, smpthd
Character (16) :: cvalue
!       .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: dinest(:), errest(:), opts(:), &
ruser(:)
Integer, Allocatable :: iopts(:), iuser(:), invalid(:), &
maxdlv(:)
!       .. Executable Statements ..
Write (nout,*) 'D01ESF Example Program Results'
```

```

Write (nout,*)
ni = 10
ndim = 4

Allocate (iopts(100),opts(100),ivalid(ni),dinest(ni),errest(ni),      &
          maxdlv(ndim))

!   Initialize option arrays.
ifail = 0
Call d01zkf('Initialize = D01ESF',iopts,100,opts,100,ifail)

!   Set any required options.
Call d01zkf('Absolute Tolerance = 0.0',iopts,100,opts,100,ifail)
Call d01zkf('Relative Tolerance = 1.0e-3',iopts,100,opts,100,ifail)
Call d01zkf('Maximum Level = 6',iopts,100,opts,100,ifail)
Call d01zkf('Index Level = 5',iopts,100,opts,100,ifail)

!   Set any required maximum dimension levels.
maxdlv(:) = 0

!   As a demonstration of safely operating with the user arrays iuser and
!   ruser when running in parallel, we will 'partition' these in the user-
!   supplied function f based on the current thread number.
!   The size of these arrays is a function of Maximum Nx and the maximum
!   allowed number of OpenMP threads.

ifail = 0
Call d01zlf('Maximum Nx',maxnx,rvalue,cvalue,optype,iopts,opts,ifail)

smpthd = x06acf()

lruser = 1 + 2*maxnx*smpthd
liuser = smpthd
Allocate (iuser(liuser),ruser(lruser))

!   iuser stores the partition indices for ruser:
iuser(1) = 2
Do j = 2, smpthd
    iuser(j) = iuser(j-1) + 2*maxnx
End Do

!   Approximate the integrals.
ifail = -1
Call d01esf(ni,ndim,f,maxdlv,dinest,errest,ivalid,iopts,opts,iuser,      &
            ruser,ifail)
Select Case (ifail)
Case (0,1,2,-1)
!   0: The result returned satisfies the requested accuracy requirements.
!   1, 2: The result returned is inaccurate for at least one integral.
!   -1: Exit was requested by setting iflag negative in f.
!       A result will be returned if at least one call to f was
!       successful.
Write (nout,99999)
Do j = 1, ni
    Write (nout,99998) j, dinest(j), errest(j), ivalid(j)
End Do
Case Default
!   If internal memory allocation failed consider reducing the options
!   'Maximum Nx' and 'Index Level', or run with fewer threads.
Write (nout,99997) ifail
End Select

99999 Format (1X,'Integral # | Estimated value | Error estimate | ',      &
            'Final state of integral')
99998 Format (1X,I11,'|',Es17.5,'|',Es16.5,'|',I8)
99997 Format (1X,'D01ESF exited with IFAIL = ',I8)
End Program d01esf

```

10.2 Program Data

None.

10.3 Program Results

D01ESF Example Program Results

Integral #	Estimated value	Error estimate	Final state of integral
1	3.83522E-02	2.39770E-05	0
2	4.01177E-01	1.69503E-05	0
3	3.95161E-01	5.66045E-06	0
4	2.58363E-02	2.30670E-05	0
5	-3.67242E-01	1.92659E-05	0
6	-4.22680E-01	2.24822E-06	0
7	-8.95077E-02	2.16953E-05	0
8	3.25958E-01	2.11959E-05	0
9	4.41739E-01	1.20901E-06	0
10	1.51388E-01	1.98894E-05	0

11 Optional Parameters

Several optional parameters in D01ESF control aspects of the algorithm, methodology used, logic or output. Their values are contained in the arrays IOPTS and OPTS; these must be initialized before calling D01ESF by first calling D01ZKF with OPTSTR set to "Initialize = D01ESF".

Each optional parameter has an associated default value; to set any of them to a non-default value, or to reset any of them to the default value, use D01ZKF. The current value of an optional parameter can be queried using D01ZLF.

The remainder of this section can be skipped if you wish to use the default values for all optional parameters.

The following is a list of the optional parameters available. A full description of each optional parameter is provided in Section 11.1.

Absolute Tolerance

Index Level

Maximum Level

Maximum Nx

Maximum Quadrature Level

Minimum Level

Quadrature Rule

Relative Tolerance

Serial Levels

Summation Precision

11.1 Description of the Optional Parameters

For each option, we give a summary line, a description of the optional parameter and details of constraints.

The summary line contains:

the keywords, where the minimum abbreviation of each keyword is underlined;

a parameter value, where the letters *a*, *i* and *r* denote options that take character, integer and real values respectively.

the default value.

The following symbol represent various machine constants:

ϵ represents the *machine precision* (see X02AJF).

All options accept the value 'DEFAULT' in order to return single options to their default states.

Keywords and character values are case insensitive, however they must be separated by at least one space.

Queryable options will return the appropriate value when queried by calling D01ZLF. They will have no effect if passed to D01ZKF.

For D01ESF the maximum length of the argument CVALUE used by D01ZLF is 15.

Absolute Tolerance r Default = $\sqrt{\epsilon}$
 $r = \epsilon_a$, the absolute tolerance required.

Index Level i Default = 4
 The maximum level at which function values are stored for later use. Larger values use increasingly more memory, and require more time to access specific elements. Lower levels result in more repeated computation. The **Maximum Quadrature Level**, m_q is the effective upper limit on **Index Level**. If $i \geq m_q$, D01ESF will use m_q as the value of **Index Level**.

Constraint: $i \geq 1$.

Maximum Level i Default = 5
 $i = L$, the maximum number of levels to evaluate.
 Constraint: $1 < i \leq 20$.

Note: the maximum allowable level in any single dimension, m_q , is governed by the **Quadrature Rule** selected. If a value greater than m_q is set, only a subset of subspaces from higher levels will be used. Should this subset be empty for a given level, computation will consider the preceding level to be the maximum level and will terminate.

Maximum Nx i Default = 128
 $i = \max n_x$, the maximum number of points to evaluate in a single call to F.
 Constraint: $1 \leq i \leq 16384$.

Maximum Quadrature Level i Queryable only
 $i = m_q$, the maximum level of the underlying one-dimensional quadrature rule (see **Quadrature Rule**).

Minimum Level i Default = 2
 The minimum number of levels which must be evaluated before an error estimate is used to determine convergence.
 Constraint: $i > 1$.

Note: if the minimum level is greater than the maximum computable level, the maximum level will be used.

Quadrature Rule a Default = Gauss–Patterson
 The underlying one-dimensional quadrature rule to be used in the construction. Open rules do not require evaluations at boundaries.

Quadrature Rule = Gauss–Patterson or GP

The interlacing Gauss–Patterson rules. Level ℓ uses $2^\ell - 1$ abscissae. All levels are open. These rules provide high order accuracy. $m_q = 9$.

Quadrature Rule = Clenshaw–Curtis or CC

The interlacing Clenshaw–Curtis rules. Level ℓ uses $2^{\ell-1} + 1$ abscissae. All levels above level 1 are closed. $m_q = 12$.

Relative Tolerance r Default = $\sqrt{\epsilon}$ $r = \epsilon_a$, the relative tolerance required.**Summation Precision** a

Default = HIGHER

Determines whether D01ESF uses working precision or higher-than-working precision to accumulate the actions over subspaces.

Summation Precision = HIGHER or H

Higher-than-working precision is used to accumulate the action over a subspace, and for the accumulation of all such actions. This is more expensive computationally, although this is probably negligible in comparison to the cost of evaluating the integrands and the overall runtime. This significantly reduces variation in the result when changing the number of threads.

Summation Precision = WORKING or W

Working precision is used to accumulate the actions over subspaces. This may provide some speedup, particularly if n_i or n_t is large. The results of parallel simulations will however be more prone to variation.

Note: the following option is relevant only to multithreaded implementations of the NAG Library.

Serial Levels i

Default = 1

$i = s_l$, the number of levels to be evaluated in serial before initializing parallelization. For relatively trivial integrands, this may need to be set greater than the default to reduce parallel overhead.
