
NAG Library Routine Document

D01FDF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

D01FDF calculates an approximation to a definite integral in up to 30 dimensions, using the method of
Sag and Szekeres (see Sag and Szekeres (1964)). The region of integration is an n-sphere, or by built-in
transformation via the unit n-cube, any product region.

2 Specification

SUBROUTINE D01FDF (NDIM, F, SIGMA, REGION, LIMIT, R0, U, RESULT, NCALLS,
IFAIL)

&

INTEGER NDIM, LIMIT, NCALLS, IFAIL
REAL (KIND=nag_wp) F, SIGMA, R0, U, RESULT
EXTERNAL F, REGION

3 Description

D01FDF calculates an approximation toZ

n-sphere of radius �

f x1; x2; . . . ; xnð Þdx1dx2 � � � dxn ð1Þ

or, more generally,
Z d1

c1

dx1 � � �
Z dn

cn

dxnf x1; . . . ; xnð Þ ð2Þ

where each ci and di may be functions of xj j < ið Þ.
The routine uses the method of Sag and Szekeres (1964), which exploits a property of the shifted
p-point trapezoidal rule, namely, that it integrates exactly all polynomials of degree < p (see Krylov
(1962)). An attempt is made to induce periodicity in the integrand by making a parameterised
transformation to the unit n-sphere. The Jacobian of the transformation and all its direct derivatives
vanish rapidly towards the surface of the unit n-sphere, so that, except for functions which have strong
singularities on the boundary, the resulting integrand will be pseudo-periodic. In addition, the variation
in the integrand can be considerably reduced, causing the trapezoidal rule to perform well.

Integrals of the form (1) are transformed to the unit n-sphere by the change of variables:

xi ¼ yi
�

r
tanh

ur

1� r2

� �

where r2 ¼
Xn
i¼1

y2i and u is an adjustable parameter.

Integrals of the form (2) are first of all transformed to the n-cube �1; 1½ �n by a linear change of
variables

xi ¼ di þ cið Þ þ di � cið Þyið Þ=2
and then to the unit sphere by a further change of variables

yi ¼ tanh
uzi
1� r

� �

where r2 ¼
Xn
i¼1

z2i and u is again an adjustable parameter.

The parameter u in these transformations determines how the transformed integrand is distributed
between the origin and the surface of the unit n-sphere. A typical value of u is 1:5. For larger u, the
integrand is concentrated toward the centre of the unit n-sphere, while for smaller u it is concentrated
toward the perimeter.

In performing the integration over the unit n-sphere by the trapezoidal rule, a displaced equidistant grid
of size h is constructed. The points of the mesh lie on concentric layers of radius

ri ¼ h

4

ffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffi
nþ 8 i� 1ð Þ

p
; i ¼ 1; 2; 3; . . . :

The routine requires you to specify an approximate maximum number of points to be used, and then
computes the largest number of whole layers to be used, subject to an upper limit of 400 layers.

In practice, the rapidly-decreasing Jacobian makes it unnecessary to include the whole unit n-sphere
and the integration region is limited by a user-specified cut-off radius r0 < 1. The grid-spacing h is
determined by r0 and the number of layers to be used. A typical value of r0 is 0:8.

Some experimentation may be required with the choice of r0 (which determines how much of the unit
n-sphere is included) and u (which determines how the transformed integrand is distributed between the
origin and surface of the unit n-sphere), to obtain best results for particular families of integrals. This
matter is discussed further in Section 9.

4 References

Krylov V I (1962) Approximate Calculation of Integrals (trans A H Stroud) Macmillan

Sag T W and Szekeres G (1964) Numerical evaluation of high-dimensional integrals Math. Comput. 18
245–253

5 Arguments

1: NDIM – INTEGER Input

On entry: n, the number of dimensions of the integral.

Constraint: 1 � NDIM � 30.

2: F – REAL (KIND=nag_wp) FUNCTION, supplied by the user. External Procedure

F must return the value of the integrand f at a given point.

The specification of F is:

FUNCTION F (NDIM, X)
REAL (KIND=nag_wp) F

INTEGER NDIM
REAL (KIND=nag_wp) X(NDIM)

1: NDIM – INTEGER Input

On entry: n, the number of dimensions of the integral.

2: XðNDIMÞ – REAL (KIND=nag_wp) array Input

On entry: the coordinates of the point at which the integrand f must be evaluated.

F must either be a module subprogram USEd by, or declared as EXTERNAL in, the (sub)
program from which D01FDF is called. Arguments denoted as Input must not be changed by this
procedure.

3: SIGMA – REAL (KIND=nag_wp) Input

On entry: indicates the region of integration.

SIGMA � 0:0
The integration is carried out over the n-sphere of radius SIGMA, centred at the origin.

SIGMA < 0:0
The integration is carried out over the product region described by REGION.

4: REGION – SUBROUTINE, supplied by the NAG Library or the user. External Procedure

If SIGMA < 0:0, REGION must evaluate the limits of integration in any dimension.

The specification of REGION is:

SUBROUTINE REGION (NDIM, X, J, C, D)

INTEGER NDIM, J
REAL (KIND=nag_wp) X(NDIM), C, D

1: NDIM – INTEGER Input

On entry: n, the number of dimensions of the integral.

2: XðNDIMÞ – REAL (KIND=nag_wp) array Input

On entry: Xð1Þ; . . . ;Xðj� 1Þ contain the current values of the first j� 1ð Þ variables,
which may be used if necessary in calculating cj and dj.

3: J – INTEGER Input

On entry: the index j for which the limits of the range of integration are required.

4: C – REAL (KIND=nag_wp) Output

On exit: the lower limit cj of the range of xj.

5: D – REAL (KIND=nag_wp) Output

On exit: the upper limit dj of the range of xj.

REGION must either be a module subprogram USEd by, or declared as EXTERNAL in, the (sub)
program from which D01FDF is called. Arguments denoted as Input must not be changed by this
procedure.

If SIGMA � 0:0, REGION is not called by D01FDF, but a dummy routine must be supplied
(D01FDV may be used).

5: LIMIT – INTEGER Input

On entry: the approximate maximum number of integrand evaluations to be used.

Constraint: LIMIT � 100.

6: R0 – REAL (KIND=nag_wp) Input

On entry: the cut-off radius on the unit n-sphere, which may be regarded as an adjustable
parameter of the method.

Suggested value: a typical value is R0 ¼ 0:8. (See also Section 9.)

Constraint: 0:0 < R0 < 1:0.

7: U – REAL (KIND=nag_wp) Input

On entry: must specify an adjustable parameter of the transformation to the unit n-sphere.

Suggested value: a typical value is U ¼ 1:5. (See also Section 9.)

Constraint: U > 0:0.

8: RESULT – REAL (KIND=nag_wp) Output

On exit: the approximation to the integral I.

9: NCALLS – INTEGER Output

On exit: the actual number of integrand evaluations used. (See also Section 9.)

10: IFAIL – INTEGER Input/Output

On entry: IFAIL must be set to 0, �1 or 1. If you are unfamiliar with this argument you should
refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is
detected, the value �1 or 1 is recommended. If the output of error messages is undesirable, then
the value 1 is recommended. Otherwise, if you are not familiar with this argument, the
recommended value is 0. When the value �1 or 1 is used it is essential to test the value of
IFAIL on exit.

On exit: IFAIL ¼ 0 unless the routine detects an error or a warning has been flagged (see
Section 6).

6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL ¼ 1

On entry, NDIM < 1,
or NDIM > 30.

IFAIL ¼ 2

On entry, LIMIT < 100.

IFAIL ¼ 3

On entry, R0 � 0:0,
or R0 � 1:0.

IFAIL ¼ 4

On entry, U � 0:0.

IFAIL ¼ �99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

No error estimate is returned, but results may be verified by repeating with an increased value of LIMIT
(provided that this causes an increase in the returned value of NCALLS).

8 Parallelism and Performance

D01FDF is not threaded in any implementation.

9 Further Comments

The time taken by D01FDF will be approximately proportional to the returned value of NCALLS,
which, except in the circumstances outlined in (b) below, will be close to the given value of LIMIT.

(a) Choice of r0 and u

If the chosen combination of r0 and u is too large in relation to the machine accuracy it is possible
that some of the points generated in the original region of integration may transform into points in
the unit n-sphere which lie too close to the boundary surface to be distinguished from it to machine
accuracy (despite the fact that r0 < 1). To be specific, the combination of r0 and u is too large if

ur0

1� r20
> 0:3465 t� 1ð Þ; if SIGMA � 0:0;

or

ur0
1� r0

> 0:3465 t� 1ð Þ; if SIGMA < 0:0;

where t is the number of bits in the mantissa of a real number.

The contribution of such points to the integral is neglected. This may be justified by appeal to the
fact that the Jacobian of the transformation rapidly approaches zero towards the surface. Neglect of
these points avoids the occurrence of overflow with integrands which are infinite on the boundary.

(b) Values of LIMIT and NCALLS

LIMIT is an approximate upper limit to the number of integrand evaluations, and may not be
chosen less than 100. There are two circumstances when the returned value of NCALLS (the actual
number of evaluations used) may be significantly less than LIMIT.

Firstly, as explained in (a), an unsuitably large combination of r0 and u may result in some of the
points being unusable. Such points are not included in the returned value of NCALLS.

Secondly, no more than 400 layers will ever be used, no matter how high LIMIT is set. This places
an effective upper limit on NCALLS as follows:

n ¼ 1 : 56
n ¼ 2 : 1252
n ¼ 3 : 23690
n ¼ 4 : 394528
n ¼ 5 : 5956906

10 Example

This example calculates the integralZ Z Z
s

dx1dx2dx3ffiffiffiffiffiffiffiffiffiffiffiffiffiffiffi
�2 � r2

p ¼ 22:2066

where s is the 3-sphere of radius �, r2 ¼ x2
1 þ x2

2 þ x2
3 and � ¼ 1:5. Both sphere-to-sphere and general

product region transformations are used. For the former, we use r0 ¼ 0:9 and u ¼ 1:5; for the latter,
r0 ¼ 0:8 and u ¼ 1:5.

10.1 Program Text

! D01FDF Example Program Text
! Mark 26 Release. NAG Copyright 2016.

Module d01fdfe_mod

! D01FDF Example Program Module:
! Parameters and User-defined Routines

! .. Use Statements ..
Use nag_library, Only: nag_wp

! .. Implicit None Statement ..
Implicit None

! .. Accessibility Statements ..
Private
Public :: f, region

! .. Parameters ..
Integer, Parameter, Public :: nout = 6

Contains
Function f(ndim,x)

! .. Function Return Value ..
Real (Kind=nag_wp) :: f

! .. Scalar Arguments ..
Integer, Intent (In) :: ndim

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (In) :: x(ndim)

! .. Intrinsic Procedures ..
Intrinsic :: abs, sqrt, sum

! .. Executable Statements ..
f = 1.0E0_nag_wp/sqrt(abs(2.25E0_nag_wp-sum(x(1:ndim)**2)))

Return

End Function f
Subroutine region(ndim,x,j,c,d)

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (Out) :: c, d
Integer, Intent (In) :: j, ndim

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (In) :: x(ndim)

! .. Intrinsic Procedures ..
Intrinsic :: abs, sqrt, sum

! .. Executable Statements ..
If (j>1) Then

d = sqrt(abs(2.25E0_nag_wp-sum(x(1:(j-1))**2)))
c = -d

Else
c = -1.5E0_nag_wp
d = 1.5E0_nag_wp

End If

Return

End Subroutine region

End Module d01fdfe_mod
Program d01fdfe

! D01FDF Example Main Program

! .. Use Statements ..
Use nag_library, Only: d01fdf, d01fdv, nag_wp
Use d01fdfe_mod, Only: f, nout, region

! .. Implicit None Statement ..
Implicit None

! .. Local Scalars ..
Real (Kind=nag_wp) :: r0, result, sigma, u
Integer :: ifail, limit, ncalls, ndim

! .. Executable Statements ..
Write (nout,*) ’D01FDF Example Program Results’

ndim = 3
limit = 8000
u = 1.5E0_nag_wp
sigma = 1.5E0_nag_wp
r0 = 0.9E0_nag_wp

ifail = 0
Call d01fdf(ndim,f,sigma,d01fdv,limit,r0,u,result,ncalls,ifail)

Write (nout,*)
Write (nout,*) ’Sphere-to-sphere transformation’
Write (nout,*)
Write (nout,99999) ’Estimated value of the integral = ’, result
Write (nout,99998) ’Number of integrand evaluations = ’, ncalls
Write (nout,*)
Write (nout,*) ’Product region transformation’

sigma = -1.0E0_nag_wp
r0 = 0.8E0_nag_wp

ifail = 0
Call d01fdf(ndim,f,sigma,region,limit,r0,u,result,ncalls,ifail)

Write (nout,*)
Write (nout,99999) ’Estimated value of the integral = ’, result
Write (nout,99998) ’Number of integrand evaluations = ’, ncalls

99999 Format (1X,A,F9.3)
99998 Format (1X,A,I5)

End Program d01fdfe

10.2 Program Data

None.

10.3 Program Results

D01FDF Example Program Results

Sphere-to-sphere transformation

Estimated value of the integral = 22.168
Number of integrand evaluations = 8026

Product region transformation

Estimated value of the integral = 22.137
Number of integrand evaluations = 8026

