
NAG Library Routine Document

D01RAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

Note: this routine uses optional parameters to define choices in the problem specification and in the
details of the algorithm. If you wish to use default settings for all of the optional parameters, you need
only read Sections 1 to 10 of this document. If, however, you wish to reset some or all of the settings
please refer to Section 11 for a detailed description of the specification of the optional parameters.

1 Purpose

D01RAF is a general purpose adaptive integrator which calculates an approximation to a vector of
definite integrals F over a finite range a; b½ �, given the vector of integrands f xð Þ.

F ¼
Z b

a

f xð Þdx

If the same subdivisions of the range are equally good for functions f1 xð Þ and f2 xð Þ, because f1 xð Þ and
f2 xð Þ have common areas of the range where they vary slowly and where they vary quickly, then we
say that f1 xð Þ and f2 xð Þ are ‘similar’. D01RAF is particularly effective for the integration of a vector of
similar functions.

2 Specification

SUBROUTINE D01RAF (IREVCM, NI, A, B, SID, NEEDI, X, LENX, NX, FM, LDFM,
DINEST, ERREST, IOPTS, OPTS, ICOM, LICOM, COM, LCOM,
IFAIL)

&
&

INTEGER IREVCM, NI, SID, NEEDI(NI), LENX, NX, LDFM,
IOPTS(100), ICOM(LICOM), LICOM, LCOM, IFAIL

&

REAL (KIND=nag_wp) A, B, X(LENX), FM(LDFM,*), DINEST(NI), ERREST(NI),
OPTS(100), COM(LCOM)

&

3 Description

D01RAF is an extension to various QUADPACK routines, including QAG, QAGS and QAGP. The
extensions made allow multiple integrands to be evaluated simultaneously, using a vectorized interface
and reverse communication.

The quadrature scheme employed by D01RAF can be chosen by you. Six Gauss–Kronrod schemes are
available. The algorithm incorporates a global acceptance criterion (as defined by Malcolm and
Simpson (1976)), optionally together with the �-algorithm (see Wynn (1956)) to perform extrapolation.
The local error estimation is described in Piessens et al. (1983).

D01RAF is the integration routine in the suite of routines D01RAF and D01RCF. It also uses optional
parameters, which can be set and queried using the routines D01ZKF and D01ZLF respectively. The
options available are described in Section 11.

First, the option arrays IOPTS and OPTS must be initialized using D01ZKF. Thereafter any required
options must be set before calling D01RAF, or the routine D01RCF.

A typical usage of this suite of routines is (in pseudo-code for clarity),

Setup phase

liopts = 100
lopts = 100
allocate(iopts(liopts),opts(lopts))
call D01ZKF(’initialize = d01raf’,iopts,liopts,opts,lopts,ifail)
call D01ZKF(’option = value’,iopts,liopts,opts,lopts,ifail)

...
call D01RCF(ni,lenxrq,ldfmrq,sdfmrq,licmin,licmax,lcmin,lcmax, &

iopts,opts,ifail)
lenx = lenxrq
ldfm = ldfmrq
sdfm = sdfmrq
licom = licmax
lcom = lcmax
allocate(icom(licom),com(lcom),x(lenx),fm(ldfm,sdfm),needi(ni), &

dinest(ni),errest(ni))

Solve phase

irevcm = 1
while irevcm 6¼0

call D01RAF(irevcm,ni,a,b,sid,needi,x,lenx,nx,fm,ldfm, &
dinest,errest,iopts,opts,icom,licom,com, &
lcom,ifail)

select case(irevcm)
case(11)

Initial solve phase
evaluate fm(1:ni,1:nx)

case(12)
Adaptive solve phase
evaluate fm(needi(1:ni)=1,1:nx)

case(0)
investigate ifail

end select
end while

Diagnostic phase

call D01ZLF(’option’,ivalue,rvalue,cvalue,optype,iopts,opts,ifail)
...

During the initial solve phase, the first estimation of the definite integral and error estimate is
constructed over the interval a; b½ �. This will have been divided into spri level 1 segments, where spri is
the number of Primary Divis ions , and wi l l use any provided break-points i f
Primary Division Mode ¼ MANUAL.

Once a complete integral estimate over a; b½ � is available, i.e., after all the estimates for the level 1
segments have been evaluated, the routine enters the adaptive phase. The estimated errors are tested
against the requested tolerances �a and �r, corresponding to the Absolute Tolerance and Relative
Tolerance respectively. Should this test fail, and additional subdivision be allowed, a segment is
selected for subdivision, and is subsequently replaced by two new segments at the next level of
refinement. How this segment is chosen may be altered by setting Prioritize Error to either favour the
segment with the maximum error, or the segment with the lowest level supporting an unacceptable
(although potentially non-maximal) error. Up to max sdiv subdivisions are allowed if sufficient memory
is provided, where max sdiv is the value of Maximum Subdivisions.

Once a sufficient number of error estimates have been constructed for a particular integral, the routine
may optionally use Extrapolation to attempt to accelerate convergence. This may significantly lower
the amount of work required for a given integration. To minimize the risk of premature convergence
from extrapolation, a safeguard �safe can be set using Extrapolation Safeguard, and the extrapolated
solution will only be considered if �safe�q � �ex , where �q and �ex are the estimated error directly from
the quadrature and from the extrapolation respectively. If extrapolation is successful for the computation
of integral j, the extrapolated solution will be returned in DINESTðjÞ on completion of D01RAF.
Otherwise the direct solution will be returned in DINESTðjÞ. This is indicated by the value of
NEEDIðjÞ on completion.

5 Arguments

Note: this routine uses reverse communication. Its use involves an initial entry, intermediate exits and
re-entries, and a final exit, as indicated by the argument IREVCM. Between intermediate exits and re-
entries, all arguments other than IREVCM, NEEDI and FM must remain unchanged.

1: IREVCM – INTEGER Input/Output

On initial entry: IREVCM ¼ 1.

IREVCM ¼ 1
Sets up data structures in ICOM and COM and starts a new integration.

Constraint: IREVCM ¼ 1 on initial entry.

On intermediate exit: IREVCM ¼ 11 or 12.

IREVCM requests the integrands fj xið Þ be evaluated for all required j 2 1; . . . ; ni as indicated by
NEEDI, and at all the points xi, for i ¼ 1; 2; . . . ; nx. Abscissae xi are provided in XðiÞ and fj xið Þ
must be returned in FMðj; iÞ.
During the initial solve phase:

IREVCM ¼ 11
Function values are required to construct the initial estimates of the definite integrals.

If NEEDIðjÞ ¼ 1, fj xið Þ must be supplied in FMðj; iÞ. This will be the case unless you have
abandoned the evaluation of specific integrals on a previous call.

If NEEDIðjÞ ¼ 0, you have previously abandoned the evaluation of integral j, and hence should
not supply the value of fj xið Þ.
DINEST and ERREST contain incomplete information during this phase. As such you should not
abandon the evaluation of any integrals during this phase unless you do not require their
estimate.

If IREVCM is set to a negative value during this phase, NEEDIðjÞ, for j ¼ 1; 2; . . . ; ni, will be
set to this negative value and IFAIL ¼ �1 will be returned.

During the adaptive solve phase:

IREVCM ¼ 12
Function values are required to improve the estimates of the definite integrals.

If NEEDIðjÞ ¼ 0, any evaluation of fj xið Þ will be discarded, so there is no need to provide them.

If NEEDIðjÞ ¼ 1, fj xið Þ must be provided in FMðj; iÞ.
If NEEDIðjÞ ¼ 2, 3 or 4, the current error estimate of integral j does not require integrand j to be
evaluated and provided in FMðj; iÞ. Should you choose to, integrand j can be evaluated in which
case NEEDIðjÞ must be set to 1.

DINEST and ERREST contain complete information during this phase.

If IREVCM is set to a negative value during this phase IFAIL ¼ 1, 2 or 3 will be returned and
the elements of NEEDI will reflect the current state of the adaptive process.

On intermediate re-entry: IREVCM should normally be left unchanged. However, if IREVCM is
set to a negative value, D01RAF will terminate, (see IREVCM ¼ 11 and IREVCM ¼ 12 above).

On final exit: IREVCM ¼ 0.

IREVCM ¼ 0
Indicates the algorithm has completed.

2: NI – INTEGER Input

On entry: ni, the number of integrands.

3: A – REAL (KIND=nag_wp) Input

On entry: a, the lower bound of the domain.

4: B – REAL (KIND=nag_wp) Input

On entry: b, the upper bound of the domain.

If b� aj j < 10�, where � is the machine precision (see X02AJF), then D01RAF will return
DINESTðjÞ ¼ ERRESTðjÞ ¼ 0:0, for j ¼ 1; 2; . . . ; ni.

5: SID – INTEGER Output

For advanced users.

On intermediate exit: SID identifies a specific set of abscissae, x, returned during the integration
process. When a new set of abscissae are generated the value of SID is incremented by 1.
Advanced users may store calculations required for an identified set x, and reuse them should
D01RAF return the same value of SID, i.e., the same set of abscissae was used.

6: NEEDIðNIÞ – INTEGER array Input/Output

On initial entry: need not be set.

On intermediate exit: NEEDIðjÞ indicates what action must be taken for integral j ¼ 1; 2; . . .ni

(see IREVCM).

NEEDIðjÞ ¼ 0
Do not provide fj xið Þ. Any provided values will be ignored.

NEEDIðjÞ ¼ 1
The values fj xið Þ must be provided in FMðj; iÞ, for i ¼ 1; 2; . . . ; nx.

NEEDIðjÞ ¼ 2
The values fj xið Þ are not required, however the error estimate for integral j is still above
the requested tolerance. If you wish to provide values for the evaluation of integral j, set
NEEDIðjÞ ¼ 1, and supply fj xið Þ in FMðj; iÞ, for i ¼ 1; 2; . . . ; nx.

NEEDIðjÞ ¼ 3
The error estimate for integral j cannot be improved to below the requested tolerance
directly, either because no more new splits may be performed due to exhaustion, or due to
the detection of extremely bad integrand behaviour. However, providing the values fj xið Þ
may still lead to some improvement, and may lead to an acceptable error estimate
indirectly using Wynn's epsilon algorithm. If you wish to provide values for the evaluation
of integral j, set NEEDIðjÞ ¼ 1, and supply fj xið Þ in FMðj; iÞ, for i ¼ 1; 2; . . . ; nx.

NEEDIðjÞ ¼ 4
The error estimate of integral j is below the requested tolerance. If you believe this to be
false, if for example the result in DINESTðjÞ is greatly different to what you may expect,
you may force the algorithm to re-evaluate this conclusion by including the evaluations of
integrand j at xi, for i ¼ 1; 2; . . . ; nx, and setting NEEDIðjÞ ¼ 1. Integral and error
estimation will be performed again during the next iteration.

On intermediate re-entry: NEEDIðjÞ may be used to indicate what action you have taken for
integral j.

NEEDIðjÞ ¼ 1
You have provided the values fj xið Þ in FMðj; iÞ, for i ¼ 1; 2; . . . ; nx.

NEEDIðjÞ < 0
You are abandoning the evaluation of integral j. The current values of DINESTðjÞ and
ERRESTðjÞ will be returned on final completion.

Otherwise you have not provided the value fj xið Þ.
On final exit: NEEDIðjÞ indicates the final state of integral j.

NEEDIðjÞ ¼ 0
The error estimate for Fj is below the requested tolerance.

NEEDIðjÞ ¼ 1
The error estimate for Fj is below the requested tolerance after extrapolation.

NEEDIðjÞ ¼ 2
The error estimate for Fj is above the requested tolerance.

NEEDIðjÞ ¼ 3
The error estimate for Fj is above the requested tolerance, and extremely bad behaviour of
integral j has been detected.

NEEDIðjÞ < 0
You prohibited further evaluation of integral j.

7: XðLENXÞ – REAL (KIND=nag_wp) array Input/Output

On initial entry: if Primary Division Mode ¼ AUTOMATIC, X need not be set. This is the
default behaviour.

If Primary Division Mode ¼ MANUAL, X is used to supply a set of initial ‘break-points’ inside
the domain of integration. Specifically, XðiÞ must contain a break-point x0

i , for
i ¼ 1; 2; . . . ; spri � 1

� �
, where spri is the number of Primary Divisions.

Constraint: if break-points are supplied, x0
i 2 a; bð Þ, x0i � a

�� �� > 10:0�, x0i � b
�� �� > 10:0�, for

i ¼ 1; 2; . . . ; spri � 1
� �

.

On intermediate exit: XðiÞ is the abscissa xi, for i ¼ 1; 2; . . . ; nx, at which the appropriate
integrals must be evaluated.

8: LENX – INTEGER Input

On entry: the dimension of the array X as declared in the (sub)program from which D01RAF is
called. Currently LENX ¼ max 122; spri � 1

� �
will be sufficient for all cases.

Constraint: LENX � lenxrq, where lenxrq is dependent upon the options currently set (see
Section 11). lenxrq is returned as LENXRQ from D01RCF.

9: NX – INTEGER Input/Output

On initial entry: need not be set.

On intermediate exit: nx, the number of abscissae at which integrands are required.

On intermediate re-entry: must not be changed.

10: FMðLDFM; �Þ – REAL (KIND=nag_wp) array Input

Note: the second dimension of the array FM must be at least sdfmrq, where sdfmrq is dependent
upon ni and the options currently set. sdfmrq is returned as SDFMRQ from D01RCF. If default
options are chosen, sdfmrq ¼ lenxrq.

On initial entry: need not be set.

On intermediate re-entry: if indicated by NEEDIðjÞ you must supply the values fj xið Þ in
FMðj; iÞ, for i ¼ 1; 2; . . . ; nx and j ¼ 1; 2; . . . ; ni.

11: LDFM – INTEGER Input

On entry: the first dimension of the array FM as declared in the (sub)program from which
D01RAF is called.

Constraint: LDFM � ldfmrq, where ldfmrq is dependent upon ni and the options currently set.
ldfmrq is returned as LDFMRQ from D01RCF. If default options are chosen, ldfmrq ¼ ni,
implying LDFM � NI.

12: DINESTðNIÞ – REAL (KIND=nag_wp) array Input/Output

DINESTðjÞ contains the current estimate of the definite integral Fj.

On initial entry: need not be set.

On intermediate re-entry: must not be altered.

On exit: contains the current estimates of the NI integrals. If IREVCM ¼ 0, this will be the final
solution.

13: ERRESTðNIÞ – REAL (KIND=nag_wp) array Input/Output

ERRESTðjÞ contains the current error estimate of the definite integral Fj.

On initial entry: need not be set.

On intermediate re-entry: must not be altered.

On exit: contains the current error estimates for the NI integrals. If IREVCM ¼ 0, ERREST
contains the final error estimates of the NI integrals.

14: IOPTSð100Þ – INTEGER array Communication Array
15: OPTSð100Þ – REAL (KIND=nag_wp) array Communication Array

The arrays IOPTS and OPTS must not be altered between calls to any of the routines D01RAF,
D01RCF, D01ZKF and D01ZLF.

16: ICOMðLICOMÞ – INTEGER array Communication Array

ICOM contains details of the integration procedure, including information on the integration of
the ni integrals over individual segments. This data is stored sequentially in the order that
segments are created. For further information see Section 9.1.

17: LICOM – INTEGER Input

On entry: the dimension of the array ICOM.

Constraint: LICOM � licmin, where licmin is dependent upon NI and the current options set.
licmin is returned as LICMIN from D01RCF. If the default options are set, then
licmin ¼ 55þ 6� NI. Larger values than licmin are recommended if you anticipate that any
integrals will require the domain to be further subdivided.

The maximum value that may be required, licmax, is returned as LICMAX from D01RCF. If
default options are chosen, except for possibly increasing the value of spri, then
licmax ¼ 50þ 5� NIþ spri þ 100

� �� 5þ NIð Þ.

D01RAF NAG Library Manual

D01RAF.6 Mark 26



18: COMðLCOMÞ – REAL (KIND=nag_wp) array Communication Array

COM contains details of the integration procedure, including information on the integration of
the ni integrals over individual segments. This data is stored sequentially in the order that
segments are created. For further information see Section 9.1.

19: LCOM – INTEGER Input

On entry: the dimension of the array COM.

Constraint: LCOM > lcmin, where lcmin is dependent upon NI, spri and the current options set.
lcmin is returned as LCMIN from D01RCF. If default options are set, then
lcmin ¼ 96þ 12� NI. Larger values are recommended if you anticipate that any integrals will
require the domain to be further subdivided.

Given the current options and arguments, the maximum value, lcmax, of LCOM that may be
required, is returned as LCMAX from D01RCF. If default options are chosen,
lcmax ¼ 94þ 9� NIþ NI=2d e þ spri þ 100

� �� 2þ NI=2d e þ 2� NIð Þ.

20: IFAIL – INTEGER Input/Output

On initial entry: IFAIL must be set to 0, �1 or 1. If you are unfamiliar with this argument you
should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is
detected, the value �1 or 1 is recommended. If the output of error messages is undesirable, then
the value 1 is recommended. Otherwise, because for this routine the values of the output
arguments may be useful even if IFAIL 6¼ 0 on exit, the recommended value is �1. When the
value �1 or 1 is used it is essential to test the value of IFAIL on exit.

On final exit: IFAIL ¼ 0 unless the routine detects an error or a warning has been flagged (see
Section 6).

6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Note: D01RAF may return useful information for one or more of the following detected errors or
warnings.

Errors or warnings detected by the routine:

IFAIL ¼ 1

At least one error estimate exceeded the requested tolerances.

IFAIL ¼ 2

Extremely bad behaviour was detected for at least one integral.

IFAIL ¼ 3

Extremely bad behaviour was detected for at least one integral. At least one other integral error
estimate was above the requested tolerance.

IFAIL ¼ 11

IREVCM had an illegal value.
On entry, IREVCM ¼ valueh i.

IFAIL ¼ 21

On entry, NI ¼ valueh i.
Constraint: NI � 1.

D01 – Quadrature D01RAF

Mark 26 D01RAF.7



IFAIL ¼ 71

On entry, Primary Division Mode ¼ MANUAL and at least one supplied break-point in X is
outside of the domain of integration.

IFAIL ¼ 81

LENX is insufficient for the chosen options.
On entry, LENX ¼ valueh i.
Constraint: LENX � valueh i.

IFAIL ¼ 111

LDFM < ldfmrq. If default options are chosen, this implies LDFM < NI.
On entry, LDFM ¼ valueh i.
Constraint: LDFM � valueh i.

IFAIL ¼ 171

LICOM is insufficient for additional subdivision.
On entry, LICOM ¼ valueh i.
Constraint: LICOM � valueh i.

IFAIL ¼ 191

LCOM is insufficient for additional subdivision.
On entry, LCOM ¼ valueh i.
Constraint: LCOM � valueh i.

IFAIL ¼ 1001

Either the option arrays IOPTS and OPTS have not been initialized for D01RAF, or they have
become corrupted.

IFAIL ¼ 1101

On entry, one of ICOM and COM has become corrupted.

IFAIL ¼ �1

Evaluation of all integrals has been stopped during the initial phase.

IFAIL ¼ �99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

D01RAF cannot guarantee, but in practice usually achieves, the following accuracy for each integral Fj:

Fj � DINESTðjÞ�� �� � tol

where

tol ¼ max �a; �r � Fj

�� ��� �
�a and �r are the error tolerances Absolute Tolerance and Relative Tolerance respectively. Moreover, it
returns ERREST, the entries of which in normal circumstances satisfy,

Fj � DINESTðjÞ�� �� � ERRESTðjÞ � tol:

8 Parallelism and Performance

D01RAF is threaded by NAG for parallel execution in multithreaded implementations of the NAG
Library.

D01RAF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor
library used by this implementation. Consult the documentation for the vendor library for further
information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the
OpenMP environment used within this routine. Please also consult the Users' Note for your
implementation for any additional implementation-specific information.

9 Further Comments

The time required by D01RAF is usually dominated by the time required to evaluate the values of the
integrands fj.

D01RAF will be most efficient if any badly behaved integrands provided have irregularities over similar
subsections of the domain. For example, evaluation of the integrals,

Z 1

0

log xð Þ
x�

1
2

x2

0
@

1
Adx

will be quite efficient, as the irregular behaviour of the first two integrands is at x ¼ 0. On the contrary,
the evaluation of the integrals,

Z 1

0

log xð Þ
log 1� xð Þ

� �
dx

will be less efficient, as the two integrands have singularities at opposite ends of the domain, which will
result in subdivisions which are only of use to one integrand. In such cases, it will be more efficient to
use two sets of calls to D01RAF.

D01RAF will flag extremely bad behaviour if a sub-interval �k with bounds a�k ; b�k½ � satisfying
b�k � a�kj j < max �a; �r � b� aj jð Þ has a local error estimate greater than the requested tolerance for at
least one integral. The values �a and �r can be set through the optional parameters Absolute Interval
Minimum and Relative Interval Minimum respectively.

9.1 Details of the Computation

This section is recommended for expert users only. It describes the contents of the arrays COM and
ICOM upon exit from D01RAF with IFAIL ¼ 0, 1, 2 or 3, and provided at least one iteration
completed, failure due to insufficient LICOM or LCOM.

The arrays ICOM and COM contain details of the integration, including various scalars, one-
dimensional arrays, and (effectively) two-dimensional arrays. The dimensions of these arrays vary

D01 – Quadrature D01RAF

Mark 26 D01RAF.9



depending on the arguments and options used and the progress of the algorithm. Here we describe some
of these details, including how and where they are stored in ICOM and COM.

Scalar quantities:

The indices in ICOM including the following scalars are available via query only options, see
Section 11.2. For example, Ildi is the integer value returned by the option Index LDI.

ldi The leading dimension of the two-dimensional integer arrays stored in ICOM detailed below.
ldi ¼ ICOMðIldiÞ.

ldr The leading dimension of the two-dimensional real arrays stored in COM detailed below.
ldr ¼ ICOMðIldrÞ.

nsdiv The number of segments that have been subdivided during the adaptive process.
nsdiv ¼ ICOMðInsdivÞ.

nseg The total number of segments formed.
nseg ¼ 2nsdiv þ spri.
nseg ¼ ICOMðInsegÞ.

dsp The reference of the first element of the array ds stored in COM.
dsp ¼ ICOMðIdspÞ.

esp The reference of the first element of the array es stored in COM.
esp ¼ ICOMðIespÞ.

evalsp The reference of the first element of the array evals stored in ICOM.
evalsp ¼ ICOMðIevalspÞ.

fcp The reference of the first element of the array fcount stored in ICOM.
fcp ¼ ICOMðIfcpÞ.

sinforp The reference of the first element of the array sinfor stored in COM.
sinforp ¼ ICOMðIsinforpÞ.

sinfoip The reference of the first element of the array sinfoi stored in ICOM.
sinfoip ¼ ICOMðIsinfoipÞ.

One-dimensional arrays:

fcountðniÞ
fcountð1Þ ¼ ICOMðfcpÞ.

fcountðjÞ contains the number of different approximations of integral j calculated, for j ¼ 1; 2; . . . ; ni.

Two-dimensional arrays:

sinfoið5; nsegÞ
sinfoið1; 1Þ ¼ ICOMðsinfoipÞ.
sinfoi contains information about the hierarchy of splitting.

sinfoið1; kÞ contains the split identifier for segment k, for k ¼ 1; 2; . . . ; nseg.

sinfoið2; kÞ contains the parent segment number of segment k (i.e., the segment was split to create
segment k), for k ¼ 1; 2; . . . ; nseg.

sinfoið3; kÞ and sinfoið4; kÞ contain the segment numbers of the two child segments formed from
segment k, if segment k has been split. If segment k has not been split, these will be negative.

sinfoið5; kÞ contains the level at which the segment exists, corresponding to na þ 1, where na is the
number of ancestor segments of segment k, for k ¼ 1; 2; . . . ; nseg. A negative level indicates that
segment k will not be split further, the level is then given by the absolute value of sinfoið5; kÞ.
sinforð2;nsegÞ

sinforð1; 1Þ ¼ COMðsinforpÞ.
sinfor contains the bounds of each segment.

sinforð1; kÞ contains the lower bound of segment k, for k ¼ 1; 2; . . . ; nseg.

sinforð2; kÞ contains the upper bound of segment k, for k ¼ 1; 2; . . . ; nseg.

evalsðni; nsegÞ
evalsð1; 1Þ ¼ ICOMðevalspÞ.

evals contains information to indicate whether an estimate of the integral j has been obtained over
segment k, and if so whether this evaluation still contributes to the direct estimate of Fj , for
j ¼ 1; 2; . . . ; ni and k ¼ 1; 2; . . . ; nseg.

evalsðj; kÞ ¼ 0 indicates that integral j has not been evaluated over segment k.

evalsðj; kÞ ¼ 1 indicates that integral j has been evaluated over segment k, and that this evaluation
contributes to the direct estimate of Fj.

evalsðj; kÞ ¼ 2 indicates that integral j has been evaluated over segment k, that this evaluation
contributes to the direct estimate of Fj, and that you have requested no further evaluation of this
integral at this segment by setting NEEDIðjÞ < 0.

evalsðj; kÞ ¼ 3 indicates that integral j has been evaluated over segment k, and this evaluation no longer
contributes to the direct estimate of Fj.

evalsðj; kÞ ¼ 4 indicates that integral j has been evaluated over segment k, that this evaluation
contributes to the direct estimate of Fj, and that this segment is too small for any further splitting to be
performed. Integral j also has a local error estimate over this segment above the requested tolerance.
Such segments cause D01RAF to return IFAIL ¼ 2 or 3, indicating extremely bad behaviour.

evalsðj; kÞ ¼ 5 indicates that integral j has been evaluated over segment k, that this evaluation
contributes to the direct estimate of Fj, and that this segment is too small for any further splitting to be
performed. The local error estimate is however below the requested tolerance.

dsðni; nsegÞ
dsð1; 1Þ ¼ COMðdspÞ.

dsðj; kÞ contains the definite integral estimate of the jth integral over the kth segment, dsj;k , provided it
has been evaluated, for j ¼ 1; 2; . . . ; ni and k ¼ 1; 2; . . . ; nseg.

esðni;nsegÞ
esð1; 1Þ ¼ COMðespÞ.

esðj; kÞ contains the definite integral error estimate of the jth integral over the kth segment, esj;k ,
provided it has been evaluated, for j ¼ 1; 2; . . . ; ni and k ¼ 1; 2; . . . ; nseg.

For each integral j, the direct approximation Dj of Fj, and its error estimate Ej, may be constructed as,

Fj � Dj ¼
P
Kj

dsj;k;

Fj �Dj

�� �� � Ej ¼
P
Kj

esj;k ;

where Kj is the set of all contributing segments, Kj ¼ k j evalsðj; kÞ ¼ 1; 2; 4 or 5; 1 � k � nsegf g.
Dj will have been returned in DINESTðjÞ, unless extrapolation was successful, as indicated by
NEEDIðjÞ.
Similarly, Ej will have been returned in ERRESTðjÞ unless extrapolation was successful, in which case
the error estimate from the extrapolation will have been returned. If for a given integral j one or more
contributing segments have unacceptable error estimates, it may be possible to improve the direct
approximation by replacing the contributions from these segments with more accurate estimates should
these be calculable by some means. Indeed for any segment �k 2 k, with lower bound a�k ¼ sinfor 1; �k

� �
and upper bound b�k ¼ sinfor 2; �k

� �
, one may alter the direct approximation Dj by the following,

dsnewj;�k � R b�k
a�k
fj xð Þ dx

Dj ¼
P
Kj

dsj;k � dsj;�k þ dsnewj;�k :

The error estimate Ej may be altered similarly.

10 Example

This example integrates

F ¼
Z �

0

x sin 2xð Þ cos 15xð Þ
x2 sin 2xð Þ cos 50xð Þ

� �
dx:

10.1 Program Text

! D01RAF Example Program Text
! Mark 26 Release. NAG Copyright 2016.

Module d01rafe_mod

! .. Use Statements ..
Use nag_library, Only: nag_wp

! .. Implicit None Statement ..
Implicit None

! .. Accessibility Statements ..
Private
Public :: display_integration_details, &

display_option
! .. Parameters ..

Integer, Parameter, Public :: nout = 6
Logical, Parameter, Public :: disp_integration_info = .True.

Contains
Subroutine display_integration_details(ni,iopts,opts,icom,licom,com, &

lcom)

! .. Use Statements ..
Use nag_library, Only: d01zlf

! .. Implicit None Statement ..
Implicit None

! .. Scalar Arguments ..
Integer, Intent (In) :: lcom, licom, ni

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (In) :: com(lcom), opts(100)
Integer, Intent (In) :: icom(licom), iopts(100)

! .. Local Scalars ..
Real (Kind=nag_wp) :: rvalue
Integer :: dsp, esp, evalsp, fcp, ifail, index, &

ldi, ldr, nsdiv, nseg, optype, &
sinfoip, sinforp

Character (16) :: cvalue
! .. Executable Statements ..

! Request communication array indices.
ifail = 0
Call d01zlf(’Index nseg’,index,rvalue,cvalue,optype,iopts,opts,ifail)
nseg = icom(index)
Call d01zlf(’Index nsdiv’,index,rvalue,cvalue,optype,iopts,opts,ifail)
nsdiv = icom(index)
Call d01zlf(’Index ldi’,index,rvalue,cvalue,optype,iopts,opts,ifail)
ldi = icom(index)
Call d01zlf(’Index ldr’,index,rvalue,cvalue,optype,iopts,opts,ifail)
ldr = icom(index)
Call d01zlf(’Index fcp’,index,rvalue,cvalue,optype,iopts,opts,ifail)
fcp = icom(index)
Call d01zlf(’Index evalsp’,index,rvalue,cvalue,optype,iopts,opts, &

ifail)
evalsp = icom(index)
Call d01zlf(’Index sinfoip’,index,rvalue,cvalue,optype,iopts,opts, &

ifail)
sinfoip = icom(index)
Call d01zlf(’Index dsp’,index,rvalue,cvalue,optype,iopts,opts,ifail)
dsp = icom(index)
Call d01zlf(’Index esp’,index,rvalue,cvalue,optype,iopts,opts,ifail)
esp = icom(index)
Call d01zlf(’Index sinforp’,index,rvalue,cvalue,optype,iopts,opts, &

ifail)

sinforp = icom(index)

Write (nout,*)
Write (nout,99999)
Write (nout,99998) ni, nseg, nsdiv
Call display_subdivision_strategy(ni,nseg,icom(fcp),icom(sinfoip), &

icom(evalsp),ldi,com(sinforp),com(dsp),com(esp),ldr)

Return
99999 Format (’ Information on integration: ’)
99998 Format (’ NI = ’,I3,’, nseg = ’,I3,’, nsdiv = ’,I3,’.’)

End Subroutine display_integration_details
Subroutine display_subdivision_strategy(ni,nseg,fcount,sinfoi,evals,ldi, &

sinfor,ds,es,ldr)

! .. Scalar Arguments ..
Integer, Intent (In) :: ldi, ldr, ni, nseg

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (In) :: ds(ldr,*), es(ldr,*), sinfor(ldr,*)
Integer, Intent (In) :: evals(ldi,*), fcount(ni), &

sinfoi(ldi,*)
! .. Local Scalars ..

Real (Kind=nag_wp) :: lbnd, ubnd
Integer :: child1, child2, j, k, level, parent, &

sid
! .. Executable Statements ..

! Display information on individual segments.
Do j = 1, ni

Write (nout,99991) j, fcount(j)
End Do
Write (nout,*)
Write (nout,99992)
Do k = 1, nseg

Write (nout,*)
sid = sinfoi(1,k)
parent = sinfoi(2,k)
child1 = sinfoi(3,k)
child2 = sinfoi(4,k)
level = sinfoi(5,k)
lbnd = sinfor(1,k)
ubnd = sinfor(2,k)
Write (nout,99999) k
Write (nout,99998) sid, parent, level
If (child1>0) Then

Write (nout,99997) child1, child2
End If
Write (nout,99996) lbnd, ubnd
Do j = 1, ni

If (evals(j,k)/=0) Then
Write (nout,99995) j, ds(j,k)
Write (nout,99994) j, es(j,k)
If (evals(j,k)==3) Then

Write (nout,99993) j
End If

End If
End Do

End Do
Return

99999 Format (’ Segment ’,I3,’.’)
99998 Format (’ Sid = ’,I3,’, Parent = ’,I3,’, Level = ’,I3,’.’)
99997 Format (’ Children = (’,I3,’,’,I3,’).’)
99996 Format (’ Bounds (’,Es11.4,’,’,Es11.4,’).’)
99995 Format (’ Integral ’,I2,’ approximation :’,1X,Es11.4,’.’)
99994 Format (’ Integral ’,I2,’ error estimate:’,1X,Es11.4,’.’)
99993 Format (’ Integral ’,I2, &

’ evaluation has been superseded by descendants.’)
99992 Format (’ Information on subdivision and evaluations over segments.’)
99991 Format (’ Integral ’,I2,’ total approximations: ’,I3,’.’)

End Subroutine display_subdivision_strategy
Subroutine display_option(optstr,optype,ivalue,rvalue,cvalue)

! Subroutine to query optype and print the appropriate option values

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: rvalue
Integer, Intent (In) :: ivalue, optype
Character (*), Intent (In) :: cvalue, optstr

! .. Executable Statements ..

Select Case (optype)
Case (1)

Write (nout,99999) optstr, ivalue
Case (2)

Write (nout,99998) optstr, rvalue
Case (3)

Write (nout,99997) optstr, cvalue
Case (4)

Write (nout,99996) optstr, ivalue, cvalue
Case (5)

Write (nout,99995) optstr, rvalue, cvalue
Case Default
End Select

Flush (nout)

Return
99999 Format (3X,A30,’ : ’,I16)
99998 Format (3X,A30,’ : ’,Es16.4)
99997 Format (3X,A30,’ : ’,12X,A16)
99996 Format (3X,A30,’ : ’,I16,3X,A16)
99995 Format (3X,A30,’ : ’,Es16.4,3X,A16)

End Subroutine display_option
End Module d01rafe_mod

Program d01rafe

! .. Use Statements ..
Use nag_library, Only: d01raf, d01rcf, d01zkf, d01zlf, nag_wp, x01aaf
Use d01rafe_mod, Only: display_integration_details, display_option, &

disp_integration_info, nout
! .. Implicit None Statement ..

Implicit None
! .. Local Scalars ..

Real (Kind=nag_wp) :: a, b, pi, rvalue
Integer :: ifail, irevcm, ivalue, j, lcmax, &

lcmin, lcom, ldfm, ldfmrq, lenx, &
lenxrq, licmax, licmin, licom, ni, &
nx, optype, sdfm, sdfmrq, sid

Character (16) :: cvalue
! .. Local Arrays ..

Real (Kind=nag_wp), Allocatable :: com(:), dinest(:), errest(:), &
fm(:,:), opts(:), x(:)

Integer, Allocatable :: icom(:), iopts(:), needi(:)
! .. Intrinsic Procedures ..

Intrinsic :: cos, sin
! .. Executable Statements ..

Continue
Write (nout,*) ’D01RAF Example Program Results’
Write (nout,*)

pi = x01aaf(pi)

! Setup phase.

! Set problem parameters
ni = 2

! Lower (a) and upper (b) bounds
a = 0.0E0_nag_wp
b = pi
Allocate (opts(100),iopts(100))

! Initialize option arrays

ifail = 0
Call d01zkf(’Initialize = d01raf’,iopts,100,opts,100,ifail)

! Set any non-default options required
Call d01zkf(’Quadrature Rule = gk41’,iopts,100,opts,100,ifail)
Call d01zkf(’Absolute Tolerance = 1.0e-7’,iopts,100,opts,100,ifail)
Call d01zkf(’Relative Tolerance = 1.0e-7’,iopts,100,opts,100,ifail)

! Determine maximum required array lengths
ifail = -1
Call d01rcf(ni,lenxrq,ldfmrq,sdfmrq,licmin,licmax,lcmin,lcmax,iopts, &

opts,ifail)
ldfm = ldfmrq
sdfm = sdfmrq
lenx = lenxrq
licom = licmax
lcom = lcmax

! Allocate remaining arrays
Allocate (icom(licom),needi(ni),com(lcom),fm(ldfm,sdfm),dinest(ni), &

errest(ni),x(lenx))

! Solve phase.
! Use D01RAF to evaluate the definite integrals of:
! f_1 = (x*sin(2*x))*cos(15*x)
! f_2 = (x*sin(2*x))*(x*cos(50*x))

! Set initial irevcm
irevcm = 1
ifail = -1

Do While (irevcm/=0)

Call d01raf(irevcm,ni,a,b,sid,needi,x,lenx,nx,fm,ldfm,dinest,errest, &
iopts,opts,icom,licom,com,lcom,ifail)

Select Case (irevcm)
Case (11)

! Initial returns.
! These will occur during the non-adaptive phase.
! All values must be supplied.
! DINEST and ERREST do not contain approximations
! over the complete interval at this stage.

! Calculate x*sin(2*x), storing the result in fm(2,1:nx) for re-use.
fm(2,1:nx) = x(1:nx)*sin(2.0E0_nag_wp*x(1:nx))

! Calculate f1
fm(1,1:nx) = fm(2,1:nx)*cos(15.0E0_nag_wp*x(1:nx))

! Complete f2 calculation.
fm(2,1:nx) = fm(2,1:nx)*x(1:nx)*cos(50.0E0_nag_wp*x(1:nx))

Case (12)
! Intermediate returns.
! These will occur during the adaptive phase.
! All requested values must be supplied.
! DINEST and ERREST do not contain approximations
! over the complete interval at this stage.

! Calculate x*sin(2*x).
fm(2,1:nx) = x(1:nx)*sin(2.0E0_nag_wp*x(1:nx))

! Calculate f1 if required.
If (needi(1)==1) Then

fm(1,1:nx) = fm(2,1:nx)*cos(15.0E0_nag_wp*x(1:nx))
End If

! Complete f2 calculation if required.
If (needi(2)==1) Then

fm(2,1:nx) = fm(2,1:nx)*x(1:nx)*cos(50.0E0_nag_wp*x(1:nx))
End If

Case (0)
! Final return. Test IFAIL.

Select Case (ifail)
Case (0:3)

! Useful information has been returned.
Case Default

! An unrecoverable error has been detected.
Go To 100

End Select
End Select

End Do

! Query some currently set options and statistics.
ifail = 0
Call d01zlf(’Quadrature rule’,ivalue,rvalue,cvalue,optype,iopts,opts, &

ifail)
Call display_option(’Quadrature rule’,optype,ivalue,rvalue,cvalue)

Call d01zlf(’Maximum Subdivisions’,ivalue,rvalue,cvalue,optype,iopts, &
opts,ifail)

Call display_option(’Maximum Subdivisions’,optype,ivalue,rvalue,cvalue)

Call d01zlf(’Extrapolation’,ivalue,rvalue,cvalue,optype,iopts,opts, &
ifail)

Call display_option(’Extrapolation’,optype,ivalue,rvalue,cvalue)

Call d01zlf(’Extrapolation Safeguard’,ivalue,rvalue,cvalue,optype,iopts, &
opts,ifail)

Call display_option(’Extrapolation safeguard’,optype,ivalue,rvalue, &
cvalue)

! Print solution
Write (nout,*)
Write (nout,99999)
Do j = 1, ni

Write (nout,99998) j, needi(j), dinest(j), errest(j)
End Do

! Investigate integration strategy
If (disp_integration_info) Then

Call display_integration_details(ni,iopts,opts,icom,licom,com,lcom)
End If

100 Continue

99999 Format (’ Integral | NEEDI | DINEST | ERREST ’)
99998 Format (2(1X,I9),2(1X,Es12.4))

End Program d01rafe

10.2 Program Data

None.

10.3 Program Results

D01RAF Example Program Results

Quadrature rule : GK41
Maximum Subdivisions : 50

Extrapolation : ON
Extrapolation safeguard : 1.0000E-12

Integral | NEEDI | DINEST | ERREST
1 0 -2.8431E-02 1.1234E-14
2 0 7.9083E-03 2.6600E-09

Information on integration:
NI = 2, nseg = 7, nsdiv = 3.
Integral 1 total approximations: 2.

Integral 2 total approximations: 4.

Information on subdivision and evaluations over segments.

Segment 1.
Sid = 1, Parent = 0, Level = 1.
Children = ( 2, 3).
Bounds ( 0.0000E+00, 3.1416E+00).
Integral 1 approximation : -2.8431E-02.
Integral 1 error estimate: 8.0372E-04.
Integral 1 evaluation has been superseded by descendants.
Integral 2 approximation : -3.6050E-01.
Integral 2 error estimate: 4.2596E+00.
Integral 2 evaluation has been superseded by descendants.

Segment 2.
Sid = 2, Parent = 1, Level = 2.
Children = ( 6, 7).
Bounds ( 0.0000E+00, 1.5708E+00).
Integral 1 approximation : -1.2285E-03.
Integral 1 error estimate: 2.8161E-15.
Integral 2 approximation : 1.9771E-03.
Integral 2 error estimate: 4.0437E-01.
Integral 2 evaluation has been superseded by descendants.

Segment 3.
Sid = 2, Parent = 1, Level = 2.
Children = ( 4, 5).
Bounds ( 1.5708E+00, 3.1416E+00).
Integral 1 approximation : -2.7202E-02.
Integral 1 error estimate: 8.4182E-15.
Integral 2 approximation : 5.9313E-03.
Integral 2 error estimate: 3.0259E+00.
Integral 2 evaluation has been superseded by descendants.

Segment 4.
Sid = 3, Parent = 3, Level = 3.
Bounds ( 1.5708E+00, 2.3562E+00).
Integral 2 approximation : 1.0922E-01.
Integral 2 error estimate: 7.9151E-10.

Segment 5.
Sid = 3, Parent = 3, Level = 3.
Bounds ( 2.3562E+00, 3.1416E+00).
Integral 2 approximation : -1.0329E-01.
Integral 2 error estimate: 1.6413E-09.

Segment 6.
Sid = 4, Parent = 2, Level = 3.
Bounds ( 0.0000E+00, 7.8540E-01).
Integral 2 approximation : 1.2343E-02.
Integral 2 error estimate: 5.2456E-11.

Segment 7.
Sid = 4, Parent = 2, Level = 3.
Bounds ( 7.8540E-01, 1.5708E+00).
Integral 2 approximation : -1.0365E-02.
Integral 2 error estimate: 1.7467E-10.

11 Optional Parameters

This section can be skipped if you wish to use the default values for all optional parameters, otherwise,
the following is a list of the optional parameters available. A full description of each optional parameter
is provided in Section 11.1.

Absolute Interval Minimum

Absolute Tolerance

Extrapolation

Extrapolation Safeguard

Maximum Subdivisions

Primary Division Mode

Primary Divisions

Prioritize Error

Quadrature Rule

Relative Interval Minimum

Relative Tolerance

The following optional parameters, see Section 11.2, may be utilized by expert users in conjunction
with the information provided in Section 9.1.

Index LDI

Index LDR

Index NSDIV

Index NSEG

Index FCP

Index EVALSP

Index DSP

Index ESP

Index SINFOIP

Index SINFORP

11.1 Description of the Optional Parameters

For each option, we give a summary line, a description of the optional parameter and details of
constraints.

The summary line contains:

the keywords, where the minimum abbreviation of each keyword is underlined;

a parameter value, where the letters a, i and r denote options that take character, integer and real
values respectively;

the default value.

The following symbol represents various machine constants:

� represents the machine precision (see X02AJF).

All options accept the value ‘DEFAULT’ in order to return single options to their default states.

Keywords and character values are case insensitive, however they must be separated by at least one
space.

Unsetable options will return the appropriate value when calling D01ZLF. They will have no effect if
passed to D01ZKF.

For D01RAF the maximum length of the argument CVALUE used by D01ZLF is 15.

Absolute Interval Minimum r Default ¼ 128:0�

r ¼ �a, the absolute lower limit for a segment to be considered for subdivision. See also Relative
Interval Minimum and Section 9.

Constraint: r � 128�.

Absolute Tolerance r Default ¼ 1024�

r ¼ �a, the absolute tolerance required. See also Relative Tolerance and Section 3.

Constraint: r � 0:0.

Extrapolation a Default ¼ ON

Activate or deactivate the use of the � algorithm (Wynn (1956)). Extrapolation often reduces the
number of iterations required to achieve the desired solution, but it can occasionally lead to premature
convergence towards an incorrect answer.

ON
Use extrapolation.

OFF
Disable extrapolation.

Extrapolation Safeguard r Default ¼ 1:0E�12

r ¼ �safe. If �q is the estimated error from the quadrature evaluation alone, and �ex is the error estimate
determined using extrapolation, then the extrapolated solution will only be accepted if �safe�q � �ex .

Maximum Subdivisions i Default ¼ 50

i ¼ max sdiv, the maximum number of subdivisions the algorithm may use in the adaptive phase,
forming at most an additional 2�max sdivð Þ segments.

Primary Divisions i Default ¼ 1

i ¼ spri, the number of initial segments of the domain a; b½ �. By default the initial segment is the entire
domain.

Constraint: 0 < i < 1000000.

Primary Division Mode a Default ¼ AUTOMATIC

Determines how the initial set of spri segments will be generated.

AUTOMATIC
D01RAF will automatically generate spri segments of equal size covering the interval a; b½ �.

MANUAL
D01RAF will use the break-points x0

i , for i ¼ 1; 2; . . . ; spri � 1, supplied in X on initial entry to
generate the initial segments covering a; b½ �. These may be supplied in any order, however it will
be more efficient to supply them in ascending (or descending if a > b) order. Repeated break-
points are allowed, although this will generate fewer initial segments.

Note: an absolute bound on the size of an initial segment of 10:0� is automatically applied in all cases,
and will result in fewer initial subdivisions being generated if automatically generated or supplied
break-points result in segments smaller than this.

Prioritize Error a Default ¼ LEVEL

Indicates how new subdivisions of segments sustaining unacceptable local errors for integrals should be
prioritized.

LEVEL
Segments with lower level with unsatisfactory error estimates will be chosen over segments with
greater error on higher levels. This will probably lead to more integrals being improved in earlier
iterations of the algorithm, and hence will probably lead to fewer repeated returns (see argument
SID), and to more integrals being satisfactorily estimated if computational exhaustion occurs.

MAXERR
The segment with the worst overall error will be split, regardless of level. This will more rapidly
improve the worst integral estimates, although it will probably result in the fewest integrals being
improved in earlier iterations, and may hence lead to more repeated returns (see argument SID),
and potentially fewer integrals satisfying the requested tolerances if computational exhaustion
occurs.

Quadrature Rule a Default ¼ GK15

The basic quadrature rule to be used during the integration. Currently 6 Gauss–Kronrod rules are
available, all identifiable by the letters GK followed by the number of points required by the Kronrod
rule. Higher order rules generally provide higher accuracy with fewer subdivisons. However, for
integrands with sharp singularities, lower order rules may be more efficient, particularly if the integrand
away from the singularity is well behaved. With higher order rules, you may need to increase the
Absolute Interval Minimum and the Relative Interval Minimum to maintain numerical difference
between the abscissae and the segment bounds.

GK15
The Gauss–Kronrod rule based on 7 Gauss points and 15 Kronrod points.

GK21
The Gauss–Kronrod rule based on 10 Gauss points and 21 Kronrod points. This is the rule used
by D01ATF

GK31
The Gauss–Kronrod rule based on 15 Gauss points and 31 Kronrod points.

GK41
The Gauss–Kronrod rule based on 20 Gauss points and 41 Kronrod points.

GK51
The Gauss–Kronrod rule based on 25 Gauss points and 51 Kronrod points.

GK61
The Gauss–Kronrod rule based on 30 Gauss points and 61 Kronrod points. This is the highest
order rule, most suitable for highly oscilliatory integrals.

Relative Interval Minimum r Default ¼ 1:0E�6

r ¼ �r, the relative factor in the lower limit, �r b� aj j, for a segment to be considered for subdivision.
See also Absolute Interval Minimum and Section 9.

Constraint: r � 0:0.

Relative Tolerance r Default ¼ ffiffi
�

p

r ¼ �r, the required relative tolerance. See also Absolute Tolerance and Section 3.

Constraint: r � 0:0.

Note: setting both �r ¼ �a ¼ 0:0 is possible, although it will most likely result in an excessive amount
of computational effort.

11.2 Diagnostic Options

These options are provided for expert users who wish to examine and modify the precise details of the
computation. They should only be used after D01RAF returns, as opposed to the options listed in
Section 11.1 which must be used before the first call to D01RAF.

Index LDI i query only

Ildi, the index of ICOM required for obtaining ldi. See Section 9.1.

Index LDR i query only

Ildr , the index of ICOM required for obtaining ldr . See Section 9.1.

Index NSDIV i query only

Insdiv , the index of ICOM required for obtaining nsdiv. See Section 9.1.

Index NSEG i query only

Inseg, the index of ICOM required for obtaining nseg. See Section 9.1.

Index FCP i query only

Ifcp, the index of ICOM required for obtaining fcp. See Section 9.1.

Index EVALSP i query only

Ievalsp, the index of ICOM required for obtaining evalsp. See Section 9.1.

Index DSP i query only

Idsp, the index of ICOM required for obtaining dsp. See Section 9.1.

Index ESP i query only

Iesp, the index of ICOM required for obtaining esp. See Section 9.1.

Index SINFOIP i query only

Isinfoip, the index of ICOM required for obtaining sinfoip. See Section 9.1.

Index SINFORP i query only

Isinforp, the index of ICOM required for obtaining sinforp. See Section 9.1.

