
NAG Library Routine Document

D02NEF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

D02NEF is a routine for integrating stiff systems of implicit ordinary differential equations coupled
with algebraic equations.

2 Specification

SUBROUTINE D02NEF (NEQ, T, TOUT, Y, YDOT, RTOL, ATOL, ITASK, RES, JAC,
ICOM, COM, LCOM, IUSER, RUSER, IFAIL)

&

INTEGER NEQ, ITASK, ICOM(50+NEQ), LCOM, IUSER(*), IFAIL
REAL (KIND=nag_wp) T, TOUT, Y(NEQ), YDOT(NEQ), RTOL(*), ATOL(*),

COM(LCOM), RUSER(*)
&

EXTERNAL RES, JAC

3 Description

D02NEF is a general purpose routine for integrating the initial value problem for a stiff system of
implicit ordinary differential equations with coupled algebraic equations written in the form

F t; y; y0ð Þ ¼ 0:

D02NEF uses the DASSL implementation of the Backward Differentiation Formulae (BDF) of orders
one to five to solve a system of the above form for y (Y) and y0 (YDOT). Values for Y and YDOT at the
initial time must be given as input. These values must be consistent, (i.e., if T, Y, YDOT are the given
initial values, they must satisfy F T;Y;YDOTð Þ ¼ 0). The routine solves the system from t ¼ T to
t ¼ TOUT.

An outline of a typical calling program for D02NEF is given below. It calls the DASSL implementation
of the BDF integrator setup routine D02MWF and the banded matrix setup routine D02NPF (if
required), and, if the integration needs to proceed, calls D02MCF before continuing the integration.

! Declarations

EXTERNAL RES, JAC
.
.
.

! Initialize the integrator
CALL D02MWF(...)

! Is the Jacobian matrix banded?
IF (BANDED) CALL D02NPF(...)

! Set DT to the required temporal resolution
! Set TEND to the final time
! Call the integrator for each temporal value:
1000 CALL D02NEF(...,RES,JAC,...)
! Continue integration?

IF (TOUT.LT.TEND .AND. ITASK.GE.0) THEN
IF (ITASK.NE.1) TOUT = MIN(TOUT+DT,TEND)

! Print solution
CALL D02MCF(...)
GO TO 1000

ENDIF
.
.
.

4 References

None.

5 Arguments

1: NEQ – INTEGER Input

On entry: the number of differential-algebraic equations to be solved.

Constraint: NEQ � 1.

2: T – REAL (KIND=nag_wp) Input/Output

On initial entry: the initial value of the independent variable, t.

On intermediate exit: t, the current value of the independent variable.

On final exit: the value of the independent variable at which the computed solution y is returned
(usually at TOUT).

3: TOUT – REAL (KIND=nag_wp) Input

On entry: the next value of t at which a computed solution is desired.

On initial entry: TOUT is used to determine the direction of integration. Integration is permitted
in either direction (see also ITASK).

Constraint: TOUT 6¼ T.

4: YðNEQÞ – REAL (KIND=nag_wp) array Input/Output

On initial entry: the vector of initial values of the dependent variables y.

On intermediate exit: the computed solution vector, y, evaluated at t ¼ T .

On final exit: the computed solution vector, evaluated at t (usually t ¼ TOUT).

5: YDOTðNEQÞ – REAL (KIND=nag_wp) array Input/Output

On initial entry: YDOT must contain approximations to the time derivatives y0 of the vector y
evaluated at the initial value of the independent variable.

On exit: the time derivatives y0 of the vector y at the last integration point.

6: RTOLð�Þ – REAL (KIND=nag_wp) array Input/Output

Note: the dimension of the array RTOL depends on the value of ITOL as set in D02MWF; it
must be at least NEQ if ITOL ¼ :TRUE: and at least 1 if ITOL ¼ :FALSE:.

On entry: the relative local error tolerance.

Constraint: RTOLðiÞ � 0:0, for i ¼ 1; 2; . . . ; n

where n ¼ NEQ when ITOL ¼ :TRUE: and n ¼ 1 otherwise.

On exit: RTOL remains unchanged unless D02NEF exits with IFAIL ¼ 16 in which case the
values may have been increased to values estimated to be appropriate for continuing the
integration.

7: ATOLð�Þ – REAL (KIND=nag_wp) array Input/Output

Note: the dimension of the array ATOL depends on the value of ITOL as set in D02MWF; it
must be at least NEQ if ITOL ¼ :TRUE: and at least 1 if ITOL ¼ :FALSE:.

On entry: the absolute local error tolerance.

Constraint: ATOLðiÞ � 0:0, for i ¼ 1; 2; . . . ; n

where n ¼ NEQ when ITOL ¼ :TRUE: and n ¼ 1 otherwise.

On exit: ATOL remains unchanged unless D02NEF exits with IFAIL ¼ 16 in which case the
values may have been increased to values estimated to be appropriate for continuing the
integration.

8: ITASK – INTEGER Input/Output

On initial entry: need not be set.

On exit: the task performed by the integrator on successful completion or an indicator that a
problem occurred during integration.

ITASK ¼ 2
The integration to TOUT was successfully completed (T ¼ TOUT) by stepping exactly to
TOUT.

ITASK ¼ 3
The integration to TOUT was successfully completed (T ¼ TOUT) by stepping past TOUT.
Y and YDOT are obtained by interpolation.

ITASK < 0
Different negative values of ITASK returned correspond to different failure exits. IFAIL
should always be checked in such cases and the corrective action taken where appropriate.

ITASK must remain unchanged between calls to D02NEF.

9: RES – SUBROUTINE, supplied by the user. External Procedure

RES must evaluate the residual

R ¼ F t; y; y0ð Þ:

The specification of RES is:

SUBROUTINE RES (NEQ, T, Y, YDOT, R, IRES, IUSER, RUSER)

INTEGER NEQ, IRES, IUSER(*)
REAL (KIND=nag_wp) T, Y(NEQ), YDOT(NEQ), R(NEQ), RUSER(*)

1: NEQ – INTEGER Input

On entry: the number of differential-algebraic equations being solved.

2: T – REAL (KIND=nag_wp) Input

On entry: t, the current value of the independent variable.

3: YðNEQÞ – REAL (KIND=nag_wp) array Input

On entry: yi, for i ¼ 1; 2; . . . ;NEQ, the current solution component.

4: YDOTðNEQÞ – REAL (KIND=nag_wp) array Input

On entry: the derivative of the solution at the current point t.

5: RðNEQÞ – REAL (KIND=nag_wp) array Output

On exit: RðiÞ must contain the ith component of R, for i ¼ 1; 2; . . . ;NEQ where

R ¼ F T;Y;YDOTð Þ:

6: IRES – INTEGER Input/Output

On entry: is always equal to zero.

On exit: IRES should normally be left unchanged. However, if an illegal value of Y is
encountered, IRES should be set to �1; D02NEF will then attempt to resolve the

problem so that illegal values of Y are not encountered. IRES should be set to �2 if you
wish to return control to the calling (sub)routine; this will cause D02NEF to exit with
IFAIL ¼ 23.

7: IUSERð�Þ – INTEGER array User Workspace
8: RUSERð�Þ – REAL (KIND=nag_wp) array User Workspace

RES is called with the arguments IUSER and RUSER as supplied to D02NEF. You
should use the arrays IUSER and RUSER to supply information to RES.

RES must either be a module subprogram USEd by, or declared as EXTERNAL in, the (sub)
program from which D02NEF is called. Arguments denoted as Input must not be changed by this
procedure.

10: JAC – SUBROUTINE, supplied by the NAG Library or the user. External Procedure

Evaluates the matrix of partial derivatives, J , where

Jij ¼ @Fi

@yj
þ CJ� @Fi

@y0j
; i; j ¼ 1; 2; . . . ;NEQ:

If this option is not required, the actual argument for JAC must be the dummy routine D02NEZ.
(D02NEZ is included in the NAG Library.) You must indicate to the integrator whether this
option is to be used by setting the argument JCEVAL appropriately in a call to the setup routine
D02MWF.

The specification of JAC is:

SUBROUTINE JAC (NEQ, T, Y, YDOT, PD, CJ, IUSER, RUSER)

INTEGER NEQ, IUSER(*)
REAL (KIND=nag_wp) T, Y(NEQ), YDOT(NEQ), PD(*), CJ, RUSER(*)

1: NEQ – INTEGER Input

On entry: the number of differential-algebraic equations being solved.

2: T – REAL (KIND=nag_wp) Input

On entry: t, the current value of the independent variable.

3: YðNEQÞ – REAL (KIND=nag_wp) array Input

On entry: yi, for i ¼ 1; 2; . . . ;NEQ, the current solution component.

4: YDOTðNEQÞ – REAL (KIND=nag_wp) array Input

On entry: the derivative of the solution at the current point t.

5: PDð�Þ – REAL (KIND=nag_wp) array Input/Output

On entry: PD is preset to zero before the call to JAC.

On exit: if the Jacobian is full then PDð j � 1ð Þ � NEQþ iÞ ¼ Jij , for i ¼ 1; 2; . . . ;NEQ
a n d j ¼ 1; 2; . . . ;NEQ; i f t h e J a c o b i a n i s b a n d e d t h e n
PDð j� 1ð Þ � 2MLþMUþ 1ð Þ þMLþMUþ i� jþ 1Þ ¼ Jij, f o r
max 1; j�MUð Þ � i � min n; jþMLð Þ; (see also in F07BDF (DGBTRF)).

6: CJ – REAL (KIND=nag_wp) Input

On entry: CJ is a scalar constant which will be defined in D02NEF.

7: IUSERð�Þ – INTEGER array User Workspace
8: RUSERð�Þ – REAL (KIND=nag_wp) array User Workspace

JAC is called with the arguments IUSER and RUSER as supplied to D02NEF. You
should use the arrays IUSER and RUSER to supply information to JAC.

JAC must either be a module subprogram USEd by, or declared as EXTERNAL in, the (sub)
program from which D02NEF is called. Arguments denoted as Input must not be changed by this
procedure.

11: ICOMð50þ NEQÞ – INTEGER array Communication Array

ICOM contains information which is usually of no interest, but is necessary for subsequent calls.
However you may find the following useful:

ICOMð22Þ
The order of the method to be attempted on the next step.

ICOMð23Þ
The order of the method used on the last step.

ICOMð26Þ
The number of steps taken so far.

ICOMð27Þ
The number of calls to RES so far.

ICOMð28Þ
The number of evaluations of the matrix of partial derivatives needed so far.

ICOMð29Þ
The total number of error test failures so far.

ICOMð30Þ
The total number of convergence test failures so far.

12: COMðLCOMÞ – REAL (KIND=nag_wp) array Communication Array

COM contains information which is usually of no interest, but is necessary for subsequent calls.
However you may find the following useful:

COMð3Þ
The step size to be attempted on the next step.

COMð4Þ
The current value of the independent variable, i.e., the farthest point integration has
reached. This will be different from T only when interpolation has been performed
(ITASK ¼ 3).

13: LCOM – INTEGER Input

On entry: the dimension of the array COM as declared in the (sub)program from which D02NEF
is called.

Constraint: LCOM � 40þ maxorder þ 4ð Þ � NEQþ NEQ� pþ q where maxorder is the
maximum order that can be used by the integration method (see MAXORD in D02MWF);
p ¼ NEQ when the Jacobian is full and p ¼ 2�MLþMUþ 1ð Þ when the Jacobian is banded;
and, q ¼ NEQ= MLþMUþ 1ð Þð Þ þ 1 when the Jacobian is to be evaluated numerically and
q ¼ 0 otherwise.

14: IUSERð�Þ – INTEGER array User Workspace
15: RUSERð�Þ – REAL (KIND=nag_wp) array User Workspace

IUSER and RUSER are not used by D02NEF, but are passed directly to RES and JAC and
should be used to pass information to these routines.

16: IFAIL – INTEGER Input/Output

On entry: IFAIL must be set to 0, �1 or 1. If you are unfamiliar with this argument you should
refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is
detected, the value �1 or 1 is recommended. If the output of error messages is undesirable, then
the value 1 is recommended. Otherwise, because for this routine the values of the output
arguments may be useful even if IFAIL 6¼ 0 on exit, the recommended value is �1. When the
value �1 or 1 is used it is essential to test the value of IFAIL on exit.

On exit: IFAIL ¼ 0 unless the routine detects an error or a warning has been flagged (see
Section 6).

6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Note: D02NEF may return useful information for one or more of the following detected errors or
warnings.

Errors or warnings detected by the routine:

IFAIL ¼ 1

On entry, NEQ ¼ valueh i.
Constraint: NEQ � 1.

IFAIL ¼ 3

On entry, T ¼ valueh i.
Constraint: TOUT 6¼ T.

TOUT is behind T in the direction of h: TOUT� T ¼ valueh i, h ¼ valueh i.
TOUT is too close to T to start integration: TOUT� T ¼ valueh i: hmin ¼ valueh i.

IFAIL ¼ 6

Some element of RTOL is less than zero.

IFAIL ¼ 7

Some element of ATOL is less than zero.

IFAIL ¼ 8

A previous call to this routine returned with ITASK ¼ valueh i and no appropriate action was
taken.

IFAIL ¼ 11

Either the initialization routine has not been called prior to the first call of this routine or a
communication array has become corrupted.

IFAIL ¼ 12

Either the initialization routine has not been called prior to the first call of this routine or a
communication array has become corrupted.

IFAIL ¼ 13

COM array is of insufficient length; length required ¼ valueh i; actual length LCOM ¼ valueh i.

IFAIL ¼ 14

All elements of RTOL and ATOL are zero.

IFAIL ¼ 15

Maximum number of steps taken on this call before reaching TOUT: T ¼ valueh i, maximum
number of steps ¼ valueh i.

IFAIL ¼ 16

Too much accuracy requested for precision of machine. RTOL and ATOL were increased by
scale factor R. Try running again with these scaled tolerances. T ¼ valueh i, R ¼ valueh i.

IFAIL ¼ 17

A solution component has become zero when a purely relative tolerance (zero absolute tolerance)
was selected for that component. T ¼ valueh i, YðI Þ ¼ valueh i for component I ¼ valueh i.

IFAIL ¼ 18

The error test failed repeatedly with hj j ¼ hmin. T ¼ valueh i. Stepsize h ¼ valueh i.

IFAIL ¼ 19

The corrector repeatedly failed to converge with hj j ¼ hmin. T ¼ valueh i. Stepsize h ¼ valueh i.

IFAIL ¼ 20

The iteration matrix is singular. T ¼ valueh i. Stepsize h ¼ valueh i.

IFAIL ¼ 21

The corrector could not converge and the error test failed repeatedly. T ¼ valueh i. Stepsize
h ¼ valueh i.

IFAIL ¼ 22

IRES was set to �1 during a call to RES and could not be resolved. T ¼ valueh i. Stepsize
h ¼ valueh i.

IFAIL ¼ 23

IRES was set to �2 during a call to RES. T ¼ valueh i. Stepsize ¼ valueh i.

IFAIL ¼ 24

The initial YDOT could not be computed. T ¼ valueh i. Stepsize h ¼ valueh i.

IFAIL ¼ 25

Repeated occurrences of input constraint violations have been detected. This could result in a
potential infinite loop: ITASK ¼ valueh i. Current violation corresponds to exit with
IFAIL ¼ valueh i.

IFAIL ¼ �99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The accuracy of the numerical solution may be controlled by a careful choice of the arguments RTOL
and ATOL. You are advised to use scalar error control unless the components of the solution are
expected to be poorly scaled. For the type of decaying solution typical of many stiff problems, relative
error control with a small absolute error threshold will be most appropriate (that is, you are advised to
choose ITOL ¼ 0 with ATOLð1Þ small but positive).

8 Parallelism and Performance

D02NEF is threaded by NAG for parallel execution in multithreaded implementations of the NAG
Library.

D02NEF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor
library used by this implementation. Consult the documentation for the vendor library for further
information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the
OpenMP environment used within this routine. Please also consult the Users' Note for your
implementation for any additional implementation-specific information.

9 Further Comments

The cost of computing a solution depends critically on the size of the differential system and to a lesser
extent on the degree of stiffness of the problem. For banded systems the cost is proportional to
NEQ� MLþMUþ 1ð Þ2, while for full systems the cost is proportional to NEQ3. Note however that
for moderately sized problems which are only mildly nonlinear the cost may be dominated by factors
proportional to NEQ� MLþMUþ 1ð Þ and NEQ2 respectively.

10 Example

For this routine two examples are presented. There is a single example program for D02NEF, with a
main program and the code to solve the two example problems given in Example 1 (EX1) and Example
2 (EX2).

Example 1 (EX1)

This example solves the well-known stiff Robertson problem written in implicit form

r1 ¼ �0:04a þ 1:0E4bc � a0
r2 ¼ 0:04a � 1:0E4bc � 3:0E7b2 � b0
r3 ¼ 3:0E7b2 � c0

with initial conditions a ¼ 1:0 and b ¼ c ¼ 0:0 over the range 0; 0:1½ � the BDF method (setup routine
D02MWF and D02NPF).

Example 2 (EX2)

This example illustrates the use of D02NEF to solve a simple algebraic problem by continuation. The
equation 4� 2yþ 0:1eyt ¼ 0 from t ¼ 0 (where y ¼ 2) to t ¼ 1.

10.1 Program Text

! D02NEF Example Program Text
! Mark 26 Release. NAG Copyright 2016.

Module d02nefe_mod

! D02NEF Example Program Module:
! Parameters and User-defined Routines

! .. Use Statements ..
Use nag_library, Only: nag_wp

! .. Implicit None Statement ..
Implicit None

! .. Accessibility Statements ..
Private
Public :: jac1, jac2, res1, res2

! .. Parameters ..
Real (Kind=nag_wp), Parameter :: alpha = 0.04_nag_wp
Real (Kind=nag_wp), Parameter :: beta = 1.0E4_nag_wp
Real (Kind=nag_wp), Parameter :: gamma = 3.0E7_nag_wp
Real (Kind=nag_wp), Parameter :: two = 2.0_nag_wp
Integer, Parameter, Public :: ml = 1, mu = 2, neq1 = 3, neq2 = 1, &

nin = 5, nout = 6
Contains

Subroutine myjac1(neq,ml,mu,t,y,ydot,pd,cj)

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: cj, t
Integer, Intent (In) :: ml, mu, neq

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Out) :: pd(2*ml+mu+1,neq)
Real (Kind=nag_wp), Intent (In) :: y(neq), ydot(neq)

! .. Local Scalars ..
Integer :: md, ms

! .. Executable Statements ..
! Main diagonal pdfull(i,i), i=1,neq

md = mu + ml + 1
pd(md,1) = -alpha - cj
pd(md,2) = -beta*y(3) - two*gamma*y(2) - cj
pd(md,3) = -cj

! 1 Subdiagonal pdfull(i+1:i), i=1,neq-1
ms = md + 1
pd(ms,1) = alpha
pd(ms,2) = two*gamma*y(2)

! First superdiagonal pdfull(i-1,i), i=2, neq
ms = md - 1
pd(ms,2) = beta*y(3)
pd(ms,3) = -beta*y(2)

! Second superdiagonal pdfull(i-2,i), i=3, neq
ms = md - 2
pd(ms,3) = beta*y(2)

Return
End Subroutine myjac1
Subroutine myjac2(neq,t,y,ydot,pd,cj)

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: cj, t
Integer, Intent (In) :: neq

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Out) :: pd(neq*neq)
Real (Kind=nag_wp), Intent (In) :: y(neq), ydot(neq)

! .. Intrinsic Procedures ..
Intrinsic :: exp

! .. Executable Statements ..
pd(1) = -two*y(1) + 0.1E0_nag_wp*t*y(1)*exp(y(1))

Return
End Subroutine myjac2

Subroutine res1(neq,t,y,ydot,r,ires,iuser,ruser)

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: t
Integer, Intent (Inout) :: ires
Integer, Intent (In) :: neq

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Out) :: r(neq)
Real (Kind=nag_wp), Intent (Inout) :: ruser(*)
Real (Kind=nag_wp), Intent (In) :: y(neq), ydot(neq)
Integer, Intent (Inout) :: iuser(*)

! .. Executable Statements ..
r(1) = -alpha*y(1) + beta*y(2)*y(3) - ydot(1)
r(2) = alpha*y(1) - beta*y(2)*y(3) - gamma*y(2)*y(2) - ydot(2)
r(3) = gamma*y(2)*y(2) - ydot(3)
Return

End Subroutine res1
Subroutine jac1(neq,t,y,ydot,pd,cj,iuser,ruser)

! .. Use Statements ..
Use nag_library, Only: d02nez

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: cj, t
Integer, Intent (In) :: neq

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Inout) :: pd(*), ruser(*)
Real (Kind=nag_wp), Intent (In) :: y(neq), ydot(neq)
Integer, Intent (Inout) :: iuser(*)

! .. Local Scalars ..
Integer :: ijac, ml, mu

! .. Executable Statements ..
ml = iuser(1)
mu = iuser(2)
ijac = iuser(3)

If (ijac==1) Then
Call myjac1(neq,ml,mu,t,y,ydot,pd,cj)

Else
Call d02nez(neq,t,y,ydot,pd,cj,iuser,ruser)

End If

Return
End Subroutine jac1
Subroutine res2(neq,t,y,ydot,r,ires,iuser,ruser)

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: t
Integer, Intent (Inout) :: ires
Integer, Intent (In) :: neq

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Out) :: r(neq)
Real (Kind=nag_wp), Intent (Inout) :: ruser(*)
Real (Kind=nag_wp), Intent (In) :: y(neq), ydot(neq)
Integer, Intent (Inout) :: iuser(*)

! .. Intrinsic Procedures ..
Intrinsic :: exp

! .. Executable Statements ..
r(1) = 4.0_nag_wp - y(1)**2 + t*0.1E0_nag_wp*exp(y(1))
Return

End Subroutine res2
Subroutine jac2(neq,t,y,ydot,pd,cj,iuser,ruser)

! .. Use Statements ..
Use nag_library, Only: d02nez

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: cj, t
Integer, Intent (In) :: neq

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Inout) :: pd(*), ruser(*)
Real (Kind=nag_wp), Intent (In) :: y(neq), ydot(neq)
Integer, Intent (Inout) :: iuser(*)

! .. Local Scalars ..
Integer :: ijac

! .. Executable Statements ..
ijac = iuser(1)

If (ijac==1) Then
Call myjac2(neq,t,y,ydot,pd,cj)

Else
Call d02nez(neq,t,y,ydot,pd,cj,iuser,ruser)

End If

Return
End Subroutine jac2

End Module d02nefe_mod
Program d02nefe

! D02NEF Example Main Program

! .. Use Statements ..
Use d02nefe_mod, Only: nout

! .. Implicit None Statement ..
Implicit None

! .. Executable Statements ..
Write (nout,*) ’D02NEF Example Program Results’

Call ex1

Call ex2

Contains
Subroutine ex1

! .. Use Statements ..
Use nag_library, Only: d02mcf, d02mwf, d02nef, d02npf, nag_wp
Use d02nefe_mod, Only: jac1, ml, mu, neq1, nin, res1

! .. Local Scalars ..
Real (Kind=nag_wp) :: h0, hmax, t, tout
Integer :: i, ifail, ijac, itask, itol, j, &

lcom, licom, maxord, neq
Character (8) :: jceval

! .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: atol(:), com(:), rtol(:), y(:), &

ydot(:)
Real (Kind=nag_wp) :: ruser(1)
Integer, Allocatable :: icom(:)
Integer :: iuser(3)

! .. Executable Statements ..
Write (nout,*)
Write (nout,*) ’D02NEF Example 1’

! Skip heading in data file
Read (nin,*)
Read (nin,*) maxord
neq = neq1
lcom = 40 + (maxord+4)*neq + (2*ml+mu+1)*neq + 2*(neq/(ml+mu+1)+1)
licom = 50 + neq
Allocate (atol(neq),com(lcom),rtol(neq),y(neq),ydot(neq),icom(licom))
Read (nin,*) ijac, itol
Read (nin,*) rtol(1:neq)
Read (nin,*) atol(1:neq)
Read (nin,*) ydot(1:neq)
If (ijac==1) Then

jceval = ’Analytic’
Else

jceval = ’Numeric’
End If

! Set initial values
Read (nin,*) y(1:neq)

! Initialize the problem, specifying that the Jacobian is to be
! evaluated analytically using the provided routine jac.

Read (nin,*) hmax, h0
Read (nin,*) t, tout

! ifail: behaviour on error exit
! =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft

ifail = 0
Call d02mwf(neq,maxord,jceval,hmax,h0,itol,icom,licom,com,lcom,ifail)

! Specify that the Jacobian is banded.

ifail = 0
Call d02npf(neq,ml,mu,icom,licom,ifail)

! Use the iuser array to pass the band dimensions through to jac.
! An alternative would be to hard code values for ml and mu in jac.

iuser(1) = ml
iuser(2) = mu
iuser(3) = ijac

Write (nout,99999)(i,i=1,neq)
Write (nout,99998) t, (y(i),i=1,neq)
itask = 0

! Obtain the solution at 5 equally spaced values of T.

loop: Do j = 1, 5
ifail = -1
Call d02nef(neq,t,tout,y,ydot,rtol,atol,itask,res1,jac1,icom,com, &

lcom,iuser,ruser,ifail)
Write (nout,99998) t, (y(i),i=1,neq)
If (ifail/=0) Then

Write (nout,99997) ifail
Exit loop

End If
tout = tout + 0.02_nag_wp
Call d02mcf(icom)

End Do loop

Write (nout,*)
Write (nout,99996) itask

99999 Format (/,1X,’ t ’,5X,3(’ Y(’,I1,’) ’))
99998 Format (1X,F8.4,3X,3(F12.6))
99997 Format (1X,’ ** D02NEF returned with IFAIL = ’,I5)
99996 Format (1X,’The integrator completed task, ITASK = ’,I4)

End Subroutine ex1
Subroutine ex2

! .. Use Statements ..
Use nag_library, Only: d02mcf, d02mwf, d02nef, nag_wp
Use d02nefe_mod, Only: jac2, neq2, nin, res2

! .. Local Scalars ..
Real (Kind=nag_wp) :: h0, hmax, t, tout
Integer :: i, ifail, ijac, itask, itol, j, &

lcom, licom, maxord, neq
Character (8) :: jceval

! .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: atol(:), com(:), rtol(:), y(:), &

ydot(:)
Real (Kind=nag_wp) :: ruser(1)
Integer, Allocatable :: icom(:)
Integer :: iuser(1)

! .. Executable Statements ..
Write (nout,*)
Write (nout,*) ’D02NEF Example 2’
Write (nout,*)
Read (nin,*)
Read (nin,*) maxord
neq = neq2
lcom = 40 + (maxord+4)*neq + neq*neq

licom = 50 + neq
Allocate (atol(neq),com(lcom),rtol(neq),y(neq),ydot(neq),icom(licom))
Read (nin,*) ijac, itol
Read (nin,*) rtol(1:neq)
Read (nin,*) atol(1:neq)
Read (nin,*) ydot(1:neq)
If (ijac==1) Then

jceval = ’Analytic’
Else

jceval = ’Numeric’
End If

! Initialize the problem, specifying that the Jacobian is to be
! evaluated analytically using the provided routine jac.

Read (nin,*) y(1:neq)
Read (nin,*) hmax, h0
Read (nin,*) t, tout

ifail = 0
Call d02mwf(neq,maxord,jceval,hmax,h0,itol,icom,licom,com,lcom,ifail)

! Use the iuser array to pass whether numerical or analytic Jacobian
! is to be used.

iuser(1) = ijac

Write (nout,99999)(i,i=1,neq)
Write (nout,99998) t, y(1:neq)
itask = 0

! Obtain the solution at 5 equally spaced values of t.

loop: Do j = 1, 5

ifail = -1
Call d02nef(neq,t,tout,y,ydot,rtol,atol,itask,res2,jac2,icom,com, &

lcom,iuser,ruser,ifail)

Write (nout,99998) t, y(1:neq)
If (ifail/=0) Then

Write (nout,99997) ifail
Exit loop

End If
tout = tout + 0.2_nag_wp
Call d02mcf(icom)

End Do loop

Write (nout,*)
Write (nout,99996) itask

99999 Format (/,1X,’ t y(’,I1,’)’)
99998 Format (1X,F8.4,3X,3(F12.6))
99997 Format (1X,’ ** D02NEF returned with IFAIL = ’,I5)
99996 Format (1X,’The integrator completed task, ITASK = ’,I4)

End Subroutine ex2
End Program d02nefe

10.2 Program Data

D02NEF Example Program Data
5 : ex1 : maxord
1 1 : ijac, itol
1.0E-3 1.0E-3 1.0E-3 : rtol
1.0E-6 1.0E-6 1.0E-6 : atol
0.0 0.0 0.0 : ydot
1.0 0.0 0.0 : y
0.0 0.0 : hmax, h0
0.0 0.02 : t, tout

5 : ex2 : maxord
1 1 : ijac, itol
0.0 : rtol
1.0E-8 : atol
0.0 : ydot
2.0 : y
0.0 0.0 : hmax, h0
0.0 0.2 : t, tout

10.3 Program Results

D02NEF Example Program Results

D02NEF Example 1

t Y(1) Y(2) Y(3)
0.0000 1.000000 0.000000 0.000000
0.0200 0.999204 0.000036 0.000760
0.0400 0.998415 0.000036 0.001549
0.0600 0.997631 0.000036 0.002333
0.0800 0.996852 0.000036 0.003112
0.1000 0.996080 0.000036 0.003884

The integrator completed task, ITASK = 3

D02NEF Example 2

t y(1)
0.0000 2.000000
0.2000 2.038016
0.4000 2.078379
0.6000 2.121462
0.8000 2.167736
1.0000 2.217821

The integrator completed task, ITASK = 3

