

NAG Library Routine Document

D02NGF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

D02NGF is a direct communication routine for integrating stiff systems of implicit ordinary differential equations coupled with algebraic equations when the Jacobian is a full matrix.

2 Specification

```
SUBROUTINE D02NGF (NEQ, LDYSAV, T, TOUT, Y, YDOT, RWORK, RTOL, ATOL,      &
                  ITOL, INFORM, RESID, YSAV, SDYSAV, JAC, WKJAC,        &
                  NWKJAC, MONITR, LDERIV, ITASK, ITRACE, IFAIL)

INTEGER              NEQ, LDYSAV, ITOL, INFORM(23), SDYSAV, NWKJAC,      &
                  ITASK, ITRACE, IFAIL
REAL (KIND=nag_wp)  T, TOUT, Y(NEQ), YDOT(NEQ), RWORK(50+4*NEQ),      &
                  RTOL(*), ATOL(*), YSAV(LDYSAV,SDYSAV),              &
                  WKJAC(NWKJAC)
LOGICAL              LDERIV(2)
EXTERNAL             RESID, JAC, MONITR
```

3 Description

D02NGF is a general purpose routine for integrating the initial value problem for a stiff system of implicit ordinary differential equations coupled with algebraic equations, written in the form

$$A(t, y)y' = g(t, y).$$

It is designed specifically for the case where the resulting Jacobian is a full matrix (see the description of JAC).

Both interval and step oriented modes of operation are available and also modes designed to permit intermediate output within an interval oriented mode.

An outline of a typical calling program for D02NGF is given below. It calls the full matrix linear algebra setup routine D02NSF, the Backward Differentiation Formula (BDF) integrator setup routine D02NVF, and its diagnostic counterpart D02NYF.

```
!      Declarations

      EXTERNAL RESID, JAC, MONITR
      .
      .
      .
      IFAIL = 0
      CALL D02NVF(..., IFAIL)
      CALL D02NSF(NEQ, NEQMAX, JCEVAL, NWKJAC, RWORK, IFAIL)
      IFAIL = -1
      CALL D02NGF(NEQ, NEQMAX, T, TOUT, Y, YDOT, RWORK, RTOL, &
                  ATOL, ITOL, INFORM, RESID, YSAVE, NY2DIM, &
                  JAC, WKJAC, NWKJAC, MONITR, LDERIV, ITASK, &
                  ITRACE, IFAIL)
      IF (IFAIL.EQ.1 .OR. IFAIL.GE.14) STOP
      IFAIL = 0
      CALL D02NYF(...)
      .
      .
      .
      STOP
      END
```

The linear algebra setup routine D02NSF and one of the integrator setup routines, D02MVF, D02NVF or D02NWF, must be called prior to the call of D02NGF. The integrator diagnostic routine D02NYF may be called after the call to D02NGF. There is also a routine, D02NZE, designed to permit you to change step size on a continuation call to D02NGF without restarting the integration process.

4 References

See the D02M–N Sub-chapter Introduction.

5 Arguments

- 1: NEQ – INTEGER *Input*
On entry: the number of differential equations to be solved.
Constraint: $\text{NEQ} \geq 1$.
- 2: LDYSAV – INTEGER *Input*
On entry: a bound on the maximum number of equations to be solved during the integration.
Constraint: $\text{LDYSAV} \geq \text{NEQ}$.
- 3: T – REAL (KIND=nag_wp) *Input/Output*
On entry: t , the value of the independent variable. The input value of T is used only on the first call as the initial point of the integration.
On exit: the value at which the computed solution y is returned (usually at TOUT).
- 4: TOUT – REAL (KIND=nag_wp) *Input/Output*
On entry: the next value of t at which a computed solution is desired. For the initial t , the input value of TOUT is used to determine the direction of integration. Integration is permitted in either direction (see also ITASK).
Constraint: $\text{TOUT} \neq \text{T}$.
On exit: normally unchanged. However, when $\text{ITASK} = 6$, then TOUT contains the value of T at which initial values have been computed without performing any integration. See descriptions of ITASK and LDERIV.
- 5: Y(NEQ) – REAL (KIND=nag_wp) array *Input/Output*
On entry: the values of the dependent variables (solution). On the first call the first NEQ elements of Y must contain the vector of initial values.
On exit: the computed solution vector, evaluated at T (usually $\text{T} = \text{TOUT}$).
- 6: YDOT(NEQ) – REAL (KIND=nag_wp) array *Input/Output*
On entry: if $\text{LDERIV}(1) = \text{.TRUE.}$, YDOT must contain approximations to the time derivatives y' of the vector y .
If $\text{LDERIV}(1) = \text{.FALSE.}$, YDOT need not be set on entry.
On exit: the time derivatives y' of the vector y at the last integration point.

- 7: RWORK(50 + 4 × NEQ) – REAL (KIND=nag_wp) array *Communication Array*
- 8: RTOL(*) – REAL (KIND=nag_wp) array *Input*
Note: the dimension of the array RTOL must be at least 1 if ITOL = 1 or 2, and at least NEQ otherwise.
On entry: the relative local error tolerance.
Constraint: RTOL(*i*) ≥ 0.0 for all relevant *i* (see ITOL).
- 9: ATOL(*) – REAL (KIND=nag_wp) array *Input*
Note: the dimension of the array ATOL must be at least 1 if ITOL = 1 or 3, and at least NEQ otherwise.
On entry: the absolute local error tolerance.
Constraint: ATOL(*i*) ≥ 0.0 for all relevant *i* (see ITOL).
- 10: ITOL – INTEGER *Input*
On entry: a value to indicate the form of the local error test. ITOL indicates to D02NGF whether to interpret either or both of RTOL or ATOL as a vector or a scalar. The error test to be satisfied is $\|e_i/w_i\| < 1.0$, where w_i is defined as follows:

ITOL	RTOL	ATOL	w_i
1	scalar	scalar	$\text{RTOL}(1) \times y_i + \text{ATOL}(1)$
2	scalar	vector	$\text{RTOL}(1) \times y_i + \text{ATOL}(i)$
3	vector	scalar	$\text{RTOL}(i) \times y_i + \text{ATOL}(1)$
4	vector	vector	$\text{RTOL}(i) \times y_i + \text{ATOL}(i)$

e_i is an estimate of the local error in y_i , computed internally, and the choice of norm to be used is defined by a previous call to an integrator setup routine.

Constraint: ITOL = 1, 2, 3 or 4.

- 11: INFORM(23) – INTEGER array *Communication Array*
- 12: RESID – SUBROUTINE, supplied by the user. *External Procedure*
 RESID must evaluate the residual

$$r = g(t, y) - A(t, y)y'$$

in one case and

$$r = -A(t, y)y'$$

in another.

The specification of RESID is:

```
SUBROUTINE RESID (NEQ, T, Y, YDOT, R, IRES)
  INTEGER          NEQ, IRES
  REAL (KIND=nag_wp) T, Y(NEQ), YDOT(NEQ), R(NEQ)
```

- 1: NEQ – INTEGER *Input*
On entry: the number of equations being solved.
- 2: T – REAL (KIND=nag_wp) *Input*
On entry: t , the current value of the independent variable.

3:	Y(NEQ) – REAL (KIND=nag_wp) array <i>On entry:</i> the value of y_i , for $i = 1, 2, \dots, \text{NEQ}$.	<i>Input</i>
4:	YDOT(NEQ) – REAL (KIND=nag_wp) array <i>On entry:</i> the value of y'_i , for $i = 1, 2, \dots, \text{NEQ}$, at t .	<i>Input</i>
5:	R(NEQ) – REAL (KIND=nag_wp) array <i>On exit:</i> R(i) must contain the i th component of r , for $i = 1, 2, \dots, \text{NEQ}$, where $r = g(t, y) - A(t, y)y' \quad (1)$ or $r = -A(t, y)y' \quad (2)$ and where the definition of r is determined by the input value of IRES.	<i>Output</i>
6:	IRES – INTEGER <i>On entry:</i> the form of the residual that must be returned in array R. IRES = -1 The residual defined in equation (2) must be returned. IRES = 1 The residual defined in equation (1) must be returned. <i>On exit:</i> should be unchanged unless one of the following actions is required of the integrator, in which case IRES should be set accordingly. IRES = 2 Indicates to the integrator that control should be passed back immediately to the calling (sub)program with the error indicator set to IFAIL = 11. IRES = 3 Indicates to the integrator that an error condition has occurred in the solution vector, its time derivative or in the value of t . The integrator will use a smaller time step to try to avoid this condition. If this is not possible, the integrator returns to the calling (sub)program with the error indicator set to IFAIL = 7. IRES = 4 Indicates to the integrator to stop its current operation and to enter MONITR immediately with argument IMON = -2.	<i>Input/Output</i>

RESID must either be a module subprogram USED by, or declared as EXTERNAL in, the (sub) program from which D02NGF is called. Arguments denoted as *Input* must **not** be changed by this procedure.

- | | | |
|-----|---|----------------------------|
| 13: | YSAV(LDYSAV, SDYSAV) – REAL (KIND=nag_wp) array | <i>Communication Array</i> |
| 14: | SDYSAV – INTEGER | <i>Input</i> |
- On entry:* the second dimension of the array YSAV as declared in the (sub)program from which D02NGF is called. An appropriate value for SDYSAV is described in the specifications of the integrator setup routines D02MVF, D02NVF and D02NWF. This value must be the same as that supplied to the integrator setup routine.
- 15: JAC – SUBROUTINE, supplied by the NAG Library or the user. *External Procedure*
- JAC must evaluate the Jacobian of the system. If this option is not required, the actual argument for JAC must be the dummy routine D02NGZ. (D02NGZ is included in the NAG Library.) You must indicate to the integrator whether this option is to be used by setting the argument JCEVAL appropriately in a call to the full linear algebra setup routine D02NSF.

First we must define the system of nonlinear equations which is solved internally by the integrator. The time derivative, y' , generated internally, has the form

$$y' = (y - z)/(hd),$$

where h is the current step size and d is an argument that depends on the integration method in use. The vector y is the current solution and the vector z depends on information from previous time steps. This means that $\frac{d}{dy}(\) = (hd)\frac{d}{dy}(\)$. The system of nonlinear equations that is solved has the form

$$A(t, y)y' - g(t, y) = 0$$

but it is solved in the form

$$r(t, y) = 0,$$

where r is the function defined by

$$r(t, y) = (hd)(A(t, y)(y - z)/(hd) - g(t, y)).$$

It is the Jacobian matrix $\frac{\partial r}{\partial y}$ that you must supply in JAC as follows:

$$\frac{\partial r_i}{\partial y_j} = a_{ij}(t, y) + (hd)\frac{\partial}{\partial y_j}\left(\sum_{k=1}^{\text{NEQ}} a_{ik}(t, y)y'_k - g_i(t, y)\right).$$

The specification of JAC is:

SUBROUTINE JAC (NEQ, T, Y, YDOT, H, D, P)

INTEGER

NEQ

REAL (KIND=nag_wp) T, Y(NEQ), YDOT(NEQ), H, D, P(NEQ, NEQ)

1: NEQ – INTEGER

Input

On entry: the number of equations being solved.

2: T – REAL (KIND=nag_wp)

Input

On entry: t , the current value of the independent variable.

3: Y(NEQ) – REAL (KIND=nag_wp) array

Input

On entry: y_i , for $i = 1, 2, \dots, \text{NEQ}$, the current solution component.

4: YDOT(NEQ) – REAL (KIND=nag_wp) array

Input

On entry: the derivative of the solution at the current point t .

5: H – REAL (KIND=nag_wp)

Input

On entry: the current step size.

6: D – REAL (KIND=nag_wp)

Input

On entry: the argument d which depends on the integration method.

7: P(NEQ, NEQ) – REAL (KIND=nag_wp) array

Input/Output

On entry: is set to zero.

On exit: $P(i, j)$ must contain $\frac{\partial r_i}{\partial y_j}$, for $i = 1, 2, \dots, \text{NEQ}$ and $j = 1, 2, \dots, \text{NEQ}$.

Only the nonzero elements of this array need be set, since it is preset to zero before the call to JAC.

JAC must either be a module subprogram USED by, or declared as EXTERNAL in, the (sub) program from which D02NGF is called. Arguments denoted as *Input* must **not** be changed by this procedure.

- 16: WKJAC(NWKJAC) – REAL (KIND=nag_wp) array *Communication Array*
 17: NWKJAC – INTEGER *Input*

On entry: the dimension of the array WKJAC as declared in the (sub)program from which D02NGF is called. This value must be the same as that supplied to the linear algebra setup routine D02NSF.

Constraint: $\text{NWKJAC} \geq \text{LDYSAV} \times (\text{LDYSAV} + 1)$.

- 18: MONITR – SUBROUTINE, supplied by the NAG Library or the user. *External Procedure*
 MONITR performs tasks requested by you. If this option is not required, then the actual argument for MONITR must be the dummy routine D02NBY. (D02NBY is included in the NAG Library.)

The specification of MONITR is:

```
SUBROUTINE MONITR (NEQ, LDYSAV, T, HLAST, HNEXT, Y, YDOT, YSAV,      &
                   R, ACOR, IMON, INLN, HMIN, HMAX, NQU)
INTEGER               NEQ, LDYSAV, IMON, INLN, NQU
REAL (KIND=nag_wp) T, HLAST, HNEXT, Y(NEQ), YDOT(NEQ),          &
                   YSAV(LDYSAV, sdysav), R(NEQ), ACOR(NEQ, 2), HMIN, &
                   HMAX
```

where *sdysav* is the numerical value of SDYSAV in the call of D02NGF.

- | | | |
|----|---|---------------------|
| 1: | NEQ – INTEGER | <i>Input</i> |
| | <i>On entry:</i> the number of equations being solved. | |
| 2: | LDYSAV – INTEGER | <i>Input</i> |
| | <i>On entry:</i> an upper bound on the number of equations to be solved. | |
| 3: | T – REAL (KIND=nag_wp) | <i>Input</i> |
| | <i>On entry:</i> the current value of the independent variable. | |
| 4: | HLAST – REAL (KIND=nag_wp) | <i>Input</i> |
| | <i>On entry:</i> the last step size successfully used by the integrator. | |
| 5: | HNEXT – REAL (KIND=nag_wp) | <i>Input/Output</i> |
| | <i>On entry:</i> the step size that the integrator proposes to take on the next step. | |
| | <i>On exit:</i> the next step size to be used. If this is different from the input value, then IMON must be set to 4. | |
| 6: | Y(NEQ) – REAL (KIND=nag_wp) array | <i>Input/Output</i> |
| | <i>On entry:</i> <i>y</i> , the values of the dependent variables evaluated at <i>t</i> . | |
| | <i>On exit:</i> these values must not be changed unless IMON is set to 2. | |
| 7: | YDOT(NEQ) – REAL (KIND=nag_wp) array | <i>Input</i> |
| | <i>On entry:</i> the time derivatives <i>y'</i> of the vector <i>y</i> . | |

- 8: YSAV(LDYSAV, *sdysav*) – REAL (KIND=nag_wp) array *Input*
On entry: workspace to enable you to carry out interpolation using either of the routines D02XJF or D02XKF.
- 9: R(NEQ) – REAL (KIND=nag_wp) array *Input*
On entry: if IMON = 0 and INLN = 3, then the first NEQ elements contain the residual vector $A(t, y)y' - g(t, y)$.
- 10: ACOR(NEQ, 2) – REAL (KIND=nag_wp) array *Input*
On entry: with IMON = 1, ACOR(*i*, 1) contains the weight used for the *i*th equation when the norm is evaluated, and ACOR(*i*, 2) contains the estimated local error for the *i*th equation. The scaled local error at the end of a timestep may be obtained by calling the real function D02ZAF as follows:
- ```

 IFAIL = 1
 ERRLOC = D02ZAF(NEQ, ACOR(1,2), ACOR(1,1), IFAIL)
 ! CHECK IFAIL BEFORE PROCEEDING

```
- 11: IMON – INTEGER *Input/Output*  
*On entry:* a flag indicating under what circumstances MONITR was called:
- IMON = -2  
 Entry from the integrator after IRES = 4 (set in RESID) caused an early termination (this facility could be used to locate discontinuities).
- IMON = -1  
 The current step failed repeatedly.
- IMON = 0  
 Entry after a call to the internal nonlinear equation solver (see INLN).
- IMON = 1  
 The current step was successful.
- On exit:* may be reset to determine subsequent action in D02NGF.
- IMON = -2  
 Integration is to be halted. A return will be made from the integrator to the calling (sub)program with IFAIL = 12.
- IMON = -1  
 Allow the integrator to continue with its own internal strategy. The integrator will try up to three restarts unless IMON  $\neq$  -1 on exit.
- IMON = 0  
 Return to the internal nonlinear equation solver, where the action taken is determined by the value of INLN (see INLN).
- IMON = 1  
 Normal exit to the integrator to continue integration.
- IMON = 2  
 Restart the integration at the current time point. The integrator will restart from order 1 when this option is used. The solution Y, provided by MONITR, will be used for the initial conditions.
- IMON = 3  
 Try to continue with the same step size and order as was to be used before the call to MONITR. HMIN and HMAX may be altered if desired.
- IMON = 4  
 Continue the integration but using a new value of HNEXT and possibly new values of HMIN and HMAX.

|     |                                                                                                                                                                                                                                                                                                                                                                |                     |
|-----|----------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------|---------------------|
| 12: | INLN – INTEGER                                                                                                                                                                                                                                                                                                                                                 | <i>Output</i>       |
|     | <i>On exit:</i> the action to be taken by the internal nonlinear equation solver when MONITR is exited with IMON = 0. By setting INLN = 3 and returning to the integrator, the residual vector is evaluated and placed in the array R, and then MONITR is called again. At present this is the only option available: INLN must not be set to any other value. |                     |
| 13: | HMIN – REAL (KIND=nag_wp)                                                                                                                                                                                                                                                                                                                                      | <i>Input/Output</i> |
|     | <i>On entry:</i> the minimum step size to be taken on the next step.<br><i>On exit:</i> the minimum step size to be used. If this is different from the input value, then IMON must be set to 3 or 4.                                                                                                                                                          |                     |
| 14: | HMAX – REAL (KIND=nag_wp)                                                                                                                                                                                                                                                                                                                                      | <i>Input/Output</i> |
|     | <i>On entry:</i> the maximum step size to be taken on the next step.<br><i>On exit:</i> the maximum step size to be used. If this is different from the input value, then IMON must be set to 3 or 4. If HMAX is set to zero, no limit is assumed.                                                                                                             |                     |
| 15: | NQU – INTEGER                                                                                                                                                                                                                                                                                                                                                  | <i>Input</i>        |
|     | <i>On entry:</i> the order of the integrator used on the last step. This is supplied to enable you to carry out interpolation using either of the routines D02XJF or D02XKF.                                                                                                                                                                                   |                     |

MONITR must either be a module subprogram USED by, or declared as EXTERNAL in, the (sub)program from which D02NGF is called. Arguments denoted as *Input* must **not** be changed by this procedure.

- 19: LDERIV(2) – LOGICAL array *Input/Output*
- On entry:* LDERIV(1) must be set to .TRUE. if you have supplied both an initial  $y$  and an initial  $y'$ . LDERIV(1) must be set to .FALSE. if only the initial  $y$  has been supplied.
- LDERIV(2) must be set to .TRUE. if the integrator is to use a modified Newton method to evaluate the initial  $y$  and  $y'$ . Note that  $y$  and  $y'$ , if supplied, are used as initial estimates. This method involves taking a small step at the start of the integration, and if ITASK = 6 on entry, T and TOUT will be set to the result of taking this small step. LDERIV(2) must be set to .FALSE. if the integrator is to use functional iteration to evaluate the initial  $y$  and  $y'$ , and if this fails a modified Newton method will then be attempted. LDERIV(2) = .TRUE. is recommended if there are implicit equations or the initial  $y$  and  $y'$  are zero.
- On exit:* LDERIV(1) is normally unchanged. However if ITASK = 6 and internal initialization was successful then LDERIV(1) = .TRUE..
- LDERIV(2) = .TRUE., if implicit equations were detected. Otherwise LDERIV(2) = .FALSE..
- 20: ITASK – INTEGER *Input*
- On entry:* the task to be performed by the integrator.
- ITASK = 1  
Normal computation of output values of  $y(t)$  at  $t = TOUT$  (by overshooting and interpolating).
- ITASK = 2  
Take one step only and return.
- ITASK = 3  
Stop at the first internal integration point at or beyond  $t = TOUT$  and return.

ITASK = 4

Normal computation of output values of  $y(t)$  at  $t = \text{TOUT}$  but without overshooting  $t = \text{TCRIT}$ . TCRIT must be specified as an option in one of the integrator setup routines before the first call to the integrator, or specified in the optional input routine before a continuation call. TCRIT may be equal to or beyond TOUT, but not before it, in the direction of integration.

ITASK = 5

Take one step only and return, without passing TCRIT. TCRIT must be specified as under ITASK = 4.

ITASK = 6

The integrator will solve for the initial values of  $y$  and  $y'$  only and then return to the calling (sub)program without doing the integration. This option can be used to check the initial values of  $y$  and  $y'$ . Functional iteration or a 'small' backward Euler method used in conjunction with a damped Newton iteration is used to calculate these values (see LDERIV). Note that if a backward Euler step is used then the value of  $t$  will have been advanced a short distance from the initial point.

**Note:** if D02NGF is recalled with a different value of ITASK (and TOUT altered), then the initialization procedure is repeated, possibly leading to different initial conditions.

*Constraint:*  $1 \leq \text{ITASK} \leq 6$ .

21: ITRACE – INTEGER

*Input*

*On entry:* the level of output that is printed by the integrator. ITRACE may take the value  $-1$ ,  $0$ ,  $1$ ,  $2$  or  $3$ .

ITRACE <  $-1$

$-1$  is assumed and similarly if ITRACE >  $3$ , then  $3$  is assumed.

ITRACE =  $-1$

No output is generated.

ITRACE =  $0$

Only warning messages are printed on the current error message unit (see X04AAF).

ITRACE >  $0$

Warning messages are printed as above, and on the current advisory message unit (see X04ABF) output is generated which details Jacobian entries, the nonlinear iteration and the time integration. The advisory messages are given in greater detail the larger the value of ITRACE.

22: IFAIL – INTEGER

*Input/Output*

*On entry:* IFAIL must be set to  $0$ ,  $-1$  or  $1$ . If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value  $-1$  or  $1$  is recommended. If the output of error messages is undesirable, then the value  $1$  is recommended. Otherwise, because for this routine the values of the output arguments may be useful even if IFAIL  $\neq 0$  on exit, the recommended value is  $-1$ . **When the value  $-1$  or  $1$  is used it is essential to test the value of IFAIL on exit.**

*On exit:* IFAIL =  $0$  unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry  $IFAIL = 0$  or  $-1$ , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$IFAIL = 1$

An illegal input was detected on entry, or after an internal call to MONITR. If ITRACE  $> -1$ , then the form of the error will be detailed on the current error message unit (see X04AAF).

$IFAIL = 2$

The maximum number of steps specified has been taken (see the description of optional inputs in the integrator setup routines and the optional input continuation routine, D02NZF).

$IFAIL = 3$

With the given values of RTOL and ATOL no further progress can be made across the integration range from the current point T. The components  $Y(1), Y(2), \dots, Y(NEQ)$  contain the computed values of the solution at the current point T.

$IFAIL = 4$

There were repeated error test failures on an attempted step, before completing the requested task, but the integration was successful as far as T. The problem may have a singularity, or the local error requirements may be inappropriate.

$IFAIL = 5$

There were repeated convergence test failures on an attempted step, before completing the requested task, but the integration was successful as far as T. This may be caused by an inaccurate Jacobian matrix or one which is incorrectly computed.

$IFAIL = 6$

Some error weight  $w_i$  became zero during the integration (see the description of ITOL). Pure relative error control ( $ATOL(i) = 0.0$ ) was requested on a variable (the  $i$ th) which has now vanished. The integration was successful as far as T.

$IFAIL = 7$

RESID set its error flag (IRES = 3) continually despite repeated attempts by the integrator to avoid this.

$IFAIL = 8$

LDERIV(1) = .FALSE. on entry but the internal initialization routine was unable to initialize  $y'$  (more detailed information may be directed to the current error message unit, see X04AAF).

$IFAIL = 9$

A singular Jacobian  $\frac{\partial r}{\partial y}$  has been encountered. You should check the problem formulation and Jacobian calculation.

$IFAIL = 10$

An error occurred during Jacobian formulation or back-substitution (a more detailed error description may be directed to the current error message unit, see X04AAF).

IFAIL = 11

RESID signalled the integrator to halt the integration and return (IRES = 2). Integration was successful as far as T.

IFAIL = 12

MONITR set IMON = -2 and so forced a return but the integration was successful as far as T.

IFAIL = 13

The requested task has been completed, but it is estimated that a small change in RTOL and ATOL is unlikely to produce any change in the computed solution. (Only applies when you are not operating in one step mode, that is when ITASK  $\neq$  2 or 5.)

IFAIL = 14

The values of RTOL and ATOL are so small that D02NGF is unable to start the integration.

IFAIL = 15

The linear algebra setup routine D02NSF was not called before the call to D02NGF.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

## 7 Accuracy

The accuracy of the numerical solution may be controlled by a careful choice of the arguments RTOL and ATOL, and to a much lesser extent by the choice of norm. You are advised to use scalar error control unless the components of the solution are expected to be poorly scaled. For the type of decaying solution typical of many stiff problems, relative error control with a small absolute error threshold will be most appropriate (that is, you are advised to choose ITOL = 1 with ATOL(1) small but positive).

## 8 Parallelism and Performance

D02NGF is not thread safe and should not be called from a multithreaded user program. Please see Section 3.12.1 in How to Use the NAG Library and its Documentation for more information on thread safety.

D02NGF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

D02NGF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The cost of computing a solution depends critically on the size of the differential system and to a lesser extent on the degree of stiffness of the problem. For D02NGF the cost is proportional to  $NEQ^3$ , though for problems which are only mildly nonlinear the cost may be dominated by factors proportional to  $NEQ^2$  except for very large problems.

In general, you are advised to choose the BDF option (setup routine D02NVF) but if efficiency is of great importance and especially if it is suspected that  $\frac{\partial}{\partial y}(A^{-1}g)$  has complex eigenvalues near the imaginary axis for some part of the integration, you should try the BLEND option (setup routine D02NWF).

## 10 Example

This example solves the well-known stiff Robertson problem written in implicit form

$$\begin{aligned} r_1 &= -0.04a + 1.0E4bc && - a' \\ r_2 &= 0.04a - 1.0E4bc - 3.0E7b^2 && - b' \\ r_3 &= && 3.0E7b^2 - c' \end{aligned}$$

with initial conditions  $a = 1.0$  and  $b = c = 0.0$  over the range  $[0, 0.1]$  with vector error control (ITOL = 4), the BDF method (setup routine D02NVF) and functional iteration. The Jacobian is calculated numerically if the functional iteration encounters difficulty and the integration is in one-step mode (ITASK = 2), with  $C^0$  interpolation to calculate the solution at intervals of 0.02 using D02XJF externally. D02NBY is used for MONITR.

### 10.1 Program Text

```
! D02NGF Example Program Text
! Mark 26 Release. NAG Copyright 2016.

Module d02ngfe_mod

! D02NGF Example Program Module:
! Parameters and User-defined Routines

! .. Use Statements ..
Use nag_library, Only: nag_wp
! .. Implicit None Statement ..
Implicit None
! .. Accessibility Statements ..
Private
Public :: resid
! .. Parameters ..
Real (Kind=nag_wp), Parameter :: alpha = 0.04_nag_wp
Real (Kind=nag_wp), Parameter :: beta = 1.0E4_nag_wp
Real (Kind=nag_wp), Parameter :: gamma = 3.0E7_nag_wp
Integer, Parameter, Public :: iset = 1, itrace = 0, neq = 3, &
 nin = 5, nout = 6
Integer, Parameter, Public :: nrw = 50 + 4*neq
Integer, Parameter, Public :: nwkjac = neq*(neq+1)
Integer, Parameter, Public :: ldysav = neq
Contains
Subroutine resid(neq,t,y,ydot,r,ires)

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: t
Integer, Intent (Inout) :: ires
Integer, Intent (In) :: neq
! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Out) :: r(neq)
Real (Kind=nag_wp), Intent (In) :: y(neq), ydot(neq)
! .. Executable Statements ..
r(1) = -ydot(1)
r(2) = -ydot(2)
```

```

 r(3) = -ydot(3)

 If (ires==1) Then
 r(1) = -alpha*y(1) + beta*y(2)*y(3) + r(1)
 r(2) = alpha*y(1) - beta*y(2)*y(3) - gamma*y(2)*y(2) + r(2)
 r(3) = gamma*y(2)*y(2) + r(3)
 End If
 Return
End Subroutine resid
End Module d02ngfe_mod

Program d02ngfe

! D02NGF Example Main Program

! .. Use Statements ..
Use nag_library, Only: d02nby, d02ngf, d02ngz, d02nsf, d02nvf, d02nyf, &
 d02xjf, nag_wp, x04abf
Use d02ngfe_mod, Only: iset, itrace, ldysav, neq, nin, nout, nrw, &
 nwkJac, resid
! .. Implicit None Statement ..
Implicit None
! .. Local Scalars ..
Real (Kind=nag_wp) :: h, h0, hmax, hmin, hu, t, tcrit, &
 tcur, tolsf, tout, xout
Integer :: i, ifail, imxer, iout, itask, itol, &
 maxord, maxstp, mxhnil, niter, nje, &
 nq, nqu, nre, nst, outchn, sdysav
Logical :: petzld
! .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: atol(:), rtol(:), rwork(:), sol(:), &
 wkjac(:), y(:), ydot(:), ysav(:, :)
Real (Kind=nag_wp) :: con(6)
Integer :: inform(23)
Logical, Allocatable :: algequ(:)
Logical :: lderiv(2)
! .. Intrinsic Procedures ..
Intrinsic :: real
! .. Executable Statements ..
Write (nout,*) 'D02NGF Example Program Results'
! Skip heading in data file
Read (nin,*)
Read (nin,*) maxord, maxstp, mxhnil
sdysav = maxord + 1
Allocate (atol(neq),rtol(neq),rwork(nrw),sol(neq),wkjac(nwkJac),y(neq), &
 ydot(neq),ysav(ldysav,sdysav),algequ(neq))
outchn = nout
Call x04abf(iset,outchn)

! Integrate to tout by overshooting tout in one step mode (itask=2)
! using B.D.F formulae with a functional iteration method.
! Default values for the array con are used. Employ vector
! tolerances and the Jacobian is evaluated internally, if necessary.
! monitr subroutine replaced by NAG dummy routine D02NBY.
! Interpolation outside D02NGF using D02XJF.

 itask = 2
 Read (nin,*) petzld
 Read (nin,*) hmin, hmax, h0, tcrit
 Read (nin,*) t, tout
 Read (nin,*) y(1:neq)
 Read (nin,*) lderiv(1:2)
 Read (nin,*) itol
 Read (nin,*) rtol(1:neq)
 Read (nin,*) atol(1:neq)
 con(1:6) = 0.0_nag_wp

! ifail: behaviour on error exit
! =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
 ifail = 0
 Call d02nvf(neq,sdysav,maxord,'Functional-iteration',petzld,con,tcrit, &

```

```

 hmin,hmax,h0,maxstp,mxhnil,'Average-L2',rwork,ifail)

! Linear algebra setup required (in case functional iteration
! encounters any difficulty).
 ifail = 0
 Call d02nsf(neq,neq,'Numerical',nwkjac,rwork,ifail)

 xout = 0.02E0_nag_wp
 iout = 1

 Write (nout,99993)(i,i=1,neq)
 Write (nout,99999) t, (y(i),i=1,neq)

! Soft fail and error messages only

loop1: Do
 ifail = -1
 Call d02ngf(neq,ldysav,t,tout,y,ydot,rwork,rtol,atol,itol,inform, &
 resid,ysav,sdysav,d02ngz,wkjac,nwkjac,d02nby,lderiv,itask,itrace, &
 ifail)

 If (ifail==0) Then

 ifail = 0
 Call d02nyf(neq,neq,hu,h,tcur,tolsf,rwork,nst,nre,nje,nqu,nq,niter, &
 imxer,algequ,inform,ifail)

loop2: Do
 If (tcur-hu<xout .And. xout<=tcur) Then

! C0 interpolation
 ifail = 0
 Call d02xjf(xout,sol,neq,ysav,ldysav,sdysav,neq,tcur,nqu,hu,h, &
 ifail)

 Write (nout,99999) xout, (sol(i),i=1,neq)
 iout = iout + 1
 xout = real(iout,kind=nag_wp)*0.02E0_nag_wp
 If (iout>=6) Then
 Write (nout,*)
 Write (nout,99997) hu, h, tcur
 Write (nout,99996) nst, nre, nje
 Write (nout,99995) nqu, nq, niter
 Write (nout,99994) ' Max err comp = ', imxer
 Exit loop1
 End If
 Else
 Exit loop2
 End If
 End Do loop2
 Else
 Write (nout,*)
 Write (nout,99998) 'Exit D02NGF with IFAIL = ', ifail, ' and T = ', &
 t
 End If
 End Do loop1

99999 Format (1X,F8.3,3(F13.5,2X))
99998 Format (1X,A,I2,A,E12.5)
99997 Format (1X,' HUSED = ',E12.5,' HNEXT = ',E12.5,' TCUR = ',E12.5)
99996 Format (1X,' NST = ',I6,' NRE = ',I6,' NJE = ',I6)
99995 Format (1X,' NQU = ',I6,' NQ = ',I6,' NITER = ',I6)
99994 Format (1X,A,I4)
99993 Format (/ ,1X,' X ',3(' Y(',I1,') '))
 End Program d02ngfe

```

## 10.2 Program Data

D02NGF Example Program Data

```

5 200 5 : maxord, maxstp, mxhnil
.FALSE. : petzld
1.0E-10 10.0 0.0 0.0 : hmin, hmax, h0, tcrit
0.0 0.1 : t, tout
1.0 0.0 0.0 : y
.FALSE. .FALSE. : lderiv
4 : itol
1.0E-4 1.0E-3 1.0E-4 : rtol
1.0E-7 1.0E-8 1.0E-7 : atol

```

## 10.3 Program Results

D02NGF Example Program Results

| X     | Y(1)    | Y(2)    | Y(3)    |
|-------|---------|---------|---------|
| 0.000 | 1.00000 | 0.00000 | 0.00000 |
| 0.020 | 0.99920 | 0.00004 | 0.00076 |
| 0.040 | 0.99841 | 0.00004 | 0.00155 |
| 0.060 | 0.99763 | 0.00004 | 0.00234 |
| 0.080 | 0.99685 | 0.00004 | 0.00311 |
| 0.100 | 0.99608 | 0.00004 | 0.00389 |

```

HUSED = 0.26516E-03 HNEXT = 0.53032E-03 TCUR = 0.10016E+00
NST = 244 NRE = 809 NJE = 0
NQU = 1 NQ = 1 NITER = 0
Max err comp = 2

```

