
NAG Library Routine Document

D02NJF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

D02NJF is a direct communication routine for integrating stiff systems of implicit ordinary differential
equations coupled with algebraic equations when the Jacobian is a sparse matrix.

2 Specification

SUBROUTINE D02NJF (NEQ, LDYSAV, T, TOUT, Y, YDOT, RWORK, RTOL, ATOL,
ITOL, INFORM, RESID, YSAV, SDYSAV, JAC, WKJAC,
NWKJAC, JACPVT, NJCPVT, MONITR, LDERIV, ITASK,
ITRACE, IFAIL)

&
&
&

INTEGER NEQ, LDYSAV, ITOL, INFORM(23), SDYSAV, NWKJAC,
JACPVT(NJCPVT), NJCPVT, ITASK, ITRACE, IFAIL

&

REAL (KIND=nag_wp) T, TOUT, Y(NEQ), YDOT(NEQ), RWORK(50+4*NEQ),
RTOL(*), ATOL(*), YSAV(LDYSAV,SDYSAV),
WKJAC(NWKJAC)

&
&

LOGICAL LDERIV(2)
EXTERNAL RESID, JAC, MONITR

3 Description

D02NJF is a general purpose routine for integrating the initial value problem for a stiff system of
implicit ordinary differential equations coupled with algebraic equations, written in the form

A t; yð Þy0 ¼ g t; yð Þ:
It is designed specifically for the case where the resulting Jacobian is a sparse matrix (see the
description of JAC).

Both interval and step oriented modes of operation are available and also modes designed to permit
intermediate output within an interval oriented mode.

An outline of a typical calling program for D02NJF is given below. It calls the sparse matrix linear
algebra setup routine D02NUF, the Backward Differentiation Formula (BDF) integrator setup routine
D02NVF, its diagnostic counterpart D02NYF, and the sparse matrix linear algebra diagnostic routine
D02NXF.

! Declarations

EXTERNAL RESID, JAC, MONITR
.
.
.

IFAIL = 0
CALL D02NVF(...,IFAIL)
CALL D02NUF(NEQ, NEQMAX, JCEVAL, NWKJAC, IA, NIA, JA, NJA, &

JACPVT, NJCPVT, SENS, U, ETA, LBLOCK, ISPLIT, &
RWORK, IFAIL)

IFAIL = -1
CALL D02NJF(NEQ, NEQMAX, T, TOUT, Y, YDOT, RWORK, RTOL, &

ATOL, ITOL, INFORM, RESID, YSAVE, NY2DIM, JAC, &
WKJAC, NWKJAC, JACPVT, NJCPVT, MONITR, LDERIV, &
ITASK, ITRACE, IFAIL)

IF(IFAIL.EQ.1 .OR. IFAIL.GE.14) STOP
IFAIL = 0
CALL D02NXF(...)
CALL D02NYF(...)

.

.

.
STOP
END

The linear algebra setup routine D02NUF and one of the integrator setup routines, D02MVF, D02NVF
or D02NWF, must be called prior to the call of D02NJF. Either or both of the integrator diagnostic
routine D02NYF, or the sparse matrix linear algebra diagnostic routine D02NXF, may be called after
the call to D02NJF. There is also a routine, D02NZF, designed to permit you to change step size on a
continuation call to D02NJF without restarting the integration process.

4 References

See the D02M–N Sub-chapter Introduction.

5 Arguments

1: NEQ – INTEGER Input

On entry: the number of differential equations to be solved.

Constraint: NEQ � 1.

2: LDYSAV – INTEGER Input

On entry: a bound on the maximum number of equations to be solved during the integration.

Constraint: LDYSAV � NEQ.

3: T – REAL (KIND=nag_wp) Input/Output

On entry: t, the value of the independent variable. The input value of T is used only on the first
call as the initial point of the integration.

On exit: the value at which the computed solution y is returned (usually at TOUT).

4: TOUT – REAL (KIND=nag_wp) Input/Output

On entry: the next value of t at which a computed solution is desired. For the initial t, the input
value of TOUT is used to determine the direction of integration. Integration is permitted in either
direction (see also ITASK).

On exit: normally unchanged. However, when ITASK ¼ 6, then TOUT contains the value of T at
which initial values have been computed without performing any integration. See descriptions of
ITASK and LDERIV.

5: YðNEQÞ – REAL (KIND=nag_wp) array Input/Output

On entry: the values of the dependent variables (solution). On the first call the first NEQ
elements of Y must contain the vector of initial values.

On exit: the computed solution vector, evaluated at T (usually T ¼ TOUT).

6: YDOTðNEQÞ – REAL (KIND=nag_wp) array Input/Output

On entry: if LDERIVð1Þ ¼ :TRUE:, YDOT must contain approximations to the time derivatives
y0 of the vector y.

If LDERIVð1Þ ¼ :FALSE:, YDOT need not be set on entry.

On exit: the time derivatives y0 of the vector y at the last integration point.

7: RWORKð50þ 4� NEQÞ – REAL (KIND=nag_wp) array Communication Array

8: RTOLð�Þ – REAL (KIND=nag_wp) array Input

Note: the dimension of the array RTOL must be at least 1 if ITOL ¼ 1 or 2, and at least NEQ
otherwise.

On entry: the relative local error tolerance.

Constraint: RTOLðiÞ � 0:0 for all relevant i (see ITOL).

9: ATOLð�Þ – REAL (KIND=nag_wp) array Input

Note: the dimension of the array ATOL must be at least 1 if ITOL ¼ 1 or 3, and at least NEQ
otherwise.

On entry: the absolute local error tolerance.

Constraint: ATOLðiÞ � 0:0 for all relevant i (see ITOL).

10: ITOL – INTEGER Input

On entry: a value to indicate the form of the local error test. ITOL indicates to D02NJF whether
to interpret either or both of RTOL or ATOL as a vector or a scalar. The error test to be satisfied
is ei=wik k < 1:0, where wi is defined as follows:

ITOL RTOL ATOL wi

1 scalar scalar RTOLð1Þ � yij j þ ATOLð1Þ
2 scalar vector RTOLð1Þ � yij j þ ATOLðiÞ
3 vector scalar RTOLðiÞ � yij j þ ATOLð1Þ
4 vector vector RTOLðiÞ � yij j þ ATOLðiÞ

ei is an estimate of the local error in yi, computed internally, and the choice of norm to be used is
defined by a previous call to an integrator setup routine.

Constraint: ITOL ¼ 1, 2, 3 or 4.

11: INFORMð23Þ – INTEGER array Communication Array

12: RESID – SUBROUTINE, supplied by the user. External Procedure

RESID must evaluate the residual

r ¼ g t; yð Þ �A t; yð Þy0

in one case and

r ¼ �A t; yð Þy0

in another.

The specification of RESID is:

SUBROUTINE RESID (NEQ, T, Y, YDOT, R, IRES)

INTEGER NEQ, IRES
REAL (KIND=nag_wp) T, Y(NEQ), YDOT(NEQ), R(NEQ)

1: NEQ – INTEGER Input

On entry: the number of equations being solved.

2: T – REAL (KIND=nag_wp) Input

On entry: t, the current value of the independent variable.

3: YðNEQÞ – REAL (KIND=nag_wp) array Input

On entry: the value of yi, for i ¼ 1; 2; . . . ;NEQ.

4: YDOTðNEQÞ – REAL (KIND=nag_wp) array Input

On entry: the value of y0i, for i ¼ 1; 2; . . . ;NEQ, at t.

5: RðNEQÞ – REAL (KIND=nag_wp) array Output

On exit: RðiÞ must contain the ith component of r, for i ¼ 1; 2; . . . ;NEQ, where

r ¼ g t; yð Þ �A t; yð Þy0 ð1Þ
or

r ¼ �A t; yð Þy0 ð2Þ
and where the definition of r is determined by the input value of IRES.

6: IRES – INTEGER Input/Output

On entry: the form of the residual that must be returned in array R.

IRES ¼ �1
The residual defined in equation (2) must be returned.

IRES ¼ 1
The residual defined in equation (1) must be returned.

On exit: should be unchanged unless one of the following actions is required of the
integrator, in which case IRES should be set accordingly.

IRES ¼ 2
Indicates to the integrator that control should be passed back immediately to the
calling (sub)program with the error indicator set to IFAIL ¼ 11.

IRES ¼ 3
Indicates to the integrator that an error condition has occurred in the solution
vector, its time derivative or in the value of t. The integrator will use a smaller
time step to try to avoid this condition. If this is not possible, the integrator
returns to the calling (sub)program with the error indicator set to IFAIL ¼ 7.

IRES ¼ 4
Indicates to the integrator to stop its current operation and to enter MONITR
immediately with argument IMON ¼ �2.

RESID must either be a module subprogram USEd by, or declared as EXTERNAL in, the (sub)
program from which D02NJF is called. Arguments denoted as Input must not be changed by this
procedure.

13: YSAVðLDYSAV;SDYSAVÞ – REAL (KIND=nag_wp) array Communication Array

14: SDYSAV – INTEGER Input

On entry: the second dimension of the array YSAV as declared in the (sub)program from which
D02NJF is called. An appropriate value for SDYSAV is described in the specifications of the
integrator setup routines D02MVF, D02NVF and D02NWF. This value must be the same as that
supplied to the integrator setup routine.

15: JAC – SUBROUTINE, supplied by the NAG Library or the user. External Procedure

JAC must evaluate the Jacobian of the system. If this option is not required, the actual argument
for JAC must be the dummy routine D02NJZ. (D02NJZ is included in the NAG Library.) You

must indicate to the integrator whether this option is to be used by setting the argument JCEVAL
appropriately in a call to the sparse linear algebra setup routine D02NUF.

First we must define the system of nonlinear equations which is solved internally by the
integrator. The time derivative, y0, generated internally, has the form

y0 ¼ y� zð Þ= hdð Þ;
where h is the current step size and d is an argument that depends on the integration method in
use. The vector y is the current solution and the vector z depends on information from previous
time steps. This means that d

dy0ð Þ ¼ hdð Þ ddyð Þ . The system of nonlinear equations that is solved

has the form

A t; yð Þy0 � g t; yð Þ ¼ 0

but it is solved in the form

r t; yð Þ ¼ 0;

where r is the function defined by

r t; yð Þ ¼ hdð Þ A t; yð Þ y� zð Þ= hdð Þ � g t; yð Þð Þ:

It is the Jacobian matrix
@r

@y
that you must supply in JAC as follows:

@ri
@yj

¼ aij t; yð Þ þ hdð Þ @

@yj

XNEQ
k¼1

aik t; yð Þy0k � gi t; yð Þ
 !

:

The specification of JAC is:

SUBROUTINE JAC (NEQ, T, Y, YDOT, H, D, J, PDJ)

INTEGER NEQ, J
REAL (KIND=nag_wp) T, Y(NEQ), YDOT(NEQ), H, D, PDJ(NEQ)

1: NEQ – INTEGER Input

On entry: the number of equations being solved.

2: T – REAL (KIND=nag_wp) Input

On entry: t, the current value of the independent variable.

3: YðNEQÞ – REAL (KIND=nag_wp) array Input

On entry: yi, for i ¼ 1; 2; . . . ;NEQ, the current solution component.

4: YDOTðNEQÞ – REAL (KIND=nag_wp) array Input

On entry: the derivative of the solution at the current point t.

5: H – REAL (KIND=nag_wp) Input

On entry: the current step size.

6: D – REAL (KIND=nag_wp) Input

On entry: the argument d which depends on the integration method.

7: J – INTEGER Input

On entry: the column of the Jacobian that JAC must return in the array PDJ.

D02 – Ordinary Differential D02NJF

Mark 26 D02NJF.5



8: PDJðNEQÞ – REAL (KIND=nag_wp) array Input/Output

On entry: is set to zero.

On exit: PDJðiÞ should be set to the i; jð Þth element of the Jacobian, where j is given by
J. Only nonzero elements of this array need be set, since it is preset to zero before the
call to JAC.

JAC must either be a module subprogram USEd by, or declared as EXTERNAL in, the (sub)
program from which D02NJF is called. Arguments denoted as Input must not be changed by this
procedure.

16: WKJACðNWKJACÞ – REAL (KIND=nag_wp) array Communication Array
17: NWKJAC – INTEGER Input

On entry: the dimension of the array WKJAC as declared in the (sub)program from which
D02NJF is called. The actual size depends on whether the sparsity structure is supplied or
whether it is to be estimated. An appropriate value for NWKJAC is described in the specification
of the linear algebra setup routine D02NUF. This value must be the same as that supplied to
D02NUF.

18: JACPVTðNJCPVTÞ – INTEGER array Communication Array
19: NJCPVT – INTEGER Input

On entry: the dimension of the array JACPVT as declared in the (sub)program from which
D02NJF is called. The actual size depends on whether the sparsity structure is supplied or
whether it is to be estimated. An appropriate value for NJCPVT is described in the specification
for the linear algebra setup routine D02NUF. This value must be same as that supplied to
D02NUF.

20: MONITR – SUBROUTINE, supplied by the NAG Library or the user. External Procedure

MONITR performs tasks requested by you. If this option is not required, then the actual
argument for MONITR must be the dummy routine D02NBY. (D02NBY is included in the NAG
Library.)

The specification of MONITR is:

SUBROUTINE MONITR (NEQ, LDYSAV, T, HLAST, HNEXT, Y, YDOT, YSAV,
R, ACOR, IMON, INLN, HMIN, HMAX, NQU)

&

INTEGER NEQ, LDYSAV, IMON, INLN, NQU
REAL (KIND=nag_wp) T, HLAST, HNEXT, Y(NEQ), YDOT(NEQ),

YSAV(LDYSAV,sdysav), R(NEQ), ACOR(NEQ,2), HMIN,
HMAX

&
&

where sdysav is the numerical value of SDYSAV in the call of D02NJF.

1: NEQ – INTEGER Input

On entry: the number of equations being solved.

2: LDYSAV – INTEGER Input

On entry: an upper bound on the number of equations to be solved.

3: T – REAL (KIND=nag_wp) Input

On entry: the current value of the independent variable.

4: HLAST – REAL (KIND=nag_wp) Input

On entry: the last step size successfully used by the integrator.

5: HNEXT – REAL (KIND=nag_wp) Input/Output

On entry: the step size that the integrator proposes to take on the next step.

On exit: the next step size to be used. If this is different from the input value, then
IMON must be set to 4.

6: YðNEQÞ – REAL (KIND=nag_wp) array Input/Output

On entry: y, the values of the dependent variables evaluated at t.

On exit: these values must not be changed unless IMON is set to 2.

7: YDOTðNEQÞ – REAL (KIND=nag_wp) array Input

On entry: the time derivatives y0 of the vector y.

8: YSAVðLDYSAV; sdysavÞ – REAL (KIND=nag_wp) array Input

On entry: workspace to enable you to carry out interpolation using either of the routines
D02XJF or D02XKF.

9: RðNEQÞ – REAL (KIND=nag_wp) array Input

On entry: if IMON ¼ 0 and INLN ¼ 3, then the first NEQ elements contain the residual
vector A t; yð Þy0 � g t; yð Þ.

10: ACORðNEQ; 2Þ – REAL (KIND=nag_wp) array Input

On entry: with IMON ¼ 1, ACORði; 1Þ contains the weight used for the ith equation
when the norm is evaluated, and ACORði; 2Þ contains the estimated local error for the
ith equation. The scaled local error at the end of a timestep may be obtained by calling
the real function D02ZAF as follows:

IFAIL = 1
ERRLOC = D02ZAF(NEQ, ACOR(1,2), ACOR(1,1), IFAIL)
! CHECK IFAIL BEFORE PROCEEDING

11: IMON – INTEGER Input/Output

On entry: a flag indicating under what circumstances MONITR was called:

IMON ¼ �2
Entry from the integrator after IRES ¼ 4 (set in RESID) caused an early
termination (this facility could be used to locate discontinuities).

IMON ¼ �1
The current step failed repeatedly.

IMON ¼ 0
Entry after a call to the internal nonlinear equation solver (see INLN).

IMON ¼ 1
The current step was successful.

On exit: may be reset to determine subsequent action in D02NJF.

IMON ¼ �2
Integration is to be halted. A return will be made from the integrator to the
calling (sub)program with IFAIL ¼ 12.

IMON ¼ �1
Allow the integrator to continue with its own internal strategy. The integrator will
try up to three restarts unless IMON 6¼ �1 on exit.

IMON ¼ 0
Return to the internal nonlinear equation solver, where the action taken is
determined by the value of INLN (see INLN).

IMON ¼ 1
Normal exit to the integrator to continue integration.

IMON ¼ 2
Restart the integration at the current time point. The integrator will restart from
order 1 when this option is used. The solution Y, provided by MONITR, will be
used for the initial conditions.

IMON ¼ 3
Try to continue with the same step size and order as was to be used before the
call to MONITR. HMIN and HMAX may be altered if desired.

IMON ¼ 4
Continue the integration but using a new value of HNEXT and possibly new
values of HMIN and HMAX.

12: INLN – INTEGER Output

On exit: the action to be taken by the internal nonlinear equation solver when MONITR
is exited with IMON ¼ 0. By setting INLN ¼ 3 and returning to the integrator, the
residual vector is evaluated and placed in the array R, and then MONITR is called
again. At present this is the only option available: INLN must not be set to any other
value.

13: HMIN – REAL (KIND=nag_wp) Input/Output

On entry: the minimum step size to be taken on the next step.

On exit: the minimum step size to be used. If this is different from the input value, then
IMON must be set to 3 or 4.

14: HMAX – REAL (KIND=nag_wp) Input/Output

On entry: the maximum step size to be taken on the next step.

On exit: the maximum step size to be used. If this is different from the input value, then
IMON must be set to 3 or 4. If HMAX is set to zero, no limit is assumed.

15: NQU – INTEGER Input

On entry: the order of the integrator used on the last step. This is supplied to enable you
to carry out interpolation using either of the routines D02XJF or D02XKF.

MONITR must either be a module subprogram USEd by, or declared as EXTERNAL in, the
(sub)program from which D02NJF is called. Arguments denoted as Input must not be changed by
this procedure.

21: LDERIVð2Þ – LOGICAL array Input/Output

On entry: LDERIVð1Þ must be set to .TRUE. if you have supplied both an initial y and an initial
y0. LDERIVð1Þ must be set to .FALSE. if only the initial y has been supplied.

LDERIVð2Þ must be set to .TRUE. if the integrator is to use a modified Newton method to
evaluate the initial y and y0. Note that y and y0, if supplied, are used as initial estimates. This
method involves taking a small step at the start of the integration, and if ITASK ¼ 6 on entry, T
and TOUT will be set to the result of taking this small step. LDERIVð2Þ must be set to .FALSE.
if the integrator is to use functional iteration to evaluate the initial y and y0, and if this fails a
modified Newton method will then be attempted. LDERIVð2Þ ¼ :TRUE: is recommended if there
are implicit equations or the initial y and y0 are zero.

D02NJF NAG Library Manual

D02NJF.8 Mark 26



On exit: LDERIVð1Þ is normally unchanged. However if ITASK ¼ 6 and internal initialization
was successful then LDERIVð1Þ ¼ :TRUE:.

LDERIVð2Þ ¼ :TRUE:, if implicit equations were detected. Otherwise LDERIVð2Þ ¼ :FALSE:.

22: ITASK – INTEGER Input

On entry: the task to be performed by the integrator.

ITASK ¼ 1
Normal computation of output values of y tð Þ at t ¼ TOUT (by overshooting and
interpolating).

ITASK ¼ 2
Take one step only and return.

ITASK ¼ 3
Stop at the first internal integration point at or beyond t ¼ TOUT and return.

ITASK ¼ 4
Normal computation of output values of y tð Þ at t ¼ TOUT but without overshooting
t ¼ TCRIT. TCRIT must be specified as an option in one of the integrator setup routines
before the first call to the integrator, or specified in the optional input routine before a
continuation call. TCRIT may be equal to or beyond TOUT, but not before it, in the
direction of integration.

ITASK ¼ 5
Take one step only and return, without passing TCRIT. TCRIT must be specified as under
ITASK ¼ 4.

ITASK ¼ 6
The integrator will solve for the initial values of y and y0 only and then return to the
calling (sub)program without doing the integration. This option can be used to check the
initial values of y and y0. Functional iteration or a ‘small’ backward Euler method used in
conjunction with a damped Newton iteration is used to calculate these values (see
LDERIV). Note that if a backward Euler step is used then the value of t will have been
advanced a short distance from the initial point.

Note: if D02NJF is recalled with a different value of ITASK (and TOUT altered), then the
initialization procedure is repeated, possibly leading to different initial conditions.

Constraint: 1 � ITASK � 6.

23: ITRACE – INTEGER Input

On entry: the level of output that is printed by the integrator. ITRACE may take the value �1, 0,
1, 2 or 3.

ITRACE < �1
�1 is assumed and similarly if ITRACE > 3, then 3 is assumed.

ITRACE ¼ �1
No output is generated.

ITRACE ¼ 0
Only warning messages are printed on the current error message unit (see X04AAF).

ITRACE > 0
Warning messages are printed as above, and on the current advisory message unit (see
X04ABF) output is generated which details Jacobian entries, the nonlinear iteration and
the time integration. The advisory messages are given in greater detail the larger the value
of ITRACE.

24: IFAIL – INTEGER Input/Output

On entry: IFAIL must be set to 0, �1 or 1. If you are unfamiliar with this argument you should
refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is
detected, the value �1 or 1 is recommended. If the output of error messages is undesirable, then
the value 1 is recommended. Otherwise, because for this routine the values of the output
arguments may be useful even if IFAIL 6¼ 0 on exit, the recommended value is �1. When the
value �1 or 1 is used it is essential to test the value of IFAIL on exit.

On exit: IFAIL ¼ 0 unless the routine detects an error or a warning has been flagged (see
Section 6).

6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL ¼ 1

An illegal input was detected on entry, or after an internal call to MONITR. If ITRACE > �1,
then the form of the error will be detailed on the current error message unit (see X04AAF).

IFAIL ¼ 2

The maximum number of steps specified has been taken (see the description of optional inputs in
the integrator setup routines and the optional input continuation routine, D02NZF).

IFAIL ¼ 3

With the given values of RTOL and ATOL no further progress can be made across the integration
range from the current point T. The components Yð1Þ;Yð2Þ; . . . ;YðNEQÞ contain the computed
values of the solution at the current point T.

IFAIL ¼ 4

There were repeated error test failures on an attempted step, before completing the requested
task, but the integration was successful as far as T. The problem may have a singularity, or the
local error requirements may be inappropriate.

IFAIL ¼ 5

There were repeated convergence test failures on an attempted step, before completing the
requested task, but the integration was successful as far as T. This may be caused by an
inaccurate Jacobian matrix or one which is incorrectly computed.

IFAIL ¼ 6

Some error weight wi became zero during the integration (see the description of ITOL). Pure
relative error control (ATOLðiÞ ¼ 0:0) was requested on a variable (the ith) which has now
vanished. The integration was successful as far as T.

IFAIL ¼ 7

RESID set its error flag (IRES ¼ 3) continually despite repeated attempts by the integrator to
avoid this.

IFAIL ¼ 8

LDERIVð1Þ ¼ :FALSE: on entry but the internal initialization routine was unable to initialize y0
(more detailed information may be directed to the current error message unit, see X04AAF).

IFAIL ¼ 9

A singular Jacobian
@r

@y
has been encountered. You should check the problem formulation and

Jacobian calculation.

IFAIL ¼ 10

An error occurred during Jacobian formulation or back-substitution (a more detailed error
description may be directed to the current error message unit, see X04AAF).

IFAIL ¼ 11

RESID signalled the integrator to halt the integration and return (IRES ¼ 2). Integration was
successful as far as T.

IFAIL ¼ 12

MONITR set IMON ¼ �2 and so forced a return but the integration was successful as far as T.

IFAIL ¼ 13

The requested task has been completed, but it is estimated that a small change in RTOL and
ATOL is unlikely to produce any change in the computed solution. (Only applies when you are
not operating in one step mode, that is when ITASK 6¼ 2 or 5.)

IFAIL ¼ 14

The values of RTOL and ATOL are so small that D02NJF is unable to start the integration.

IFAIL ¼ 15

The linear algebra setup routine D02NUF was not called before the call to D02NJF.

IFAIL ¼ �99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The accuracy of the numerical solution may be controlled by a careful choice of the arguments RTOL
and ATOL, and to a much lesser extent by the choice of norm. You are advised to use scalar error
control unless the components of the solution are expected to be poorly scaled. For the type of decaying
solution typical of many stiff problems, relative error control with a small absolute error threshold will
be most appropriate (that is, you are advised to choose ITOL ¼ 1 with ATOLð1Þ small but positive).

8 Parallelism and Performance

D02NJF is not thread safe and should not be called from a multithreaded user program. Please see
Section 3.12.1 in How to Use the NAG Library and its Documentation for more information on thread
safety.

D02NJF is threaded by NAG for parallel execution in multithreaded implementations of the NAG
Library.

D02NJF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor
library used by this implementation. Consult the documentation for the vendor library for further
information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the
OpenMP environment used within this routine. Please also consult the Users' Note for your
implementation for any additional implementation-specific information.

9 Further Comments

Since numerical stability and memory are often conflicting requirements when solving ordinary
differential systems where the Jacobian matrix is sparse we provide a diagnostic routine, D02NXF,
whose aim is to inform you how much memory is required to solve the problem and to give you some
indicators of numerical stability.

In general, you are advised to choose the BDF option (setup routine D02NVF) but if efficiency is of

great importance and especially if it is suspected that
@

@y
A�1gð Þ has complex eigenvalues near the

imaginary axis for some part of the integration, you should try the BLEND option (setup routine
D02NWF).

10 Example

This example solves the well-known stiff Robertson problem written as a mixed differential/algebraic
system in implicit form

r1 ¼ aþ bþ c� 1:0

r2 ¼ 0:04a� 1:0E4bc� 3:0E7b2 � b0

r3 ¼ 3:0E7b2 � c0

exploiting the fact that, from the initial conditions a ¼ 1:0 and b ¼ c ¼ 0:0, we know that aþ bþ c ¼ 1
for all time. We integrate over the range 0; 10:0½ � with vector relative error control and scalar absolute
error control (ITOL ¼ 3) and using the BDF integrator (setup routine D02NVF) and a modified Newton
method. The Jacobian is evaluated, in turn, using the `A' (Analytical) and `F' (Full information) options.
We provide a monitor routine to terminate the integration when the value of the component a falls
below 0:9.

10.1 Program Text

! D02NJF Example Program Text
! Mark 26 Release. NAG Copyright 2016.

Module d02njfe_mod

! D02NJF Example Program Module:
! Parameters and User-defined Routines

! .. Use Statements ..
Use nag_library, Only: nag_wp

! .. Implicit None Statement ..
Implicit None

! .. Accessibility Statements ..
Private
Public :: jac, monitr, resid

! .. Parameters ..
Real (Kind=nag_wp), Parameter :: alpha = 0.04_nag_wp
Real (Kind=nag_wp), Parameter :: beta = 1.0E4_nag_wp
Real (Kind=nag_wp), Parameter :: gamma = 3.0E7_nag_wp
Real (Kind=nag_wp), Parameter :: one = 1.0_nag_wp

Real (Kind=nag_wp), Parameter :: two = 2.0_nag_wp
Real (Kind=nag_wp), Parameter :: zero = 0.0_nag_wp
Integer, Parameter, Public :: iset = 1, itrace = 0, neq = 3
Integer, Parameter, Public :: nia = neq + 1
Integer, Parameter, Public :: nin = 5, nout = 6
Integer, Parameter, Public :: nrw = 50 + 4*neq
Integer, Parameter, Public :: ldysav = neq
Integer, Parameter :: nelts = 8
Integer, Parameter, Public :: nja = nelts
Integer, Parameter, Public :: njcpvt = 20*neq + 12*nelts
Integer, Parameter, Public :: nwkjac = 4*neq + 12*nelts

Contains
Subroutine resid(neq,t,y,ydot,r,ires)

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: t
Integer, Intent (Inout) :: ires
Integer, Intent (In) :: neq

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Out) :: r(neq)
Real (Kind=nag_wp), Intent (In) :: y(neq), ydot(neq)

! .. Executable Statements ..
r(1) = zero
r(2) = -ydot(2)
r(3) = -ydot(3)
If (ires==1) Then

r(1) = y(1) + y(2) + y(3) - one + r(1)
r(2) = alpha*y(1) - beta*y(2)*y(3) - gamma*y(2)*y(2) + r(2)
r(3) = gamma*y(2)*y(2) + r(3)

End If
Return

End Subroutine resid

Subroutine jac(neq,t,y,ydot,h,d,j,pdj)

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: d, h, t
Integer, Intent (In) :: j, neq

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Inout) :: pdj(neq)
Real (Kind=nag_wp), Intent (In) :: y(neq), ydot(neq)

! .. Local Scalars ..
Real (Kind=nag_wp) :: hxd

! .. Executable Statements ..
! 8 nonzero elements in total

hxd = h*d
If (j==1) Then

pdj(1) = zero - hxd*(one)
pdj(2) = zero - hxd*(alpha)

! note: pdj(3) is zero
Else If (j==2) Then

pdj(1) = zero - hxd*(one)
pdj(2) = one - hxd*(-beta*y(3)-two*gamma*y(2))
pdj(3) = zero - hxd*(two*gamma*y(2))

Else If (j==3) Then
pdj(1) = zero - hxd*(one)
pdj(2) = zero - hxd*(-beta*y(2))
pdj(3) = one - hxd*(zero)

End If
Return

End Subroutine jac

Subroutine monitr(neq,ldysav,t,hlast,hnext,y,ydot,ysav,r,acor,imon,inln, &
hmin,hmax,nqu)

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: hlast, t
Real (Kind=nag_wp), Intent (Inout) :: hmax, hmin, hnext
Integer, Intent (Inout) :: imon
Integer, Intent (Out) :: inln
Integer, Intent (In) :: ldysav, neq, nqu

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (In) :: acor(neq,2), r(neq), ydot(neq), &

ysav(ldysav,*)
Real (Kind=nag_wp), Intent (Inout) :: y(neq)

! .. Executable Statements ..
inln = 3
If (y(1)<=0.9_nag_wp) Then

imon = -2
End If
Return

End Subroutine monitr
End Module d02njfe_mod

Program d02njfe

! D02NJF Example Main Program

! .. Use Statements ..
Use nag_library, Only: d02njf, d02nuf, d02nvf, d02nxf, d02nyf, nag_wp, &

x04abf
Use d02njfe_mod, Only: iset, itrace, jac, ldysav, monitr, neq, nia, nin, &

nja, njcpvt, nout, nrw, nwkjac, resid
! .. Implicit None Statement ..

Implicit None
! .. Local Scalars ..

Real (Kind=nag_wp) :: eta, h, h0, hmax, hmin, hu, sens, t, &
tcrit, tcur, tinit, tolsf, tout, u

Integer :: i, icall, icase, ifail, igrow, &
imxer, isplit, isplt, itask, itol, &
liwreq, liwusd, lrwreq, lrwusd, &
maxord, maxstp, mxhnil, nblock, ngp, &
niter, nje, nlu, nnz, nq, nqu, nre, &
nst, outchn, sdysav

Logical :: lblock, petzld
! .. Local Arrays ..

Real (Kind=nag_wp), Allocatable :: atol(:), rtol(:), rwork(:), &
wkjac(:), y(:), ydot(:), yinit(:), &
ysav(:,:)

Real (Kind=nag_wp) :: con(6)
Integer, Allocatable :: ia(:), ja(:), jacpvt(:)
Integer :: inform(23)
Logical, Allocatable :: algequ(:)
Logical :: lderiv(2)

! .. Executable Statements ..
Write (nout,*) ’D02NJF Example Program Results’

! Skip heading in data file
Read (nin,*)
Read (nin,*) maxord, maxstp, mxhnil
sdysav = maxord + 1
Allocate (atol(neq),rtol(neq),rwork(nrw),wkjac(nwkjac),y(neq), &

yinit(neq),ydot(neq),ysav(ldysav,sdysav),ia(nia),ja(nja), &
jacpvt(njcpvt),algequ(neq))

Read (nin,*) ia(1:nia)
Read (nin,*) ja(1:nja)

outchn = nout
Call x04abf(iset,outchn)

! Two cases. In both cases:
! integrate to tout by overshooting (itask=1);
! use B.D.F formulae with a Newton method;
! use the Petzold error test (differential algebraic system);
! use default values for the array con;
! employ vector relative tolerance and scalar absolute tolerance.
! the Jacobian is supplied by jac;
! the monitr routine is used to force a return when y(1) < 0.9.

Read (nin,*) hmin, hmax, h0, tcrit
Read (nin,*) eta, sens, u
Read (nin,*) lblock
Read (nin,*) tinit, tout

Read (nin,*) itol, isplt
Read (nin,*) yinit(1:neq)
Read (nin,*) rtol(1:neq)
Read (nin,*) atol(1)

con(1:6) = 0.0_nag_wp
itask = 1
petzld = .True.

cases: Do icase = 1, 2

! Initialize
t = tinit
isplit = isplt
y(1:neq) = yinit(1:neq)
lderiv(1:2) = .False.

ifail = 0
Call d02nvf(neq,sdysav,maxord,’Newton’,petzld,con,tcrit,hmin,hmax,h0, &

maxstp,mxhnil,’Average-L2’,rwork,ifail)
Write (nout,*)

Select Case (icase)
Case (1)

! First case. The Jacobian structure is determined internally by
! calls to jac.

ifail = 0
Call d02nuf(neq,neq,’Analytical’,nwkjac,ia,nia,ja,nja,jacpvt,njcpvt, &

sens,u,eta,lblock,isplit,rwork,ifail)
Write (nout,*) ’ Analytic Jacobian, structure not supplied’

Case (2)
! Second case. The Jacobian structure is supplied.

ifail = 0
Call d02nuf(neq,neq,’Full info’,nwkjac,ia,nia,ja,nja,jacpvt,njcpvt, &

sens,u,eta,lblock,isplit,rwork,ifail)
Write (nout,*) ’ Analytic Jacobian, structure supplied’

End Select

Write (nout,99988)(i,i=1,neq)
Write (nout,99999) t, (y(i),i=1,neq)

! Soft fail and error messages only

ifail = 1
Call d02njf(neq,ldysav,t,tout,y,ydot,rwork,rtol,atol,itol,inform, &

resid,ysav,sdysav,jac,wkjac,nwkjac,jacpvt,njcpvt,monitr,lderiv, &
itask,itrace,ifail)

If (ifail==0 .Or. ifail==12) Then
Write (nout,99999) t, (y(i),i=1,neq)

ifail = 0
Call d02nyf(neq,neq,hu,h,tcur,tolsf,rwork,nst,nre,nje,nqu,nq,niter, &

imxer,algequ,inform,ifail)

Write (nout,*)
Write (nout,99997) hu, h, tcur
Write (nout,99996) nst, nre, nje
Write (nout,99995) nqu, nq, niter
Write (nout,99994) ’ Max err comp = ’, imxer
icall = 0

Call d02nxf(icall,liwreq,liwusd,lrwreq,lrwusd,nlu,nnz,ngp,isplit, &
igrow,lblock,nblock,inform)

Write (nout,*)
Write (nout,99993) liwreq, liwusd
Write (nout,99992) lrwreq, lrwusd
Write (nout,99991) nlu, nnz
Write (nout,99990) ngp, isplit
Write (nout,99989) igrow, nblock

Else If (ifail==10) Then
icall = 1

Call d02nxf(icall,liwreq,liwusd,lrwreq,lrwusd,nlu,nnz,ngp,isplit, &
igrow,lblock,nblock,inform)

Write (nout,*)
Write (nout,99993) liwreq, liwusd
Write (nout,99992) lrwreq, lrwusd

Else
Write (nout,*)
Write (nout,99998) ’Exit D02NJF with IFAIL = ’, ifail, ’ and T = ’, &

t
End If

End Do cases

99999 Format (1X,F8.3,3(F13.5,2X))
99998 Format (1X,A,I5,A,E12.5)
99997 Format (1X,’ HUSED = ’,E12.5,’ HNEXT = ’,E12.5,’ TCUR = ’,E12.5)
99996 Format (1X,’ NST = ’,I6,’ NRE = ’,I6,’ NJE = ’,I6)
99995 Format (1X,’ NQU = ’,I6,’ NQ = ’,I6,’ NITER = ’,I6)
99994 Format (1X,A,I4)
99993 Format (1X,’ NJCPVT (required ’,I4,’ used ’,I8,’)’)
99992 Format (1X,’ NWKJAC (required ’,I4,’ used ’,I8,’)’)
99991 Format (1X,’ No. of LU-decomps ’,I4,’ No. of nonzeros ’,I8)
99990 Format (1X,’ No. of FCN calls to form Jacobian ’,I4,’ Try ISPLIT ’,I4)
99989 Format (1X,’ Growth est ’,I8,’ No. of blocks on diagonal ’,I4)
99988 Format (/,1X,’ X ’,3(’ Y(’,I1,’) ’))

End Program d02njfe

10.2 Program Data

D02NJF Example Program Data
5 200 5 : maxord, maxstp, mxhnil
1 3 6 9 : ia
1 2 1 2 3 1 2 3 : ja
1.0E-10 10.0 0.0 0.0 : hmin, hmax, h0, tcrit
1.0E-4 1.0E-6 0.1 : eta, sens, u
.TRUE. : lblock
0.0 10.0 : t, tout
3 0 : itol, isplit
1.0 0.0 0.0 : y
1.0E-4 1.0E-3 1.0E-4 : rtol
1.0E-7 : atol

10.3 Program Results

D02NJF Example Program Results

Analytic Jacobian, structure not supplied

X Y(1) Y(2) Y(3)
0.000 1.00000 0.00000 0.00000
4.957 0.89208 0.00002 0.10790

HUSED = 0.59971E+00 HNEXT = 0.59971E+00 TCUR = 0.49566E+01
NST = 52 NRE = 132 NJE = 12
NQU = 4 NQ = 4 NITER = 117
Max err comp = 3

NJCPVT (required 105 used 156)
NWKJAC (required 34 used 79)
No. of LU-decomps 12 No. of nonzeros 9
No. of FCN calls to form Jacobian 0 Try ISPLIT 73
Growth est 1034 No. of blocks on diagonal 1

Analytic Jacobian, structure supplied

X Y(1) Y(2) Y(3)
0.000 1.00000 0.00000 0.00000

4.957 0.89208 0.00002 0.10790

HUSED = 0.59971E+00 HNEXT = 0.59971E+00 TCUR = 0.49566E+01
NST = 52 NRE = 131 NJE = 12
NQU = 4 NQ = 4 NITER = 117
Max err comp = 3

NJCPVT (required 99 used 156)
NWKJAC (required 31 used 79)
No. of LU-decomps 12 No. of nonzeros 8
No. of FCN calls to form Jacobian 0 Try ISPLIT 72
Growth est 1034 No. of blocks on diagonal 1

 0

 0.2

 0.4

 0.6

 0.8

 1

 1.2

 0.001  0.01  0.1  1  10
0.00000

0.00001

0.00001

0.00002

0.00002

0.00003

0.00003

0.00004

So
lu

ti
on

 (
a,

c)

So
lu

ti
on

 (
b)

x(logscale)

Example Program
Stiff DAE with Sparse Jacobian

Stiff Robertson Problem using BDF with Newton Iterations

a

b

c

D02 – Ordinary Differential D02NJF

Mark 26 D02NJF.17 (last)


	D02NJF
	1 Purpose
	2 Specification
	3 Description
	4 References
	5 Arguments
	NEQ
	LDYSAV
	T
	TOUT
	Y
	YDOT
	RWORK
	RTOL
	ATOL
	ITOL
	INFORM
	RESID
	NEQ
	T
	Y
	YDOT
	R
	IRES

	YSAV
	SDYSAV
	JAC
	NEQ
	T
	Y
	YDOT
	H
	D
	J
	PDJ

	WKJAC
	NWKJAC
	JACPVT
	NJCPVT
	MONITR
	NEQ
	LDYSAV
	T
	HLAST
	HNEXT
	Y
	YDOT
	YSAV
	R
	ACOR
	IMON
	INLN
	HMIN
	HMAX
	NQU

	LDERIV
	ITASK
	ITRACE
	IFAIL

	6 Error Indicators and Warnings
	IFAIL=1
	IFAIL=2
	IFAIL=3
	IFAIL=4
	IFAIL=5
	IFAIL=6
	IFAIL=7
	IFAIL=8
	IFAIL=9
	IFAIL=10
	IFAIL=11
	IFAIL=12
	IFAIL=13
	IFAIL=14
	IFAIL=15
	IFAIL=-99
	IFAIL=-399
	IFAIL=-999

	7 Accuracy
	8 Parallelism and Performance
	9 Further Comments
	10 Example
	10.1 Program Text
	10.2 Program Data
	10.3 Program Results


	NAG Fortran Library Manual, Mark 26
	Copyright Statement
	Foreword
	Introduction
	How to Use the NAG Library and its Documentation
	Mark 26 NAG Fortran Library News
	Implementation-specific Details for Users
	Advice on Replacement Calls for Withdrawn/Superseded Routines
	Code Contributors
	Support from NAG
	Index

	Chapters of the Library
	A00 - Library Identification
	A00 Chapter Contents
	A00 Chapter Introduction

	A02 - Complex Arithmetic
	A02 Chapter Contents
	A02 Chapter Introduction

	C02 - Zeros of Polynomials
	C02 Chapter Contents
	C02 Chapter Introduction

	C05 - Roots of One or More Transcendental Equations
	C05 Chapter Contents
	C05 Chapter Introduction

	C06 - Summation of Series
	C06 Chapter Contents
	C06 Chapter Introduction

	C09 - Wavelet Transforms
	C09 Chapter Contents
	C09 Chapter Introduction

	D01 - Quadrature
	D01 Chapter Contents
	D01 Chapter Introduction

	D02 - Ordinary Differential Equations
	D02 Chapter Contents
	D02 Chapter Introduction
	D02M-N Sub-chapter Introduction

	D03 - Partial Differential Equations
	D03 Chapter Contents
	D03 Chapter Introduction

	D04 - Numerical Differentiation
	D04 Chapter Contents
	D04 Chapter Introduction

	D05 - Integral Equations
	D05 Chapter Contents
	D05 Chapter Introduction

	D06 - Mesh Generation
	D06 Chapter Contents
	D06 Chapter Introduction

	E01 - Interpolation
	E01 Chapter Contents
	E01 Chapter Introduction

	E02 - Curve and Surface Fitting
	E02 Chapter Contents
	E02 Chapter Introduction

	E04 - Minimizing or Maximizing a Function
	E04 Chapter Contents
	E04 Chapter Introduction

	E05 - Global Optimization of a Function
	E05 Chapter Contents
	E05 Chapter Introduction

	F - Linear Algebra
	F Chapter Contents
	F Chapter Introduction

	F01 - Matrix Operations, Including Inversion
	F01 Chapter Contents
	F01 Chapter Introduction

	F02 - Eigenvalues and Eigenvectors
	F02 Chapter Contents
	F02 Chapter Introduction

	F03 - Determinants
	F03 Chapter Contents
	F03 Chapter Introduction

	F04 - Simultaneous Linear Equations
	F04 Chapter Contents
	F04 Chapter Introduction

	F05 - Orthogonalization
	F05 Chapter Contents
	F05 Chapter Introduction

	F06 - Linear Algebra Support Routines
	F06 Chapter Contents
	F06 Chapter Introduction

	F07 - Linear Equations (LAPACK)
	F07 Chapter Contents
	F07 Chapter Introduction

	F08 - Least Squares and Eigenvalue Problems (LAPACK)
	F08 Chapter Contents
	F08 Chapter Introduction

	F11 - Large Scale Linear Systems
	F11 Chapter Contents
	F11 Chapter Introduction

	F12 - Large Scale Eigenproblems
	F12 Chapter Contents
	F12 Chapter Introduction

	F16 - Further Linear Algebra Support Routines
	F16 Chapter Contents
	F16 Chapter Introduction

	G01 - Simple Calculations on Statistical Data
	G01 Chapter Contents
	G01 Chapter Introduction

	G02 - Correlation and Regression Analysis
	G02 Chapter Contents
	G02 Chapter Introduction

	G03 - Multivariate Methods
	G03 Chapter Contents
	G03 Chapter Introduction

	G04 - Analysis of Variance
	G04 Chapter Contents
	G04 Chapter Introduction

	G05 - Random Number Generators
	G05 Chapter Contents
	G05 Chapter Introduction

	G07 - Univariate Estimation
	G07 Chapter Contents
	G07 Chapter Introduction

	G08 - Nonparametric Statistics
	G08 Chapter Contents
	G08 Chapter Introduction

	G10 - Smoothing in Statistics
	G10 Chapter Contents
	G10 Chapter Introduction

	G11 - Contingency Table Analysis
	G11 Chapter Contents
	G11 Chapter Introduction

	G12 - Survival Analysis
	G12 Chapter Contents
	G12 Chapter Introduction

	G13 - Time Series Analysis
	G13 Chapter Contents
	G13 Chapter Introduction

	H - Operations Research
	H Chapter Contents
	H Chapter Introduction

	M01 - Sorting and Searching
	M01 Chapter Contents
	M01 Chapter Introduction

	S - Approximations of Special Functions
	S Chapter Contents
	S Chapter Introduction

	X01 - Mathematical Constants
	X01 Chapter Contents
	X01 Chapter Introduction

	X02 - Machine Constants
	X02 Chapter Contents
	X02 Chapter Introduction

	X03 - Inner Products
	X03 Chapter Contents
	X03 Chapter Introduction

	X04 - Input/Output Utilities
	X04 Chapter Contents
	X04 Chapter Introduction

	X05 - Date and Time Utilities
	X05 Chapter Contents
	X05 Chapter Introduction

	X06 - OpenMP Utilities
	X06 Chapter Contents
	X06 Chapter Introduction

	X07 - IEEE Arithmetic
	X07 Chapter Contents
	X07 Chapter Introduction




