
NAG Library Routine Document

D02TGF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

D02TGF solves a system of linear ordinary differential equations by least squares fitting of a series of
Chebyshev polynomials using collocation.

2 Specification

SUBROUTINE D02TGF (N, M, L, X0, X1, K1, KP, C, LDC, COEFF, BDYC, W, LW,
IW, LIW, IFAIL)

&

INTEGER N, M(N), L(N), K1, KP, LDC, LW, IW(LIW), LIW, IFAIL
REAL (KIND=nag_wp) X0, X1, C(LDC,N), W(LW)
EXTERNAL COEFF, BDYC

3 Description

D02TGF calculates an approximate solution of a linear or linearized system of ordinary differential
equations as a Chebyshev series. Suppose there are n differential equations for n variables
y1; y2; . . . ; yn, over the range x0; x1ð Þ. Let the ith equation be

Xmiþ1

j¼1

Xn

k¼1

fi
kj xð Þy j�1ð Þ

k xð Þ ¼ ri xð Þ

where y
jð Þ
k xð Þ ¼ djyk xð Þ

dxj . COEFF evaluates the coefficients fikj xð Þ and the right-hand side ri xð Þ for each i,
1 � i � n, at any point x. The boundary conditions may be applied either at the end points or at
intermediate points; they are written in the same form as the differential equations, and specified by
BDYC. For example the jth boundary condition out of those associated with the ith differential
equation takes the form

Xliþ1

j¼1

Xn

k¼1

fij
kj xij
� �

y
j�1ð Þ
k xij

� � ¼ rij xij
� �

;

where xij lies between x0 and x1. It is assumed in this routine that certain of the boundary conditions
are associated with each differential equation. This is for your convenience; the grouping does not
affect the results.

The degree of the polynomial solution must be the same for all variables. You specify the degree
required, k1 � 1, and the number of collocation points, kp, in the range. The routine sets up a system of
linear equations for the Chebyshev coefficients, with n equations for each collocation point and one for
each boundary condition. The collocation points are chosen at the extrema of a shifted Chebyshev
polynomial of degree kp � 1. The boundary conditions are satisfied exactly, and the remaining equations
are solved by a least squares method. The result produced is a set of Chebyshev coefficients for the n
functions y1; y2; . . . ; yn, with the range normalized to �1; 1½ �.
E02AKF can be used to evaluate the components of the solution at any point on the range x0; x1½ � (see
Section 10 for an example). E02AHF and E02AJF may be used to obtain Chebyshev series
representations of derivatives and integrals (respectively) of the components of the solution.

4 References

Picken S M (1970) Algorithms for the solution of differential equations in Chebyshev-series by the
selected points method Report Math. 94 National Physical Laboratory

5 Arguments

1: N – INTEGER Input

On entry: n, the number of differential equations in the system.

Constraint: N � 1.

2: MðNÞ – INTEGER array Input

On entry: MðiÞ must be set to the highest order derivative occurring in the ith equation, for
i ¼ 1; 2; . . . ; n.

Constraint: MðiÞ � 1, for i ¼ 1; 2; . . . ; n.

3: LðNÞ – INTEGER array Input

On entry: LðiÞ must be set to the number of boundary conditions associated with the ith equation,
for i ¼ 1; 2; . . . ; n.

Constraint: LðiÞ � 0, for i ¼ 1; 2; . . . ; n.

4: X0 – REAL (KIND=nag_wp) Input

On entry: the left-hand boundary, x0.

5: X1 – REAL (KIND=nag_wp) Input

On entry: the right-hand boundary, x1.

Constraint: X1 > X0.

6: K1 – INTEGER Input

On entry: the number of coefficients, k1, to be returned in the Chebyshev series representation of
the solution (hence, the degree of the polynomial approximation is K1� 1).

Constraint: K1 � 1þ max
1�i�N

MðiÞ.

7: KP – INTEGER Input

On entry: the number of collocation points to be used, kp.

Constraint: N� KP � N� K1þ
XN

i¼1

LðiÞ.

8: CðLDC;NÞ – REAL (KIND=nag_wp) array Output

On exit: the kth column of C contains the computed Chebyshev coefficients of the kth component
of the solution, yk; that is, the computed solution is:

yk ¼
Xk1

i¼1

Cði; kÞTi�1 xð Þ; 1 � k � n;

where Ti xð Þ is the Chebyshev polynomial of the first kind and
P

denotes that the first
coefficient, Cð1; kÞ, is halved.

D02TGF NAG Library Manual

D02TGF.2 Mark 26



9: LDC – INTEGER Input

On entry: the first dimension of the array C as declared in the (sub)program from which D02TGF
is called.

Constraint: LDC � K1.

10: COEFF – SUBROUTINE, supplied by the user. External Procedure

COEFF defines the system of differential equations (see Section 3). It must evaluate the
coefficient functions fi

kj xð Þ and the right-hand side function ri xð Þ of the ith equation at a given
point. Only nonzero entries of the array A and RHS need be specifically assigned, since all
elements are set to zero by D02TGF before calling COEFF.

The specification of COEFF is:

SUBROUTINE COEFF (X, I, A, IA, IA1, RHS)

INTEGER I, IA, IA1
REAL (KIND=nag_wp) X, A(IA,IA1), RHS

Important: the dimension declaration for A must contain the variable IA, not an integer
constant.

1: X – REAL (KIND=nag_wp) Input

On entry: x, the point at which the functions must be evaluated.

2: I – INTEGER Input

On entry: the equation for which the coefficients and right-hand side are to be
evaluated.

3: AðIA; IA1Þ – REAL (KIND=nag_wp) array Input/Output

On entry: all elements of A are set to zero.

On exit: Aðk; jÞ must contain the value fi
kj xð Þ, for 1 � k � n, 1 � j � mi þ 1.

4: IA – INTEGER Input
5: IA1 – INTEGER Input

On entry: the first dimension of the array A and the second dimension of the array A as
declared in the (sub)program from which D02TGF is called.

6: RHS – REAL (KIND=nag_wp) Input/Output

On entry: is set to zero.

On exit: it must contain the value ri xð Þ.

COEFF must either be a module subprogram USEd by, or declared as EXTERNAL in, the (sub)
program from which D02TGF is called. Arguments denoted as Input must not be changed by this
procedure.

11: BDYC – SUBROUTINE, supplied by the user. External Procedure

BDYC defines the boundary conditions (see Section 3). It must evaluate the coefficient functions
fij
kj and right-hand side function rij in the jth boundary condition associated with the ith

equation, at the point xij at which the boundary condition is applied. Only nonzero entries of the
array A and RHS need be specifically assigned, since all elements are set to zero by D02TGF
before calling BDYC.

The specification of BDYC is:

SUBROUTINE BDYC (X, I, J, A, IA, IA1, RHS)

INTEGER I, J, IA, IA1
REAL (KIND=nag_wp) X, A(IA,IA1), RHS

Important: the dimension declaration for A must contain the variable IA, not an integer
constant.

1: X – REAL (KIND=nag_wp) Output

On exit: xij, the value at which the boundary condition is applied.

2: I – INTEGER Input

On entry: the differential equation with which the condition is associated.

3: J – INTEGER Input

On entry: the boundary condition for which the coefficients and right-hand side are to
be evaluated.

4: AðIA; IA1Þ – REAL (KIND=nag_wp) array Input/Output

On entry: all elements of A are set to zero.

On exit: the value fij
kj xijð Þ, for 1 � k � n, 1 � j � mi þ 1.

5: IA – INTEGER Input
6: IA1 – INTEGER Input

On entry: the first dimension of the array A and the second dimension of the array A as
declared in the (sub)program from which D02TGF is called.

7: RHS – REAL (KIND=nag_wp) Input/Output

On entry: is set to zero.

On exit: the value rij xijð Þ.

BDYC must either be a module subprogram USEd by, or declared as EXTERNAL in, the (sub)
program from which D02TGF is called. Arguments denoted as Input must not be changed by this
procedure.

12: WðLWÞ – REAL (KIND=nag_wp) array Workspace
13: LW – INTEGER Input

On entry: the dimension of the array W as declared in the (sub)program from which D02TGF is
called.

Constraint: LW � 2� N� KPþ NLð Þ � N� K1þ 1ð Þ þ 7� N� K1, where NL ¼
Xn

i¼1

LðiÞ .

14: IWðLIWÞ – INTEGER array Workspace
15: LIW – INTEGER Input

On entry: the dimension of the array IW as declared in the (sub)program from which D02TGF is
called.

Constraint: LIW � N� K1þ 1.

16: IFAIL – INTEGER Input/Output

On entry: IFAIL must be set to 0, �1 or 1. If you are unfamiliar with this argument you should
refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is
detected, the value �1 or 1 is recommended. If the output of error messages is undesirable, then
the value 1 is recommended. Otherwise, if you are not familiar with this argument, the
recommended value is 0. When the value �1 or 1 is used it is essential to test the value of
IFAIL on exit.

On exit: IFAIL ¼ 0 unless the routine detects an error or a warning has been flagged (see
Section 6).

6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL ¼ 1

On entry, N < 1,
or MðiÞ < 1 for some i,
or LðiÞ < 0 for some i,
or X0 � X1,
or K1 < 1þMðiÞ for some i,

or N� KP < N� K1þ
Xn

i¼1

LðiÞ,
or LDC < K1.

IFAIL ¼ 2

On entry, LW is too small (see Section 5),
or LIW < N� K1.

IFAIL ¼ 3

Either the boundary conditions are not linearly independent, or the rank of the matrix of
equations for the coefficients is less than the number of unknowns. Increasing KP may overcome
this latter problem.

IFAIL ¼ 4

The least squares routine F04AMF has failed to correct the first approximate solution (see
F04AMF). Increasing KP may remove this difficulty.

IFAIL ¼ �99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

Estimates of the accuracy of the solution may be obtained by using the checks described in Section 9.
The Chebyshev coefficients are calculated by a stable numerical method.

8 Parallelism and Performance

D02TGF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor
library used by this implementation. Consult the documentation for the vendor library for further
information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the
OpenMP environment used within this routine. Please also consult the Users' Note for your
implementation for any additional implementation-specific information.

9 Further Comments

The time taken by D02TGF depends on the complexity of the system of differential equations, the
degree of the polynomial solution and the number of matching points.

If the number of matching points kp is equal to the number of coefficients k1 minus the average number

of boundary conditions 1
n

Xn

i¼1

li , then the least squares solution reduces to simple solution of linear

equations and true collocation results. The accuracy of the solution may be checked by repeating the
calculation with different values of k1. If the Chebyshev coefficients decrease rapidly, the size of the
last two or three gives an indication of the error. If they do not decrease rapidly, it may be desirable to
use a different method. Note that the Chebyshev coefficients are calculated for the range normalized to
�1; 1½ �.
Generally the number of boundary conditions required is equal to the sum of the orders of the n
differential equations. However, in some cases fewer boundary conditions are needed, because the
assumption of a polynomial solution is equivalent to one or more boundary conditions (since it excludes
singular solutions).

A system of nonlinear differential equations must be linearized before using the routine. The
calculation is repeated iteratively. On each iteration the linearized equation is used. In the example in
Section 10, the y variables are to be determined at the current iteration whilst the z variables correspond
to the solution determined at the previous iteration, (or the initial approximation on the first iteration).
For a starting approximation, we may take, say, a linear function, and set up the appropriate Chebyshev
coefficients before starting the iteration. For example, if y1 ¼ axþ b in the range x0; x1ð Þ, we set B, the
array of coefficients,

B 1; 1ð Þ ¼ a� x0 þ x1ð Þ þ 2� b,

B 1; 2ð Þ ¼ a� x1 � x0ð Þ=2,
and the remainder of the entries to zero.

In some cases a better initial approximation may be needed and can be obtained by using E02ADF or
E02AFF to obtain a Chebyshev series for an approximate solution. The coefficients of the current
iterate must be communicated to COEFF and BDYC, for example using global variables. (See
Section 10.) The convergence of the (Newton) iteration cannot be guaranteed in general, though it is
usually satisfactory from a good starting approximation.

10 Example

This example solves the nonlinear system

2y01 þ y22 � 1
� �

y1 þ y2 ¼ 0;

2y002 � y01 ¼ 0;

in the range �1; 1ð Þ, with y1 ¼ 0, y2 ¼ 3, y02 ¼ 0 at x ¼ �1.

Suppose an approximate solution is z1, z2 such that y1 � z1, y2 � z2: then the first equation gives, on
linearizing,

2y01 þ z22 � 1
� �

y1 þ 2z1z2 þ 1ð Þy2 ¼ 2z1z
2
2:

The starting approximation is taken to be z1 ¼ 0, z2 ¼ 3. In the program below, the array B is used to
hold the coefficients of the previous iterate (or of the starting approximation). We iterate until the
Chebyshev coefficients converge to five figures. E02AKF is used to calculate the solution from its
Chebyshev coefficients.

10.1 Program Text

! D02TGF Example Program Text
! Mark 26 Release. NAG Copyright 2016.

Module d02tgfe_mod

! D02TGF Example Program Module:
! Parameters and User-defined Routines

! .. Use Statements ..
Use nag_library, Only: nag_wp

! .. Implicit None Statement ..
Implicit None

! .. Accessibility Statements ..
Private
Public :: bdyc, coeff

! .. Parameters ..
Real (Kind=nag_wp), Parameter, Public :: coeff_tol = 1.0E-5_nag_wp
Real (Kind=nag_wp), Parameter :: one = 1.0_nag_wp
Real (Kind=nag_wp), Parameter :: two = 2.0_nag_wp
Integer, Parameter, Public :: n = 2, nin = 5, nout = 6

! .. Local Scalars ..
Real (Kind=nag_wp), Public, Save :: x0, x1
Integer, Public, Save :: k1

! .. Local Arrays ..
Real (Kind=nag_wp), Allocatable, Public, Save :: b(:,:)
Integer, Public, Save :: l(n) = (/1,2/)
Integer, Public, Save :: m(n) = (/1,2/)

Contains
Subroutine coeff(x,i,a,ia,ia1,rhs)

! .. Use Statements ..
Use nag_library, Only: e02akf

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (Inout) :: rhs
Real (Kind=nag_wp), Intent (In) :: x
Integer, Intent (In) :: i, ia, ia1

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Inout) :: a(ia,ia1)

! .. Local Scalars ..
Real (Kind=nag_wp) :: z1, z2
Integer :: ifail

! .. Executable Statements ..
If (i<=1) Then

! Evaluate z1, z2 at x using previous coeffs b.
ifail = 0
Call e02akf(k1,x0,x1,b(1,1),1,k1,x,z1,ifail)

Call e02akf(k1,x0,x1,b(1,2),1,k1,x,z2,ifail)

a(1,1) = z2*z2 - one
a(1,2) = two
a(2,1) = two*z1*z2 + one
rhs = two*z1*z2*z2

Else
a(1,2) = -one
a(2,3) = two

End If
Return

End Subroutine coeff
Subroutine bdyc(x,i,j,a,ia,ia1,rhs)

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (Inout) :: rhs
Real (Kind=nag_wp), Intent (Out) :: x
Integer, Intent (In) :: i, ia, ia1, j

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Inout) :: a(ia,ia1)

! .. Executable Statements ..
x = -one
a(i,j) = one
If (i==2 .And. j==1) Then

rhs = 3.0_nag_wp
End If
Return

End Subroutine bdyc
End Module d02tgfe_mod
Program d02tgfe

! D02TGF Example Main Program

! .. Use Statements ..
Use nag_library, Only: d02tgf, e02akf, nag_wp
Use d02tgfe_mod, Only: b, bdyc, coeff, coeff_tol, k1, l, m, n, nin, &

nout, x0, x1
! .. Implicit None Statement ..

Implicit None
! .. Local Scalars ..

Real (Kind=nag_wp) :: emax, x, xinc
Integer :: i, ia1, ifail, iter, j, k, kp, ldc, &

liw, lsum, lw, mimax
! .. Local Arrays ..

Real (Kind=nag_wp), Allocatable :: c(:,:), w(:), y(:)
Integer, Allocatable :: iw(:)

! .. Intrinsic Procedures ..
Intrinsic :: abs, max, real, sum

! .. Executable Statements ..
Write (nout,*) ’D02TGF Example Program Results’

! Skip heading in data file
Read (nin,*)
Read (nin,*) mimax, kp
Read (nin,*) x0, x1

lsum = sum(l(1:n))
k1 = mimax + 1
ldc = k1
liw = n*k1
lw = 2*(n*kp+lsum)*(n*k1+1) + 7*n*k1
Allocate (b(k1,n),c(ldc,n),w(lw),y(n),iw(liw))

! Initialize coefficients b(:,:) such that z1 = 0 and z2 = 3.
b(1:k1,1:n) = 0.0_nag_wp
b(1,2) = 6.0_nag_wp

! Iterate until coefficients of linearized systems converge.
iter = 0
emax = 1.0_nag_wp

iters: Do While (emax>=coeff_tol)
iter = iter + 1

Write (nout,*)
Write (nout,99999) ’ Iteration’, iter, ’ Chebyshev coefficients are’

! ifail: behaviour on error exit
! =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft

ifail = 0
Call d02tgf(n,m,l,x0,x1,k1,kp,c,ldc,coeff,bdyc,w,lw,iw,liw,ifail)

Write (nout,99998)(c(1:k1,j),j=1,n)
emax = 0.0_nag_wp
Do j = 1, n

Do i = 1, k1
emax = max(emax,abs(c(i,j)-b(i,j)))

End Do
End Do
b(1:k1,1:n) = c(1:k1,1:n)

End Do iters

Deallocate (b)

! Print solution on uniform mesh.
k = 9
ia1 = 1
Write (nout,*)
Write (nout,99999) ’Solution evaluated at’, k, ’ equally spaced points’
Write (nout,*)
Write (nout,99997) ’ X ’, (j,j=1,n)
xinc = (x1-x0)/real(k-1,kind=nag_wp)
x = x0
Do i = 1, k

Do j = 1, n
ifail = 0
Call e02akf(k1,x0,x1,c(1,j),ia1,k1,x,y(j),ifail)

End Do
Write (nout,99996) x, (y(j),j=1,n)
x = x + xinc

End Do

99999 Format (1X,A,I3,A)
99998 Format (1X,9F8.4)
99997 Format (1X,A,2(’ Y(’,I1,’)’))
99996 Format (1X,3F10.4)

End Program d02tgfe

10.2 Program Data

D02TGF Example Program Data
8 15 : mimax, kp
-1.0 1.0 : x0, x1

10.3 Program Results

D02TGF Example Program Results

Iteration 1 Chebyshev coefficients are
-0.5659 -0.1162 0.0906 -0.0468 0.0196 -0.0069 0.0021 -0.0006 0.0001
5.7083 -0.1642 -0.0087 0.0059 -0.0025 0.0009 -0.0003 0.0001 -0.0000

Iteration 2 Chebyshev coefficients are
-0.6338 -0.1599 0.0831 -0.0445 0.0193 -0.0071 0.0023 -0.0006 0.0001
5.6881 -0.1792 -0.0144 0.0053 -0.0023 0.0008 -0.0003 0.0001 -0.0000

Iteration 3 Chebyshev coefficients are
-0.6344 -0.1604 0.0828 -0.0446 0.0193 -0.0071 0.0023 -0.0006 0.0001
5.6880 -0.1793 -0.0145 0.0053 -0.0023 0.0008 -0.0003 0.0001 -0.0000

Iteration 4 Chebyshev coefficients are
-0.6344 -0.1604 0.0828 -0.0446 0.0193 -0.0071 0.0023 -0.0006 0.0001
5.6880 -0.1793 -0.0145 0.0053 -0.0023 0.0008 -0.0003 0.0001 -0.0000

Solution evaluated at 9 equally spaced points

X Y(1) Y(2)
-1.0000 0.0000 3.0000
-0.7500 -0.2372 2.9827
-0.5000 -0.3266 2.9466
-0.2500 -0.3640 2.9032
0.0000 -0.3828 2.8564
0.2500 -0.3951 2.8077
0.5000 -0.4055 2.7577
0.7500 -0.4154 2.7064
1.0000 -0.4255 2.6538

