
NAG Library Routine Document

D02TZF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

D02TZF returns information about the solution of a general two-point boundary value problem
computed by D02TLF.

2 Specification

SUBROUTINE D02TZF (MXMESH, NMESH, MESH, IPMESH, ERMX, IERMX, IJERMX,
RCOMM, ICOMM, IFAIL)

&

INTEGER MXMESH, NMESH, IPMESH(MXMESH), IERMX, IJERMX,
ICOMM(*), IFAIL

&

REAL (KIND=nag_wp) MESH(MXMESH), ERMX, RCOMM(*)

3 Description

D02TZF and its associated routines (D02TLF, D02TVF, D02TXF and D02TYF) solve the two-point
boundary value problem for a nonlinear mixed order system of ordinary differential equations

First, D02TVF must be called to specify the initial mesh, error requirements and other details. Then,
D02TLF can be used to solve the boundary value problem. After successful computation, D02TZF can
be used to ascertain details about the final mesh. D02TYF can be used to compute the approximate
solution anywhere on the interval a; b½ � using interpolation.

The routines are based on modified versions of the codes COLSYS and COLNEW (see Ascher et al.
(1979) and Ascher and Bader (1987)). A comprehensive treatment of the numerical solution of
boundary value problems can be found in Ascher et al. (1988) and Keller (1992).

5 Arguments

1: MXMESH – INTEGER Input

On entry: the maximum number of points allowed in the mesh.

Constraint: this must be identical to the value supplied for the argument MXMESH in the prior
call to D02TVF.

2: NMESH – INTEGER Output

On exit: the number of points in the mesh last used by D02TLF.

3: MESHðMXMESHÞ – REAL (KIND=nag_wp) array Output

On exit: MESHðiÞ contains the ith point of the mesh last used by D02TLF, for
i ¼ 1; 2; . . . ;NMESH. MESHð1Þ will contain a and MESHðNMESHÞ will contain b. The
remaining elements of MESH are not initialized.

4: IPMESHðMXMESHÞ – INTEGER array Output

On exit: IPMESHðiÞ specifies the nature of the point MESHðiÞ, for i ¼ 1; 2; . . . ;NMESH, in the
final mesh computed by D02TLF.

IPMESHðiÞ ¼ 1
Indicates that the ith point is a fixed point and was used by the solver before an
extrapolation-like error test.

IPMESHðiÞ ¼ 2
Indicates that the ith point was used by the solver before an extrapolation-like error test.

IPMESHðiÞ ¼ 3
Indicates that the ith point was used by the solver only as part of an extrapolation-like
error test.

The remaining elements of IPMESH are initialized to �1.

See Section 9 for advice on how these values may be used in conjunction with a continuation
process.

5: ERMX – REAL (KIND=nag_wp) Output

On exit: an estimate of the maximum error in the solution computed by D02TLF, that is

ERMX ¼ max
yi � vik k

1:0þ vik kð Þ
where vi is the approximate solution for the ith solution component. If D02TLF returned
successfully with IFAIL ¼ 0, then ERMX will be less than TOLSðIJERMXÞ where TOLS
contains the error requirements as specified in Sections 3 and 5 in D02TVF.

If D02TLF returned with IFAIL ¼ 5, then ERMX will be greater than TOLSðIJERMXÞ.

If D02TLF returned any other value for IFAIL then an error estimate is not available and ERMX
is initialized to 0:0.

6: IERMX – INTEGER Output

On exit: indicates the mesh sub-interval where the value of ERMX has been computed, that is
MESHðIERMXÞ;MESHðIERMXþ 1Þ½ �.
If an estimate of the error is not available then IERMX is initialized to 0.

7: IJERMX – INTEGER Output

On exit: indicates the component i ( ¼ IJERMX) of the solution for which ERMX has been
computed, that is the approximation of yi on MESHðIERMXÞ;MESHðIERMXþ 1Þ½ � is estimated
to have the largest error of all components yi over mesh sub-intervals defined by MESH.

If an estimate of the error is not available then IJERMX is initialized to 0.

8: RCOMMð�Þ – REAL (KIND=nag_wp) array Communication Array

Note: the dimension of this array is dictated by the requirements of associated functions that
must have been previously called. This array must be the same array passed as argument
RCOMM in the previous call to D02TLF.

On entry: this must be the same array as supplied to D02TLF and must remain unchanged
between calls.

On exit: contains information about the solution for use on subsequent calls to associated
routines.

9: ICOMMð�Þ – INTEGER array Communication Array

Note: the dimension of this array is dictated by the requirements of associated functions that
must have been previously called. This array must be the same array passed as argument
ICOMM in the previous call to D02TLF.

On entry: this must be the same array as supplied to D02TLF and must remain unchanged
between calls.

On exit: contains information about the solution for use on subsequent calls to associated
routines.

10: IFAIL – INTEGER Input/Output

On entry: IFAIL must be set to 0, �1 or 1. If you are unfamiliar with this argument you should
refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is
detected, the value �1 or 1 is recommended. If the output of error messages is undesirable, then
the value 1 is recommended. Otherwise, because for this routine the values of the output
arguments may be useful even if IFAIL 6¼ 0 on exit, the recommended value is �1. When the
value �1 or 1 is used it is essential to test the value of IFAIL on exit.

On exit: IFAIL ¼ 0 unless the routine detects an error or a warning has been flagged (see
Section 6).

6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Note: D02TZF may return useful information for one or more of the following detected errors or
warnings.

Errors or warnings detected by the routine:

IFAIL ¼ 1

On entry, an illegal value for MXMESH was specified, or an invalid call to D02TZF was made,
for example without a previous call to the solver routine D02TLF.

IFAIL ¼ 2

The solver routine D02TLF did not converge to a solution or did not satisfy the error
requirements. The last mesh computed by D02TLF has been returned by D02TZF. This mesh
should be treated with extreme caution as nothing can be said regarding its quality or suitability
for any subsequent computation.

IFAIL ¼ �99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

Not applicable.

8 Parallelism and Performance

D02TZF is not threaded in any implementation.

9 Further Comments

Note that:

if D02TLF returned IFAIL ¼ 0, 4 or 5 then it will always be the case that
IPMESHð1Þ ¼ IPMESHðNMESHÞ ¼ 1;

if D02TLF returned IFAIL ¼ 0 or 5 then it will always be the case that IPMESHðiÞ ¼ 3, for
i ¼ 2; 4; . . . ;NMESH� 1 (even i) and IPMESHðiÞ ¼ 1 or 2, for i ¼ 3; 5; . . . ;NMESH� 2 (odd
i);

if D02TLF returned IFAIL ¼ 4 then it will always be the case that IPMESHðiÞ ¼ 1 or 2, for
i ¼ 2; 3; . . . ;NMESH� 1.

If D02TZF returns IFAIL ¼ 0, then examination of the mesh may provide assistance in determining a
suitable starting mesh for D02TVF in any subsequent attempts to solve similar problems.

If the problem being treated by D02TLF is one of a series of related problems (for example, as part of a
continuation process), then the values of IPMESH and MESH may be suitable as input arguments to
D02TXF. Using the mesh points not involved in the extrapolation error test is usually appropriate.
IPMESH and MESH should be passed unchanged to D02TXF but NMESH should be replaced by
NMESHþ 1ð Þ=2.
If D02TZF returns IFAIL ¼ 2, nothing can be said regarding the quality of the mesh returned. However,
it may be a useful starting mesh for D02TVF in any subsequent attempts to solve the same problem.

If D02TLF returns IFAIL ¼ 5, this corresponds to the solver requiring more than MXMESH mesh
points to satisfy the error requirements. If MXMESH can be increased and the preceding call to
D02TLF was not part, or was the first part, of a continuation process then the values in MESH may
provide a suitable mesh with which to initialize a subsequent attempt to solve the same problem. If it is
not possible to provide more mesh points then relaxing the error requirements by setting
TOLSðIJERMXÞ to ERMX might lead to a successful solution. It may be necessary to reset the
other components of TOLS. Note that resetting the tolerances can lead to a different sequence of
meshes being computed and hence to a different solution being computed.

10 Example

The following example is used to illustrate the use of fixed mesh points, simple continuation and
numerical approximation of a Jacobian. See also D02TLF, D02TVF, D02TXF and D02TYF, for the
illustration of other facilities.

Consider the Lagerstrom–Cole equation

y00 ¼ y� yy0ð Þ=�
with the boundary conditions

y 0ð Þ ¼ � y 1ð Þ ¼ �; ð1Þ
where � is small and positive. The nature of the solution depends markedly on the values of �; �. See
Cole (1968).

We choose � ¼ �1
3; � ¼ 1

3 for which the solution is known to have corner layers at x ¼ 1
3;

2
3 . We choose

an initial mesh of seven points 0:0; 0:15; 0:3; 0:5; 0:7; 0:85; 1:0½ � and ensure that the points x ¼ 0:3; 0:7
near the corner layers are fixed, that is the corresponding elements of the array IPMESH are set to 1.
First we compute the solution for � ¼ 1:0E�4 using in GUESS the initial approximation
y xð Þ ¼ �þ � � �ð Þx which satisfies the boundary conditions. Then we use simple continuation to
compute the solution for � ¼ 1:0E�5. We use the suggested values for NMESH, IPMESH and MESH in
the call to D02TXF prior to the continuation call, that is only every second point of the preceding mesh
is used and the fixed mesh points are retained.

Although the analytic Jacobian for this system is easy to evaluate, for illustration the procedure FJAC
uses central differences and calls to FFUN to compute a numerical approximation to the Jacobian.

10.1 Program Text

! D02TZF Example Program Text
! Mark 26 Release. NAG Copyright 2016.

Module d02tzfe_mod

! D02TZF Example Program Module:
! Parameters and User-defined Routines

! .. Use Statements ..
Use nag_library, Only: nag_wp

! .. Implicit None Statement ..
Implicit None

! .. Accessibility Statements ..
Private
Public :: ffun, fjac, gafun, gajac, gbfun, &

gbjac, guess
! .. Parameters ..

Integer, Parameter, Public :: mmax = 2, neq = 1, nin = 5, &
nlbc = 1, nout = 6, nrbc = 1

! .. Local Scalars ..
Real (Kind=nag_wp), Public, Save :: alpha, beta, eps

! .. Local Arrays ..
Integer, Public, Save :: m(1) = (/2/)

Contains
Subroutine ffun(x,y,neq,m,f,iuser,ruser)

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: x
Integer, Intent (In) :: neq

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Out) :: f(neq)
Real (Kind=nag_wp), Intent (Inout) :: ruser(*)
Real (Kind=nag_wp), Intent (In) :: y(neq,0:*)
Integer, Intent (Inout) :: iuser(*)
Integer, Intent (In) :: m(neq)

! .. Executable Statements ..
f(1) = (y(1,0)-y(1,0)*y(1,1))/eps
Return

End Subroutine ffun
Subroutine fjac(x,y,neq,m,dfdy,iuser,ruser)

! .. Use Statements ..
Use nag_library, Only: x02ajf

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: x
Integer, Intent (In) :: neq

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Inout) :: dfdy(neq,neq,0:*), ruser(*)
Real (Kind=nag_wp), Intent (In) :: y(neq,0:*)
Integer, Intent (Inout) :: iuser(*)
Integer, Intent (In) :: m(neq)

! .. Local Scalars ..
Real (Kind=nag_wp) :: epsh, fac, ptrb
Integer :: i, j, k

! .. Local Arrays ..
Real (Kind=nag_wp) :: f1(1), f2(1), yp(1,0:3)

! .. Intrinsic Procedures ..
Intrinsic :: abs, max, sqrt

! .. Executable Statements ..
epsh = 100.0_nag_wp*x02ajf()
fac = sqrt(x02ajf())
Do i = 1, neq

Do j = 0, m(i) - 1
yp(i,j) = y(i,j)

End Do
End Do
Do i = 1, neq

Do j = 0, m(i) - 1
ptrb = max(epsh,fac*abs(y(i,j)))
yp(i,j) = y(i,j) + ptrb
Call ffun(x,yp,neq,m,f1,iuser,ruser)
yp(i,j) = y(i,j) - ptrb
Call ffun(x,yp,neq,m,f2,iuser,ruser)
Do k = 1, neq

dfdy(k,i,j) = 0.5_nag_wp*(f1(k)-f2(k))/ptrb
End Do
yp(i,j) = y(i,j)

End Do
End Do
Return

End Subroutine fjac
Subroutine gafun(ya,neq,m,nlbc,ga,iuser,ruser)

! .. Scalar Arguments ..
Integer, Intent (In) :: neq, nlbc

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Out) :: ga(nlbc)
Real (Kind=nag_wp), Intent (Inout) :: ruser(*)
Real (Kind=nag_wp), Intent (In) :: ya(neq,0:*)
Integer, Intent (Inout) :: iuser(*)
Integer, Intent (In) :: m(neq)

! .. Executable Statements ..
ga(1) = ya(1,0) - alpha
Return

End Subroutine gafun
Subroutine gbfun(yb,neq,m,nrbc,gb,iuser,ruser)

! .. Scalar Arguments ..
Integer, Intent (In) :: neq, nrbc

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Out) :: gb(nrbc)
Real (Kind=nag_wp), Intent (Inout) :: ruser(*)
Real (Kind=nag_wp), Intent (In) :: yb(neq,0:*)
Integer, Intent (Inout) :: iuser(*)
Integer, Intent (In) :: m(neq)

! .. Executable Statements ..
gb(1) = yb(1,0) - beta
Return

End Subroutine gbfun
Subroutine gajac(ya,neq,m,nlbc,dgady,iuser,ruser)

! .. Parameters ..
Real (Kind=nag_wp), Parameter :: one = 1.0_nag_wp

! .. Scalar Arguments ..
Integer, Intent (In) :: neq, nlbc

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Inout) :: dgady(nlbc,neq,0:*), ruser(*)
Real (Kind=nag_wp), Intent (In) :: ya(neq,0:*)
Integer, Intent (Inout) :: iuser(*)
Integer, Intent (In) :: m(neq)

! .. Executable Statements ..
dgady(1,1,0) = one
Return

End Subroutine gajac
Subroutine gbjac(yb,neq,m,nrbc,dgbdy,iuser,ruser)

! .. Parameters ..
Real (Kind=nag_wp), Parameter :: one = 1.0_nag_wp

! .. Scalar Arguments ..
Integer, Intent (In) :: neq, nrbc

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Inout) :: dgbdy(nrbc,neq,0:*), ruser(*)
Real (Kind=nag_wp), Intent (In) :: yb(neq,0:*)
Integer, Intent (Inout) :: iuser(*)
Integer, Intent (In) :: m(neq)

! .. Executable Statements ..
dgbdy(1,1,0) = one
Return

End Subroutine gbjac
Subroutine guess(x,neq,m,y,dym,iuser,ruser)

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: x
Integer, Intent (In) :: neq

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Out) :: dym(neq)
Real (Kind=nag_wp), Intent (Inout) :: ruser(*), y(neq,0:*)
Integer, Intent (Inout) :: iuser(*)
Integer, Intent (In) :: m(neq)

! .. Executable Statements ..
y(1,0) = alpha + (beta-alpha)*x
y(1,1) = (beta-alpha)
dym(1) = 0.0_nag_wp
Return

End Subroutine guess
End Module d02tzfe_mod
Program d02tzfe

! D02TZF Example Main Program

! .. Use Statements ..
Use nag_library, Only: d02tlf, d02tvf, d02txf, d02tyf, d02tzf, nag_wp
Use d02tzfe_mod, Only: alpha, beta, eps, ffun, fjac, gafun, gajac, &

gbfun, gbjac, guess, m, mmax, neq, nin, nlbc, &
nout, nrbc

! .. Implicit None Statement ..
Implicit None

! .. Local Scalars ..

Real (Kind=nag_wp) :: ermx
Integer :: i, iermx, ifail, ijermx, j, licomm, &

lrcomm, mxmesh, ncol, nmesh
Logical :: failed

! .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: mesh(:), rcomm(:), tol(:), y(:,:)
Real (Kind=nag_wp) :: ruser(1)
Integer, Allocatable :: icomm(:), ipmesh(:)
Integer :: iuser(2)

! .. Executable Statements ..
Write (nout,*) ’D02TZF Example Program Results’
Write (nout,*)

! Skip heading in data file
Read (nin,*)
Read (nin,*) ncol, nmesh, mxmesh
Allocate (mesh(mxmesh),tol(neq),y(neq,0:mmax-1),ipmesh(mxmesh))

Read (nin,*) alpha, beta, eps
Read (nin,*) mesh(1:nmesh)
Read (nin,*) ipmesh(1:nmesh)
Read (nin,*) tol(1:neq)

! Workspace query to get size of rcomm and icomm
ifail = 0
Call d02tvf(neq,m,nlbc,nrbc,ncol,tol,mxmesh,nmesh,mesh,ipmesh,ruser,0, &

iuser,2,ifail)
lrcomm = iuser(1)
licomm = iuser(2)
Allocate (rcomm(lrcomm),icomm(licomm))

! Initialize
ifail = 0
Call d02tvf(neq,m,nlbc,nrbc,ncol,tol,mxmesh,nmesh,mesh,ipmesh,rcomm, &

lrcomm,icomm,licomm,ifail)

eps = 0.1_nag_wp*eps

contn: Do j = 1, 2
Write (nout,99997) tol(1), eps

! Solve
ifail = -1
Call d02tlf(ffun,fjac,gafun,gbfun,gajac,gbjac,guess,rcomm,icomm,iuser, &

ruser,ifail)
failed = ifail /= 0

! Extract mesh.
ifail = -1
Call d02tzf(mxmesh,nmesh,mesh,ipmesh,ermx,iermx,ijermx,rcomm,icomm, &

ifail)

! Print mesh statistics.
Write (nout,99996) nmesh, ermx, iermx, ijermx

If (failed) Then
Exit contn

End If

! Print solution at every second point on final mesh.
Write (nout,99999)
Do i = 1, nmesh, 2

ifail = -1
Call d02tyf(mesh(i),y,neq,mmax,rcomm,icomm,ifail)
Write (nout,99998) mesh(i), y(1,0), y(1,1)

End Do

If (j==1) Then
! Halve final mesh for new initial mesh and set up for continuation.

nmesh = (nmesh+1)/2
ifail = 0
Call d02txf(mxmesh,nmesh,mesh,ipmesh,rcomm,icomm,ifail)

! Reduce continuation parameter.
eps = 0.1_nag_wp*eps

End If
End Do contn

99999 Format (/,’ Solution and derivative at every second point:’,/, &
’ x u u’’’)

99998 Format (’ ’,F8.4,2F11.5)
99997 Format (/,/,’ Tolerance = ’,E8.1,’ EPS = ’,E10.3)
99996 Format (/,’ Used a mesh of ’,I4,’ points’,/,’ Maximum error = ’,E10.2, &

’ in interval ’,I4,’ for component ’,I4)
End Program d02tzfe

10.2 Program Data

D02TZF Example Program Data
5 7 50 : ncol, nmesh, mxmesh

-0.333333333333333333333
0.333333333333333333333 0.001 : alpha, beta, eps
0.0 0.15 0.3 0.5 0.7 0.85 1.0 : mesh(1:nmesh)
1 2 1 2 1 2 1 : ipmesh(1:nmesh)

1.0E-5 : tol

10.3 Program Results

D02TZF Example Program Results

Tolerance = 0.1E-04 EPS = 0.100E-03

Used a mesh of 25 points
Maximum error = 0.21E-05 in interval 16 for component 1

Solution and derivative at every second point:
x u u’

0.0000 -0.33333 1.00000
0.0750 -0.25833 1.00000
0.1500 -0.18333 1.00000
0.2250 -0.10833 1.00002
0.3000 -0.03332 1.00372
0.4000 -0.00001 0.00084
0.5000 -0.00000 0.00000
0.6000 0.00001 0.00084
0.7000 0.03332 1.00372
0.7750 0.10833 1.00002
0.8500 0.18333 1.00000
0.9250 0.25833 1.00000
1.0000 0.33333 1.00000

Tolerance = 0.1E-04 EPS = 0.100E-04

Used a mesh of 49 points
Maximum error = 0.21E-05 in interval 32 for component 1

Solution and derivative at every second point:
x u u’

0.0000 -0.33333 1.00014
0.0375 -0.29583 1.00018
0.0750 -0.25833 1.00022
0.1125 -0.22083 1.00029
0.1500 -0.18333 1.00040
0.1875 -0.14583 1.00059
0.2250 -0.10833 1.00098
0.2625 -0.07083 1.00202
0.3000 -0.03333 1.00745
0.3500 -0.00001 0.00354
0.4000 -0.00000 0.00000
0.4500 -0.00000 0.00000

0.5000 -0.00000 0.00000
0.5500 0.00000 0.00000
0.6000 0.00000 0.00000
0.6500 0.00001 0.00354
0.7000 0.03333 1.00745
0.7375 0.07083 1.00202
0.7750 0.10833 1.00098
0.8125 0.14583 1.00059
0.8500 0.18333 1.00040
0.8875 0.22083 1.00029
0.9250 0.25833 1.00022
0.9625 0.29583 1.00018
1.0000 0.33333 1.00014

