
NAG Library Routine Document

D03UBF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

D03UBF performs at each call one iteration of the Strongly Implicit Procedure. It is used to calculate
on successive calls a sequence of approximate corrections to the current estimate of the solution when
solving a system of simultaneous algebraic equations for which the iterative update matrix is of seven-
point molecule form on a three-dimensional topologically-rectangular mesh. (‘Topological’ means that a
polar grid r; �ð Þ, for example, can be used as it is equivalent to a rectangular box.)

2 Specification

SUBROUTINE D03UBF (N1, N2, N3, LDA, SDA, A, B, C, D, E, F, G, APARAM, IT,
R, WRKSP1, WRKSP2, WRKSP3, IFAIL)

&

INTEGER N1, N2, N3, LDA, SDA, IT, IFAIL
REAL (KIND=nag_wp) A(LDA,SDA,N3), B(LDA,SDA,N3), C(LDA,SDA,N3),

D(LDA,SDA,N3), E(LDA,SDA,N3), F(LDA,SDA,N3),
G(LDA,SDA,N3), APARAM, R(LDA,SDA,N3),
WRKSP1(LDA,SDA,N3), WRKSP2(LDA,SDA,N3),
WRKSP3(LDA,SDA,N3)

&
&
&
&

3 Description

Given a set of simultaneous equations

Mt ¼ q ð1Þ
(which could be nonlinear) derived, for example, from a finite difference representation of a three-
dimensional elliptic partial differential equation and its boundary conditions, the solution t may be
obtained iteratively from a starting approximation t 1ð Þ by the formulae

r nð Þ ¼ q �Mt nð Þ

Ms nð Þ ¼ r nð Þ

t nþ1ð Þ ¼ t nð Þ þ s nð Þ:

Thus r nð Þ is the residual of the nth approximate solution t nð Þ, and s nð Þ is the update change vector.

D03UBF determines the approximate change vector s corresponding to a given residual r, i.e., it
determines an approximate solution to a set of equations

Ms ¼ r ð2Þ
where M is a square n1 � n2 � n3ð Þ by n1 � n2 � n3ð Þ matrix and r is a known vector of length
n1 � n2 � n3ð Þ. The set of equations (2) must be of seven-diagonal form

aijksij;k�1 þ bijksi;j�1;k þ cijksi�1;jk þ dijksijk þ eijksiþ1;jk þ fijksi;jþ1;k þ gijksij;kþ1 ¼ rijk

for i ¼ 1; 2; . . . ; n1, j ¼ 1; 2; . . . ; n2 and k ¼ 1; 2; . . . ; n3, provided that dijk 6¼ 0:0. Indeed, if dijk ¼ 0:0,
then the equation is assumed to be

sijk ¼ rijk:

The calling program supplies the current residual r at each iteration and the coefficients of the seven-
point molecule system of equations on which the update procedure is based. The routine performs one
iteration, using the approximate LU factorization of the Strongly Implicit Procedure with the necessary
acceleration argument adjustment, to calculate the approximate solution s of the set of equations (2).

The change s overwrites the residual array for return to the calling program. The calling program must
combine this change stored in r with the old approximation to obtain the new approximate solution for
t. It must then recalculate the residuals and, if the accuracy requirements have not been satisfied,
commence the next iterative cycle.

Clearly there is no requirement that the iterative update matrix passed in the form of the seven-diagonal
element arrays A, B, C, D, E, F and G is the same as that used to calculate the residuals, and therefore
the one governing the problem. However, the convergence may be impaired if they are not equal.
Indeed, if the system of equations (1) is not precisely of the seven-diagonal form illustrated above but
has a few additional terms, then the methods of deferred or defect correction can be employed. The
residual is calculated by the calling program using the full system of equations, but the update formula
is based on a seven-diagonal system (2) of the form given above. For example, the solution of a system
of e l even -d i agona l equa t i ons each invo lv ing the combina t i on o f t e rms wi th
ti�1;j�1;k; ti�1;j;k; ti;j�1;k; ti;j;k�1 and tijk could use the seven-diagonal coefficients on which to base the
update, provided these incorporate the major features of the equations.

Problems in topologically non-rectangular box-shaped regions can be solved using the routine by
surrounding the region with a circumscribing topologically rectangular box. The equations for the nodal
values external to the region of interest are set to zero (i.e., dijk ¼ rijk ¼ 0) and the boundary conditions
are incorporated into the equations for the appropriate nodes.

If there is no better initial approximation when starting the iterative cycle, one can use an array of all
zeros as the initial approximation from which the first set of residuals are determined.

The routine can be used to solve linear elliptic equations in which case the arrays A, B, C, D, E, F and
G and the quantities q will be unchanged during the iterative cycles, or for solving nonlinear elliptic
equations in which case some or all of these arrays may require updating as each new approximate
solution is derived. Depending on the nonlinearity, some under-relaxation of the coefficients and/or
source terms may be needed during their recalculation using the new estimates of the solution (see
Jacobs (1972)).

The routine can also be used to solve each step of a time-dependent parabolic equation in three space
dimensions. The solution at each time step can be expressed in terms of an elliptic equation if the
Crank–Nicolson or other form of implicit time integration is used.

Neither diagonal dominance, nor positive-definiteness, of the matrix M or of the update matrix formed
from the arrays A, B, C, D, E, F and G is necessary to ensure convergence.

For problems in which the solution is not unique, in the sense that an arbitrary constant can be added to
the solution (for example Poisson's equation with all Neumann boundary conditions), the calling
program should subtract a typical nodal value from the whole solution t at every iteration to keep
rounding errors to a minimum for those cases when convergence is slow. For such problems there is
generally an associated compatibility condition. For the example mentioned this compatibility condition
equates the total net source within the region (i.e., the source integrated over the region) with the total
net outflow across the boundaries defined by the Neumann conditions (i.e., the normal derivative
integrated along the whole boundary). It is very important that the algebraic equations derived to model
such a problem accurately implement the compatibility condition. If they do not, a net source or sink is
very likely to be represented by the set of algebraic equations and no steady-state solution of the
equations exists.

4 References

Ames W F (1977) Nonlinear Partial Differential Equations in Engineering (2nd Edition) Academic
Press

Jacobs D A H (1972) The strongly implicit procedure for the numerical solution of parabolic and
elliptic partial differential equations Note RD/L/N66/72 Central Electricity Research Laboratory

Stone H L (1968) Iterative solution of implicit approximations of multi-dimensional partial differential
equations SIAM J. Numer. Anal. 5 530–558

Weinstein H G, Stone H L and Kwan T V (1969) Iterative procedure for solution of systems of
parabolic and elliptic equations in three dimensions Industrial and Engineering Chemistry
Fundamentals 8 281–287

5 Arguments

1: N1 – INTEGER Input

On entry: the number of nodes in the first coordinate direction, n1.

Constraint: N1 > 1.

2: N2 – INTEGER Input

On entry: the number of nodes in the second coordinate direction, n2.

Constraint: N2 > 1.

3: N3 – INTEGER Input

On entry: the number of nodes in the third coordinate direction, n3.

Constraint: N3 > 1.

4: LDA – INTEGER Input

On entry: the first dimension of the arrays A, B, C, D, E, F, G, R, WRKSP1, WRKSP2 and
WRKSP3 as declared in the (sub)program from which D03UBF is called.

Constraint: LDA � N1.

5: SDA – INTEGER Input

On entry: the second dimension of the arrays A, B, C, D, E, F, G, R, WRKSP1, WRKSP2 and
WRKSP3 as declared in the (sub)program from which D03UBF is called.

Constraint: SDA � N2.

6: AðLDA; SDA;N3Þ – REAL (KIND=nag_wp) array Input

On entry: Aði; j; kÞ must contain the coefficient of sij;k�1 in the i; j; kð Þth equation of the system
(2), for i ¼ 1; 2; . . . ;N1, j ¼ 1; 2; . . . ;N2 and k ¼ 1; 2; . . . ;N3. The elements of A, for k ¼ 1,
must be zero after incorporating the boundary conditions, since they involve nodal values from
outside the box.

7: BðLDA; SDA;N3Þ – REAL (KIND=nag_wp) array Input

On entry: Bði; j; kÞ must contain the coefficient of si;j�1;k in the i; j; kð Þth equation of the system
(2), for i ¼ 1; 2; . . . ;N1, j ¼ 1; 2; . . . ;N2 and k ¼ 1; 2; . . . ;N3. The elements of B, for j ¼ 1,
must be zero after incorporating the boundary conditions, since they involve nodal values from
outside the box.

8: CðLDA; SDA;N3Þ – REAL (KIND=nag_wp) array Input

On entry: Cði; j; kÞ must contain the coefficient of si�1;j;k in the i; j; kð Þth equation of the system
(2), for i ¼ 1; 2; . . . ;N1, j ¼ 1; 2; . . . ;N2 and k ¼ 1; 2; . . . ;N3. The elements of C, for i ¼ 1, must
be zero after incorporating the boundary conditions, since they involve nodal values from outside
the box.

9: DðLDA; SDA;N3Þ – REAL (KIND=nag_wp) array Input

On entry: Dði; j; kÞ must contain the coefficient of sijk , the ‘central’ term, in the i; j; kð Þth
equation of the system (2), for i ¼ 1; 2; . . . ;N1, j ¼ 1; 2; . . . ;N2 and k ¼ 1; 2; . . . ;N3. The
elements of D are checked to ensure that they are nonzero. If any element is found to be zero, the
corresponding algebraic equation is assumed to be sijk ¼ rijk . This feature can be used to define
the equations for nodes at which, for example, Dirichlet boundary conditions are applied, or for
nodes external to the problem of interest, by setting Dði; j; kÞ ¼ 0:0 at appropriate points. The
corresponding value of rijk is set equal to the appropriate value, namely the difference between
the prescribed value of tijk and the current value in the Dirichlet case, or zero at an external
point.

10: EðLDA;SDA;N3Þ – REAL (KIND=nag_wp) array Input

On entry: Eði; j; kÞ must contain the coefficient of siþ1;j;k in the i; j; kð Þth equation of the system
(2), for i ¼ 1; 2; . . . ;N1, j ¼ 1; 2; . . . ;N2 and k ¼ 1; 2; . . . ;N3. The elements of E, for i ¼ N1,
must be zero after incorporating the boundary conditions, since they involve nodal values from
outside the box.

11: FðLDA;SDA;N3Þ – REAL (KIND=nag_wp) array Input

On entry: Fði; j; kÞ must contain the coefficient of si;jþ1;k in the i; j; kð Þth equation of the system
(2), for i ¼ 1; 2; . . . ;N1, j ¼ 1; 2; . . . ;N2 and k ¼ 1; 2; . . . ;N3. The elements of F, for j ¼ N2,
must be zero after incorporating the boundary conditions, since they involve nodal values from
outside the box.

12: GðLDA; SDA;N3Þ – REAL (KIND=nag_wp) array Input

On entry: Gði; j; kÞ must contain the coefficient of si;j;kþ1 in the i; j; kð Þth equation of the system
(2), for i ¼ 1; 2; . . . ;N1, j ¼ 1; 2; . . . ;N2 and k ¼ 1; 2; . . . ;N3. The elements of G, for k ¼ N3,
must be zero after incorporating the boundary conditions, since they involve nodal values from
outside the box.

13: APARAM – REAL (KIND=nag_wp) Input

On entry: the iteration acceleration factor. A value of 1:0 is adequate for most typical problems.
However, if convergence is slow, the value can be reduced, typically to 0:2 or 0:1. If divergence
is obtained, the value can be increased, typically to 2:0, 5:0 or 10:0.

Constraint: 0:0 < APARAM � N1� 1ð Þ2 þ N2� 1ð Þ2 þ N3� 1ð Þ2
� �

=3:0.

14: IT – INTEGER Input

On entry: the iteration number. It must be initialized, but not necessarily to 1, before the first call,
and should be incremented by one in the calling program for each subsequent call. The routine
uses this counter to select the appropriate acceleration argument from a sequence of nine, each
one being used twice in succession. (Note that the acceleration argument depends on the value of
APARAM.)

15: RðLDA;SDA;N3Þ – REAL (KIND=nag_wp) array Input/Output

On entry: the current residual rijk on the right-hand side of the i; j; kð Þth equation of the system
(2), for i ¼ 1; 2; . . . ;N1, j ¼ 1; 2; . . . ;N2 and k ¼ 1; 2; . . . ;N3.

On exit: these residuals are overwritten by the corresponding components of the solution s of the
system (2), i.e., the changes to be made to the vector t to reduce the residuals supplied.

16: WRKSP1ðLDA; SDA;N3Þ – REAL (KIND=nag_wp) array Workspace
17: WRKSP2ðLDA; SDA;N3Þ – REAL (KIND=nag_wp) array Workspace
18: WRKSP3ðLDA; SDA;N3Þ – REAL (KIND=nag_wp) array Workspace

19: IFAIL – INTEGER Input/Output

On entry: IFAIL must be set to 0, �1 or 1. If you are unfamiliar with this argument you should
refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is
detected, the value �1 or 1 is recommended. If the output of error messages is undesirable, then
the value 1 is recommended. Otherwise, if you are not familiar with this argument, the
recommended value is 0. When the value �1 or 1 is used it is essential to test the value of
IFAIL on exit.

On exit: IFAIL ¼ 0 unless the routine detects an error or a warning has been flagged (see
Section 6).

6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL ¼ 1

On entry, N1 < 2,
or N2 < 2,
or N3 < 2.

IFAIL ¼ 2

On entry, LDA < N1,
or SDA < N2.

IFAIL ¼ 3

On entry, APARAM � 0:0.

IFAIL ¼ 4

On entry, APARAM > N1� 1ð Þ2 þ N2� 1ð Þ2 þ N3� 1ð Þ2
� �

=3:0.

IFAIL ¼ �99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The improvement in accuracy for each iteration, i.e., on each call, depends on the size of the system
and on the condition of the update matrix characterised by the seven-diagonal coefficient arrays. The
ultimate accuracy obtainable depends on the above factors and on the machine precision. However,
since D03UBF works with residuals and the update vector, the calling program can, in most cases
where at each iteration all the residuals are usually of about the same size, calculate the residuals from
extended precision values of the function, source term and equation coefficients if greater accuracy is
required. The rate of convergence obtained with the Strongly Implicit Procedure is not always smooth
because of the cyclic use of nine acceleration arguments. The convergence may become slow with very
large problems. The final accuracy obtained can be judged approximately from the rate of convergence
determined from the changes to the dependent variable t and in particular the change on the last
iteration.

8 Parallelism and Performance

D03UBF is not threaded in any implementation.

9 Further Comments

The time taken is approximately proportional to N1� N2� N3 for each call.

When used with deferred or defect correction, the residual is calculated in the calling program from a
different system of equations to those represented by the seven-point molecule coefficients used by
D03UBF as the basis of the iterative update procedure. When using deferred correction the overall rate
of convergence depends not only on the items detailed in Section 7 but also on the difference between
the two coefficient matrices used.

Convergence may not always be obtained when the problem is very large and/or the coefficients of the
equations have widely disparate values. The latter case may be associated with an ill-conditioned
matrix.

10 Example

This example solves Laplace's equation in a rectangular box with a non-uniform grid spacing in the x, y
and z coordinate directions and with Dirichlet boundary conditions specifying the function on the
surfaces of the box equal to

e 1:0þxð Þ=y n2ð Þ � cos
ffiffiffi
2

p
y=y n2ð Þ

� �
� e �1:0�zð Þ=y n2ð Þ:

Note that this is the same problem as that solved in the example for D03ECF. The differences in the
maximum residuals obtained at each iteration between the two test runs are explained by the fact that in
D03ECF the residual at each node is normalized by dividing by the central coefficient, whereas this
normalization has not been used in the example program for D03UBF.

10.1 Program Text

Program d03ubfe

! D03UBF Example Program Text

! Mark 26 Release. NAG Copyright 2016.

! .. Use Statements ..
Use nag_library, Only: d03ubf, nag_wp

! .. Implicit None Statement ..
Implicit None

! .. Parameters ..
Real (Kind=nag_wp), Parameter :: one = 1.0_nag_wp
Real (Kind=nag_wp), Parameter :: two = 2.0_nag_wp
Real (Kind=nag_wp), Parameter :: zero = 0.0_nag_wp
Integer, Parameter :: nin = 5, nout = 6

D03UBF NAG Library Manual

D03UBF.6 Mark 26



! .. Local Scalars ..
Real (Kind=nag_wp) :: adel, aparam, ares, delmax, delmn, &

resmax, resmn, root2, x1, x2, y1, &
y2, yy, z1, z2

Integer :: i, ifail, it, j, k, lda, n1, n2, n3, &
nits, sda

! .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: a(:,:,:), b(:,:,:), c(:,:,:), &

d(:,:,:), e(:,:,:), f(:,:,:), &
g(:,:,:), q(:,:,:), r(:,:,:), &
t(:,:,:), wrksp1(:,:,:), &
wrksp2(:,:,:), wrksp3(:,:,:), x(:), &
y(:), z(:)

! .. Intrinsic Procedures ..
Intrinsic :: abs, cos, exp, max, real, sqrt

! .. Executable Statements ..
Write (nout,*) ’D03UBF Example Program Results’
Write (nout,*)

! Skip heading in data file
Read (nin,*)
Read (nin,*) n1, n2, n3, nits
lda = n1
sda = n2
Allocate (a(lda,sda,n3),b(lda,sda,n3),c(lda,sda,n3),d(lda,sda,n3), &

e(lda,sda,n3),f(lda,sda,n3),g(lda,sda,n3),q(lda,sda,n3),r(lda,sda,n3), &
t(lda,sda,n3),wrksp1(lda,sda,n3),wrksp2(lda,sda,n3), &
wrksp3(lda,sda,n3),x(n1),y(n2),z(n3))

Read (nin,*) x(1:n1)
Read (nin,*) y(1:n2)
Read (nin,*) z(1:n3)
root2 = sqrt(two)
aparam = one

! Set up difference equation coefficients, source terms and
! initial approximation

a(1:n1,1:n2,1:n3) = zero
b(1:n1,1:n2,1:n3) = zero
c(1:n1,1:n2,1:n3) = zero
e(1:n1,1:n2,1:n3) = zero
f(1:n1,1:n2,1:n3) = zero
g(1:n1,1:n2,1:n3) = zero
q(1:n1,1:n2,1:n3) = zero
t(1:n1,1:n2,1:n3) = zero

! Specification for internal nodes
Do k = 2, n3 - 1

a(2:n1-1,2:n2-1,k) = two/((z(k)-z(k-1))*(z(k+1)-z(k-1)))
g(2:n1-1,2:n2-1,k) = two/((z(k+1)-z(k))*(z(k+1)-z(k-1)))

End Do
Do j = 2, n2 - 1

b(2:n1-1,j,2:n3-1) = two/((y(j)-y(j-1))*(y(j+1)-y(j-1)))
f(2:n1-1,j,2:n3-1) = two/((y(j+1)-y(j))*(y(j+1)-y(j-1)))

End Do
Do i = 2, n1 - 1

c(i,2:n2-1,2:n3-1) = two/((x(i)-x(i-1))*(x(i+1)-x(i-1)))
e(i,2:n2-1,2:n3-1) = two/((x(i+1)-x(i))*(x(i+1)-x(i-1)))

End Do
d(1:n1,1:n2,1:n3) = -a(1:n1,1:n2,1:n3) - b(1:n1,1:n2,1:n3) - &

c(1:n1,1:n2,1:n3) - e(1:n1,1:n2,1:n3) - f(1:n1,1:n2,1:n3) - &
g(1:n1,1:n2,1:n3)

! Specification for boundary nodes
yy = one/y(n2)
x1 = (x(1)+one)*yy
x2 = (x(n1)+one)*yy
Do j = 1, n2

y1 = root2*y(j)*yy
q(1,j,1:n3) = exp(x1)*cos(y1)*exp((-z(1:n3)-one)*yy)
q(n1,j,1:n3) = exp(x2)*cos(y1)*exp((-z(1:n3)-one)*yy)

End Do
y1 = root2*y(1)*yy
y2 = root2*y(n2)*yy
Do i = 1, n1

x1 = (x(i)+one)*yy
q(i,1,1:n3) = exp(x1)*cos(y1)*exp((-z(1:n3)-one)*yy)
q(i,n2,1:n3) = exp(x1)*cos(y2)*exp((-z(1:n3)-one)*yy)

End Do
z1 = (-z(1)-one)*yy
z2 = (-z(n3)-one)*yy
Do i = 1, n1

x1 = (x(i)+one)*yy
q(i,1:n2,1) = exp(x1)*cos(root2*y(1:n2)*yy)*exp(z1)
q(i,1:n2,n3) = exp(x1)*cos(root2*y(1:n2)*yy)*exp(z2)

End Do
! Iterative loop

Do it = 1, nits
resmax = zero
resmn = zero
Do k = 1, n3

Do j = 1, n2
Do i = 1, n1

If (d(i,j,k)/=zero) Then
! Seven point molecule formula

r(i,j,k) = q(i,j,k) - a(i,j,k)*t(i,j,k-1) - &
b(i,j,k)*t(i,j-1,k) - c(i,j,k)*t(i-1,j,k) - &
d(i,j,k)*t(i,j,k) - e(i,j,k)*t(i+1,j,k) - &
f(i,j,k)*t(i,j+1,k) - g(i,j,k)*t(i,j,k+1)

Else
! Explicit equation

r(i,j,k) = q(i,j,k) - t(i,j,k)
End If
ares = abs(r(i,j,k))
resmax = max(resmax,ares)
resmn = resmn + ares

End Do
End Do

End Do
resmn = resmn/(real(n1*n2*n3,kind=nag_wp))

! ifail: behaviour on error exit
! =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft

ifail = 0
Call d03ubf(n1,n2,n3,lda,sda,a,b,c,d,e,f,g,aparam,it,r,wrksp1,wrksp2, &

wrksp3,ifail)

If (it==1) Then
Write (nout,99997) ’Iteration’, ’Residual’, ’Change’
Write (nout,99996) ’No’, ’Max.’, ’Mean’, ’Max.’, ’Mean’

End If

! Update the dependent variable
delmax = zero
delmn = zero
Do k = 1, n3

Do j = 1, n2
Do i = 1, n1

t(i,j,k) = t(i,j,k) + r(i,j,k)
adel = abs(r(i,j,k))
delmax = max(delmax,adel)
delmn = delmn + adel

End Do
End Do

End Do
delmn = delmn/(real(n1*n2*n3,kind=nag_wp))
Write (nout,99999) it, resmax, resmn, delmax, delmn

! Convergence tests here if required
End Do

! End of iterative loop
Write (nout,*)
Write (nout,*) ’Table of calculated function values’
Write (nout,99995)
Write (nout,*)
Write (nout,99998)((k,j,(i,t(i,j,k),i=1,n1),j=1,n2),k=1,n3)

99999 Format (1X,I5,4(2X,E11.4))
99998 Format ((1X,I1,I3,1X,4(1X,I3,2X,F8.3)))
99997 Format (1X,A,6X,A,19X,A)
99996 Format (2X,A,7X,A,8X,A,11X,A,6X,A,/)
99995 Format (1X,’K J’,2X,4(1X,’(I T )’))

End Program d03ubfe

10.2 Program Data

D03UBF Example Program Data
4 5 6 10 : n1, n2, n3, nits
0.0 1.0 3.0 6.0 : x
0.0 1.0 3.0 6.0 10.0 : y
0.0 1.0 3.0 6.0 10.0 15.0 : z

10.3 Program Results

D03UBF Example Program Results

Iteration Residual Change
No Max. Mean Max. Mean

1 0.1822E+01 0.4847E+00 0.1822E+01 0.6173E+00
2 0.8585E-02 0.9369E-03 0.1970E-01 0.1895E-02
3 0.3168E-02 0.7783E-04 0.1496E-02 0.5819E-04
4 0.4085E-04 0.2179E-05 0.3848E-04 0.1931E-05
5 0.7820E-05 0.3999E-06 0.5481E-05 0.2312E-06
6 0.2246E-06 0.1524E-07 0.2333E-06 0.1093E-07
7 0.2219E-07 0.1669E-08 0.2222E-07 0.9131E-09
8 0.2841E-08 0.1820E-09 0.1969E-08 0.9337E-10
9 0.6696E-09 0.4762E-10 0.5873E-09 0.2450E-10

10 0.7848E-10 0.4908E-11 0.5863E-10 0.2671E-11

Table of calculated function values
K J (I T ) (I T ) (I T ) (I T )

1 1 1 1.000 2 1.105 3 1.350 4 1.822
1 2 1 0.990 2 1.094 3 1.336 4 1.804
1 3 1 0.911 2 1.007 3 1.230 4 1.661
1 4 1 0.661 2 0.731 3 0.892 4 1.205
1 5 1 0.156 2 0.172 3 0.211 4 0.284
2 1 1 0.905 2 1.000 3 1.221 4 1.649
2 2 1 0.896 2 0.990 3 1.210 4 1.632
2 3 1 0.825 2 0.912 3 1.114 4 1.503
2 4 1 0.598 2 0.662 3 0.809 4 1.090
2 5 1 0.141 2 0.156 3 0.190 4 0.257
3 1 1 0.741 2 0.819 3 1.000 4 1.350
3 2 1 0.733 2 0.811 3 0.991 4 1.336
3 3 1 0.675 2 0.747 3 0.913 4 1.230
3 4 1 0.490 2 0.543 3 0.664 4 0.892
3 5 1 0.116 2 0.128 3 0.156 4 0.211
4 1 1 0.549 2 0.607 3 0.741 4 1.000
4 2 1 0.543 2 0.601 3 0.734 4 0.990
4 3 1 0.500 2 0.554 3 0.677 4 0.911
4 4 1 0.363 2 0.402 3 0.492 4 0.661
4 5 1 0.086 2 0.095 3 0.116 4 0.156
5 1 1 0.368 2 0.407 3 0.497 4 0.670
5 2 1 0.364 2 0.403 3 0.492 4 0.664
5 3 1 0.335 2 0.371 3 0.454 4 0.611
5 4 1 0.243 2 0.270 3 0.330 4 0.443
5 5 1 0.057 2 0.063 3 0.077 4 0.105
6 1 1 0.223 2 0.247 3 0.301 4 0.407
6 2 1 0.221 2 0.244 3 0.298 4 0.403
6 3 1 0.203 2 0.225 3 0.274 4 0.371
6 4 1 0.148 2 0.163 3 0.199 4 0.269
6 5 1 0.035 2 0.038 3 0.047 4 0.063

Example Program
Solution of Three-dimensional Laplace’s Equation in a Box

Solutions in the xy-plane for Various Values of z

