NAG Library Routine Document

E02AFF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

E02AFF computes the coefficients of a polynomial, in its Chebyshev series form, which interpolates (passes exactly through) data at a special set of points. Least squares polynomial approximations can also be obtained.

2 Specification

```
SUBROUTINE E02AFF (NPLUS1, F, A, IFAIL)
INTEGER            NPLUS1, IFAIL
REAL (KIND=nag_wp) F(NPLUS1), A(NPLUS1)
```

3 Description

E02AFF computes the coefficients a_j, for $j = 1, 2, \ldots, n+1$, in the Chebyshev series

$$\tfrac{1}{2}a_1 T_0(\bar{x}) + a_2 T_1(\bar{x}) + a_3 T_2(\bar{x}) + \cdots + a_{n+1} T_n(\bar{x}),$$

which interpolates the data f_r at the points

$$\bar{x}_r = \cos((r-1)\pi/n), \quad r = 1, 2, \ldots, n+1.$$

Here $T_j(\bar{x})$ denotes the Chebyshev polynomial of the first kind of degree j with argument $\bar{x}$. The use of these points minimizes the risk of unwanted fluctuations in the polynomial and is recommended when the data abscissae can be chosen by you, e.g., when the data is given as a graph. For further advantages of this choice of points, see Clenshaw (1962).

In terms of your original variables, x say, the values of x at which the data f_r are to be provided are

$$x_r = \tfrac{1}{2}(x_{\max} - x_{\min})\cos(\pi(r-1)/n) + \tfrac{1}{2}(x_{\max} + x_{\min}), \quad r = 1, 2, \ldots, n+1$$

where $x_{\max}$ and $x_{\min}$ are respectively the upper and lower ends of the range of x over which you wish to interpolate.

Truncation of the resulting series after the term involving a_{i+1}, say, yields a least squares approximation to the data. This approximation, $p(\bar{x})$, say, is the polynomial of degree i which minimizes

$$\tfrac{1}{2}\epsilon_1^2 + \epsilon_2^2 + \epsilon_3^2 + \cdots + \epsilon_n^2 + \tfrac{1}{2}\epsilon_{n+1}^2,$$

where the residual $\epsilon_r = p(\bar{x}_r) - f_r$, for $r = 1, 2, \ldots, n+1$.

The method employed is based on the application of the three-term recurrence relation due to Clenshaw (1955) for the evaluation of the defining expression for the Chebyshev coefficients (see, for example, Clenshaw (1962)). The modifications to this recurrence relation suggested by Reinsch and Gentleman (see Gentleman (1969)) are used to give greater numerical stability.

For further details of the algorithm and its use see Cox (1974) and Cox and Hayes (1973).

Subsequent evaluation of the computed polynomial, perhaps truncated after an appropriate number of terms, should be carried out using E02AEF.

4 References

Clenshaw C W (1955) A note on the summation of Chebyshev series *Math. Tables Aids Comput.* **9** 118–120

Clenshaw C W (1962) Chebyshev Series for Mathematical Functions *Mathematical tables* HMSO

Cox M G (1974) A data-fitting package for the non-specialist user *Software for Numerical Mathematics* (ed D J Evans) Academic Press

Cox M G and Hayes J G (1973) Curve fitting: a guide and suite of algorithms for the non-specialist user *NPL Report NAC26* National Physical Laboratory

Gentleman W M (1969) An error analysis of Goertzel's (Watt's) method for computing Fourier coefficients *Comput. J.* **12** 160–165

5 Arguments

1: NPLUS1 – INTEGER *Input*

On entry: the number $n+1$ of data points (one greater than the degree n of the interpolating polynomial).

Constraint: NPLUS1 ≥ 2.

2: F(NPLUS1) – REAL (KIND=nag_wp) array *Input*

On entry: for $r = 1, 2, \ldots, n+1$, F(r) must contain f_r the value of the dependent variable (ordinate) corresponding to the value

$$\bar{x}_r = \cos(\pi(r-1)/n)$$

of the independent variable (abscissa) $\bar{x}$, or equivalently to the value

$$x(r) = \tfrac{1}{2}(x_{\max} - x_{\min})\cos(\pi(r-1)/n) + \tfrac{1}{2}(x_{\max} + x_{\min})$$

of your original variable x. Here $x_{\max}$ and $x_{\min}$ are respectively the upper and lower ends of the range over which you wish to interpolate.

3: A(NPLUS1) – REAL (KIND=nag_wp) array *Output*

On exit: A(j) is the coefficient a_j in the interpolating polynomial, for $j = 1, 2, \ldots, n+1$.

4: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, NPLUS1 < 2.

IFAIL $= -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The rounding errors committed are such that the computed coefficients are exact for a slightly perturbed set of ordinates $f_r + \delta f_r$. The ratio of the sum of the absolute values of the δf_r to the sum of the absolute values of the f_r is less than a small multiple of $(n+1)\epsilon$, where ϵ is the ***machine precision***.

8 Parallelism and Performance

E02AFF is not threaded in any implementation.

9 Further Comments

The time taken is approximately proportional to $(n+1)^2 + 30$.

For choice of degree when using the routine for least squares approximation, see Section 3.2 in the E02 Chapter Introduction.

10 Example

Determine the Chebyshev coefficients of the polynomial which interpolates the data $\bar{x}_r, f_r$, for $r = 1, 2, \ldots, 11$, where $\bar{x}_r = \cos(\pi \times (r-1)/10)$ and $f_r = e^{\bar{x}_r}$. Evaluate, for comparison with the values of f_r, the resulting Chebyshev series at $\bar{x}_r$, for $r = 1, 2, \ldots, 11$.

The example program supplied is written in a general form that will enable polynomial interpolations of arbitrary data at the cosine points $\cos(\pi \times (r-1)/n)$, for $r = 1, 2, \ldots, n+1$, to be obtained for any n ($=$ NPLUS1 $- 1$). Note that E02AEF is used to evaluate the interpolating polynomial. The program is self-starting in that any number of datasets can be supplied.

10.1 Program Text

```
    Program e02affe

!     E02AFF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: e02aef, e02aff, nag_wp, x01aaf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: fit, pi, piby2n
      Integer                          :: i, ifail, j, n, nplus1, r
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: a(:), f(:), xcap(:)
!     .. Intrinsic Procedures ..
      Intrinsic                        :: real, sin
!     .. Executable Statements ..
      Write (nout,*) 'E02AFF Example Program Results'

!     Skip heading in data file
      Read (nin,*)

      Read (nin,*) n
      nplus1 = n + 1
      Allocate (a(nplus1),f(nplus1),xcap(nplus1))

      piby2n = 0.5E0_nag_wp*x01aaf(pi)/real(n,kind=nag_wp)

      Read (nin,*)(f(r),r=1,nplus1)

      Do r = 1, nplus1
        i = r - 1

!       The following method of evaluating  XCAP = cos(PI*I/N)
!       ensures that the computed value has a small relative error
!       and, moreover, is bounded in modulus by unity for all
!       I = 0, 1, ..., N.  (It is assumed that the sine routine
!       produces a result with a small relative error for values
!       of the argument between  -PI/4  and  PI/4).

        If (4*i<=n) Then
          xcap(i+1) = 1.0E0_nag_wp - 2.0E0_nag_wp*sin(piby2n*real(i,kind=      &
            nag_wp))**2
        Else If (4*i>3*n) Then
          xcap(i+1) = 2.0E0_nag_wp*sin(piby2n*real(n-i,kind=nag_wp))**2 -      &
            1.0E0_nag_wp
        Else
          xcap(i+1) = sin(piby2n*real(n-2*i,kind=nag_wp))
        End If

      End Do

      ifail = 0
      Call e02aff(nplus1,f,a,ifail)

      Write (nout,*)
      Write (nout,*) '         Chebyshev'
      Write (nout,*) '  J   coefficient A(J)'
      Write (nout,99998)(j,a(j),j=1,nplus1)
      Write (nout,*)
      Write (nout,*) '  R    Abscissa   Ordinate      Fit'

      Do r = 1, nplus1

        ifail = 0
        Call e02aef(nplus1,a,xcap(r),fit,ifail)
```

```
        Write (nout,99999) r, xcap(r), f(r), fit
      End Do

99999 Format (1X,I3,3F11.4)
99998 Format (1X,I3,F14.7)
    End Program e02affe
```

10.2 Program Data

```
E02AFF Example Program Data
  10
     2.7182
     2.5884
     2.2456
     1.7999
     1.3620
     1.0000
     0.7341
     0.5555
     0.4452
     0.3863
     0.3678
```

10.3 Program Results

```
 E02AFF Example Program Results

          Chebyshev
   J   coefficient A(J)
   1     2.5320000
   2     1.1303095
   3     0.2714893
   4     0.0443462
   5     0.0055004
   6     0.0005400
   7     0.0000307
   8    -0.0000006
   9    -0.0000004
  10     0.0000049
  11    -0.0000200

   R    Abscissa   Ordinate      Fit
   1     1.0000     2.7182     2.7182
   2     0.9511     2.5884     2.5884
   3     0.8090     2.2456     2.2456
   4     0.5878     1.7999     1.7999
   5     0.3090     1.3620     1.3620
   6     0.0000     1.0000     1.0000
   7    -0.3090     0.7341     0.7341
   8    -0.5878     0.5555     0.5555
   9    -0.8090     0.4452     0.4452
  10    -0.9511     0.3863     0.3863
  11    -1.0000     0.3678     0.3678
```


Example Program
Interpolating Chebyshev Polynomial
P(x)
x
|ex–P(x)| (error)
Chebyshev polynomial fit
|ex–P(x)|
2.5
2
1.5
1
0.5
0
-1
-0.5
0
0.5
1
0.00015
0.0001
5e-05