

NAG Library Routine Document

E02AJF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

E02AJF determines the coefficients in the Chebyshev series representation of the indefinite integral of a polynomial given in Chebyshev series form.

2 Specification

```
SUBROUTINE E02AJF (NP1, XMIN, XMAX, A, IA1, LA, QATM1, AINTC, IAIN1, &
                  LAINT, IFAIL)
INTEGER          NP1, IA1, LA, IAIN1, LAINT, IFAIL
REAL (KIND=nag_wp) XMIN, XMAX, A(LA), QATM1, AINTC(LAINT)
```

3 Description

E02AJF forms the polynomial which is the indefinite integral of a given polynomial. Both the original polynomial and its integral are represented in Chebyshev series form. If supplied with the coefficients a_i , for $i = 0, 1, \dots, n$, of a polynomial $p(x)$ of degree n , where

$$p(x) = \frac{1}{2}a_0 + a_1T_1(\bar{x}) + \dots + a_nT_n(\bar{x}),$$

the routine returns the coefficients a'_i , for $i = 0, 1, \dots, n+1$, of the polynomial $q(x)$ of degree $n+1$, where

$$q(x) = \frac{1}{2}a'_0 + a'_1T_1(\bar{x}) + \dots + a'_{n+1}T_{n+1}(\bar{x}),$$

and

$$q(x) = \int p(x)dx.$$

Here $T_j(\bar{x})$ denotes the Chebyshev polynomial of the first kind of degree j with argument \bar{x} . It is assumed that the normalized variable \bar{x} in the interval $[-1, +1]$ was obtained from your original variable x in the interval $[x_{\min}, x_{\max}]$ by the linear transformation

$$\bar{x} = \frac{2x - (x_{\max} + x_{\min})}{x_{\max} - x_{\min}}$$

and that you require the integral to be with respect to the variable x . If the integral with respect to \bar{x} is required, set $x_{\max} = 1$ and $x_{\min} = -1$.

Values of the integral can subsequently be computed, from the coefficients obtained, by using E02AKF.

The method employed is that of Chebyshev series (see Chapter 8 of Modern Computing Methods (1961)), modified for integrating with respect to x . Initially taking $a_{n+1} = a_{n+2} = 0$, the routine forms successively

$$a'_i = \frac{a_{i-1} - a_{i+1}}{2i} \times \frac{x_{\max} - x_{\min}}{2}, \quad i = n+1, n, \dots, 1.$$

The constant coefficient a'_0 is chosen so that $q(x)$ is equal to a specified value, QATM1, at the lower end point of the interval on which it is defined, i.e., $\bar{x} = -1$, which corresponds to $x = x_{\min}$.

4 References

Modern Computing Methods (1961) Chebyshev-series *NPL Notes on Applied Science* **16** (2nd Edition) HMSO

5 Arguments

- 1: NP1 – INTEGER *Input*
On entry: $n + 1$, where n is the degree of the given polynomial $p(x)$. Thus NP1 is the number of coefficients in this polynomial.
Constraint: $NP1 \geq 1$.

- 2: XMIN – REAL (KIND=nag_wp) *Input*
 3: XMAX – REAL (KIND=nag_wp) *Input*
On entry: the lower and upper end points respectively of the interval $[x_{\min}, x_{\max}]$. The Chebyshev series representation is in terms of the normalized variable \bar{x} , where

$$\bar{x} = \frac{2x - (x_{\max} + x_{\min})}{x_{\max} - x_{\min}}.$$
Constraint: $XMAX > XMIN$.

- 4: A(LA) – REAL (KIND=nag_wp) array *Input*
On entry: the Chebyshev coefficients of the polynomial $p(x)$. Specifically, element $i \times IA1 + 1$ of A must contain the coefficient a_i , for $i = 0, 1, \dots, n$. Only these $n + 1$ elements will be accessed.
 Unchanged on exit, but see AINTC, below.

- 5: IA1 – INTEGER *Input*
On entry: the index increment of A. Most frequently the Chebyshev coefficients are stored in adjacent elements of A, and IA1 must be set to 1. However, if for example, they are stored in $A(1), A(4), A(7), \dots$, then the value of IA1 must be 3. See also Section 9.
Constraint: $IA1 \geq 1$.

- 6: LA – INTEGER *Input*
On entry: the dimension of the array A as declared in the (sub)program from which E02AJF is called.
Constraint: $LA \geq 1 + (NP1 - 1) \times IA1$.

- 7: QATM1 – REAL (KIND=nag_wp) *Input*
On entry: the value that the integrated polynomial is required to have at the lower end point of its interval of definition, i.e., at $\bar{x} = -1$ which corresponds to $x = x_{\min}$. Thus, QATM1 is a constant of integration and will normally be set to zero by you.

- 8: AINTC(LAINT) – REAL (KIND=nag_wp) array *Output*
On exit: the Chebyshev coefficients of the integral $q(x)$. (The integration is with respect to the variable x , and the constant coefficient is chosen so that $q(x_{\min})$ equals QATM1). Specifically, element $i \times IAIN1 + 1$ of AINTC contains the coefficient a'_i , for $i = 0, 1, \dots, n + 1$. A call of the routine may have the array name AINTC the same as A, provided that note is taken of the order in which elements are overwritten when choosing starting elements and increments IA1 and IAIN1: i.e., the coefficients, a_0, a_1, \dots, a_{i-2} must be intact after coefficient a'_i is stored. In particular it is possible to overwrite the a_i entirely by having $IA1 = IAIN1$, and the actual array for A and AINTC identical.

- 9: IAIN1 – INTEGER *Input*
On entry: the index increment of AINTC. Most frequently the Chebyshev coefficients are required in adjacent elements of AINTC, and IAIN1 must be set to 1. However, if, for example, they are to be stored in AINTC(1), AINTC(4), AINTC(7), ..., then the value of IAIN1 must be 3. See also Section 9.
Constraint: $IAIN1 \geq 1$.
- 10: LAINT – INTEGER *Input*
On entry: the dimension of the array AINTC as declared in the (sub)program from which E02AJF is called.
Constraint: $LAINT \geq 1 + (NP1) \times IAIN1$.
- 11: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.
For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $NP1 < 1$,
or $XMAX \leq XMIN$,
or $IA1 < 1$,
or $LA \leq (NP1 - 1) \times IA1$,
or $IAIN1 < 1$,
or $LAINT \leq NP1 \times IAIN1$.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

In general there is a gain in precision in numerical integration, in this case associated with the division by $2i$ in the formula quoted in Section 3.

8 Parallelism and Performance

E02AJF is not threaded in any implementation.

9 Further Comments

The time taken is approximately proportional to $n + 1$.

The increments IA1, IAIN1 are included as arguments to give a degree of flexibility which, for example, allows a polynomial in two variables to be integrated with respect to either variable without rearranging the coefficients.

10 Example

Suppose a polynomial has been computed in Chebyshev series form to fit data over the interval $[-0.5, 2.5]$. The following program evaluates the integral of the polynomial from 0.0 to 2.0. (For the purpose of this example, XMIN, XMAX and the Chebyshev coefficients are simply supplied in DATA statements. Normally a program would read in or generate data and compute the fitted polynomial).

10.1 Program Text

```

Program e02ajfe

!      E02AJF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: e02ajf, e02akf, nag_wp
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Real (Kind=nag_wp), Parameter      :: xmax = 2.5E0_nag_wp
      Real (Kind=nag_wp), Parameter      :: xmin = -0.5E0_nag_wp
      Integer, Parameter                  :: nout = 6, np1 = 7
      Integer, Parameter                  :: la = np1
      Integer, Parameter                  :: laint = np1 + 1
      Real (Kind=nag_wp), Parameter      :: a(la) = (/2.53213E0_nag_wp,      &
1.13032E0_nag_wp,0.27150E0_nag_wp,    &
0.04434E0_nag_wp,0.00547E0_nag_wp,    &
0.00054E0_nag_wp,0.00004E0_nag_wp/)

!      .. Local Scalars ..
      Real (Kind=nag_wp)                 :: ra, rb, res, xa, xb
      Integer                             :: ifail
!      .. Local Arrays ..
      Real (Kind=nag_wp)                 :: aintc(laint)
!      .. Executable Statements ..
      Write (nout,*) 'E02AJF Example Program Results'

      ifail = 0
      Call e02ajf(np1,xmin,xmax,a,1,la,0.0E0_nag_wp,aintc,1,laint,ifail)

      xa = 0.0E0_nag_wp
      xb = 2.0E0_nag_wp

      ifail = 0
      Call e02akf(np1+1,xmin,xmax,aintc,1,laint,xa,ra,ifail)

      ifail = 0
      Call e02akf(np1+1,xmin,xmax,aintc,1,laint,xb,rb,ifail)

```

```
res = rb - ra

Write (nout,*)
Write (nout,99999) 'Value of definite integral is ', res

99999 Format (1X,A,F10.4)
End Program e02ajfe
```

10.2 Program Data

None.

10.3 Program Results

E02AJF Example Program Results

Value of definite integral is 2.1515
