
NAG Library Routine Document

E04GBF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

E04GBF is a comprehensive quasi-Newton algorithm for finding an unconstrained minimum of a sum
of squares of m nonlinear functions in n variables m � nð Þ. First derivatives are required.

The routine is intended for functions which have continuous first and second derivatives (although it
will usually work even if the derivatives have occasional discontinuities).

2 Specification

SUBROUTINE E04GBF (M, N, LSQLIN, LSQFUN, LSQMON, IPRINT, MAXCAL, ETA,
XTOL, STEPMX, X, FSUMSQ, FVEC, FJAC, LDFJAC, S, V,
LDV, NITER, NF, IW, LIW, W, LW, IFAIL)

&
&

INTEGER M, N, IPRINT, MAXCAL, LDFJAC, LDV, NITER, NF,
IW(LIW), LIW, LW, IFAIL

&

REAL (KIND=nag_wp) ETA, XTOL, STEPMX, X(N), FSUMSQ, FVEC(M),
FJAC(LDFJAC,N), S(N), V(LDV,N), W(LW)

&

EXTERNAL LSQLIN, LSQFUN, LSQMON

3 Description

E04GBF is essentially identical to the subroutine LSQFDQ in the NPL Algorithms Library. It is
applicable to problems of the form:

MinimizeF xð Þ ¼
Xm
i¼1

fi xð Þ½ �2

where x ¼ x1; x2; . . . ; xnð ÞT and m � n. (The functions fi xð Þ are often referred to as ‘residuals’.)

You must supply a subroutine to calculate the values of the fi xð Þ and their first derivatives
@fi
@xj

at any

point x.

From a starting point x 1ð Þ supplied by you, the routine generates a sequence of points x 2ð Þ; x 3ð Þ; . . .,
which is intended to converge to a local minimum of F xð Þ. The sequence of points is given by

x kþ1ð Þ ¼ x kð Þ þ � kð Þp kð Þ

where the vector p kð Þ is a direction of search, and � kð Þ is chosen such that F x kð Þ þ � kð Þp kð Þ� �
is

approximately a minimum with respect to � kð Þ.

The vector p kð Þ used depends upon the reduction in the sum of squares obtained during the last iteration.
If the sum of squares was sufficiently reduced, then p kð Þ is the Gauss–Newton direction; otherwise the
second derivatives of the fi xð Þ are taken into account using a quasi-Newton updating scheme.

The method is designed to ensure that steady progress is made whatever the starting point, and to have
the rapid ultimate convergence of Newton's method.

4 References

Gill P E and Murray W (1978) Algorithms for the solution of the nonlinear least squares problem SIAM
J. Numer. Anal. 15 977–992

5 Arguments

1: M – INTEGER Input
2: N – INTEGER Input

On entry: the number m of residuals, fi xð Þ, and the number n of variables, xj.

Constraint: 1 � N � M.

3: LSQLIN – SUBROUTINE, supplied by the NAG Library. External Procedure

LSQLIN enables you to specify whether the linear minimizations (i.e., minimizations of
F x kð Þ þ � kð Þp kð Þ� �

with respect to � kð Þ) are to be performed by a routine which just requires the
evaluation of the fi xð Þ (E04FCV), or by a routine which also requires the first derivatives of the
fi xð Þ (E04HEV).

It will often be possible to evaluate the first derivatives of the residuals in about the same amount
of computer time that is required for the evaluation of the residuals themselves – if this is so
then E04GBF should be called with routine E04HEV as the argument LSQLIN. However, if the
evaluation of the derivatives takes more than about 4 times as long as the evaluation of the
residuals, then E04FCV will usually be preferable. If in doubt, use E04HEV as it is slightly more
robust.

Whichever subroutine is used, must be declared as EXTERNAL in the subroutine from which
E04GBF is called.

4: LSQFUN – SUBROUTINE, supplied by the user. External Procedure

LSQFUN must calculate the vector of values fi xð Þ and Jacobian matrix of first derivatives
@fi
@xj

at

any point x. (However, if you do not wish to calculate the residuals or first derivatives at a
particular x, there is the option of setting an argument to cause E04GBF to terminate
immediately.)

The specification of LSQFUN is:

SUBROUTINE LSQFUN (IFLAG, M, N, XC, FVEC, FJAC, LDFJAC, IW, LIW,
W, LW)

&

INTEGER IFLAG, M, N, LDFJAC, IW(LIW), LIW, LW
REAL (KIND=nag_wp) XC(N), FVEC(M), FJAC(LDFJAC,N), W(LW)

Important: the dimension declaration for FJAC must contain the variable LDFJAC, not an
integer constant.

1: IFLAG – INTEGER Input/Output

On entry: will be set to 0, 1 or 2.

IFLAG ¼ 0
Indicates that only the residuals need to be evaluated

IFLAG ¼ 1
Indicates that only the Jacobian matrix needs to be evaluated

IFLAG ¼ 2
Indicates that both the residuals and the Jacobian matrix must be calculated.

If E04HEV is used as E04GBF's LSQLIN, LSQFUN will always be called with IFLAG
set to 2.

On exit: if it is not possible to evaluate the fi xð Þ or their first derivatives at the point
given in XC (or if it is wished to stop the calculations for any other reason), you should
reset IFLAG to some negative number and return control to E04GBF. E04GBF will
then terminate immediately, with IFAIL set to your setting of IFLAG.

2: M – INTEGER Input

On entry: m, the number of residuals.

3: N – INTEGER Input

On entry: n, the number of variables.

4: XCðNÞ – REAL (KIND=nag_wp) array Input

On entry: the point x at which the values of the fi and the
@fi
@xj

are required.

5: FVECðMÞ – REAL (KIND=nag_wp) array Output

On exit: unless IFLAG ¼ 1 on entry, or IFLAG is reset to a negative number, then
FVECðiÞ must contain the value of fi at the point x, for i ¼ 1; 2; . . . ;m.

6: FJACðLDFJAC;NÞ – REAL (KIND=nag_wp) array Output

On exit: unless IFLAG ¼ 0 on entry, or IFLAG is reset to a negative number, then

FJACði; jÞ must contain the value of
@fi
@xj

at the point x, for i ¼ 1; 2; . . . ;m and

j ¼ 1; 2; . . . ; n.

7: LDFJAC – INTEGER Input

On entry: the first dimension of the array FJAC, set to m by E04GBF.

8: IWðLIWÞ – INTEGER array Workspace
9: LIW – INTEGER Input
10: WðLWÞ – REAL (KIND=nag_wp) array Workspace
11: LW – INTEGER Input

LSQFUN is called with E04GBF's arguments IW, LIW, W, LW as these arguments.
They are present so that, when other library routines require the solution of a
minimization subproblem, constants needed for the evaluation of residuals can be
passed through IW and W. Similarly, you could pass quantities to LSQFUN from the
segment which calls E04GBF by using partitions of IW and W beyond those used as
workspace by E04GBF. However, because of the danger of mistakes in partitioning, it is
recommended that you should pass information to LSQFUN via COMMON global
variables and not use IW or W at all. In any case you must not change the elements of
IW and W used as workspace by E04GBF.

LSQFUN must either be a module subprogram USEd by, or declared as EXTERNAL in, the
(sub)program from which E04GBF is called. Arguments denoted as Input must not be changed
by this procedure.

Note: LSQFUN should be tested separately before being used in conjunction with E04GBF.

5: LSQMON – SUBROUTINE, supplied by the NAG Library or the user. External Procedure

If IPRINT � 0, you must supply LSQMON which is suitable for monitoring the minimization
process. LSQMON must not change the values of any of its arguments.

If IPRINT < 0, the NAG Library dummy routine E04FDZ can be used as LSQMON.

The specification of LSQMON is:

SUBROUTINE LSQMON (M, N, XC, FVEC, FJAC, LDFJAC, S, IGRADE,
NITER, NF, IW, LIW, W, LW)

&

INTEGER M, N, LDFJAC, IGRADE, NITER, NF, IW(LIW),
LIW, LW

&

REAL (KIND=nag_wp) XC(N), FVEC(M), FJAC(LDFJAC,N), S(N), W(LW)

Important: the dimension declaration for FJAC must contain the variable LDFJAC, not an
integer constant.

1: M – INTEGER Input

On entry: m, the numbers of residuals.

2: N – INTEGER Input

On entry: n, the numbers of variables.

3: XCðNÞ – REAL (KIND=nag_wp) array Input

On entry: the coordinates of the current point x.

4: FVECðMÞ – REAL (KIND=nag_wp) array Input

On entry: the values of the residuals fi at the current point x.

5: FJACðLDFJAC;NÞ – REAL (KIND=nag_wp) array Input

On entry: FJACði; jÞ contains the value of
@fi
@xj

at the current point x, for i ¼ 1; 2; . . . ;m

and j ¼ 1; 2; . . . ; n.

6: LDFJAC – INTEGER Input

On entry: the first dimension of the array FJAC as declared in the (sub)program from
which E04GBF is called.

7: SðNÞ – REAL (KIND=nag_wp) array Input

On entry: the singular values of the current Jacobian matrix. Thus S may be useful as
information about the structure of your problem.

8: IGRADE – INTEGER Input

On entry: E04GBF estimates the dimension of the subspace for which the Jacobian
matrix can be used as a valid approximation to the curvature (see Gill and Murray
(1978)). This estimate is called the grade of the Jacobian matrix, and IGRADE gives its
current value.

9: NITER – INTEGER Input

On entry: the number of iterations which have been performed in E04GBF.

10: NF – INTEGER Input

On entry: the number of evaluations of the residuals. (If E04HEV is used as LSQLIN,
NF is also the number of evaluations of the Jacobian matrix.)

11: IWðLIWÞ – INTEGER array Workspace
12: LIW – INTEGER Input
13: WðLWÞ – REAL (KIND=nag_wp) array Workspace
14: LW – INTEGER Input

As in LSQFUN, these arguments correspond to the arguments IW, LIW, W, LW of
E04GBF. They are included in LSQMON's argument list primarily for when E04GBF is
called by other library routines.

LSQMON must either be a module subprogram USEd by, or declared as EXTERNAL in, the
(sub)program from which E04GBF is called. Arguments denoted as Input must not be changed
by this procedure.

Note: you should normally print the sum of squares of residuals, so as to be able to examine the
sequence of values of F xð Þ mentioned in Section 7. It is usually helpful to also print XC, the
gradient of the sum of squares, NITER and NF.

6: IPRINT – INTEGER Input

On entry: the frequency with which LSQMON is to be called.

IPRINT > 0
LSQMON is called once every IPRINT iterations and just before exit from E04GBF.

IPRINT ¼ 0
LSQMON is just called at the final point.

IPRINT < 0
LSQMON is not called at all.

IPRINT should normally be set to a small positive number.

Suggested value: IPRINT ¼ 1.

7: MAXCAL – INTEGER Input

On entry: enables you to limit the number of times that LSQFUN is called by E04GBF. There
will be an error exit (see Section 6) after MAXCAL calls of LSQFUN.

Suggested value:

MAXCAL ¼ 75� n if E04FCV is used as LSQLIN,

MAXCAL ¼ 50� n if E04HEV is used as LSQLIN.

Constraint: MAXCAL � 1.

8: ETA – REAL (KIND=nag_wp) Input

On entry: every iteration of E04GBF involves a linear minimization (i.e., minimization of
F x kð Þ þ � kð Þp kð Þ� �

with respect to � kð Þ). ETA specifies how accurately these linear minimizations

are to be performed. The minimum with respect to � kð Þ will be located more accurately for small
values of ETA (say, 0:01) than for large values (say, 0:9).

Although accurate linear minimizations will generally reduce the number of iterations performed
by E04GBF, they will increase the number of calls of LSQFUN made every iteration. On balance
it is usually more efficient to perform a low accuracy minimization.

Suggested value:

ETA ¼ 0:9 if N > 1 and E04HEV is used as LSQLIN,

ETA ¼ 0:5 if N > 1 and E04FCV is uses as LSQLIN,

ETA ¼ 0:0 if N ¼ 1.

Constraint: 0:0 � ETA < 1:0.

9: XTOL – REAL (KIND=nag_wp) Input

On entry: the accuracy in x to which the solution is required.

If xtrue is the true value of x at the minimum, then xsol, the estimated position before a normal
exit, is such that

xsol � xtruek k < XTOL� 1:0þ xtruek kð Þ;

where yk k ¼
ffiffiffiffiffiffiffiffiffiffiffiffiXn
j¼1

y2j

s
. For example, if the elements of xsol are not much larger than 1:0 in

modulus and if XTOL ¼ 1:0E�5, then xsol is usually accurate to about five decimal places. (For
further details see Section 7.)

If F xð Þ and the variables are scaled roughly as described in Section 9 and � is the machine
precision, then a setting of order XTOL ¼ ffiffi

�
p

will usually be appropriate. If XTOL is set to 0:0
or some positive value less than 10�, E04GBF will use 10� instead of XTOL, since 10� is
probably the smallest reasonable setting.

Constraint: XTOL � 0:0.

10: STEPMX – REAL (KIND=nag_wp) Input

On entry: an estimate of the Euclidean distance between the solution and the starting point
supplied by you. (For maximum efficiency, a slight overestimate is preferable.)

E04GBF will ensure that, for each iteration,

Xn
j¼1

x
kð Þ
j � x

k�1ð Þ
j

� �2
� STEPMXð Þ2

where k is the iteration number. Thus, if the problem has more than one solution, E04GBF is
most likely to find the one nearest to the starting point. On difficult problems, a realistic choice
can prevent the sequence of x kð Þ entering a region where the problem is ill-behaved and can help
avoid overflow in the evaluation of F xð Þ. However, an underestimate of STEPMX can lead to
inefficiency.

Suggested value: STEPMX ¼ 100000:0.

Constraint: STEPMX � XTOL.

11: XðNÞ – REAL (KIND=nag_wp) array Input/Output

On entry: XðjÞ must be set to a guess at the jth component of the position of the minimum, for
j ¼ 1; 2; . . . ; n.

On exit: the final point x kð Þ. Thus, if IFAIL ¼ 0 on exit, XðjÞ is the jth component of the
estimated position of the minimum.

12: FSUMSQ – REAL (KIND=nag_wp) Output

On exit: the value of F xð Þ, the sum of squares of the residuals fi xð Þ, at the final point given in X.

13: FVECðMÞ – REAL (KIND=nag_wp) array Output

On exit: the value of the residual fi xð Þ at the final point given in X, for i ¼ 1; 2; . . . ;m.

14: FJACðLDFJAC;NÞ – REAL (KIND=nag_wp) array Output

On exit: the value of the first derivative
@fi
@xj

evaluated at the final point given in X, for

i ¼ 1; 2; . . . ;m and j ¼ 1; 2; . . . ; n.

15: LDFJAC – INTEGER Input

On entry: the first dimension of the array FJAC as declared in the (sub)program from which
E04GBF is called.

Constraint: LDFJAC � M.

16: SðNÞ – REAL (KIND=nag_wp) array Output

On exit: the singular values of the Jacobian matrix at the final point. Thus S may be useful as
information about the structure of your problem.

17: VðLDV;NÞ – REAL (KIND=nag_wp) array Output

On exit: the matrix V associated with the singular value decomposition

J ¼ USV T

of the Jacobian matrix at the final point, stored by columns. This matrix may be useful for
statistical purposes, since it is the matrix of orthonormalized eigenvectors of JTJ .

18: LDV – INTEGER Input

On entry: the first dimension of the array V as declared in the (sub)program from which E04GBF
is called.

Constraint: LDV � N.

19: NITER – INTEGER Output

On exit: the number of iterations which have been performed in E04GBF.

20: NF – INTEGER Output

On exit: the number of times that the residuals have been evaluated (i.e., the number of calls of
LSQFUN). If E04HEV is used as LSQLIN, NF is also the number of times that the Jacobian
matrix has been evaluated.

21: IWðLIWÞ – INTEGER array Communication Array
22: LIW – INTEGER Input

On entry: the dimension of the array IW as declared in the (sub)program from which E04GBF is
called.

Constraint: LIW � 1.

23: WðLWÞ – REAL (KIND=nag_wp) array Communication Array
24: LW – INTEGER Input

On entry: the dimension of the array W as declared in the (sub)program from which E04GBF is
called.

Constraints:

if N > 1, LW � 7� NþM� Nþ 2�Mþ N� N;
if N ¼ 1, LW � 9þ 3�M.

25: IFAIL – INTEGER Input/Output

On entry: IFAIL must be set to 0, �1 or 1. If you are unfamiliar with this argument you should
refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is
detected, the value �1 or 1 is recommended. If the output of error messages is undesirable, then
the value 1 is recommended. Otherwise, because for this routine the values of the output
arguments may be useful even if IFAIL 6¼ 0 on exit, the recommended value is �1. When the
value �1 or 1 is used it is essential to test the value of IFAIL on exit.

On exit: IFAIL ¼ 0 unless the routine detects an error or a warning has been flagged (see
Section 6).

6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Note: E04GBF may return useful information for one or more of the following detected errors or
warnings.

Errors or warnings detected by the routine:

IFAIL < 0

A negative value of IFAIL indicates an exit from E04GBF because you have set IFLAG negative
in LSQFUN. The value of IFAIL will be the same as your setting of IFLAG.

IFAIL ¼ 1

On entry, N < 1,
or M < N,
or MAXCAL < 1,
or ETA < 0:0,
or ETA � 1:0,
or XTOL < 0:0,
or STEPMX < XTOL,
or LDFJAC < M,
or LDV < N,
or LIW < 1,
or LW < 7� NþM� Nþ 2�Mþ N� N when N > 1,
or LW < 9þ 3�M when N ¼ 1.

When this exit occurs, no values will have been assigned to FSUMSQ, or to the elements of
FVEC, FJAC, S or V.

IFAIL ¼ 2

There have been MAXCAL calls of LSQFUN. If steady reductions in the sum of squares, F xð Þ,
were monitored up to the point where this exit occurred, then the exit probably occurred simply
because MAXCAL was set too small, so the calculations should be restarted from the final point
held in X. This exit may also indicate that F xð Þ has no minimum.

IFAIL ¼ 3

The conditions for a minimum have not all been satisfied, but a lower point could not be found.
This could be because XTOL has been set so small that rounding errors in the evaluation of the
residuals and derivatives make attainment of the convergence conditions impossible. See
Section 7 for further information.

IFAIL ¼ 4

The method for computing the singular value decomposition of the Jacobian matrix has failed to
converge in a reasonable number of sub-iterations. It may be worth applying E04GBF again
starting with an initial approximation which is not too close to the point at which the failure
occurred.

IFAIL ¼ �99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

The values IFAIL ¼ 2, 3 or 4 may also be caused by mistakes in LSQFUN, by the formulation of the
problem or by an awkward function. If there are no such mistakes it is worth restarting the calculations
from a different starting point (not the point at which the failure occurred) in order to avoid the region
which caused the failure.

7 Accuracy

A successful exit (IFAIL ¼ 0) is made from E04GBF when (B1, B2 and B3) or B4 or B5 hold, where

B1 � � kð Þ � p kð Þ�� �� < XTOLþ �ð Þ � 1:0þ x kð Þ�� ��� �
B2 � F kð Þ � F k�1ð Þ�� �� < XTOLþ �ð Þ2 � 1:0þ F kð Þ� �
B3 � g kð Þ�� �� < �1=3 � 1:0þ F kð Þ� �
B4 � F kð Þ < �2

B5 � g kð Þ�� �� < ��
ffiffiffiffiffiffiffiffiffi
F kð Þ

p� �1=2

and where :k k and � are as defined in XTOL, and F kð Þ and g kð Þ are the values of F xð Þ and its vector of
first derivatives at x kð Þ.

If IFAIL ¼ 0, then the vector in X on exit, xsol, is almost certainly an estimate of xtrue, the position of
the minimum to the accuracy specified by XTOL.

If IFAIL ¼ 3, then xsol may still be a good estimate of xtrue, but to verify this you should make the
following checks. If

(a) the sequence F x kð Þ� �	 

converges to F xsolð Þ at a superlinear or a fast linear rate, and

(b) g xsolð ÞTg xsolð Þ < 10�where T denotes transpose, then it is almost certain that xsol is a close
approximation to the minimum.

When (b) is true, then usually F xsolð Þ is a close approximation to F xtrueð Þ. The values of F x kð Þ� �
can

be calculated in LSQMON, and the vector g xsolð Þ can be calculated from the contents of FVEC and
FJAC on exit from E04GBF.

Further suggestions about confirmation of a computed solution are given in the E04 Chapter
Introduction.

8 Parallelism and Performance

E04GBF is threaded by NAG for parallel execution in multithreaded implementations of the NAG
Library.

E04GBF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor
library used by this implementation. Consult the documentation for the vendor library for further
information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the
OpenMP environment used within this routine. Please also consult the Users' Note for your
implementation for any additional implementation-specific information.

9 Further Comments

The number of iterations required depends on the number of variables, the number of residuals, the
behaviour of F xð Þ, the accuracy demanded and the distance of the starting point from the solution. The
number of multiplications performed per iteration of E04GBF varies, but for m � n is approximately
n�m2 þO n3

� �
. In addition, each iteration makes at least one call of LSQFUN. So, unless the

residuals and their derivatives can be evaluated very quickly, the run time will be dominated by the
time spent in LSQFUN.

Ideally, the problem should be scaled so that, at the solution, F xð Þ and the corresponding values of the
xj are each in the range �1;þ1ð Þ, and so that at points one unit away from the solution F xð Þ differs
from its value at the solution by approximately one unit. This will usually imply that the Hessian matrix
of F xð Þ at the solution is well-conditioned. It is unlikely that you will be able to follow these
recommendations very closely, but it is worth trying (by guesswork), as sensible scaling will reduce the
difficulty of the minimization problem, so that E04GBF will take less computer time.

When the sum of squares represents the goodness-of-fit of a nonlinear model to observed data, elements
of the variance-covariance matrix of the estimated regression coefficients can be computed by a
subsequent call to E04YCF, using information returned in the arrays S and V. See E04YCF for further
details.

10 Example

This example finds least squares estimates of x1, x2 and x3 in the model

y ¼ x1 þ t1
x2t2 þ x3t3

using the 15 sets of data given in the following table.

y t1 t2 t3
0:14 1:0 15:0 1:0
0:18 2:0 14:0 2:0
0:22 3:0 13:0 3:0
0:25 4:0 12:0 4:0
0:29 5:0 11:0 5:0
0:32 6:0 10:0 6:0
0:35 7:0 9:0 7:0
0:39 8:0 8:0 8:0
0:37 9:0 7:0 7:0
0:58 10:0 6:0 6:0
0:73 11:0 5:0 5:0
0:96 12:0 4:0 4:0
1:34 13:0 3:0 3:0
2:10 14:0 2:0 2:0
4:39 15:0 1:0 1:0

Before calling E04GBF, the program calls E04YAF to check LSQFUN. It uses 0:5; 1:0; 1:5ð Þ as the
initial guess at the position of the minimum.

10.1 Program Text

! E04GBF Example Program Text
! Mark 26 Release. NAG Copyright 2016.

Module e04gbfe_mod

! E04GBF Example Program Module:
! Parameters and User-defined Routines

! .. Use Statements ..
Use nag_library, Only: nag_wp

! .. Implicit None Statement ..
Implicit None

! .. Accessibility Statements ..
Private
Public :: lsqfun, lsqgrd, lsqmon

! .. Parameters ..
Real (Kind=nag_wp), Parameter :: one = 1.0_nag_wp
Real (Kind=nag_wp), Parameter :: two = 2.0_nag_wp
Real (Kind=nag_wp), Parameter :: zero = 0.0_nag_wp
Integer, Parameter :: inc1 = 1

Integer, Parameter, Public :: liw = 1, m = 15, n = 3, nin = 5, &
nout = 6, nt = 3

Integer, Parameter, Public :: ldfjac = m
Integer, Parameter, Public :: ldv = n
Integer, Parameter, Public :: lw = 7*n + m*n + 2*m + n*n
Character (1), Parameter :: trans = ’T’

! .. Local Arrays ..
Real (Kind=nag_wp), Public, Save :: t(m,nt), y(m)

Contains
Subroutine lsqgrd(m,n,fvec,fjac,ldfjac,g)

! Routine to evaluate gradient of the sum of squares

! .. Use Statements ..
Use nag_library, Only: dgemv

! .. Scalar Arguments ..
Integer, Intent (In) :: ldfjac, m, n

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (In) :: fjac(ldfjac,n), fvec(m)
Real (Kind=nag_wp), Intent (Out) :: g(n)

! .. Executable Statements ..
! The NAG name equivalent of dgemv is f06paf

Call dgemv(trans,m,n,one,fjac,ldfjac,fvec,inc1,zero,g,inc1)

g(1:n) = two*g(1:n)

Return

End Subroutine lsqgrd
Subroutine lsqfun(iflag,m,n,xc,fvec,fjac,ldfjac,iw,liw,w,lw)

! Routine to evaluate the residuals and their 1st derivatives.
! This routine is also suitable for use when E04FCV is used as
! LSQLIN, since it can deal with IFLAG = 0 as well as IFLAG = 2.

! .. Scalar Arguments ..
Integer, Intent (Inout) :: iflag
Integer, Intent (In) :: ldfjac, liw, lw, m, n

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Inout) :: fjac(ldfjac,n), w(lw)
Real (Kind=nag_wp), Intent (Out) :: fvec(m)
Real (Kind=nag_wp), Intent (In) :: xc(n)
Integer, Intent (Inout) :: iw(liw)

! .. Local Scalars ..
Real (Kind=nag_wp) :: denom, dummy
Integer :: i

! .. Executable Statements ..
Do i = 1, m

denom = xc(2)*t(i,2) + xc(3)*t(i,3)
fvec(i) = xc(1) + t(i,1)/denom - y(i)

If (iflag/=0) Then
fjac(i,1) = one
dummy = -one/(denom*denom)
fjac(i,2) = t(i,1)*t(i,2)*dummy
fjac(i,3) = t(i,1)*t(i,3)*dummy

End If

End Do

Return

End Subroutine lsqfun
Subroutine lsqmon(m,n,xc,fvec,fjac,ldfjac,s,igrade,niter,nf,iw,liw,w,lw)

! Monitoring routine

! .. Use Statements ..
Use nag_library, Only: ddot

! .. Parameters ..
Integer, Parameter :: ndec = 3

! .. Scalar Arguments ..
Integer, Intent (In) :: igrade, ldfjac, liw, lw, m, n, nf, &

niter
! .. Array Arguments ..

Real (Kind=nag_wp), Intent (In) :: fjac(ldfjac,n), fvec(m), s(n), &
xc(n)

Real (Kind=nag_wp), Intent (Inout) :: w(lw)
Integer, Intent (Inout) :: iw(liw)

! .. Local Scalars ..
Real (Kind=nag_wp) :: fsumsq, gtg
Integer :: j

! .. Local Arrays ..
Real (Kind=nag_wp) :: g(ndec)

! .. Executable Statements ..
! The NAG name equivalent of ddot is f06eaf

fsumsq = ddot(m,fvec,inc1,fvec,inc1)

Call lsqgrd(m,n,fvec,fjac,ldfjac,g)

gtg = ddot(n,g,inc1,g,inc1)

Write (nout,*)
Write (nout,*) &

’ Itn F evals SUMSQ GTG Grade’
Write (nout,99999) niter, nf, fsumsq, gtg, igrade
Write (nout,*)
Write (nout,*) &

’ X G Singular values’

Write (nout,99998)(xc(j),g(j),s(j),j=1,n)

Return

99999 Format (1X,I4,6X,I5,6X,1P,E13.5,6X,1P,E9.1,6X,I3)
99998 Format (1X,1P,E13.5,10X,1P,E9.1,10X,1P,E9.1)

End Subroutine lsqmon
End Module e04gbfe_mod
Program e04gbfe

! E04GBF Example Main Program

! .. Use Statements ..
Use nag_library, Only: e04gbf, e04hev, e04yaf, nag_wp, x02ajf
Use e04gbfe_mod, Only: ldfjac, ldv, liw, lsqfun, lsqgrd, lsqmon, lw, m, &

n, nin, nout, nt, t, y
! .. Implicit None Statement ..

Implicit None
! .. Local Scalars ..

Real (Kind=nag_wp) :: eta, fsumsq, stepmx, xtol
Integer :: i, ifail, iprint, maxcal, nf, niter

! .. Local Arrays ..
Real (Kind=nag_wp) :: fjac(ldfjac,n), fvec(m), g(n), s(n), &

v(ldv,n), w(lw), x(n)
Integer :: iw(liw)

! .. Intrinsic Procedures ..
Intrinsic :: sqrt

! .. Executable Statements ..
Write (nout,*) ’E04GBF Example Program Results’

! Skip heading in data file
Read (nin,*)

! Observations of TJ (J = 1, 2, ..., nt) are held in T(I, J)
! (I = 1, 2, . . . , m)

Do i = 1, m
Read (nin,*) y(i), t(i,1:nt)

End Do

! Check LSQFUN by calling E04YAF at an arbitrary point

x(1:nt) = (/0.19_nag_wp,-1.34_nag_wp,0.88_nag_wp/)

ifail = 0
Call e04yaf(m,n,lsqfun,x,fvec,fjac,ldfjac,iw,liw,w,lw,ifail)

! Continue setting parameters for E04GBF

! Set IPRINT to 1 to obtain output from LSQMON at each iteration
iprint = -1

maxcal = 50*n

! Since E04HEV is being used as LSQLIN, we set ETA to 0.9

eta = 0.9_nag_wp

xtol = 10.0_nag_wp*sqrt(x02ajf())

! We estimate that the minimum will be within 10 units of the
! starting point

stepmx = 10.0_nag_wp

! Set up the starting point

x(1:nt) = (/0.5_nag_wp,1.0_nag_wp,1.5_nag_wp/)

ifail = -1
Call e04gbf(m,n,e04hev,lsqfun,lsqmon,iprint,maxcal,eta,xtol,stepmx,x, &

fsumsq,fvec,fjac,ldfjac,s,v,ldv,niter,nf,iw,liw,w,lw,ifail)

Select Case (ifail)
Case (0,2:)

Write (nout,*)
Write (nout,99999) ’On exit, the sum of squares is’, fsumsq
Write (nout,99999) ’at the point’, x(1:n)

Call lsqgrd(m,n,fvec,fjac,ldfjac,g)

Write (nout,99998) ’The corresponding gradient is’, g(1:n)
Write (nout,*) ’ (machine dependent)’
Write (nout,*) ’and the residuals are’
Write (nout,99997) fvec(1:m)

End Select

99999 Format (1X,A,3F12.4)
99998 Format (1X,A,1P,3E12.3)
99997 Format (1X,1P,E9.1)

End Program e04gbfe

10.2 Program Data

E04GBF Example Program Data
0.14 1.0 15.0 1.0
0.18 2.0 14.0 2.0
0.22 3.0 13.0 3.0
0.25 4.0 12.0 4.0
0.29 5.0 11.0 5.0
0.32 6.0 10.0 6.0
0.35 7.0 9.0 7.0
0.39 8.0 8.0 8.0
0.37 9.0 7.0 7.0
0.58 10.0 6.0 6.0
0.73 11.0 5.0 5.0
0.96 12.0 4.0 4.0
1.34 13.0 3.0 3.0
2.10 14.0 2.0 2.0
4.39 15.0 1.0 1.0

10.3 Program Results

E04GBF Example Program Results

On exit, the sum of squares is 0.0082
at the point 0.0824 1.1330 2.3437
The corresponding gradient is 1.199E-09 -1.865E-11 1.807E-11

(machine dependent)
and the residuals are
-5.9E-03
-2.7E-04
2.7E-04
6.5E-03

-8.2E-04
-1.3E-03
-4.5E-03
-2.0E-02
8.2E-02

-1.8E-02
-1.5E-02
-1.5E-02
-1.1E-02
-4.2E-03
6.8E-03

