
NAG Library Routine Document

E04JYF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

E04JYF is an easy-to-use quasi-Newton algorithm for finding a minimum of a function
F x1; x2; . . . ; xnð Þ, subject to fixed upper and lower bounds of the independent variables
x1; x2; . . . ; xn, using function values only.

It is intended for functions which are continuous and which have continuous first and second
derivatives (although it will usually work even if the derivatives have occasional discontinuities).

2 Specification

SUBROUTINE E04JYF (N, IBOUND, FUNCT1, BL, BU, X, F, IW, LIW, W, LW,
IUSER, RUSER, IFAIL)

&

INTEGER N, IBOUND, IW(LIW), LIW, LW, IUSER(*), IFAIL
REAL (KIND=nag_wp) BL(N), BU(N), X(N), F, W(LW), RUSER(*)
EXTERNAL FUNCT1

3 Description

E04JYF is applicable to problems of the form:

MinimizeF x1; x2; . . . ; xnð Þ subject to lj � xj � uj; j ¼ 1; 2; . . . ; n

when derivatives of F xð Þ are unavailable.

Special provision is made for problems which actually have no bounds on the xj, problems which have
only non-negativity bounds and problems in which l1 ¼ l2 ¼ � � � ¼ ln and u1 ¼ u2 ¼ � � � ¼ un. You must
supply a subroutine to calculate the value of F xð Þ at any point x.

From a starting point you supplied there is generated, on the basis of estimates of the gradient and the
curvature of F xð Þ, a sequence of feasible points which is intended to converge to a local minimum of
the constrained function. An attempt is made to verify that the final point is a minimum.

A typical iteration starts at the current point x where nz (say) variables are free from both their bounds.
The projected gradient vector gz, whose elements are finite difference approximations to the derivatives
of F xð Þ with respect to the free variables, is known. A unit lower triangular matrix L and a diagonal
matrix D (both of dimension nz), such that LDLT is a positive definite approximation of the matrix of
second derivatives with respect to the free variables (i.e., the projected Hessian) are also held. The
equations

LDLTpz ¼ �gz

are solved to give a search direction pz, which is expanded to an n-vector p by an insertion of
appropriate zero elements. Then � is found such that F xþ �pð Þ is approximately a minimum (subject
to the fixed bounds) with respect to �; x is replaced by xþ �p, and the matrices L and D are updated
so as to be consistent with the change produced in the estimated gradient by the step �p. If any variable
actually reaches a bound during the search along p, it is fixed and nz is reduced for the next iteration.
Most iterations calculate gz using forward differences, but central differences are used when they seem
necessary.

There are two sets of convergence criteria – a weaker and a stronger. Whenever the weaker criteria are
satisfied, the Lagrange multipliers are estimated for all the active constraints. If any Lagrange multiplier
estimate is significantly negative, then one of the variables associated with a negative Lagrange
multiplier estimate is released from its bound and the next search direction is computed in the extended

subspace (i.e., nz is increased). Otherwise minimization continues in the current subspace provided that
this is practicable. When it is not, or when the stronger convergence criteria are already satisfied, then,
if one or more Lagrange multiplier estimates are close to zero, a slight perturbation is made in the
values of the corresponding variables in turn until a lower function value is obtained. The normal
algorithm is then resumed from the perturbed point.

If a saddle point is suspected, a local search is carried out with a view to moving away from the saddle
point. A local search is also performed when a point is found which is thought to be a constrained
minimum.

4 References

Gill P E and Murray W (1976) Minimization subject to bounds on the variables NPL Report NAC 72
National Physical Laboratory

5 Arguments

1: N – INTEGER Input

On entry: the number n of independent variables.

Constraint: N � 1.

2: IBOUND – INTEGER Input

On entry: indicates whether the facility for dealing with bounds of special forms is to be used.

It must be set to one of the following values:

IBOUND ¼ 0
If you are supplying all the lj and uj individually.

IBOUND ¼ 1
If there are no bounds on any xj.

IBOUND ¼ 2
If all the bounds are of the form 0 � xj.

IBOUND ¼ 3
If l1 ¼ l2 ¼ . . . ¼ ln and u1 ¼ u2 ¼ . . . ¼ un.

3: FUNCT1 – SUBROUTINE, supplied by the user. External Procedure

You must supply FUNCT1 to calculate the value of the function F xð Þ at any point x. It should be
tested separately before being used with E04JYF (see the E04 Chapter Introduction).

The specification of FUNCT1 is:

SUBROUTINE FUNCT1 (N, XC, FC, IUSER, RUSER)

INTEGER N, IUSER(*)
REAL (KIND=nag_wp) XC(N), FC, RUSER(*)

1: N – INTEGER Input

On entry: the number n of variables.

2: XCðNÞ – REAL (KIND=nag_wp) array Input

On entry: the point x at which the function value is required.

3: FC – REAL (KIND=nag_wp) Output

On exit: the value of the function F at the current point x.

4: IUSERð�Þ – INTEGER array User Workspace
5: RUSERð�Þ – REAL (KIND=nag_wp) array User Workspace

FUNCT1 is called with the arguments IUSER and RUSER as supplied to E04JYF. You
should use the arrays IUSER and RUSER to supply information to FUNCT1.

FUNCT1 must either be a module subprogram USEd by, or declared as EXTERNAL in, the (sub)
program from which E04JYF is called. Arguments denoted as Input must not be changed by this
procedure.

4: BLðNÞ – REAL (KIND=nag_wp) array Input/Output

On entry: the lower bounds lj.

If IBOUND is set to 0, you must set BLðjÞ to lj , for j ¼ 1; 2; . . . ; n. (If a lower bound is not
specified for a particular xj, the corresponding BLðjÞ should be set to �106.)

If IBOUND is set to 3, you must set BLð1Þ to l1; E04JYF will then set the remaining elements of
BL equal to BLð1Þ.
On exit: the lower bounds actually used by E04JYF.

5: BUðNÞ – REAL (KIND=nag_wp) array Input/Output

On entry: the upper bounds uj.

If IBOUND is set to 0, you must set BUðjÞ to uj , for j ¼ 1; 2; . . . ; n. (If an upper bound is not
specified for a particular xj, the corresponding BUðjÞ should be set to 106.)

If IBOUND is set to 3, you must set BUð1Þ to u1; E04JYF will then set the remaining elements
of BU equal to BUð1Þ.
On exit: the upper bounds actually used by E04JYF.

6: XðNÞ – REAL (KIND=nag_wp) array Input/Output

On entry: XðjÞ must be set to an estimate of the jth component of the position of the minimum,
for j ¼ 1; 2; . . . ; n.

On exit: the lowest point found during the calculations. Thus, if IFAIL ¼ 0 on exit, XðjÞ is the
jth component of the position of the minimum.

7: F – REAL (KIND=nag_wp) Output

On exit: the value of F xð Þ corresponding to the final point stored in X.

8: IWðLIWÞ – INTEGER array Output

On exit: if IFAIL ¼ 0, 3 or 5, the first N elements of IW contain information about which
variables are currently on their bounds and which are free. Specifically, if xi is:

– fixed on its upper bound, IWðiÞ is �1;

– fixed on its lower bound, IWðiÞ is �2;

– effectively a constant (i.e., lj ¼ uj), IWðiÞ is �3;

– free, IWðiÞ gives its position in the sequence of free variables.

In addition, IWðNþ 1Þ contains the number of free variables (i.e., nz). The rest of the array is
used as workspace.

9: LIW – INTEGER Input

On entry: the dimension of the array IW as declared in the (sub)program from which E04JYF is
called.

Constraint: LIW � Nþ 2.

10: WðLWÞ – REAL (KIND=nag_wp) array Output

On exit: if IFAIL ¼ 0, 3 or 5, WðiÞ contains a finite difference approximation to the ith element
of the projected gradient vector gz, for i ¼ 1; 2; . . . ;N. In addition, WðNþ 1Þ contains an estimate
of the condition number of the projected Hessian matrix (i.e., k). The rest of the array is used as
workspace.

11: LW – INTEGER Input

On entry: the dimension of the array W as declared in the (sub)program from which E04JYF is
called.

Constraint: LW � max N� N� 1ð Þ=2þ 12� N; 13ð Þ.

12: IUSERð�Þ – INTEGER array User Workspace
13: RUSERð�Þ – REAL (KIND=nag_wp) array User Workspace

IUSER and RUSER are not used by E04JYF, but are passed directly to FUNCT1 and should be
used to pass information to this routine.

14: IFAIL – INTEGER Input/Output

On entry: IFAIL must be set to 0, �1 or 1. If you are unfamiliar with this argument you should
refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is
detected, the value �1 or 1 is recommended. If the output of error messages is undesirable, then
the value 1 is recommended. Otherwise, because for this routine the values of the output
arguments may be useful even if IFAIL 6¼ 0 on exit, the recommended value is �1. When the
value �1 or 1 is used it is essential to test the value of IFAIL on exit.

On exit: IFAIL ¼ 0 unless the routine detects an error or a warning has been flagged (see
Section 6).

6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Note: E04JYF may return useful information for one or more of the following detected errors or
warnings.

Errors or warnings detected by the routine:

IFAIL ¼ 1

On entry, N < 1,
or IBOUND < 0,
or IBOUND > 3,
or IBOUND ¼ 0 and BLðjÞ > BUðjÞ for some j,
or IBOUND ¼ 3 and BLð1Þ > BUð1Þ,
or LIW < Nþ 2,
or LW < max 13; 12� Nþ N� N� 1ð Þ=2ð Þ.

IFAIL ¼ 2

There have been 400� n function evaluations, yet the algorithm does not seem to be converging.
The calculations can be restarted from the final point held in X. The error may also indicate that
F xð Þ has no minimum.

IFAIL ¼ 3

The conditions for a minimum have not all been met but a lower point could not be found and
the algorithm has failed.

IFAIL ¼ 4

An overflow has occurred during the computation. This is an unlikely failure, but if it occurs you
should restart at the latest point given in X.

IFAIL ¼ 5
IFAIL ¼ 6
IFAIL ¼ 7
IFAIL ¼ 8

There is some doubt about whether the point x found by E04JYF is a minimum. The degree of
confidence in the result decreases as IFAIL increases. Thus, when IFAIL ¼ 5 it is probable that
the final x gives a good estimate of the position of a minimum, but when IFAIL ¼ 8 it is very
unlikely that the routine has found a minimum.

IFAIL ¼ 9

In the search for a minimum, the modulus of one of the variables has become very large � 106
� �

.
This indicates that there is a mistake in FUNCT1, that your problem has no finite solution, or that
the problem needs rescaling (see Section 9).

IFAIL ¼ 10

The computed set of forward-difference intervals (stored in Wð9� Nþ 1Þ;Wð9� Nþ 2Þ; . . . ;
Wð10� NÞ) is such that XðiÞ þWð9� Nþ iÞ � XðiÞ for some i.

This is an unlikely failure, but if it occurs you should attempt to select another starting point.

IFAIL ¼ �99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

If you are dissatisfied with the result (e.g., because IFAIL ¼ 5, 6, 7 or 8), it is worth restarting the
calculations from a different starting point (not the point at which the failure occurred) in order to avoid
the region which caused the failure. If persistent trouble occurs and the gradient can be calculated, it
may be advisable to change to a routine which uses gradients (see the E04 Chapter Introduction).

7 Accuracy

A successful exit (IFAIL ¼ 0) is made from E04JYF when (B1, B2 and B3) or B4 hold, and the local
search confirms a minimum, where

B1 	 � kð Þ � p kð Þ�� �� < xtol þ
ffiffi
�

pð Þ � 1:0þ x kð Þ�� ��� �

B2 	 F kð Þ � F k�1ð Þ�� �� < x2tol þ �
� �� 1:0þ F kð Þ�� ��� �

B3 	 g kð Þ
z

�� �� < �1=3 þ xtol

� �� 1:0þ F kð Þ�� ��� �

B4 	 g kð Þ
z

�� �� < 0:01� ffiffi
�

p
.

(Quantities with superscript k are the values at the kth iteration of the quantities mentioned in Section 3,
xtol ¼ 100

ffiffi
�

p
, � is the machine precision and :k k denotes the Euclidean norm. The vector gz is returned

in the array W.)

If IFAIL ¼ 0, then the vector in X on exit, xsol, is almost certainly an estimate of the position of the
minimum, xtrue, to the accuracy specified by xtol.

If IFAIL ¼ 3 or 5, xsol may still be a good estimate of xtrue, but the following checks should be made.
Let k denote an estimate of the condition number of the projected Hessian matrix at xsol. (The value of
k is returned in WðNþ 1Þ). If
(i) the sequence F x kð Þ� �� �

converges to F xsolð Þ at a superlinear or a fast linear rate,

(ii) gz xxolð Þk k2 < 10:0� �, and

(iii) k < 1:0= gz xsolð Þk k,
then it is almost certain that xsol is a close approximation to the position of a minimum. When (ii) is
true, then usually F xsolð Þ is a close approximation to F xtrueð Þ.
When a successful exit is made then, for a computer with a mantissa of t decimals, one would expect to
get about t=2� 1 decimals accuracy in x and about t� 1 decimals accuracy in F , provided the problem
is reasonably well scaled.

8 Parallelism and Performance

E04JYF is not threaded in any implementation.

9 Further Comments

The number of iterations required depends on the number of variables, the behaviour of F xð Þ and the
distance of the starting point from the solution. The number of operations performed in an iteration of
E04JYF is roughly proportional to n2. In addition, each iteration makes at least mþ 1 calls of
FUNCT1, where m is the number of variables not fixed on bounds. So, unless F xð Þ can be evaluated
very quickly, the run time will be dominated by the time spent in FUNCT1.

Ideally the problem should be scaled so that at the solution the value of F xð Þ and the corresponding
values of x1; x2; . . . ; xn are each in the range �1;þ1ð Þ, and so that at points a unit distance away from
the solution, F is approximately a unit value greater than at the minimum. It is unlikely that you will be
able to follow these recommendations very closely, but it is worth trying (by guesswork), as sensible
scaling will reduce the difficulty of the minimization problem, so that E04JYF will take less computer
time.

10 Example

To minimize

F ¼ x1 þ 10x2ð Þ2 þ 5 x3 � x4ð Þ2 þ x2 � 2x3ð Þ4 þ 10 x1 � x4ð Þ4

subject to

1 � x1 � 3
�2 � x2 � 0
1 � x4 � 3;

starting from the initial guess 3;�1; 0; 1ð Þ.

10.1 Program Text

! E04JYF Example Program Text
! Mark 26 Release. NAG Copyright 2016.

Module e04jyfe_mod

! E04JYF Example Program Module:
! Parameters and User-defined Routines

! .. Use Statements ..
Use nag_library, Only: nag_wp

! .. Implicit None Statement ..
Implicit None

! .. Accessibility Statements ..
Private
Public :: funct1

! .. Parameters ..
Integer, Parameter, Public :: n = 4, nout = 6
Integer, Parameter, Public :: liw = n + 2
Integer, Parameter, Public :: lw = n*(n-1)/2 + 12*n

Contains
Subroutine funct1(n,xc,fc,iuser,ruser)

! Routine to evaluate objective function.

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (Out) :: fc
Integer, Intent (In) :: n

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Inout) :: ruser(*)
Real (Kind=nag_wp), Intent (In) :: xc(n)
Integer, Intent (Inout) :: iuser(*)

! .. Local Scalars ..
Real (Kind=nag_wp) :: x1, x2, x3, x4

! .. Executable Statements ..
x1 = xc(1)
x2 = xc(2)
x3 = xc(3)
x4 = xc(4)
fc = (x1+10.0_nag_wp*x2)**2 + 5.0_nag_wp*(x3-x4)**2 + &

(x2-2.0_nag_wp*x3)**4 + 10.0_nag_wp*(x1-x4)**4

Return

End Subroutine funct1
End Module e04jyfe_mod
Program e04jyfe

! E04JYF Example Main Program

! .. Use Statements ..
Use nag_library, Only: e04jyf, nag_wp
Use e04jyfe_mod, Only: funct1, liw, lw, n, nout

! .. Implicit None Statement ..
Implicit None

! .. Local Scalars ..
Real (Kind=nag_wp) :: f
Integer :: ibound, ifail

! .. Local Arrays ..
Real (Kind=nag_wp) :: bl(n), bu(n), ruser(n), w(lw), x(n)
Integer :: iuser(n), iw(6)

! .. Executable Statements ..
Write (nout,*) ’E04JYF Example Program Results’

Flush (nout)

x(1:n) = (/3.0_nag_wp,-1.0_nag_wp,0.0_nag_wp,1.0_nag_wp/)
ibound = 0

! X(3) is unconstrained, so we set BL(3) to a large negative
! number and BU(3) to a large positive number.

bl(1:n) = (/1.0_nag_wp,-2.0_nag_wp,-1.0E6_nag_wp,1.0_nag_wp/)
bu(1:n) = (/3.0_nag_wp,0.0_nag_wp,1.0E6_nag_wp,3.0_nag_wp/)

ifail = -1
Call e04jyf(n,ibound,funct1,bl,bu,x,f,iw,liw,w,lw,iuser,ruser,ifail)

Select Case (ifail)
Case (0,2:)

Write (nout,*)
Write (nout,99999) ’Function value on exit is ’, f
Write (nout,99998) ’at the point’, x(1:n)

End Select

99999 Format (1X,A,F8.4)
99998 Format (1X,A,4F9.4)

End Program e04jyfe

10.2 Program Data

None.

10.3 Program Results

E04JYF Example Program Results
** It is probable that a local minimum has been found,
** but it cannot be guaranteed.
** ABNORMAL EXIT from NAG Library routine E04JYF: IFAIL = 5
** NAG soft failure - control returned

Function value on exit is 2.4338
at the point 1.0000 -0.0852 0.4093 1.0000

