
NAG Library Routine Document

E04KZF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

E04KZF is an easy-to-use modified Newton algorithm for finding a minimum of a function
F x1; x2; . . . ; xnð Þ, subject to fixed upper and lower bounds on the independent variables x1; x2; . . . ; xn,
when first derivatives of F are available. It is intended for functions which are continuous and which
have continuous first and second derivatives (although it will usually work even if the derivatives have
occasional discontinuities).

2 Specification

SUBROUTINE E04KZF (N, IBOUND, FUNCT2, BL, BU, X, F, G, IW, LIW, W, LW,
IUSER, RUSER, IFAIL)

&

INTEGER N, IBOUND, IW(LIW), LIW, LW, IUSER(*), IFAIL
REAL (KIND=nag_wp) BL(N), BU(N), X(N), F, G(N), W(LW), RUSER(*)
EXTERNAL FUNCT2

3 Description

E04KZF is applicable to problems of the form:

MinimizeF x1; x2; . . . ; xnð Þ subject to lj � xj � uj; j ¼ 1; 2; . . . ; n

when first derivatives are known.

Special provision is made for problems which actually have no bounds on the xj, problems which have
only non-negativity bounds, and problems in which l1 ¼ l2 ¼ � � � ¼ ln and u1 ¼ u2 ¼ � � � ¼ un. You
must supply a subroutine to calculate the values of F xð Þ and its first derivatives at any point x.

From a starting point you supplied there is generated, on the basis of estimates of the gradient of the
curvature of F xð Þ, a sequence of feasible points which is intended to converge to a local minimum of
the constrained function.

4 References

Gill P E and Murray W (1976) Minimization subject to bounds on the variables NPL Report NAC 72
National Physical Laboratory

5 Arguments

1: N – INTEGER Input

On entry: the number n of independent variables.

Constraint: N � 1.

2: IBOUND – INTEGER Input

On entry: indicates whether the facility for dealing with bounds of special forms is to be used. It
must be set to one of the following values:

IBOUND ¼ 0
If you are supplying all the lj and uj individually.

IBOUND ¼ 1
If there are no bounds on any xj.

IBOUND ¼ 2
If all the bounds are of the form 0 � xj.

IBOUND ¼ 3
If l1 ¼ l2 ¼ � � � ¼ ln and u1 ¼ u2 ¼ � � � ¼ un.

Constraint: 0 � IBOUND � 3.

3: FUNCT2 – SUBROUTINE, supplied by the user. External Procedure

You must supply this routine to calculate the values of the function F xð Þ and its first derivatives
@F

@xj
at any point x. It should be tested separately before being used in conjunction with E04KZF

(see Chapter E04).

The specification of FUNCT2 is:

SUBROUTINE FUNCT2 (N, XC, FC, GC, IUSER, RUSER)

INTEGER N, IUSER(*)
REAL (KIND=nag_wp) XC(N), FC, GC(N), RUSER(*)

1: N – INTEGER Input

On entry: the number n of variables.

2: XCðNÞ – REAL (KIND=nag_wp) array Input

On entry: the point x at which the function and derivatives are required.

3: FC – REAL (KIND=nag_wp) Output

On exit: the value of the function F at the current point x,

4: GCðNÞ – REAL (KIND=nag_wp) array Output

On exit: GCðjÞ must be set to the value of the first derivative
@F

@xj
at the point x, for

j ¼ 1; 2; . . . ; n.

5: IUSERð�Þ – INTEGER array User Workspace
6: RUSERð�Þ – REAL (KIND=nag_wp) array User Workspace

FUNCT2 is called with the arguments IUSER and RUSER as supplied to E04KZF. You
should use the arrays IUSER and RUSER to supply information to FUNCT2.

FUNCT2 must either be a module subprogram USEd by, or declared as EXTERNAL in, the (sub)
program from which E04KZF is called. Arguments denoted as Input must not be changed by this
procedure.

4: BLðNÞ – REAL (KIND=nag_wp) array Input/Output

On entry: the lower bounds lj.

If IBOUND is set to 0, you must set BLðjÞ to lj , for j ¼ 1; 2; . . . ; n. (If a lower bound is not
specified for a particular xj, the corresponding BLðjÞ should be set to �106.)

If IBOUND is set to 3, you must set BLð1Þ to l1; E04KZF will then set the remaining elements of
BL equal to BLð1Þ.
On exit: the lower bounds actually used by E04KZF.

5: BUðNÞ – REAL (KIND=nag_wp) array Input/Output

On entry: the upper bounds uj.

If IBOUND is set to 0, you must set BUðjÞ to uj , for j ¼ 1; 2; . . . ; n. (If an upper bound is not
specified for a particular xj, the corresponding BUðjÞ should be set to 106.)

If IBOUND is set to 3, you must set BUð1Þ to u1; E04KZF will then set the remaining elements
of BU equal to BUð1Þ.
On exit: the upper bounds actually used by E04KZF.

6: XðNÞ – REAL (KIND=nag_wp) array Input/Output

On entry: XðjÞ must be set to a guess at the jth component of the position of the minimum, for
j ¼ 1; 2; . . . ; n. The routine checks the gradient at the starting point, and is more likely to detect
any error in your programming if the initial XðjÞ are nonzero and mutually distinct.

On exit: the lowest point found during the calculations of the position of the minimum.

7: F – REAL (KIND=nag_wp) Output

On exit: the value of F xð Þ corresponding to the final point stored in X.

8: GðNÞ – REAL (KIND=nag_wp) array Output

On exit: the value of
@F

@xj
corresponding to the final point stored in X, for j ¼ 1; 2; . . . ; n; the

value of GðjÞ for variables not on a bound should normally be close to zero.

9: IWðLIWÞ – INTEGER array Workspace
10: LIW – INTEGER Input

On entry: the dimension of the array IW as declared in the (sub)program from which E04KZF is
called.

Constraint: LIW � Nþ 2.

11: WðLWÞ – REAL (KIND=nag_wp) array Workspace
12: LW – INTEGER Input

On entry: the dimension of the array W as declared in the (sub)program from which E04KZF is
called.

Constraint: LW � max N� Nþ 7ð Þ; 10ð Þ.

13: IUSERð�Þ – INTEGER array User Workspace
14: RUSERð�Þ – REAL (KIND=nag_wp) array User Workspace

IUSER and RUSER are not used by E04KZF, but are passed directly to FUNCT2 and should be
used to pass information to this routine.

15: IFAIL – INTEGER Input/Output

On entry: IFAIL must be set to 0, �1 or 1. If you are unfamiliar with this argument you should
refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is
detected, the value �1 or 1 is recommended. If the output of error messages is undesirable, then
the value 1 is recommended. Otherwise, because for this routine the values of the output
arguments may be useful even if IFAIL 6¼ 0 on exit, the recommended value is �1. When the
value �1 or 1 is used it is essential to test the value of IFAIL on exit.

On exit: IFAIL ¼ 0 unless the routine detects an error or a warning has been flagged (see
Section 6).

E04 – Minimizing or Maximizing a Function E04KZF

Mark 26 E04KZF.3



6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Note: E04KZF may return useful information for one or more of the following detected errors or
warnings.

Errors or warnings detected by the routine:

IFAIL ¼ 1

On entry, N < 1,
or IBOUND < 0,
or IBOUND > 3,
or IBOUND ¼ 0 and BLðjÞ > BUðjÞ for some j,
or IBOUND ¼ 3 and BLð1Þ > BUð1Þ,
or LIW < Nþ 2,
or LW < max 10;N� Nþ 7ð Þð Þ.

IFAIL ¼ 2

There has been a large number of function evaluations, yet the algorithm does not seem to be
converging. The calculations can be restarted from the final point held in X. The error may also
indicate that F xð Þ has no minimum.

IFAIL ¼ 3

The conditions for a minimum have not all been met but a lower point could not be found and
the algorithm has failed.

IFAIL ¼ 4

Not used. (This value of the argument is included to make the significance of IFAIL ¼ 5 etc.
consistent in the easy-to-use routines.)

IFAIL ¼ 5
IFAIL ¼ 6
IFAIL ¼ 7
IFAIL ¼ 8

There is some doubt about whether the point x found by E04KZF is a minimum. The degree of
confidence in the result decreases as IFAIL increases. Thus, when IFAIL ¼ 5 it is probable that
the final x gives a good estimate of the position of a minimum, but when IFAIL ¼ 8 it is very
unlikely that the routine has found a minimum.

IFAIL ¼ 9

In the search for a minimum, the modulus of one of the variables has become very large � 106
� �

.
This indicates that there is a mistake in FUNCT2, that your problem has no finite solution, or that
the problem needs rescaling (see Section 9).

IFAIL ¼ 10

It is very likely that you have made an error in forming the gradient.

IFAIL ¼ �99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

If you are dissatisfied with the result (e.g., because IFAIL ¼ 5, 6, 7 or 8), it is worth restarting the
calculations from a different starting point (not the point at which the failure occurred) in order to avoid
the region which caused the failure. If persistent trouble occurs and it is possible to calculate second
derivatives it may be advisable to change to a routine which uses second derivatives (see the E04
Chapter Introduction).

7 Accuracy

When a successful exit is made then, for a computer with a mantissa of t decimals, one would expect to
get about t=2� 1 decimals accuracy in x and about t� 1 decimals accuracy in F , provided the problem
is reasonably well scaled.

8 Parallelism and Performance

E04KZF is not threaded in any implementation.

9 Further Comments

The number of iterations required depends on the number of variables, the behaviour of F xð Þ and the
distance of the starting point from the solution. The number of operations performed in an iteration of
E04KZF is roughly proportional to n3 þO n2

� �
. In addition, each iteration makes at least mþ 1 calls of

FUNCT2 where m is the number of variables not fixed on bounds. So unless F xð Þ and the gradient
vector can be evaluated very quickly, the run time will be dominated by the time spent in FUNCT2.

Ideally the problem should be scaled so that at the solution the value of F xð Þ and the corresponding
values of x1; x2; . . . ; xn are in the range �1;þ1ð Þ, and so that at points a unit distance away from the
solution, F is approximately a unit value greater than at the minimum. It is unlikely that you will be
able to follow these recommendations very closely, but it is worth trying (by guesswork), as sensible
scaling will reduce the difficulty of the minimization problem, so that E04KZF will take less computer
time.

10 Example

A program to minimize

F ¼ x1 þ 10x2ð Þ2 þ 5 x3 � x4ð Þ2 þ x2 � 2x3ð Þ4 þ 10 x1 � x4ð Þ4

subject to

1 � x1 � 3
�2 � x2 � 0
1 � x4 � 3

starting from the initial guess 3;�1; 0; 1ð Þ.
In practice, it is worth trying to make FUNCT2 as efficient as possible. This has not been done in the
example program for reasons of clarity.

10.1 Program Text

! E04KZF Example Program Text
! Mark 26 Release. NAG Copyright 2016.

Module e04kzfe_mod

! E04KZF Example Program Module:
! Parameters and User-defined Routines

! .. Use Statements ..
Use nag_library, Only: nag_wp

! .. Implicit None Statement ..
Implicit None

! .. Accessibility Statements ..
Private
Public :: funct2

! .. Parameters ..
Integer, Parameter, Public :: n = 4, nout = 6
Integer, Parameter, Public :: liw = n + 2
Integer, Parameter, Public :: lw = n*(n+7)

Contains
Subroutine funct2(n,xc,fc,gc,iuser,ruser)

! Routine to evaluate objective function and its 1st derivatives.

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (Out) :: fc
Integer, Intent (In) :: n

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Out) :: gc(n)
Real (Kind=nag_wp), Intent (Inout) :: ruser(*)
Real (Kind=nag_wp), Intent (In) :: xc(n)
Integer, Intent (Inout) :: iuser(*)

! .. Local Scalars ..
Real (Kind=nag_wp) :: x1, x2, x3, x4

! .. Executable Statements ..
x1 = xc(1)
x2 = xc(2)
x3 = xc(3)
x4 = xc(4)
fc = (x1+10.0_nag_wp*x2)**2 + 5.0_nag_wp*(x3-x4)**2 + &

(x2-2.0_nag_wp*x3)**4 + 10.0_nag_wp*(x1-x4)**4
gc(1) = 2.0_nag_wp*(x1+10.0_nag_wp*x2) + 40.0_nag_wp*(x1-x4)**3
gc(2) = 20.0_nag_wp*(x1+10.0_nag_wp*x2) + 4.0_nag_wp*(x2-2.0_nag_wp*x3 &

)**3
gc(3) = 10.0_nag_wp*(x3-x4) - 8.0_nag_wp*(x2-2.0_nag_wp*x3)**3
gc(4) = -10.0_nag_wp*(x3-x4) - 40.0_nag_wp*(x1-x4)**3

Return

End Subroutine funct2
End Module e04kzfe_mod
Program e04kzfe

! E04KZF Example Main Program

! .. Use Statements ..
Use nag_library, Only: e04kzf, nag_wp
Use e04kzfe_mod, Only: funct2, liw, lw, n, nout

! .. Implicit None Statement ..
Implicit None

! .. Local Scalars ..
Real (Kind=nag_wp) :: f
Integer :: ibound, ifail

! .. Local Arrays ..
Real (Kind=nag_wp) :: bl(n), bu(n), g(n), ruser(1), w(lw), &

x(n)
Integer :: iuser(1), iw(liw)

! .. Executable Statements ..
Write (nout,*) ’E04KZF Example Program Results’
Flush (nout)

E04KZF NAG Library Manual

E04KZF.6 Mark 26



x(1:n) = (/3.0_nag_wp,-1.0_nag_wp,0.0_nag_wp,1.0_nag_wp/)

ibound = 0

! X(3) is unconstrained, so we set BL(3) to a large negative
! number and BU(3) to a large positive number.

bl(1:n) = (/1.0_nag_wp,-2.0_nag_wp,-1.0E6_nag_wp,1.0_nag_wp/)
bu(1:n) = (/3.0_nag_wp,0.0_nag_wp,1.0E6_nag_wp,3.0_nag_wp/)

ifail = -1
Call e04kzf(n,ibound,funct2,bl,bu,x,f,g,iw,liw,w,lw,iuser,ruser,ifail)

Select Case (ifail)
Case (0,2:)

Write (nout,*)
Write (nout,99999) ’Function value on exit is ’, f
Write (nout,99999) ’at the point’, x(1:n)
Write (nout,*) ’the corresponding (machine dependent) gradient is’
Write (nout,99998) g(1:n)

End Select

99999 Format (1X,A,4F12.4)
99998 Format (13X,4E12.4)

End Program e04kzfe

10.2 Program Data

None.

10.3 Program Results

E04KZF Example Program Results
** It is probable that a local minimum has been found,
** but it cannot be guaranteed.
** ABNORMAL EXIT from NAG Library routine E04KZF: IFAIL = 5
** NAG soft failure - control returned

Function value on exit is 2.4338
at the point 1.0000 -0.0852 0.4093 1.0000
the corresponding (machine dependent) gradient is

0.2953E+00 -0.5870E-09 0.1176E-08 0.5907E+01

