

NAG Library Routine Document

F01KFF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F01KFF computes the Fréchet derivative $L(A, E)$ of the p th power (where p is real) of the complex n by n matrix A applied to the complex n by n matrix E . The principal matrix power A^p is also returned.

2 Specification

```
SUBROUTINE F01KFF (N, A, LDA, E, LDE, P, IFAIL)
  INTEGER          N, LDA, LDE, IFAIL
  REAL (KIND=nag_wp) P
  COMPLEX (KIND=nag_wp) A(LDA,*), E(LDE,*)
```

3 Description

For a matrix A with no eigenvalues on the closed negative real line, A^p ($p \in \mathbb{R}$) can be defined as

$$A^p = \exp(p \log(A))$$

where $\log(A)$ is the principal logarithm of A (the unique logarithm whose spectrum lies in the strip $\{z : -\pi < \text{Im}(z) < \pi\}$). If A is nonsingular but has negative real eigenvalues, the principal logarithm is not defined, but a non-principal p th power can be defined by using a non-principal logarithm.

The Fréchet derivative of the matrix p th power of A is the unique linear mapping $E \mapsto L(A, E)$ such that for any matrix E

$$(A+E)^p - A^p - L(A, E) = o(\|E\|).$$

The derivative describes the first-order effect of perturbations in A on the matrix power A^p .

F01KFF uses the algorithms of Higham and Lin (2011) and Higham and Lin (2013) to compute A^p and $L(A, E)$. The real number p is expressed as $p = q + r$ where $q \in (-1, 1)$ and $r \in \mathbb{Z}$. Then $A^p = A^q A^r$. The integer power A^r is found using a combination of binary powering and, if necessary, matrix inversion. The fractional power A^q is computed using a Schur decomposition, a Padé approximant and the scaling and squaring method. The Padé approximant is differentiated in order to obtain the Fréchet derivative of A^q and $L(A, E)$ is then computed using a combination of the chain rule and the product rule for Fréchet derivatives.

4 References

Higham N J (2008) *Functions of Matrices: Theory and Computation* SIAM, Philadelphia, PA, USA

Higham N J and Lin L (2011) A Schur–Padé algorithm for fractional powers of a matrix *SIAM J. Matrix Anal. Appl.* **32(3)** 1056–1078

Higham N J and Lin L (2013) An improved Schur–Padé algorithm for fractional powers of a matrix and their Fréchet derivatives *SIAM J. Matrix Anal. Appl.* **34(3)** 1341–1360

5 Arguments

- 1: N – INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.

- 2: $A(LDA, *)$ – COMPLEX (KIND=nag_wp) array *Input/Output*
Note: the second dimension of the array A must be at least N .
On entry: the n by n matrix A .
On exit: the n by n principal matrix p th power, A^p . Alternatively if $IFAIL = 1$, a non-principal p th power is returned.
- 3: LDA – INTEGER *Input*
On entry: the first dimension of the array A as declared in the (sub)program from which F01KFF is called.
Constraint: $LDA \geq N$.
- 4: $E(LDE, *)$ – COMPLEX (KIND=nag_wp) array *Input/Output*
Note: the second dimension of the array E must be at least N .
On entry: the n by n matrix E .
On exit: the Fréchet derivative $L(A, E)$.
- 5: LDE – INTEGER *Input*
On entry: the first dimension of the array E as declared in the (sub)program from which F01KFF is called.
Constraint: $LDE \geq N$.
- 6: P – REAL (KIND=nag_wp) *Input*
On entry: the required power of A .
- 7: $IFAIL$ – INTEGER *Input/Output*
On entry: $IFAIL$ must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.
For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of $IFAIL$ on exit.**
On exit: $IFAIL = 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry $IFAIL = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$IFAIL = 1$

A has eigenvalues on the negative real line. The principal p th power is not defined in this case, so a non-principal power was returned.

$IFAIL = 2$

A is singular so the p th power cannot be computed.

IFAIL = 3

A^p has been computed using an IEEE double precision Padé approximant, although the arithmetic precision is higher than IEEE double precision.

IFAIL = 4

An unexpected internal error occurred. This failure should not occur and suggests that the routine has been called incorrectly.

IFAIL = -1

On entry, $N = \langle value \rangle$.
Constraint: $N \geq 0$.

IFAIL = -3

On entry, $LDA = \langle value \rangle$ and $N = \langle value \rangle$.
Constraint: $LDA \geq N$.

IFAIL = -5

On entry, $LDE = \langle value \rangle$ and $N = \langle value \rangle$.
Constraint: $LDE \geq N$.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

For a normal matrix A (for which $A^H A = A A^H$), the Schur decomposition is diagonal and the computation of the fractional part of the matrix power reduces to evaluating powers of the eigenvalues of A and then constructing A^p using the Schur vectors. This should give a very accurate result. In general, however, no error bounds are available for the algorithm. See Higham and Lin (2011) and Higham and Lin (2013) for details and further discussion.

If the condition number of the matrix power is required then F01KEF should be used.

8 Parallelism and Performance

F01KFF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F01KFF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The complex allocatable memory required by the algorithm is approximately $6 \times n^2$.

The cost of the algorithm is $O(n^3)$ floating-point operations; see Higham and Lin (2011) and Higham and Lin (2013).

If the matrix p th power alone is required, without the Fréchet derivative, then F01FQF should be used. If the condition number of the matrix power is required then F01KEF should be used. The real analogue of this routine is F01JFF.

10 Example

This example finds A^p and the Fréchet derivative of the matrix power $L(A, E)$, where $p = 0.2$,

$$A = \begin{pmatrix} 2 & 3 & 2 & 1+3i \\ 2+i & 1 & 1 & 2+i \\ 0+i & 2+2i & 0+2i & 0+4i \\ 3 & 0+i & 3 & 1 \end{pmatrix} \quad \text{and} \quad E = \begin{pmatrix} 0+i & 3 & 2 & 1+3i \\ 0+i & 1 & 3+3i & 0+i \\ 0+i & 2+2i & 0+2i & 0 \\ 2 & 0+i & 1 & 1 \end{pmatrix}.$$

10.1 Program Text

```

Program f01kffe

!      F01KFF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: f01kff, nag_wp, x04daf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: p
      Integer                      :: i, ifail, lda, lde, n
!      .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: a(:,,:), e(:,,:)
!      .. Executable Statements ..
      Write (nout,*) 'F01KFF Example Program Results'
      Write (nout,*)
      Flush (nout)
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, p
      lda = n
      lde = n
      Allocate (a(lda,n))
      Allocate (e(lde,n))
!      Read A from data file
      Read (nin,*)(a(i,1:n),i=1,n)
!      Read E from data file
      Read (nin,*)(e(i,1:n),i=1,n)

!      ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0

!      Find A^p and L_p(A,E)
      Call f01kff(n,a,lda,e,lde,p,ifail)

!      Print solution
      Call x04daf('General',' ',n,n,a,lda,'A^p',ifail)
      Write (nout,*)
      Call x04daf('General',' ',n,n,e,lde,'L_p(A,E)',ifail)

End Program f01kffe

```

10.2 Program Data

F01KFF Example Program Data

```

4      0.2                               :Values of N and P

(2.0,0.0)  (3.0,0.0)  (2.0,0.0)  (1.0,3.0)
(2.0,1.0)  (1.0,0.0)  (1.0,0.0)  (2.0,1.0)
(0.0,1.0)  (2.0,2.0)  (0.0,2.0)  (0.0,4.0)
(3.0,0.0)  (0.0,1.0)  (3.0,0.0)  (1.0,0.0)  :End of matrix A

(0.0,1.0)  (3.0,0.0)  (2.0,0.0)  (1.0,3.0)
(0.0,1.0)  (1.0,0.0)  (3.0,3.0)  (0.0,1.0)
(0.0,1.0)  (2.0,2.0)  (0.0,2.0)  (0.0,0.0)
(2.0,0.0)  (0.0,1.0)  (1.0,0.0)  (1.0,0.0)  :End of matrix E

```

10.3 Program Results

F01KFF Example Program Results

```

A^p
      1      2      3      4
1      1.2029      0.0810      0.2374     -0.0520
      -0.0424      0.0428     -0.1718      0.0976

2      0.1311      1.1054     -0.0757      0.2308
      -0.0378      0.1091      0.0066      0.1373

3     -0.0305      0.4878      1.0822     -0.1050
      -0.1948      0.2846      0.2620      0.3131

4      0.3401     -0.3005      0.1838      1.2347
      0.1792     -0.0857     -0.0261     -0.1571

```

```

L_p(A,E)
      1      2      3      4
1      0.0980 -0.0980      0.0410      0.0136
      -0.0926      0.2759 -0.2629      0.1853

2     -0.0644 -0.2093      0.4315      0.1337
      0.3359 -0.3976      0.0395 -0.0976

3      0.1912      0.2279 -0.0963 -0.0925
      0.0032      0.3308      0.1146 -0.3254

4     -0.0907 -0.0153      0.1299      0.2238
      0.1255 -0.4022      0.0694      0.1179

```
