

# NAG Library Routine Document

## F01QJF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

### 1 Purpose

F01QJF finds the  $RQ$  factorization of the real  $m$  by  $n$  ( $m \leq n$ ) matrix  $A$ , so that  $A$  is reduced to upper triangular form by means of orthogonal transformations from the right.

### 2 Specification

```
SUBROUTINE F01QJF (M, N, A, LDA, ZETA, IFAIL)
  INTEGER          M, N, LDA, IFAIL
  REAL (KIND=nag_wp) A(LDA,*), ZETA(M)
```

### 3 Description

The  $m$  by  $n$  matrix  $A$  is factorized as

$$A = \begin{pmatrix} R & 0 \end{pmatrix} P^T \quad \text{when } m < n,$$

$$A = RP^T \quad \text{when } m = n,$$

where  $P$  is an  $n$  by  $n$  orthogonal matrix and  $R$  is an  $m$  by  $m$  upper triangular matrix.

$P$  is given as a sequence of Householder transformation matrices

$$P = P_m \dots P_2 P_1,$$

the  $(m - k + 1)$ th transformation matrix,  $P_k$ , being used to introduce zeros into the  $k$ th row of  $A$ .  $P_k$  has the form

$$P_k = I - u_k u_k^T,$$

where

$$u_k = \begin{pmatrix} w_k \\ \zeta_k \\ 0 \\ z_k \end{pmatrix},$$

$\zeta_k$  is a scalar,  $w_k$  is an  $(k - 1)$  element vector and  $z_k$  is an  $(n - m)$  element vector.  $u_k$  is chosen to annihilate the elements in the  $k$ th row of  $A$ .

The vector  $u_k$  is returned in the  $k$ th element of ZETA and in the  $k$ th row of A, such that  $\zeta_k$  is in ZETA( $k$ ), the elements of  $w_k$  are in A( $k, 1$ ), ..., A( $k, k - 1$ ) and the elements of  $z_k$  are in A( $k, m + 1$ ), ..., A( $k, n$ ). The elements of  $R$  are returned in the upper triangular part of A.

### 4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

Wilkinson J H (1965) *The Algebraic Eigenvalue Problem* Oxford University Press, Oxford

## 5 Arguments

- 1: M – INTEGER *Input*  
*On entry:*  $m$ , the number of rows of the matrix  $A$ .  
 When  $M = 0$  then an immediate return is effected.  
*Constraint:*  $M \geq 0$ .
  
- 2: N – INTEGER *Input*  
*On entry:*  $n$ , the number of columns of the matrix  $A$ .  
*Constraint:*  $N \geq M$ .
  
- 3: A(LDA,\*) – REAL (KIND=nag\_wp) array *Input/Output*  
**Note:** the second dimension of the array  $A$  must be at least  $\max(1, N)$ .  
*On entry:* the leading  $m$  by  $n$  part of the array  $A$  must contain the matrix to be factorized.  
*On exit:* the  $m$  by  $m$  upper triangular part of  $A$  will contain the upper triangular matrix  $R$ , and the  $m$  by  $m$  strictly lower triangular part of  $A$  and the  $m$  by  $(n - m)$  rectangular part of  $A$  to the right of the upper triangular part will contain details of the factorization as described in Section 3.
  
- 4: LDA – INTEGER *Input*  
*On entry:* the first dimension of the array  $A$  as declared in the (sub)program from which F01QJF is called.  
*Constraint:*  $LDA \geq \max(1, M)$ .
  
- 5: ZETA(M) – REAL (KIND=nag\_wp) array *Output*  
*On exit:*  $ZETA(k)$  contains the scalar  $\zeta_k$  for the  $(m - k + 1)$ th transformation. If  $P_k = I$  then  $ZETA(k) = 0.0$ , otherwise  $ZETA(k)$  contains  $\zeta_k$  as described in Section 3 and  $\zeta_k$  is always in the range  $(1.0, \sqrt{2.0})$ .
  
- 6: IFAIL – INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.  
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**  
*On exit:* IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = -1

On entry,  $M < 0$ ,  
 or  $N < M$ ,  
 or  $LDA < M$ .

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

## 7 Accuracy

The computed factors  $R$  and  $P$  satisfy the relation

$$\begin{pmatrix} R & 0 \end{pmatrix} P^T = A + E,$$

where

$$\|E\| \leq c\epsilon\|A\|,$$

$\epsilon$  is the **machine precision** (see X02AJF),  $c$  is a modest function of  $m$  and  $n$ , and  $\|\cdot\|$  denotes the spectral (two) norm.

## 8 Parallelism and Performance

F01QJF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The approximate number of floating-point operations is given by  $2 \times m^2(3n - m)/3$ .

The first  $k$  rows of the orthogonal matrix  $P^T$  can be obtained by calling F01QKF, which overwrites the  $k$  rows of  $P^T$  on the first  $k$  rows of the array A.  $P^T$  is obtained by the call:

```
IFAIL = 0
CALL F01QKF('Separate',M,N,K,A,LDA,ZETA,WORK,IFAIL)
```

WORK must be a  $\max(m - 1, k - m, 1)$  element array. If K is larger than M, then A must have been declared to have at least K rows.

Operations involving the matrix  $R$  can readily be performed by the Level 2 BLAS routines F06PFF (DTRMV) and F06PJF (DTRSV) (see Chapter F06), but note that no test for near singularity of  $R$  is incorporated into F06PJF (DTRSV). If  $R$  is singular, or nearly singular then F02WUF can be used to determine the singular value decomposition of  $R$ .

## 10 Example

This example obtains the  $RQ$  factorization of the 3 by 5 matrix

$$A = \begin{pmatrix} 2.0 & 2.0 & 1.6 & 2.0 & 1.2 \\ 2.5 & 2.5 & -0.4 & -0.5 & -0.3 \\ 2.5 & 2.5 & 2.8 & 0.5 & -2.9 \end{pmatrix}.$$

### 10.1 Program Text

```

Program f01qjfe

!      F01QJF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: f01qjff, nag_wp, x04cbf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Integer                     :: i, ifail, lda, m, n
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: a(:,,:), zeta(:)
      Character (1)                :: dummy(1)
!      .. Executable Statements ..
      Write (nout,*) 'F01QJF Example Program Results'
      Write (nout,*)
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) m, n
      lda = n
      Allocate (a(lda,n),zeta(m))
      Read (nin,*)(a(i,1:n),i=1,m)

!      ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
!      Find the RQ factorization of A
      Call f01qjff(m,n,a,lda,zeta,ifail)

      Write (nout,*) 'RQ factorization of A'
      Write (nout,*)
      Write (nout,*) 'Vector ZETA'
      Write (nout,99999) zeta(1:m)
      Write (nout,*)
      Flush (nout)
      Call x04cbf('G',' ',m,n,a,lda,'F8.4', &
        'Matrix A after factorization (R is in left-hand upper triangle)','N', &
        dummy,'N',dummy,80,0,ifail)

99999 Format (5(1X,F8.4))
End Program f01qjfe

```

### 10.2 Program Data

```

F01QJF Example Program Data
  3      5      : m, n
  2.0    2.0    1.6    2.0    1.2
  2.5    2.5   -0.4   -0.5   -0.3
  2.5    2.5    2.8    0.5   -2.9 : a

```

### 10.3 Program Results

F01QJF Example Program Results

RQ factorization of A

Vector ZETA

1.0092    1.2981    1.2329

Matrix A after factorization (R is in left-hand upper triangle)

-3.1446 -1.0705 -2.2283   0.6333   0.7619

0.5277 -2.8345 -2.2283 -0.1662   0.0945

0.3766   0.3766 -5.3852   0.0753 -0.4368

---