NAG Library Routine Document

F04AEF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F04AEF calculates the accurate solution of a set of real linear equations with multiple right-hand sides using an LU factorization with partial pivoting, and iterative refinement.

2 Specification

```
SUBROUTINE F04AEF (A, LDA, B, LDB, N, M, C, LDC, WKSPCE, AA, LDAA, BB,     &
                   LDBB, IFAIL)

INTEGER            LDA, LDB, N, M, LDC, LDAA, LDBB, IFAIL
REAL (KIND=nag_wp) A(LDA,*), B(LDB,*), C(LDC,*), WKSPCE(max(1,N)),         &
                   AA(LDAA,*), BB(LDBB,*)
```

3 Description

Given a set of real linear equations $AX = B$, the routine first computes an LU factorization of A with partial pivoting, $PA = LU$, where P is a permutation matrix, L is lower triangular and U is unit upper triangular. An approximation to X is found by forward and backward substitution. The residual matrix $R = B - AX$ is then calculated using ***additional precision***, and a correction D to X is found by solving $LUD = PR$. X is replaced by $X + D$ and this iterative refinement of the solution is repeated until full machine accuracy has been obtained.

4 References

Wilkinson J H and Reinsch C (1971) *Handbook for Automatic Computation II, Linear Algebra* Springer–Verlag

5 Arguments

1: A(LDA, $*$) – REAL (KIND=nag_wp) array *Input*

Note: the second dimension of the array A must be at least max(1, N).

On entry: the n by n matrix A.

2: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F04AEF is called.

Constraint: LDA $\geq$ max(1, N).

3: B(LDB, $*$) – REAL (KIND=nag_wp) array *Input*

Note: the second dimension of the array B must be at least max(1, M).

On entry: the n by m right-hand side matrix B.

4: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F04AEF is called.

Constraint: LDB $\geq \max(1, \mathrm{N})$.

5: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

6: M – INTEGER *Input*

On entry: m, the number of right-hand sides.

Constraint: M ≥ 0.

7: C(LDC, $*$) – REAL (KIND=nag_wp) array *Output*

Note: the second dimension of the array C must be at least $\max(1, \mathrm{M})$.

On exit: the n by m solution matrix X.

8: LDC – INTEGER *Input*

On entry: the first dimension of the array C as declared in the (sub)program from which F04AEF is called.

Constraint: LDC $\geq \max(1, \mathrm{N})$.

9: WKSPCE($\max(1, \mathrm{N})$) – REAL (KIND=nag_wp) array *Workspace*

10: AA(LDAA, $*$) – REAL (KIND=nag_wp) array *Output*

Note: the second dimension of the array AA must be at least $\max(1, \mathrm{N})$.

On exit: the triangular factors L and U, except that the unit diagonal elements of U are not stored.

11: LDAA – INTEGER *Input*

On entry: the first dimension of the array AA as declared in the (sub)program from which F04AEF is called.

Constraint: LDAA $\geq \max(1, \mathrm{N})$.

12: BB(LDBB, $*$) – REAL (KIND=nag_wp) array *Output*

Note: the second dimension of the array BB must be at least $\max(1, \mathrm{M})$.

On exit: the final n by m residual matrix $R = B - AX$.

13: LDBB – INTEGER *Input*

On entry: the first dimension of the array BB as declared in the (sub)program from which F04AEF is called.

Constraint: LDBB $\geq \max(1, \mathrm{N})$.

14: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then

the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

The matrix A is singular, possibly due to rounding errors.

IFAIL $= 2$

Iterative refinement fails to improve the solution, i.e., the matrix A is too ill-conditioned.

IFAIL $= 3$

On entry, N < 0,
or M < 0,
or LDA $<$ max(1, N),
or LDB $<$ max(1, N),
or LDC $<$ max(1, N),
or LDAA $<$ max(1, N),
or LDBB $<$ max(1, N).

IFAIL $= -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The computed solutions should be correct to full machine accuracy. For a detailed error analysis see page 107 of Wilkinson and Reinsch (1971).

8 Parallelism and Performance

F04AEF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F04AEF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The time taken by F04AEF is approximately proportional to n^3.

If there is only one right-hand side, it is simpler to use F04ATF.

10 Example

This example solves the set of linear equations $AX = B$ where

$$A = \begin{pmatrix} 33 & 16 & 72 \\ -24 & -10 & -57 \\ -8 & -4 & -17 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} -359 \\ 281 \\ 85 \end{pmatrix}.$$

10.1 Program Text

```
    Program f04aefe

!     F04AEF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: f04aef, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter                :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                           :: i, ifail, lda, ldaa, ldb, ldbb, ldc, &
                                           m, n
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable   :: a(:,:), aa(:,:), b(:,:), bb(:,:),    &
                                           c(:,:), wkspce(:)
!     .. Executable Statements ..
      Write (nout,*) 'F04AEF Example Program Results'
      Write (nout,*)
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n
      lda = n
      ldaa = n
      ldb = n
      ldbb = n
      ldc = n
      m = 1
      Allocate (a(lda,n),aa(ldaa,n),b(ldb,1),bb(ldbb,1),c(ldc,1),wkspce(n))
      Read (nin,*)(a(i,1:n),i=1,n), (b(i,1),i=1,n)

!     ifail: behaviour on error exit
!             =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
      Call f04aef(a,lda,b,ldb,n,m,c,ldc,wkspce,aa,ldaa,bb,ldbb,ifail)

      Write (nout,*) ' Solution'
      Write (nout,99999)(c(i,1),i=1,n)

99999 Format (1X,F9.4)
    End Program f04aefe
```

10.2 Program Data

```
F04AEF Example Program Data
  3                    : n
   33   16   72
  -24  -10  -57
   -8   -4  -17
 -359  281   85         : matrices A and B
```

10.3 Program Results

```
 F04AEF Example Program Results

  Solution
    1.0000
   -2.0000
   -5.0000
```