NAG Library Routine Document

F04ATF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F04ATF calculates the accurate solution of a set of real linear equations with a single right-hand side, using an LU factorization with partial pivoting, and iterative refinement.

2 Specification

```
SUBROUTINE F04ATF (A, LDA, B, N, C, AA, LDAA, WKS1, WKS2, IFAIL)
INTEGER            LDA, N, LDAA, IFAIL
REAL (KIND=nag_wp) A(LDA,*), B(*), C(N), AA(LDAA,N), WKS1(N), WKS2(N)
```

3 Description

Given a set of real linear equations, $Ax = b$, the routine first computes an LU factorization of A with partial pivoting, $PA = LU$, where P is a permutation matrix, L is lower triangular and U is unit upper triangular. An approximation to x is found by forward and backward substitution in $Ly = Pb$ and $Ux = y$. The residual vector $r = b - Ax$ is then calculated using ***additional precision***, and a correction d to x is found by solving $LUd = r$. x is replaced by $x + d$, and this iterative refinement of the solution is repeated until full machine accuracy is obtained.

4 References

Wilkinson J H and Reinsch C (1971) *Handbook for Automatic Computation II, Linear Algebra* Springer–Verlag

5 Arguments

1: A(LDA, $*$) – REAL (KIND=nag_wp) array *Input*

Note: the second dimension of the array A must be at least max(1, N).

On entry: the n by n matrix A.

2: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F04ATF is called.

Constraint: LDA $\geq$ max(1, N).

3: B($*$) – REAL (KIND=nag_wp) array *Input*

Note: the dimension of the array B must be at least max(1, N).

On entry: the right-hand side vector b.

4: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N $\geq$ 0.

5: C(N) – REAL (KIND=nag_wp) array *Output*

On exit: the solution vector x.

6: AA(LDAA, N) – REAL (KIND=nag_wp) array *Output*

Note: the second dimension of the array AA must be at least $\max(1, \mathrm{N})$.

On exit: the triangular factors L and U, except that the unit diagonal elements of U are not stored.

7: LDAA – INTEGER *Input*

On entry: the first dimension of the array AA as declared in the (sub)program from which F04ATF is called.

Constraint: LDAA $\geq \max(1, \mathrm{N})$.

8: WKS1(N) – REAL (KIND=nag_wp) array *Workspace*

9: WKS2(N) – REAL (KIND=nag_wp) array *Workspace*

10: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

The matrix A is singular, possibly due to rounding errors.

IFAIL $= 2$

Iterative refinement fails to improve the solution, i.e., the matrix A is too ill-conditioned.

IFAIL $= 3$

On entry, N < 0,
or LDA $< \max(1, \mathrm{N})$,
or LDAA $< \max(1, \mathrm{N})$.

IFAIL $= -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The computed solutions should be correct to full machine accuracy. For a detailed error analysis see page 107 of Wilkinson and Reinsch (1971).

8 Parallelism and Performance

F04ATF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F04ATF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The time taken by F04ATF is approximately proportional to n^3.

The routine **must not** be called with the same name for arguments B and C.

10 Example

This example solves the set of linear equations $Ax = b$ where

$$A = \begin{pmatrix} 33 & 16 & 72 \\ -24 & -10 & -57 \\ -8 & -4 & -17 \end{pmatrix} \quad \text{and} \quad b = \begin{pmatrix} -359 \\ 281 \\ 85 \end{pmatrix}.$$

10.1 Program Text

```
    Program f04atfe

!     F04ATF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: f04atf, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: i, ifail, lda, ldaa, n
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: a(:,:), aa(:,:), b(:), c(:),         &
                                          wks1(:), wks2(:)
!     .. Executable Statements ..
      Write (nout,*) 'F04ATF Example Program Results'
```

```
      Write (nout,*)
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n
      lda = n
      ldaa = n
      Allocate (a(lda,n),aa(ldaa,n),b(n),c(n),wks1(n),wks2(n))
      Read (nin,*)(a(i,1:n),i=1,n), b(1:n)

!     ifail: behaviour on error exit
!             =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
      Call f04atf(a,lda,b,n,c,aa,ldaa,wks1,wks2,ifail)

      Write (nout,*) ' Solution'
      Write (nout,99999) c(1:n)

99999 Format (1X,F9.4)
    End Program f04atfe
```

10.2 Program Data

```
F04ATF Example Program Data
  3              : n
   33   16   72
  -24  -10  -57
   -8   -4  -17
 -359  281   85   : matrices A and B
```

10.3 Program Results

```
 F04ATF Example Program Results

  Solution
    1.0000
   -2.0000
   -5.0000
```