
NAG Library Routine Document

F04BFF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

F04BFF computes the solution to a real system of linear equations AX ¼ B, where A is an n by n
symmetric positive definite band matrix of band width 2kþ 1, and X and B are n by r matrices. An
estimate of the condition number of A and an error bound for the computed solution are also returned.

2 Specification

SUBROUTINE F04BFF (UPLO, N, KD, NRHS, AB, LDAB, B, LDB, RCOND, ERRBND,
IFAIL)

&

INTEGER N, KD, NRHS, LDAB, LDB, IFAIL
REAL (KIND=nag_wp) AB(LDAB,*), B(LDB,*), RCOND, ERRBND
CHARACTER(1) UPLO

3 Description

The Cholesky factorization is used to factor A as A ¼ UTU , if UPLO ¼ U , or A ¼ LLT, if
UPLO ¼ L , where U is an upper triangular band matrix with k superdiagonals, and L is a lower
triangular band matrix with k subdiagonals. The factored form of A is then used to solve the system of
equations AX ¼ B.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A,
Hammarling S, McKenney A and Sorensen D (1999) LAPACK Users' Guide (3rd Edition) SIAM,
Philadelphia http://www.netlib.org/lapack/lug

Higham N J (2002) Accuracy and Stability of Numerical Algorithms (2nd Edition) SIAM, Philadelphia

5 Arguments

1: UPLO – CHARACTER(1) Input

On entry: if UPLO ¼ U , the upper triangle of the matrix A is stored.

If UPLO ¼ L , the lower triangle of the matrix A is stored.

Constraint: UPLO ¼ U or L .

2: N – INTEGER Input

On entry: the number of linear equations n, i.e., the order of the matrix A.

Constraint: N � 0.

3: KD – INTEGER Input

On entry: the number of superdiagonals k (and the number of subdiagonals) of the band matrix
A.

Constraint: KD � 0.

4: NRHS – INTEGER Input

On entry: the number of right-hand sides r, i.e., the number of columns of the matrix B.

Constraint: NRHS � 0.

5: ABðLDAB; �Þ – REAL (KIND=nag_wp) array Input/Output

Note: the second dimension of the array AB must be at least max 1;Nð Þ.
On entry: the n by n symmetric band matrix A. The upper or lower triangular part of the
symmetric matrix is stored in the first KDþ 1 rows of the array. The jth column of A is stored in
the jth column of the array AB as follows:

The matrix is stored in rows 1 to kþ 1, more precisely,

if UPLO ¼ U , the elements of the upper triangle of A within the band must be stored with
element Aij in ABðkþ 1þ i� j; jÞ for max 1; j� kð Þ � i � j;

if UPLO ¼ L , the elements of the lower triangle of A within the band must be stored with
element Aij in ABð1þ i� j; jÞ for j � i � min n; jþ kð Þ:

See Section 9 below for further details.

On exit: if IFAIL ¼ 0 or Nþ 1, the factor U or L from the Cholesky factorization A ¼ UTU or
A ¼ LLT, in the same storage format as A.

6: LDAB – INTEGER Input

On entry: the first dimension of the array AB as declared in the (sub)program from which
F04BFF is called.

Constraint: LDAB � KDþ 1.

7: BðLDB; �Þ – REAL (KIND=nag_wp) array Input/Output

Note: the second dimension of the array B must be at least max 1;NRHSð Þ.
On entry: the n by r matrix of right-hand sides B.

On exit: if IFAIL ¼ 0 or Nþ 1, the n by r solution matrix X.

8: LDB – INTEGER Input

On entry: the first dimension of the array B as declared in the (sub)program from which F04BFF
is called.

Constraint: LDB � max 1;Nð Þ.

9: RCOND – REAL (KIND=nag_wp) Output

On exit: if IFAIL ¼ 0 or Nþ 1, an estimate of the reciprocal of the condition number of the

matrix A, computed as RCOND ¼ 1= Ak k1 A�1
�� ��

1

� �
.

10: ERRBND – REAL (KIND=nag_wp) Output

On exit: if IFAIL ¼ 0 or Nþ 1, an estimate of the forward error bound for a computed solution
x̂, such that x̂� xk k1= xk k1 � ERRBND, where x̂ is a column of the computed solution returned
in the array B and x is the corresponding column of the exact solution X. If RCOND is less than
machine precision, then ERRBND is returned as unity.

11: IFAIL – INTEGER Input/Output

On entry: IFAIL must be set to 0, �1 or 1. If you are unfamiliar with this argument you should
refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is
detected, the value �1 or 1 is recommended. If the output of error messages is undesirable, then
the value 1 is recommended. Otherwise, if you are not familiar with this argument, the
recommended value is 0. When the value �1 or 1 is used it is essential to test the value of
IFAIL on exit.

On exit: IFAIL ¼ 0 unless the routine detects an error or a warning has been flagged (see
Section 6).

6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL > 0 and IFAIL � N

The principal minor of order valueh i of the matrix A is not positive definite. The factorization
has not been completed and the solution could not be computed.

IFAIL ¼ Nþ 1

A solution has been computed, but RCOND is less than machine precision so that the matrix A
is numerically singular.

IFAIL ¼ �1

On entry, UPLO not one of `U' or `u' or `L' or `l': UPLO ¼ valueh i.

IFAIL ¼ �2

On entry, N ¼ valueh i.
Constraint: N � 0.

IFAIL ¼ �3

On entry, KD ¼ valueh i.
Constraint: KD � 0.

IFAIL ¼ �4

On entry, NRHS ¼ valueh i.
Constraint: NRHS � 0.

IFAIL ¼ �6

On entry, LDAB ¼ valueh i and KD ¼ valueh i.
Constraint: LDAB � KDþ 1.

IFAIL ¼ �8

On entry, LDB ¼ valueh i and N ¼ valueh i.
Constraint: LDB � max 1;Nð Þ.

IFAIL ¼ �99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �999

Dynamic memory allocation failed.

The integer allocatable memory required is N, and the real allocatable memory required is
3� N. Allocation failed before the solution could be computed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The computed solution for a single right-hand side, x̂, satisfies an equation of the form

Aþ Eð Þx̂ ¼ b;

where

Ek k1 ¼ O �ð Þ Ak k1
and � is the machine precision. An approximate error bound for the computed solution is given by

x̂� xk k1
xk k1

� � Að Þ Ek k1
Ak k1

;

where � Að Þ ¼ A�1
�� ��

1
Ak k1, the condition number of A with respect to the solution of the linear

equations. F04BFF uses the approximation Ek k1 ¼ � Ak k1 to estimate ERRBND. See Section 4.4 of
Anderson et al. (1999) for further details.

8 Parallelism and Performance

F04BFF is threaded by NAG for parallel execution in multithreaded implementations of the NAG
Library.

F04BFF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor
library used by this implementation. Consult the documentation for the vendor library for further
information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the
OpenMP environment used within this routine. Please also consult the Users' Note for your
implementation for any additional implementation-specific information.

9 Further Comments

The band storage scheme for the array AB is illustrated by the following example, when n ¼ 6, k ¼ 2,
and UPLO ¼ U :

On entry:

� � a13 a24 a35 a46
� a12 a23 a34 a45 a56
a11 a22 a33 a44 a55 a66

On exit:

� � u13 u24 u35 u46
� u12 u23 u34 u45 u56
u11 u22 u33 u44 u55 u66

Similarly, if UPLO ¼ L the format of AB is as follows:

On entry:

a11 a22 a33 a44 a55 a66
a21 a32 a43 a54 a65 �
a31 a42 a53 a64 � �

On exit:

l11 l22 l33 l44 l55 l66
l21 l32 l43 l54 l65 �
l31 l42 l53 l64 � �

Array elements marked � need not be set and are not referenced by the routine.

Assuming that n � k, the total number of floating-point operations required to solve the equations
AX ¼ B is approximately n kþ 1ð Þ2 for the factorization and 4nkr for the solution following the
factorization. The condition number estimation typically requires between four and five solves and
never more than eleven solves, following the factorization.

In practice the condition number estimator is very reliable, but it can underestimate the true condition
number; see Section 15.3 of Higham (2002) for further details.

The complex analogue of F04BFF is F04CFF.

10 Example

This example solves the equations

AX ¼ B;

where A is the symmetric positive definite band matrix

A ¼
5:49 2:68 0 0
2:68 5:63 �2:39 0

0 �2:39 2:60 �2:22
0 0 �2:22 5:17

0
B@

1
CA and B ¼

22:09 5:10
9:31 30:81

�5:24 �25:82
11:83 22:90

0
B@

1
CA:

An estimate of the condition number of A and an approximate error bound for the computed solutions
are also printed.

10.1 Program Text

Program f04bffe

! F04BFF Example Program Text

! Mark 26 Release. NAG Copyright 2016.

! .. Use Statements ..
Use nag_library, Only: f04bff, nag_wp, x04caf

! .. Implicit None Statement ..
Implicit None

! .. Parameters ..
Integer, Parameter :: nin = 5, nout = 6
Character (1), Parameter :: uplo = ’U’

! .. Local Scalars ..
Real (Kind=nag_wp) :: errbnd, rcond
Integer :: i, ierr, ifail, j, kd, ldab, ldb, n, &

nrhs
! .. Local Arrays ..

Real (Kind=nag_wp), Allocatable :: ab(:,:), b(:,:)
! .. Intrinsic Procedures ..

Intrinsic :: max, min
! .. Executable Statements ..

Write (nout,*) ’F04BFF Example Program Results’
Write (nout,*)
Flush (nout)

! Skip heading in data file

Read (nin,*)
Read (nin,*) n, kd, nrhs
ldab = kd + 1
ldb = n
Allocate (ab(ldab,n),b(ldb,nrhs))

! Read the upper or lower triangular part of the band matrix A
! from data file

If (uplo==’U’) Then
Do i = 1, n

Read (nin,*)(ab(kd+1+i-j,j),j=i,min(n,i+kd))
End Do

Else If (uplo==’L’) Then
Do i = 1, n

Read (nin,*)(ab(1+i-j,j),j=max(1,i-kd),i)
End Do

End If

! Read B from data file
Read (nin,*)(b(i,1:nrhs),i=1,n)

! Solve the equations AX = B for X

! ifail: behaviour on error exit
! =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft

ifail = 1
Call f04bff(uplo,n,kd,nrhs,ab,ldab,b,ldb,rcond,errbnd,ifail)

If (ifail==0) Then
! Print solution, estimate of condition number and approximate
! error bound

ierr = 0
Call x04caf(’General’,’ ’,n,nrhs,b,ldb,’Solution’,ierr)

Write (nout,*)
Write (nout,*) ’Estimate of condition number’
Write (nout,99999) 1.0E0_nag_wp/rcond
Write (nout,*)
Write (nout,*) ’Estimate of error bound for computed solutions’
Write (nout,99999) errbnd

Else If (ifail==n+1) Then
! Matrix A is numerically singular. Print estimate of
! reciprocal of condition number and solution

Write (nout,*)
Write (nout,*) ’Estimate of reciprocal of condition number’
Write (nout,99999) rcond
Write (nout,*)
Flush (nout)

ierr = 0
Call x04caf(’General’,’ ’,n,nrhs,b,ldb,’Solution’,ierr)

Else If (ifail>0 .And. ifail<=n) Then
! The matrix A is not positive definite to working precision

Write (nout,99998) ’The leading minor of order ’, ifail, &
’ is not positive definite’

Else
Write (nout,99997) ifail

End If

99999 Format (6X,1P,E9.1)
99998 Format (1X,A,I3,A)
99997 Format (1X,’ ** F04BFF returned with IFAIL = ’,I5)

End Program f04bffe

10.2 Program Data

F04BFF Example Program Data

4 1 2 : n, kd, nrhs

5.49 2.68
5.63 -2.39

2.60 -2.22
5.17 : matrix A

22.09 5.10
9.31 30.81

-5.24 -25.82
11.83 22.90 : matrix B

10.3 Program Results

F04BFF Example Program Results

Solution
1 2

1 5.0000 -2.0000
2 -2.0000 6.0000
3 -3.0000 -1.0000
4 1.0000 4.0000

Estimate of condition number
7.4E+01

Estimate of error bound for computed solutions
8.2E-15

