

NAG Library Routine Document

F04CBF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F04CBF computes the solution to a complex system of linear equations $AX = B$, where A is an n by n band matrix, with k_l subdiagonals and k_u superdiagonals, and X and B are n by r matrices. An estimate of the condition number of A and an error bound for the computed solution are also returned.

2 Specification

```
SUBROUTINE F04CBF (N, KL, KU, NRHS, AB, LDAB, IPIV, B, LDB, RCOND,      &
                  ERRBND, IFAIL)
```

```
INTEGER          N, KL, KU, NRHS, LDAB, IPIV(N), LDB, IFAIL
REAL (KIND=nag_wp) RCOND, ERRBND
COMPLEX (KIND=nag_wp) AB(LDAB,*), B(LDB,*)
```

3 Description

The LU decomposition with partial pivoting and row interchanges is used to factor A as $A = PLU$, where P is a permutation matrix, L is the product of permutation matrices and unit lower triangular matrices with k_l subdiagonals, and U is upper triangular with $(k_l + k_u)$ superdiagonals. The factored form of A is then used to solve the system of equations $AX = B$.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Higham N J (2002) *Accuracy and Stability of Numerical Algorithms* (2nd Edition) SIAM, Philadelphia

5 Arguments

- 1: N – INTEGER *Input*
On entry: the number of linear equations n , i.e., the order of the matrix A .
Constraint: $N \geq 0$.
- 2: KL – INTEGER *Input*
On entry: the number of subdiagonals k_l , within the band of A .
Constraint: $KL \geq 0$.
- 3: KU – INTEGER *Input*
On entry: the number of superdiagonals k_u , within the band of A .
Constraint: $KU \geq 0$.
- 4: NRHS – INTEGER *Input*
On entry: the number of right-hand sides r , i.e., the number of columns of the matrix B .
Constraint: $NRHS \geq 0$.

- 5: AB(LDAB,*) – COMPLEX (KIND=nag_wp) array Input/Output
Note: the second dimension of the array AB must be at least $\max(1, N)$.
On entry: the n by n matrix A .
The matrix is stored in rows $k_l + 1$ to $2k_l + k_u + 1$; the first k_l rows need not be set, more precisely, the element A_{ij} must be stored in
$$AB(k_l + k_u + 1 + i - j, j) = A_{ij} \quad \text{for } \max(1, j - k_u) \leq i \leq \min(n, j + k_l).$$
See Section 9 for further details.
On exit: if IFAIL ≥ 0 , AB is overwritten by details of the factorization.
The upper triangular band matrix U , with $k_l + k_u$ superdiagonals, is stored in rows 1 to $k_l + k_u + 1$ of the array, and the multipliers used to form the matrix L are stored in rows $k_l + k_u + 2$ to $2k_l + k_u + 1$.
- 6: LDAB – INTEGER Input
On entry: the first dimension of the array AB as declared in the (sub)program from which F04CBF is called.
Constraint: $LDAB \geq 2 \times KL + KU + 1$.
- 7: IPIV(N) – INTEGER array Output
On exit: if IFAIL ≥ 0 , the pivot indices that define the permutation matrix P ; at the i th step row i of the matrix was interchanged with row IPIV(i). IPIV(i) = i indicates a row interchange was not required.
- 8: B(LDB,*) – COMPLEX (KIND=nag_wp) array Input/Output
Note: the second dimension of the array B must be at least $\max(1, NRHS)$.
On entry: the n by r matrix of right-hand sides B .
On exit: if IFAIL = 0 or $N + 1$, the n by r solution matrix X .
- 9: LDB – INTEGER Input
On entry: the first dimension of the array B as declared in the (sub)program from which F04CBF is called.
Constraint: $LDB \geq \max(1, N)$.
- 10: RCOND – REAL (KIND=nag_wp) Output
On exit: if IFAIL ≥ 0 , an estimate of the reciprocal of the condition number of the matrix A , computed as $RCOND = \left(\|A\|_1 \|A^{-1}\|_1 \right)^{-1}$.
- 11: ERRBND – REAL (KIND=nag_wp) Output
On exit: if IFAIL = 0 or $N + 1$, an estimate of the forward error bound for a computed solution \hat{x} , such that $\|\hat{x} - x\|_1 / \|x\|_1 \leq ERRBND$, where \hat{x} is a column of the computed solution returned in the array B and x is the corresponding column of the exact solution X . If RCOND is less than **machine precision**, then ERRBND is returned as unity.
- 12: IFAIL – INTEGER Input/Output
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.
For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the

recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL > 0 and IFAIL \leq N

Diagonal element $\langle value \rangle$ of the upper triangular factor is zero. The factorization has been completed, but the solution could not be computed.

IFAIL = N + 1

A solution has been computed, but RCOND is less than *machine precision* so that the matrix A is numerically singular.

IFAIL = -1

On entry, N = $\langle value \rangle$.
Constraint: N \geq 0.

IFAIL = -2

On entry, KL = $\langle value \rangle$.
Constraint: KL \geq 0.

IFAIL = -3

On entry, KU = $\langle value \rangle$.
Constraint: KU \geq 0.

IFAIL = -4

On entry, NRHS = $\langle value \rangle$.
Constraint: NRHS \geq 0.

IFAIL = -6

On entry, LDAB = $\langle value \rangle$, KL = $\langle value \rangle$ and KU = $\langle value \rangle$.
Constraint: LDAB \geq 2 \times KL + KU + 1.

IFAIL = -9

On entry, LDB = $\langle value \rangle$ and N = $\langle value \rangle$.
Constraint: LDB \geq max(1, N).

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

The real allocatable memory required is N , and the complex allocatable memory required is $2 \times N$. In this case the factorization and the solution X have been computed, but RCOND and ERRBND have not been computed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The computed solution for a single right-hand side, \hat{x} , satisfies an equation of the form

$$(A + E)\hat{x} = b,$$

where

$$\|E\|_1 = O(\epsilon)\|A\|_1$$

and ϵ is the *machine precision*. An approximate error bound for the computed solution is given by

$$\frac{\|\hat{x} - x\|_1}{\|x\|_1} \leq \kappa(A) \frac{\|E\|_1}{\|A\|_1},$$

where $\kappa(A) = \|A^{-1}\|_1\|A\|_1$, the condition number of A with respect to the solution of the linear equations. F04CBF uses the approximation $\|E\|_1 = \epsilon\|A\|_1$ to estimate ERRBND. See Section 4.4 of Anderson *et al.* (1999) for further details.

8 Parallelism and Performance

F04CBF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F04CBF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The band storage scheme for the array AB is illustrated by the following example, when $n = 6$, $k_l = 1$, and $k_u = 2$. Storage of the band matrix A in the array AB:

*	*	*	+	+	+
*	*	a_{13}	a_{24}	a_{35}	a_{46}
*	a_{12}	a_{23}	a_{34}	a_{45}	a_{56}
a_{11}	a_{22}	a_{33}	a_{44}	a_{55}	a_{66}
a_{21}	a_{32}	a_{43}	a_{54}	a_{65}	*

Array elements marked * need not be set and are not referenced by the routine. Array elements marked + need not be set, but are defined on exit from the routine and contain the elements u_{14} , u_{25} and u_{36} .

The total number of floating-point operations required to solve the equations $AX = B$ depends upon the pivoting required, but if $n \gg k_l + k_u$ then it is approximately bounded by $O(nk_l(k_l + k_u))$ for the factorization and $O(n(2k_l + k_u), r)$ for the solution following the factorization. The condition number estimation typically requires between four and five solves and never more than eleven solves, following the factorization.

In practice the condition number estimator is very reliable, but it can underestimate the true condition number; see Section 15.3 of Higham (2002) for further details.

The real analogue of F04CBF is F04BBF.

10 Example

This example solves the equations

$$AX = B,$$

where A is the band matrix

$$A = \begin{pmatrix} -1.65 + 2.26i & -2.05 - 0.85i & 0.97 - 2.84i & 0 \\ 0.00 + 6.30i & -1.48 - 1.75i & -3.99 + 4.01i & 0.59 - 0.48i \\ 0 & -0.77 + 2.83i & -1.06 + 1.94i & 3.33 - 1.04i \\ 0 & 0 & 4.48 - 1.09i & -0.46 - 1.72i \end{pmatrix}$$

and

$$B = \begin{pmatrix} -1.06 + 21.50i & 12.85 + 2.84i \\ -22.72 - 53.90i & -70.22 + 21.57i \\ 28.24 - 38.60i & -20.73 - 1.23i \\ -34.56 + 16.73i & 26.01 + 31.97i \end{pmatrix}.$$

An estimate of the condition number of A and an approximate error bound for the computed solutions are also printed.

10.1 Program Text

```

Program f04cbfe

!      F04CBF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: f04cbf, nag_wp, x04dbf, x04dff
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: errbnd, rcond
      Integer                     :: i, ierr, ifail, j, k, kl, ku, ldab, &
                                   ldb, n, nrhs
!      .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: ab(:,,:), b(:,,:)
      Integer, Allocatable           :: ipiv(:)
      Character (1)                  :: clabs(1), rlabs(1)
!      .. Intrinsic Procedures ..
      Intrinsic                      :: max, min
!      .. Executable Statements ..
      Write (nout,*) 'F04CBF Example Program Results'
      Write (nout,*)
      Flush (nout)
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, kl, ku, nrhs
      ldab = 2*kl + ku + 1
      ldb = n
      Allocate (ab(ldab,n),b(ldb,nrhs),ipiv(n))
!      Read A and B from data file

      k = kl + ku + 1
      Read (nin,*)((ab(k+i-j,j),j=max(i-kl,1),min(i+ku,n)),i=1,n)
      Read (nin,*)(b(i,1:nrhs),i=1,n)

!      Solve the equations AX = B for X

!      ifail: behaviour on error exit

```

```

!           =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
ifail = 1
Call f04cbf(n,kl,ku,nrhs,ab,ldab,ipiv,b,ldb,rcond,errbnd,ifail)

If (ifail==0) Then
!       Print solution, estimate of condition number and approximate
!       error bound

        ierr = 0
        Call x04dbf('General',' ',n,nrhs,b,ldb,'Bracketed','F7.4','Solution', &
            'Integer',rlabs,'Integer',clabs,80,0,ierr)

        Write (nout,*)
        Write (nout,*) 'Estimate of condition number'
        Write (nout,99999) 1.0E0_nag_wp/rcond
        Write (nout,*)
        Write (nout,*) 'Estimate of error bound for computed solutions'
        Write (nout,99999) errbnd
    Else If (ifail==n+1) Then
!       Matrix A is numerically singular. Print estimate of
!       reciprocal of condition number and solution
        Write (nout,*)
        Write (nout,*) 'Estimate of reciprocal of condition number'
        Write (nout,99999) rcond
        Write (nout,*)
        Flush (nout)

        ierr = 0
        Call x04dbf('General',' ',n,nrhs,b,ldb,'Bracketed','F7.4','Solution', &
            'Integer',rlabs,'Integer',clabs,80,0,ierr)

    Else If (ifail>0 .And. ifail<=n) Then
!       The upper triangular matrix U is exactly singular. Print
!       details of factorization
        Write (nout,*)
        Flush (nout)

        ierr = 0
        Call x04dff(n,n,kl,kl+ku,ab,ldab,'Bracketed','F7.4', &
            'Details of factorization','None',rlabs,'None',clabs,80,0,ierr)

!       Print pivot indices
        Write (nout,*)
        Write (nout,*) 'Pivot indices'
        Write (nout,99998) ipiv(1:n)
    Else
        Write (nout,99997) ifail
    End If

99999 Format (4X,1P,E9.1)
99998 Format ((1X,7I11))
99997 Format (1X,' ** F04CBF returned with IFAIL = ',I5)
End Program f04cbfe

```

10.2 Program Data

F04CBF Example Program Data

```

      4           1           2           2           : n, kl, ku, nrhs

( -1.65,  2.26) ( -2.05, -0.85) (  0.97, -2.84)
(  0.00,  6.30) ( -1.48, -1.75) ( -3.99,  4.01) (  0.59, -0.48)
              ( -0.77,  2.83) ( -1.06,  1.94) (  3.33, -1.04)
                      (  4.48, -1.09) ( -0.46, -1.72) : matrix A

( -1.06, 21.50) ( 12.85,  2.84)
(-22.72,-53.90) (-70.22, 21.57)
( 28.24,-38.60) (-20.73, -1.23)
(-34.56, 16.73) ( 26.01, 31.97)                      : matrix B

```

10.3 Program Results

F04CBF Example Program Results

Solution

	1	2
1	(-3.0000, 2.0000)	(1.0000, 6.0000)
2	(1.0000,-7.0000)	(-7.0000,-4.0000)
3	(-5.0000, 4.0000)	(3.0000, 5.0000)
4	(6.0000,-8.0000)	(-8.0000, 2.0000)

Estimate of condition number
1.0E+02

Estimate of error bound for computed solutions
1.2E-14
