

# NAG Library Routine Document

## F07AGF (DGECON)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F07AGF (DGECON) estimates the condition number of a real matrix  $A$ , where  $A$  has been factorized by F07ADF (DGETRF).

### 2 Specification

```
SUBROUTINE F07AGF (NORM, N, A, LDA, ANORM, RCOND, WORK, IWORK, INFO)
  INTEGER          N, LDA, IWORK(N), INFO
  REAL (KIND=nag_wp) A(LDA,*), ANORM, RCOND, WORK(4*N)
  CHARACTER(1)     NORM
```

The routine may be called by its LAPACK name *dgecon*.

### 3 Description

F07AGF (DGECON) estimates the condition number of a real matrix  $A$ , in either the 1-norm or the  $\infty$ -norm:

$$\kappa_1(A) = \|A\|_1 \|A^{-1}\|_1 \quad \text{or} \quad \kappa_\infty(A) = \|A\|_\infty \|A^{-1}\|_\infty.$$

Note that  $\kappa_\infty(A) = \kappa_1(A^T)$ .

Because the condition number is infinite if  $A$  is singular, the routine actually returns an estimate of the **reciprocal** of the condition number.

The routine should be preceded by a call to F06RAF to compute  $\|A\|_1$  or  $\|A\|_\infty$ , and a call to F07ADF (DGETRF) to compute the  $LU$  factorization of  $A$ . The routine then uses Higham's implementation of Hager's method (see Higham (1988)) to estimate  $\|A^{-1}\|_1$  or  $\|A^{-1}\|_\infty$ .

### 4 References

Higham N J (1988) FORTRAN codes for estimating the one-norm of a real or complex matrix, with applications to condition estimation *ACM Trans. Math. Software* **14** 381–396

### 5 Arguments

- 1: NORM – CHARACTER(1) *Input*  
*On entry:* indicates whether  $\kappa_1(A)$  or  $\kappa_\infty(A)$  is estimated.  
 NORM = '1' or 'O'  
      $\kappa_1(A)$  is estimated.  
 NORM = 'I'  
      $\kappa_\infty(A)$  is estimated.  
*Constraint:* NORM = '1', 'O' or 'I'.
- 2: N – INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .

- 3: A(LDA,\*) – REAL (KIND=nag\_wp) array *Input*  
**Note:** the second dimension of the array A must be at least  $\max(1, N)$ .  
*On entry:* the *LU* factorization of A, as returned by F07ADF (DGETRF).
- 4: LDA – INTEGER *Input*  
*On entry:* the first dimension of the array A as declared in the (sub)program from which F07AGF (DGECON) is called.  
*Constraint:*  $LDA \geq \max(1, N)$ .
- 5: ANORM – REAL (KIND=nag\_wp) *Input*  
*On entry:* if NORM = '1' or 'O', the 1-norm of the **original** matrix A.  
 If NORM = 'I', the  $\infty$ -norm of the **original** matrix A.  
 ANORM may be computed by calling F06RAF with the same value for the argument NORM.  
 ANORM must be computed either **before** calling F07ADF (DGETRF) or else from a **copy** of the original matrix A (see Section 10).  
*Constraint:*  $ANORM \geq 0.0$ .
- 6: RCOND – REAL (KIND=nag\_wp) *Output*  
*On exit:* an estimate of the reciprocal of the condition number of A. RCOND is set to zero if exact singularity is detected or the estimate underflows. If RCOND is less than **machine precision**, A is singular to working precision.
- 7: WORK(4 × N) – REAL (KIND=nag\_wp) array *Workspace*
- 8: IWORK(N) – INTEGER array *Workspace*
- 9: INFO – INTEGER *Output*  
*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If INFO =  $-i$ , argument  $i$  had an illegal value. An explanatory message is output, and execution of the program is terminated.

## 7 Accuracy

The computed estimate RCOND is never less than the true value  $\rho$ , and in practice is nearly always less than  $10\rho$ , although examples can be constructed where RCOND is much larger.

## 8 Parallelism and Performance

F07AGF (DGECON) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

A call to F07AGF (DGECON) involves solving a number of systems of linear equations of the form  $Ax = b$  or  $A^T x = b$ ; the number is usually 4 or 5 and never more than 11. Each solution involves approximately  $2n^2$  floating-point operations but takes considerably longer than a call to F07AEF (DGETRS) with one right-hand side, because extra care is taken to avoid overflow when  $A$  is approximately singular.

The complex analogue of this routine is F07AUF (ZGECON).

## 10 Example

This example estimates the condition number in the 1-norm of the matrix  $A$ , where

$$A = \begin{pmatrix} 1.80 & 2.88 & 2.05 & -0.89 \\ 5.25 & -2.95 & -0.95 & -3.80 \\ 1.58 & -2.69 & -2.90 & -1.04 \\ -1.11 & -0.66 & -0.59 & 0.80 \end{pmatrix}.$$

Here  $A$  is nonsymmetric and must first be factorized by F07ADF (DGETRF). The true condition number in the 1-norm is 152.16.

### 10.1 Program Text

```

Program f07agfe

!      F07AGF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: dgecon, dgetrf, dlange => f06raf, nag_wp, x02ajf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
      Character (1), Parameter   :: norm = '1'
!      .. Local Scalars ..
      Real (Kind=nag_wp)         :: anorm, rcond
      Integer                    :: i, info, lda, n
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: a(:, :), work(:)
      Integer, Allocatable        :: ipiv(:), iwork(:)
!      .. Executable Statements ..
      Write (nout,*) 'F07AGF Example Program Results'
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n
      lda = n
      Allocate (a(lda,n),work(4*n),ipiv(n),iwork(n))

!      Read A from data file

      Read (nin,*)(a(i,1:n),i=1,n)

!      Compute norm of A

!      f06raf is the NAG name equivalent of the LAPACK auxiliary dlange
      anorm = dlange(norm,n,n,a,lda,work)

!      Factorize A
!      The NAG name equivalent of dgetrf is f07adf
      Call dgetrf(n,n,a,lda,ipiv,info)

      Write (nout,*)
      If (info==0) Then

```

```

!      Estimate condition number

!      The NAG name equivalent of dgecon is f07agf
      Call dgecon(norm,n,a,lda,anorm,rcond,work,iwork,info)

      If (rcond>=x02ajf()) Then
        Write (nout,99999) 'Estimate of condition number =',      &
          1.0E0_nag_wp/rcond
      Else
        Write (nout,*) 'A is singular to working precision'
      End If
    Else
      Write (nout,*) 'The factor U is singular'
    End If

99999 Format (1X,A,1P,E10.2)
      End Program f07agfe

```

## 10.2 Program Data

```

F07AGF Example Program Data
4                               :Value of N
1.80   2.88   2.05  -0.89
5.25  -2.95  -0.95  -3.80
1.58  -2.69  -2.90  -1.04
-1.11 -0.66 -0.59   0.80   :End of matrix A

```

## 10.3 Program Results

F07AGF Example Program Results

Estimate of condition number = 1.52E+02

---