

# NAG Library Routine Document

## F07ARF (ZGETRF)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F07ARF (ZGETRF) computes the  $LU$  factorization of a complex  $m$  by  $n$  matrix.

### 2 Specification

```
SUBROUTINE F07ARF (M, N, A, LDA, IPIV, INFO)
  INTEGER          M, N, LDA, IPIV(min(M,N)), INFO
  COMPLEX (KIND=nag_wp) A(LDA,*)
```

The routine may be called by its LAPACK name *zgetrf*.

### 3 Description

F07ARF (ZGETRF) forms the  $LU$  factorization of a complex  $m$  by  $n$  matrix  $A$  as  $A = PLU$ , where  $P$  is a permutation matrix,  $L$  is lower triangular with unit diagonal elements (lower trapezoidal if  $m > n$ ) and  $U$  is upper triangular (upper trapezoidal if  $m < n$ ). Usually  $A$  is square ( $m = n$ ), and both  $L$  and  $U$  are triangular. The routine uses partial pivoting, with row interchanges.

### 4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

### 5 Arguments

- 1: M – INTEGER *Input*  
*On entry:*  $m$ , the number of rows of the matrix  $A$ .  
*Constraint:*  $M \geq 0$ .
- 2: N – INTEGER *Input*  
*On entry:*  $n$ , the number of columns of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 3: A(LDA,\*) – COMPLEX (KIND=nag\_wp) array *Input/Output*  
**Note:** the second dimension of the array  $A$  must be at least  $\max(1, N)$ .  
*On entry:* the  $m$  by  $n$  matrix  $A$ .  
*On exit:* the factors  $L$  and  $U$  from the factorization  $A = PLU$ ; the unit diagonal elements of  $L$  are not stored.
- 4: LDA – INTEGER *Input*  
*On entry:* the first dimension of the array  $A$  as declared in the (sub)program from which F07ARF (ZGETRF) is called.  
*Constraint:*  $LDA \geq \max(1, M)$ .

- 5: IPIV(min(M,N)) – INTEGER array *Output*  
*On exit:* the pivot indices that define the permutation matrix. At the  $i$ th step, if  $\text{IPIV}(i) > i$  then row  $i$  of the matrix  $A$  was interchanged with row  $\text{IPIV}(i)$ , for  $i = 1, 2, \dots, \min(m, n)$ .  $\text{IPIV}(i) \leq i$  indicates that, at the  $i$ th step, a row interchange was not required.
- 6: INFO – INTEGER *Output*  
*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If INFO =  $-i$ , argument  $i$  had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

Element  $\langle \text{value} \rangle$  of the diagonal is exactly zero. The factorization has been completed, but the factor  $U$  is exactly singular, and division by zero will occur if it is used to solve a system of equations.

## 7 Accuracy

The computed factors  $L$  and  $U$  are the exact factors of a perturbed matrix  $A + E$ , where

$$|E| \leq c(\min(m, n))\epsilon P|L||U|,$$

$c(n)$  is a modest linear function of  $n$ , and  $\epsilon$  is the *machine precision*.

## 8 Parallelism and Performance

F07ARF (ZGETRF) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F07ARF (ZGETRF) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The total number of real floating-point operations is approximately  $\frac{8}{3}n^3$  if  $m = n$  (the usual case),  $\frac{4}{3}n^2(3m - n)$  if  $m > n$  and  $\frac{4}{3}m^2(3n - m)$  if  $m < n$ .

A call to this routine with  $m = n$  may be followed by calls to the routines:

F07ASF (ZGETRS) to solve  $AX = B$ ,  $A^T X = B$  or  $A^H X = B$ ;

F07AUF (ZGECON) to estimate the condition number of  $A$ ;

F07AWF (ZGETRI) to compute the inverse of  $A$ .

The real analogue of this routine is F07ADF (DGETRF).

## 10 Example

This example computes the  $LU$  factorization of the matrix  $A$ , where

$$A = \begin{pmatrix} -1.34 + 2.55i & 0.28 + 3.17i & -6.39 - 2.20i & 0.72 - 0.92i \\ -0.17 - 1.41i & 3.31 - 0.15i & -0.15 + 1.34i & 1.29 + 1.38i \\ -3.29 - 2.39i & -1.91 + 4.42i & -0.14 - 1.35i & 1.72 + 1.35i \\ 2.41 + 0.39i & -0.56 + 1.47i & -0.83 - 0.69i & -1.96 + 0.67i \end{pmatrix}.$$

### 10.1 Program Text

Program f07arfe

```
!      F07ARF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: nag_wp, x04dbf, zgetrf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Integer                     :: i, ifail, info, lda, m, n
!      .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: a(:, :)
      Integer, Allocatable        :: ipiv(:)
      Character (1)               :: clabs(1), rlabs(1)
!      .. Intrinsic Procedures ..
      Intrinsic                   :: min
!      .. Executable Statements ..
      Write (nout,*) 'F07ARF Example Program Results'
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) m, n
      lda = m
      Allocate (a(lda,n),ipiv(n))

!      Read A from data file

      Read (nin,*)(a(i,1:n),i=1,m)

!      Factorize A

!      The NAG name equivalent of zgetrf is f07arf
      Call zgetrf(m,n,a,lda,ipiv,info)

!      Print details of factorization

      Write (nout,*)
      Flush (nout)

!      ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
      Call x04dbf('General',' ',m,n,a,lda,'Bracketed','F7.4',
        'Details of factorization','Integer',rlabs,'Integer',clabs,80,0,ifail) &

!      Print pivot indices

      Write (nout,*)
      Write (nout,*) 'IPIV'
      Write (nout,99999) ipiv(1:min(m,n))

      If (info/=0) Then
```

```

      Write (nout,*) 'The factor U is singular'
    End If

99999 Format ((1X,I12,3I18))
    End Program f07arfe

```

## 10.2 Program Data

F07ARF Example Program Data

```

  4  4                                     :Values of M and N
(-1.34, 2.55) ( 0.28, 3.17) (-6.39,-2.20) ( 0.72,-0.92)
(-0.17,-1.41) ( 3.31,-0.15) (-0.15, 1.34) ( 1.29, 1.38)
(-3.29,-2.39) (-1.91, 4.42) (-0.14,-1.35) ( 1.72, 1.35)
( 2.41, 0.39) (-0.56, 1.47) (-0.83,-0.69) (-1.96, 0.67) :End of matrix A

```

## 10.3 Program Results

F07ARF Example Program Results

Details of factorization

```

      1          2          3          4
1  (-3.2900,-2.3900) (-1.9100, 4.4200) (-0.1400,-1.3500) ( 1.7200, 1.3500)
2  ( 0.2376, 0.2560) ( 4.8952,-0.7114) (-0.4623, 1.6966) ( 1.2269, 0.6190)
3  (-0.1020,-0.7010) (-0.6691, 0.3689) (-5.1414,-1.1300) ( 0.9983, 0.3850)
4  (-0.5359, 0.2707) (-0.2040, 0.8601) ( 0.0082, 0.1211) ( 0.1482,-0.1252)

```

IPIV

```

      3          2          3          4

```

---