
NAG Library Routine Document

F07BBF (DGBSVX)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

F07BBF (DGBSVX) uses the LU factorization to compute the solution to a real system of linear
equations

AX ¼ B or ATX ¼ B;

where A is an n by n band matrix with kl subdiagonals and ku superdiagonals, and X and B are n by r
matrices. Error bounds on the solution and a condition estimate are also provided.

2 Specification

SUBROUTINE F07BBF (FACT, TRANS, N, KL, KU, NRHS, AB, LDAB, AFB, LDAFB,
IPIV, EQUED, R, C, B, LDB, X, LDX, RCOND, FERR, BERR,
WORK, IWORK, INFO)

&
&

INTEGER N, KL, KU, NRHS, LDAB, LDAFB, IPIV(*), LDB, LDX,
IWORK(N), INFO

&

REAL (KIND=nag_wp) AB(LDAB,*), AFB(LDAFB,*), R(*), C(*), B(LDB,*),
X(LDX,*), RCOND, FERR(NRHS), BERR(NRHS),
WORK(max(1,3*N))

&
&

CHARACTER(1) FACT, TRANS, EQUED

The routine may be called by its LAPACK name dgbsvx.

3 Description

F07BBF (DGBSVX) performs the following steps:

1. Equilibration

The linear system to be solved may be badly scaled. However, the system can be equilibrated as a
first stage by setting FACT ¼ E . In this case, real scaling factors are computed and these factors
then determine whether the system is to be equilibrated. Equilibrated forms of the systems
AX ¼ B and ATX ¼ B are

DRADCð Þ D�1
C X

� � ¼ DRB

and

DRADCð ÞT D�1
R X

� � ¼ DCB;

respectively, where DR and DC are diagonal matrices, with positive diagonal elements, formed
from the computed scaling factors.

When equilibration is used, A will be overwritten by DRADC and B will be overwritten by DRB
(or DCB when the solution of ATX ¼ B is sought).

2. Factorization

The matrix A, or its scaled form, is copied and factored using the LU decomposition

A ¼ PLU;

where P is a permutation matrix, L is a unit lower triangular matrix, and U is upper triangular.

This stage can be by-passed when a factored matrix (with scaled matrices and scaling factors) are
supplied; for example, as provided by a previous call to F07BBF (DGBSVX) with the same matrix
A.

3. Condition Number Estimation

The LU factorization of A determines whether a solution to the linear system exists. If some
diagonal element of U is zero, then U is exactly singular, no solution exists and the routine returns
with a failure. Otherwise the factorized form of A is used to estimate the condition number of the
matrix A. If the reciprocal of the condition number is less than machine precision then a warning
code is returned on final exit.

4. Solution

The (equilibrated) system is solved for X (D�1
C X or D�1

R X) using the factored form of A
(DRADC).

5. Iterative Refinement

Iterative refinement is applied to improve the computed solution matrix and to calculate error
bounds and backward error estimates for the computed solution.

6. Construct Solution Matrix X

If equilibration was used, the matrix X is premultiplied by DC (if TRANS ¼ N ) or DR (if
TRANS ¼ T or C ) so that it solves the original system before equilibration.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A,
Hammarling S, McKenney A and Sorensen D (1999) LAPACK Users' Guide (3rd Edition) SIAM,
Philadelphia http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) Matrix Computations (3rd Edition) Johns Hopkins University
Press, Baltimore

Higham N J (2002) Accuracy and Stability of Numerical Algorithms (2nd Edition) SIAM, Philadelphia

5 Arguments

1: FACT – CHARACTER(1) Input

On entry: specifies whether or not the factorized form of the matrix A is supplied on entry, and if
not, whether the matrix A should be equilibrated before it is factorized.

FACT ¼ F
AFB and IPIV contain the factorized form of A. If EQUED 6¼ N , the matrix A has been
equilibrated with scaling factors given by R and C. AB, AFB and IPIV are not modified.

FACT ¼ N
The matrix A will be copied to AFB and factorized.

FACT ¼ E
The matrix A will be equilibrated if necessary, then copied to AFB and factorized.

Constraint: FACT ¼ F , N or E .

2: TRANS – CHARACTER(1) Input

On entry: specifies the form of the system of equations.

TRANS ¼ N
AX ¼ B (No transpose).

TRANS ¼ T or C
ATX ¼ B (Transpose).

Constraint: TRANS ¼ N , T or C .

3: N – INTEGER Input

On entry: n, the number of linear equations, i.e., the order of the matrix A.

Constraint: N � 0.

4: KL – INTEGER Input

On entry: kl, the number of subdiagonals within the band of the matrix A.

Constraint: KL � 0.

5: KU – INTEGER Input

On entry: ku, the number of superdiagonals within the band of the matrix A.

Constraint: KU � 0.

6: NRHS – INTEGER Input

On entry: r, the number of right-hand sides, i.e., the number of columns of the matrix B.

Constraint: NRHS � 0.

7: ABðLDAB; �Þ – REAL (KIND=nag_wp) array Input/Output

Note: the second dimension of the array AB must be at least max 1;Nð Þ.
On entry: the n by n coefficient matrix A.

The matrix is stored in rows 1 to kl þ ku þ 1, more precisely, the element Aij must be stored in

ABðku þ 1þ i� j; jÞ for max 1; j� kuð Þ � i � min n; jþ klð Þ:
See Section 9 for further details.

If FACT ¼ F and EQUED 6¼ N , A must have been equilibrated by the scaling factors in R and/
or C.

On exit: if FACT ¼ F or N , or if FACT ¼ E and EQUED ¼ N , AB is not modified.

If EQUED 6¼ N then, if no constraints are violated, A is scaled as follows:

if EQUED ¼ R , A ¼ DrA;

if EQUED ¼ C , A ¼ ADc;

if EQUED ¼ B , A ¼ DrADc.

8: LDAB – INTEGER Input

On entry: the first dimension of the array AB as declared in the (sub)program from which
F07BBF (DGBSVX) is called.

Constraint: LDAB � KLþ KUþ 1.

9: AFBðLDAFB; �Þ – REAL (KIND=nag_wp) array Input/Output

Note: the second dimension of the array AFB must be at least max 1;Nð Þ.
On entry: if FACT ¼ N or E , AFB need not be set.

If FACT ¼ F , details of the LU factorization of the n by n band matrix A, as computed by
F07BDF (DGBTRF).

F07 – Linear Equations (LAPACK) F07BBF

Mark 26 F07BBF.3



The upper triangular band matrix U , with kl þ ku superdiagonals, is stored in rows 1 to
kl þ ku þ 1 of the array, and the multipliers used to form the matrix L are stored in rows
kl þ ku þ 2 to 2kl þ ku þ 1.

If EQUED 6¼ N , AFB is the factorized form of the equilibrated matrix A.

On exit: if FACT ¼ F , AFB is unchanged from entry.

Otherwise, if no constraints are violated, then if FACT ¼ N , AFB returns details of the LU
factorization of the band matrix A, and if FACT ¼ E , AFB returns details of the LU
factorization of the equilibrated band matrix A (see the description of AB for the form of the
equilibrated matrix).

10: LDAFB – INTEGER Input

On entry: the first dimension of the array AFB as declared in the (sub)program from which
F07BBF (DGBSVX) is called.

Constraint: LDAFB � 2� KLþ KUþ 1.

11: IPIVð�Þ – INTEGER array Input/Output

Note: the dimension of the array IPIV must be at least max 1;Nð Þ.
On entry: if FACT ¼ N or E , IPIV need not be set.

If FACT ¼ F , IPIV contains the pivot indices from the factorization A ¼ LU , as computed by
F07BDF (DGBTRF); row i of the matrix was interchanged with row IPIVðiÞ.
On exit: if FACT ¼ F , IPIV is unchanged from entry.

Otherwise, if no constraints are violated, IPIV contains the pivot indices that define the
permutation matrix P ; at the ith step row i of the matrix was interchanged with row IPIVðiÞ.
IPIVðiÞ ¼ i indicates a row interchange was not required.

If FACT ¼ N , the pivot indices are those corresponding to the factorization A ¼ LU of the
original matrix A.

If FACT ¼ E , the pivot indices are those corresponding to the factorization of A ¼ LU of the
equilibrated matrix A.

12: EQUED – CHARACTER(1) Input/Output

On entry: if FACT ¼ N or E , EQUED need not be set.

If FACT ¼ F , EQUED must specify the form of the equilibration that was performed as follows:

if EQUED ¼ N , no equilibration;

if EQUED ¼ R , row equilibration, i.e., A has been premultiplied by DR;

if EQUED ¼ C , column equilibration, i.e., A has been postmultiplied by DC;

if EQUED ¼ B , both row and column equilibration, i.e., A has been replaced by DRADC.

On exit: if FACT ¼ F , EQUED is unchanged from entry.

Otherwise, if no constraints are violated, EQUED specifies the form of equilibration that was
performed as specified above.

Constraint: if FACT ¼ F , EQUED ¼ N , R , C or B .

13: Rð�Þ – REAL (KIND=nag_wp) array Input/Output

Note: the dimension of the array R must be at least max 1;Nð Þ.
On entry: if FACT ¼ N or E , R need not be set.

If FACT ¼ F and EQUED ¼ R or B , R must contain the row scale factors for A, DR; each
element of R must be positive.

On exit: if FACT ¼ F , R is unchanged from entry.

Otherwise, if no constraints are violated and EQUED ¼ R or B , R contains the row scale
factors for A, DR, such that A is multiplied on the left by DR; each element of R is positive.

14: Cð�Þ – REAL (KIND=nag_wp) array Input/Output

Note: the dimension of the array C must be at least max 1;Nð Þ.
On entry: if FACT ¼ N or E , C need not be set.

If FACT ¼ F or EQUED ¼ C or B , C must contain the column scale factors for A, DC ; each
element of C must be positive.

On exit: if FACT ¼ F , C is unchanged from entry.

Otherwise, if no constraints are violated and EQUED ¼ C or B , C contains the row scale
factors for A, DC ; each element of C is positive.

15: BðLDB; �Þ – REAL (KIND=nag_wp) array Input/Output

Note: the second dimension of the array B must be at least max 1;NRHSð Þ.
On entry: the n by r right-hand side matrix B.

On exit: if EQUED ¼ N , B is not modified.

If TRANS ¼ N and EQUED ¼ R or B , B is overwritten by DRB.

If TRANS ¼ T or C and EQUED ¼ C or B , B is overwritten by DCB.

16: LDB – INTEGER Input

On entry: the first dimension of the array B as declared in the (sub)program from which F07BBF
(DGBSVX) is called.

Constraint: LDB � max 1;Nð Þ.

17: XðLDX; �Þ – REAL (KIND=nag_wp) array Output

Note: the second dimension of the array X must be at least max 1;NRHSð Þ.
On exit: if INFO ¼ 0 or Nþ 1, the n by r solution matrix X to the original system of equations.
Note that the arrays A and B are modified on exit if EQUED 6¼ N , and the solution to the
equilibrated system is D�1

C X if TRANS ¼ N and EQUED ¼ C or B , or D�1
R X if TRANS ¼ T

or C and EQUED ¼ R or B .

18: LDX – INTEGER Input

On entry: the first dimension of the array X as declared in the (sub)program from which F07BBF
(DGBSVX) is called.

Constraint: LDX � max 1;Nð Þ.

19: RCOND – REAL (KIND=nag_wp) Output

On exit: if no constraints are violated, an estimate of the reciprocal condition number of the

matrix A (after equilibration if that is performed), computed as RCOND ¼ 1:0= Ak k1 A�1
�� ��

1

� �
.

20: FERRðNRHSÞ – REAL (KIND=nag_wp) array Output

On exit: if INFO ¼ 0 or Nþ 1, an estimate of the forward error bound for each computed
solution vector, such that x̂j � xj

�� ��
1= xj

�� ��
1 � FERRðjÞ where x̂j is the jth column of the

computed solution returned in the array X and xj is the corresponding column of the exact
solution X. The estimate is as reliable as the estimate for RCOND, and is almost always a slight
overestimate of the true error.

21: BERRðNRHSÞ – REAL (KIND=nag_wp) array Output

On exit: if INFO ¼ 0 or Nþ 1, an estimate of the component-wise relative backward error of
each computed solution vector x̂j (i.e., the smallest relative change in any element of A or B that
makes x̂j an exact solution).

22: WORKðmax 1; 3� Nð ÞÞ – REAL (KIND=nag_wp) array Output

On exit: if INFO ¼ 0, WORKð1Þ contains the reciprocal pivot growth factor max aij
�� ��=max uij

�� ��.
If WORKð1Þ is much less than 1, then the stability of the LU factorization of the (equilibrated)
matrix A could be poor. This also means that the solution X, condition estimator RCOND, and
forward error bound FERR could be unreliable. If the factorization fails with
INFO > 0 and INFO � N, WORKð1Þ contains the reciprocal pivot growth factor for the leading
INFO columns of A.

23: IWORKðNÞ – INTEGER array Workspace

24: INFO – INTEGER Output

On exit: INFO ¼ 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO ¼ �i, argument i had an illegal value. An explanatory message is output, and execution
of the program is terminated.

INFO > 0 and INFO � N

Element valueh i of the diagonal is exactly zero. The factorization has been completed, but the
factor U is exactly singular, so the solution and error bounds could not be computed.
RCOND ¼ 0:0 is returned.

INFO ¼ Nþ 1

U is nonsingular, but RCOND is less than machine precision, meaning that the matrix is singular
to working precision. Nevertheless, the solution and error bounds are computed because there are
a number of situations where the computed solution can be more accurate than the value of
RCOND would suggest.

7 Accuracy

For each right-hand side vector b, the computed solution x̂ is the exact solution of a perturbed system of
equations Aþ Eð Þx̂ ¼ b, where

Ej j � c nð Þ�P Lj j Uj j;
c nð Þ is a modest linear function of n, and � is the machine precision. See Section 9.3 of Higham (2002)
for further details.

If x is the true solution, then the computed solution x̂ satisfies a forward error bound of the form

x� x̂k k1
x̂k k1

� wc cond A; x̂; bð Þ

where cond A; x̂; bð Þ ¼ A�1
�� �� Aj j x̂j j þ bj jð Þ�� ��

1= x̂k k1 � cond Að Þ ¼ A�1
�� �� Aj j�� ��

1 � �1 Að Þ. If x̂ is the
jth column of X, then wc is returned in BERRðjÞ and a bound on x� x̂k k1= x̂k k1 is returned in
FERRðjÞ. See Section 4.4 of Anderson et al. (1999) for further details.

8 Parallelism and Performance

F07BBF (DGBSVX) is threaded by NAG for parallel execution in multithreaded implementations of the
NAG Library.

F07BBF (DGBSVX) makes calls to BLAS and/or LAPACK routines, which may be threaded within the
vendor library used by this implementation. Consult the documentation for the vendor library for further
information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the
OpenMP environment used within this routine. Please also consult the Users' Note for your
implementation for any additional implementation-specific information.

9 Further Comments

The band storage scheme for the array AB is illustrated by the following example, when n ¼ 6, kl ¼ 1,
and ku ¼ 2. Storage of the band matrix A in the array AB:

� � a13 a24 a35 a46
� a12 a23 a34 a45 a56

a11 a22 a33 a44 a55 a66
a21 a32 a43 a54 a65 �

The total number of floating-point operations required to solve the equations AX ¼ B depends upon the
pivoting required, but if n � kl þ ku then it is approximately bounded by O nkl kl þ kuð Þð Þ for the
factorization and O n 2kl þ kuð Þrð Þ for the solution following the factorization. The condition number
estimation typically requires between four and five solves and never more than eleven solves, following
the factorization. The solution is then refined, and the errors estimated, using iterative refinement; see
F07BHF (DGBRFS) for information on the floating-point operations required.

In practice the condition number estimator is very reliable, but it can underestimate the true condition
number; see Section 15.3 of Higham (2002) for further details.

The complex analogue of this routine is F07BPF (ZGBSVX).

10 Example

This example solves the equations

AX ¼ B;

where A is the band matrix

A ¼
�0:23 2:54 �3:66 0
�6:98 2:46 �2:73 �2:13
0 2:56 2:46 4:07
0 0 �4:78 �3:82

0
B@

1
CA and B ¼

4:42 �36:01
27:13 �31:67
�6:14 �1:16
10:50 �25:82

0
B@

1
CA:

Estimates for the backward errors, forward errors, condition number and pivot growth are also output,
together with information on the equilibration of A.

10.1 Program Text

Program f07bbfe

! F07BBF Example Program Text

! Mark 26 Release. NAG Copyright 2016.

! .. Use Statements ..
Use nag_library, Only: dgbsvx, nag_wp, x04caf

! .. Implicit None Statement ..
Implicit None

! .. Parameters ..
Integer, Parameter :: nin = 5, nout = 6

! .. Local Scalars ..
Real (Kind=nag_wp) :: rcond
Integer :: i, ifail, info, j, k, kl, ku, ldab, &

ldafb, ldb, ldx, n, nrhs
Character (1) :: equed

! .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: ab(:,:), afb(:,:), b(:,:), berr(:), &

c(:), ferr(:), r(:), work(:), x(:,:)
Integer, Allocatable :: ipiv(:), iwork(:)

! .. Intrinsic Procedures ..
Intrinsic :: max, min

! .. Executable Statements ..
Write (nout,*) ’F07BBF Example Program Results’
Write (nout,*)
Flush (nout)

! Skip heading in data file
Read (nin,*)
Read (nin,*) n, nrhs, kl, ku
ldb = n
ldx = n
ldab = kl + ku + 1
ldafb = ldab + kl
Allocate (ab(ldab,n),afb(ldafb,n),b(ldb,nrhs),berr(nrhs),c(n), &

ferr(nrhs),r(n),work(3*n),x(ldx,nrhs),ipiv(n),iwork(n))

! Read the band matrix A and B from data file

k = ku + 1
Read (nin,*)((ab(k+i-j,j),j=max(i-kl,1),min(i+ku,n)),i=1,n)
Read (nin,*)(b(i,1:nrhs),i=1,n)

! Solve the equations AX = B for X
! The NAG name equivalent of dgbsvx is f07bbf

Call dgbsvx(’Equilibration’,’No transpose’,n,kl,ku,nrhs,ab,ldab,afb, &
ldafb,ipiv,equed,r,c,b,ldb,x,ldx,rcond,ferr,berr,work,iwork,info)

If ((info==0) .Or. (info==n+1)) Then

! Print solution, error bounds, condition number, the form
! of equilibration and the pivot growth factor

! ifail: behaviour on error exit
! =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft

ifail = 0
Call x04caf(’General’,’ ’,n,nrhs,x,ldx,’Solution(s)’,ifail)

Write (nout,*)
Write (nout,*) ’Backward errors (machine-dependent)’
Write (nout,99999) berr(1:nrhs)
Write (nout,*)
Write (nout,*) ’Estimated forward error bounds (machine-dependent)’
Write (nout,99999) ferr(1:nrhs)
Write (nout,*)
Write (nout,*) ’Estimate of reciprocal condition number’
Write (nout,99999) rcond
Write (nout,*)
If (equed==’N’) Then

Write (nout,*) ’A has not been equilibrated’
Else If (equed==’R’) Then

Write (nout,*) ’A has been row scaled as diag(R)*A’
Else If (equed==’C’) Then

Write (nout,*) ’A has been column scaled as A*diag(C)’
Else If (equed==’B’) Then

Write (nout,*) &
’A has been row and column scaled as diag(R)*A*diag(C)’

End If
Write (nout,*)
Write (nout,*) ’Estimate of reciprocal pivot growth factor’
Write (nout,99999) work(1)

If (info==n+1) Then

Write (nout,*)
Write (nout,*) ’The matrix A is singular to working precision’

End If
Else

Write (nout,99998) ’The (’, info, ’,’, info, ’)’, &
’ element of the factor U is zero’

End If

99999 Format ((3X,1P,7E11.1))
99998 Format (1X,A,I3,A,I3,A,A)

End Program f07bbfe

10.2 Program Data

F07BBF Example Program Data
4 2 1 2 :Values of N, NRHS, KL and KU

-0.23 2.54 -3.66
-6.98 2.46 -2.73 -2.13

2.56 2.46 4.07
-4.78 -3.82 :End of matrix A

4.42 -36.01
27.13 -31.67
-6.14 -1.16
10.50 -25.82 :End of matrix B

10.3 Program Results

F07BBF Example Program Results

Solution(s)
1 2

1 -2.0000 1.0000
2 3.0000 -4.0000
3 1.0000 7.0000
4 -4.0000 -2.0000

Backward errors (machine-dependent)
1.1E-16 9.9E-17

Estimated forward error bounds (machine-dependent)
1.6E-14 1.9E-14

Estimate of reciprocal condition number
1.8E-02

A has not been equilibrated

Estimate of reciprocal pivot growth factor
1.0E+00

