NAG Library Routine Document

F07BDF (DGBTRF)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07BDF (DGBTRF) computes the LU factorization of a real m by n band matrix.

2 Specification

```
SUBROUTINE F07BDF (M, N, KL, KU, AB, LDAB, IPIV, INFO)
INTEGER            M, N, KL, KU, LDAB, IPIV(min(M,N)), INFO
REAL (KIND=nag_wp) AB(LDAB,*)
```

The routine may be called by its LAPACK name ***dgbtrf***.

3 Description

F07BDF (DGBTRF) forms the LU factorization of a real m by n band matrix A using partial pivoting, with row interchanges. Usually $m = n$, and then, if A has k_l nonzero subdiagonals and k_u nonzero superdiagonals, the factorization has the form $A = PLU$, where P is a permutation matrix, L is a lower triangular matrix with unit diagonal elements and at most k_l nonzero elements in each column, and U is an upper triangular band matrix with $k_l + k_u$ superdiagonals.

Note that L is not a band matrix, but the nonzero elements of L can be stored in the same space as the subdiagonal elements of A. U is a band matrix but with k_l additional superdiagonals compared with A. These additional superdiagonals are created by the row interchanges.

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Arguments

1: M – INTEGER *Input*

On entry: m, the number of rows of the matrix A.

Constraint: $M \geq 0$.

2: N – INTEGER *Input*

On entry: n, the number of columns of the matrix A.

Constraint: $N \geq 0$.

3: KL – INTEGER *Input*

On entry: k_l, the number of subdiagonals within the band of the matrix A.

Constraint: $KL \geq 0$.

4: KU – INTEGER *Input*

On entry: k_u, the number of superdiagonals within the band of the matrix A.

Constraint: $KU \geq 0$.

5: AB(LDAB, $*$) – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array AB must be at least max(1, N).

On entry: the m by n matrix A.

The matrix is stored in rows $k_l + 1$ to $2k_l + k_u + 1$; the first k_l rows need not be set, more precisely, the element A_{ij} must be stored in

$$\text{AB}(k_l + k_u + 1 + i - j, j) = A_{ij} \quad \text{for } \max(1, j - k_u) \le i \le \min(m, j + k_l).$$

See Section 9 in F07BAF (DGBSV) for further details.

On exit: if INFO ≥ 0, AB is overwritten by details of the factorization.

The upper triangular band matrix U, with $k_l + k_u$ superdiagonals, is stored in rows 1 to $k_l + k_u + 1$ of the array, and the multipliers used to form the matrix L are stored in rows $k_l + k_u + 2$ to $2k_l + k_u + 1$.

6: LDAB – INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F07BDF (DGBTRF) is called.

Constraint: LDAB $\geq 2 \times$ KL + KU + 1.

7: IPIV(min(M, N)) – INTEGER array *Output*

On exit: the pivot indices that define the permutation matrix. At the ith step, if IPIV(i) $> i$ then row i of the matrix A was interchanged with row IPIV(i), for $i = 1, 2, \ldots, \min(m, n)$. IPIV($i$) $\leq i$ indicates that, at the ith step, a row interchange was not required.

8: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO $= -i$, argument i had an illegal value.

If INFO $= -999$, dynamic memory allocation failed. See Section 3.7 in How to Use the NAG Library and its Documentation for further information. An explanatory message is output, and execution of the program is terminated.

INFO > 0

Element $\langle value \rangle$ of the diagonal is exactly zero. The factorization has been completed, but the factor U is exactly singular, and division by zero will occur if it is used to solve a system of equations.

7 Accuracy

The computed factors L and U are the exact factors of a perturbed matrix $A + E$, where

$$|E| \leq c(k)\epsilon P|L||U|,$$

$c(k)$ is a modest linear function of $k = k_l + k_u + 1$, and ϵ is the ***machine precision***. This assumes $k \ll \min(m, n)$.

8 Parallelism and Performance

F07BDF (DGBTRF) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F07BDF (DGBTRF) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations varies between approximately $2nk_l(k_u+1)$ and $2nk_l(k_l+k_u+1)$, depending on the interchanges, assuming $m=n\gg k_l$ and $n\gg k_u$.

A call to F07BDF (DGBTRF) may be followed by calls to the routines:

F07BEF (DGBTRS) to solve $AX=B$ or $A^{\mathrm{T}}X=B$;

F07BGF (DGBCON) to estimate the condition number of A.

The complex analogue of this routine is F07BRF (ZGBTRF).

10 Example

This example computes the LU factorization of the matrix A, where

$$A=\begin{pmatrix}-0.23 & 2.54 & -3.66 & 0.00\\ -6.98 & 2.46 & -2.73 & -2.13\\ 0.00 & 2.56 & 2.46 & 4.07\\ 0.00 & 0.00 & -4.78 & -3.82\end{pmatrix}.$$

Here A is treated as a band matrix with one subdiagonal and two superdiagonals.

10.1 Program Text

```
    Program f07bdfe

!     F07BDF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: dgbtrf, nag_wp, x04cef
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: i, ifail, info, j, k, kl, ku, ldab,  &
                                          m, n
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: ab(:,:)
      Integer, Allocatable             :: ipiv(:)
!     .. Intrinsic Procedures ..
      Intrinsic                        :: max, min
!     .. Executable Statements ..
      Write (nout,*) 'F07BDF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) m, n, kl, ku
      ldab = 2*kl + ku + 1
      Allocate (ab(ldab,n),ipiv(n))

!     Read A from data file

      k = kl + ku + 1
      Read (nin,*)((ab(k+i-j,j),j=max(i-kl,1),min(i+ku,n)),i=1,m)
```

```
!     Factorize A
!     The NAG name equivalent of dgbtrf is f07bdf
      Call dgbtrf(m,n,kl,ku,ab,ldab,ipiv,info)

!     Print details of factorization

      Write (nout,*)
      Flush (nout)

!     ifail: behaviour on error exit
!             =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
      Call x04cef(m,n,kl,kl+ku,ab,ldab,'Details of factorization',ifail)

!     Print pivot indices

      Write (nout,*)
      Write (nout,*) 'IPIV'
      Write (nout,99999) ipiv(1:min(m,n))

      If (info/=0) Then
        Write (nout,*) 'The factor U is singular'
      End If

99999 Format ((3X,7I11))
    End Program f07bdfe
```

10.2 Program Data

```
F07BDF Example Program Data
  4  4  1  2                      :Values of M, N, KL and KU
 -0.23   2.54  -3.66
 -6.98   2.46  -2.73  -2.13
         2.56   2.46   4.07
                -4.78  -3.82   :End of matrix A
```

10.3 Program Results

```
 F07BDF Example Program Results

 Details of factorization
             1          2          3          4
 1     -6.9800     2.4600    -2.7300    -2.1300
 2      0.0330     2.5600     2.4600     4.0700
 3                 0.9605    -5.9329    -3.8391
 4                            0.8057    -0.7269

 IPIV
             2          3          3          4
```