NAG Library Routine Document

F07CGF (DGTCON)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07CGF (DGTCON) estimates the reciprocal condition number of a real n by n tridiagonal matrix A, using the LU factorization returned by F07CDF (DGTTRF).

2 Specification

```
SUBROUTINE F07CGF (NORM, N, DL, D, DU, DU2, IPIV, ANORM, RCOND, WORK,      &
                   IWORK, INFO)

INTEGER            N, IPIV(*), IWORK(N), INFO
REAL (KIND=nag_wp) DL(*), D(*), DU(*), DU2(*), ANORM, RCOND, WORK(2*N)
CHARACTER(1)       NORM
```

The routine may be called by its LAPACK name ***dgtcon***.

3 Description

F07CGF (DGTCON) should be preceded by a call to F07CDF (DGTTRF), which uses Gaussian elimination with partial pivoting and row interchanges to factorize the matrix A as

$$A = PLU,$$

where P is a permutation matrix, L is unit lower triangular with at most one nonzero subdiagonal element in each column, and U is an upper triangular band matrix, with two superdiagonals. F07CGF (DGTCON) then utilizes the factorization to estimate either $\left\|A^{-1}\right\|_1$ or $\left\|A^{-1}\right\|_\infty$, from which the estimate of the reciprocal of the condition number of A, $1/\kappa(A)$ is computed as either

$$1/\kappa_1(A) = 1/\left(\|A\|_1\left\|A^{-1}\right\|_1\right)$$

or

$$1/\kappa_\infty(A) = 1/\left(\|A\|_\infty\left\|A^{-1}\right\|_\infty\right).$$

$1/\kappa(A)$ is returned, rather than $\kappa(A)$, since when A is singular $\kappa(A)$ is infinite.

Note that $\kappa_\infty(A) = \kappa_1(A^{\mathrm{T}})$.

4 References

Higham N J (2002) *Accuracy and Stability of Numerical Algorithms* (2nd Edition) SIAM, Philadelphia

5 Arguments

1: NORM – CHARACTER(1) *Input*

On entry: specifies the norm to be used to estimate $\kappa(A)$.

NORM = '1' or 'O'
  Estimate $\kappa_1(A)$.

NORM = 'I'
Estimate $\kappa_\infty(A)$.

Constraint: NORM = '1', 'O' or 'I'.

2: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: $N \geq 0$.

3: DL($*$) – REAL (KIND=nag_wp) array *Input*

Note: the dimension of the array DL must be at least $\max(1, N - 1)$.

On entry: must contain the $(n - 1)$ multipliers that define the matrix L of the LU factorization of A.

4: D($*$) – REAL (KIND=nag_wp) array *Input*

Note: the dimension of the array D must be at least $\max(1, N)$.

On entry: must contain the n diagonal elements of the upper triangular matrix U from the LU factorization of A.

5: DU($*$) – REAL (KIND=nag_wp) array *Input*

Note: the dimension of the array DU must be at least $\max(1, N - 1)$.

On entry: must contain the $(n - 1)$ elements of the first superdiagonal of U.

6: DU2($*$) – REAL (KIND=nag_wp) array *Input*

Note: the dimension of the array DU2 must be at least $\max(1, N - 2)$.

On entry: must contain the $(n - 2)$ elements of the second superdiagonal of U.

7: IPIV($*$) – INTEGER array *Input*

Note: the dimension of the array IPIV must be at least $\max(1, N)$.

On entry: must contain the n pivot indices that define the permutation matrix P. At the ith step, row i of the matrix was interchanged with row IPIV(i), and IPIV(i) must always be either i or $(i + 1)$, IPIV(i) $= i$ indicating that a row interchange was not performed.

8: ANORM – REAL (KIND=nag_wp) *Input*

On entry: if NORM = '1' or 'O', the 1-norm of the **original** matrix A.

If NORM = 'I', the ∞-norm of the **original** matrix A.

ANORM may be computed by calling F06RNF with the same value for the argument NORM.

ANORM must be computed either **before** calling F07CDF (DGTTRF) or else from a **copy** of the original matrix A (see Section 10).

Constraint: ANORM ≥ 0.0.

9: RCOND – REAL (KIND=nag_wp) *Output*

On exit: contains an estimate of the reciprocal condition number.

10: WORK($2 \times N$) – REAL (KIND=nag_wp) array *Workspace*

11: IWORK(N) – INTEGER array *Workspace*

12: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO $= -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

In practice the condition number estimator is very reliable, but it can underestimate the true condition number; see Section 15.3 of Higham (2002) for further details.

8 Parallelism and Performance

F07CGF (DGTCON) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The condition number estimation typically requires between four and five solves and never more than eleven solves, following the factorization. The total number of floating-point operations required to perform a solve is proportional to n.

The complex analogue of this routine is F07CUF (ZGTCON).

10 Example

This example estimates the condition number in the 1-norm of the tridiagonal matrix A given by

$$A = \begin{pmatrix} 3.0 & 2.1 & 0 & 0 & 0 \\ 3.4 & 2.3 & -1.0 & 0 & 0 \\ 0 & 3.6 & -5.0 & 1.9 & 0 \\ 0 & 0 & 7.0 & -0.9 & 8.0 \\ 0 & 0 & 0 & -6.0 & 7.1 \end{pmatrix}.$$

10.1 Program Text

```
    Program f07cgfe

!     F07CGF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: dgtcon, dgttrf, f06rnf, nag_wp, x02ajf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: anorm, rcond
      Integer                          :: info, n
!     .. Local Arrays ..
```

```
      Real (Kind=nag_wp), Allocatable  :: d(:), dl(:), du(:), du2(:), work(:)
      Integer, Allocatable             :: ipiv(:), iwork(:)
!     .. Executable Statements ..
      Write (nout,*) 'F07CGF Example Program Results'
      Write (nout,*)
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n

      Allocate (d(n),dl(n-1),du(n-1),du2(n-2),work(2*n),ipiv(n),iwork(n))

!     Read the tridiagonal matrix A from data file

      Read (nin,*) du(1:n-1)
      Read (nin,*) d(1:n)
      Read (nin,*) dl(1:n-1)

!     Compute the 1-norm of A
      anorm = f06rnf('1-norm',n,dl,d,du)

!     Factorize the tridiagonal matrix A
!     The NAG name equivalent of dgttrf is f07cdf
      Call dgttrf(n,dl,d,du,du2,ipiv,info)

      If (info==0) Then

!       Estimate the condition number of A
!       The NAG name equivalent of dgtcon is f07cgf
        Call dgtcon('1-norm',n,dl,d,du,du2,ipiv,anorm,rcond,work,iwork,info)

!       Print the estimated condition number

        If (rcond>=x02ajf()) Then
          Write (nout,99999) 'Estimate of condition number = ',              &
            1.0_nag_wp/rcond
        Else
          Write (nout,99999) 'A is singular to working precision. RCOND = ',  &
            rcond
        End If

      Else
        Write (nout,99998) 'The (', info, ',', info, ')',                    &
          ' element of the factor U is zero'
      End If

99999 Format (1X,A,1P,E10.2)
99998 Format (1X,A,I3,A,I3,A,A)
    End Program f07cgfe
```

10.2 Program Data

```
F07CGF Example Program Data
   5                           :Value of N
        2.1  -1.0   1.9   8.0
  3.0   2.3  -5.0  -0.9   7.1
  3.4   3.6   7.0  -6.0        :End of matrix A
```

10.3 Program Results

```
 F07CGF Example Program Results

 Estimate of condition number =   9.27E+01
```