

NAG Library Routine Document

F07CRF (ZGTTRF)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F07CRF (ZGTTRF) computes the *LU* factorization of a complex n by n tridiagonal matrix A .

2 Specification

```
SUBROUTINE F07CRF (N, DL, D, DU, DU2, IPIV, INFO)
  INTEGER          N, IPIV(N), INFO
  COMPLEX (KIND=nag_wp) DL(*), D(*), DU(*), DU2(N-2)
```

The routine may be called by its LAPACK name *zgttrf*.

3 Description

F07CRF (ZGTTRF) uses Gaussian elimination with partial pivoting and row interchanges to factorize the matrix A as

$$A = PLU,$$

where P is a permutation matrix, L is unit lower triangular with at most one nonzero subdiagonal element in each column, and U is an upper triangular band matrix, with two superdiagonals.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

5 Arguments

- 1: N – INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.
- 2: DL(*) – COMPLEX (KIND=nag_wp) array *Input/Output*
Note: the dimension of the array DL must be at least $\max(1, N - 1)$.
On entry: must contain the $(n - 1)$ subdiagonal elements of the matrix A .
On exit: is overwritten by the $(n - 1)$ multipliers that define the matrix L of the *LU* factorization of A .
- 3: D(*) – COMPLEX (KIND=nag_wp) array *Input/Output*
Note: the dimension of the array D must be at least $\max(1, N)$.
On entry: must contain the n diagonal elements of the matrix A .
On exit: is overwritten by the n diagonal elements of the upper triangular matrix U from the *LU* factorization of A .

- 4: DU(*) – COMPLEX (KIND=nag_wp) array *Input/Output*
Note: the dimension of the array DU must be at least $\max(1, N - 1)$.
On entry: must contain the $(n - 1)$ superdiagonal elements of the matrix A .
On exit: is overwritten by the $(n - 1)$ elements of the first superdiagonal of U .
- 5: DU2(N - 2) – COMPLEX (KIND=nag_wp) array *Output*
On exit: contains the $(n - 2)$ elements of the second superdiagonal of U .
- 6: IPIV(N) – INTEGER array *Output*
On exit: contains the n pivot indices that define the permutation matrix P . At the i th step, row i of the matrix was interchanged with row IPIV(i). IPIV(i) will always be either i or $(i + 1)$, IPIV(i) = i indicating that a row interchange was not performed.
- 7: INFO – INTEGER *Output*
On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

Element $\langle value \rangle$ of the diagonal is exactly zero. The factorization has been completed, but the factor U is exactly singular, and division by zero will occur if it is used to solve a system of equations.

7 Accuracy

The computed factorization satisfies an equation of the form

$$A + E = PLU,$$

where

$$\|E\|_{\infty} = O(\epsilon)\|A\|_{\infty}$$

and ϵ is the *machine precision*.

Following the use of this routine, F07CSF (ZGTTRS) can be used to solve systems of equations $AX = B$ or $A^T X = B$ or $A^H X = B$, and F07CUF (ZGTCON) can be used to estimate the condition number of A .

8 Parallelism and Performance

F07CRF (ZGTTRF) is not threaded in any implementation.

9 Further Comments

The total number of floating-point operations required to factorize the matrix A is proportional to n .

The real analogue of this routine is F07CDF (DGTTRF).

10 Example

This example factorizes the tridiagonal matrix A given by

$$A = \begin{pmatrix} -1.3 + 1.3i & 2.0 - 1.0i & 0 & 0 & 0 \\ 1.0 - 2.0i & -1.3 + 1.3i & 2.0 + 1.0i & 0 & 0 \\ 0 & 1.0 + 1.0i & -1.3 + 3.3i & -1.0 + 1.0i & 0 \\ 0 & 0 & 2.0 - 3.0i & -0.3 + 4.3i & 1.0 - 1.0i \\ 0 & 0 & 0 & 1.0 + 1.0i & -3.3 + 1.3i \end{pmatrix}.$$

10.1 Program Text

```

Program f07crfe

!      F07CRF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: nag_wp, zgtrf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Integer                     :: info, n
!      .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: d(:), dl(:), du(:), du2(:)
      Integer, Allocatable        :: ipiv(:)
!      .. Executable Statements ..
      Write (nout,*) 'F07CRF Example Program Results'
      Write (nout,*)
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n

      Allocate (d(n),dl(n-1),du(n-1),du2(n-2),ipiv(n))

!      Read the tridiagonal matrix A from data file

      Read (nin,*) du(1:n-1)
      Read (nin,*) d(1:n)
      Read (nin,*) dl(1:n-1)

!      Factorize the tridiagonal matrix A
!      The NAG name equivalent of zgtrf is f07crf
      Call zgtrf(n,dl,d,du,du2,ipiv,info)

      If (info>0) Then
         Write (nout,99999) 'The (', info, ',', info, ')',
           ' element of the factor U is zero'
      End If

!      Print details of the factorization

      Write (nout,*) 'Details of factorization'
      Write (nout,*)
      Write (nout,*) ' Second superdiagonal of U'
      Write (nout,99998) du2(1:n-2)
      Write (nout,*)
      Write (nout,*) ' First superdiagonal of U'
      Write (nout,99998) du(1:n-1)
      Write (nout,*)
      Write (nout,*) ' Main diagonal of U'
      Write (nout,99998) d(1:n)
      Write (nout,*)
      Write (nout,*) ' Multipliers'
      Write (nout,99998) dl(1:n-1)
      Write (nout,*)

```

```

      Write (nout,*) ' Vector of interchanges'
      Write (nout,99997) ipiv(1:n)

99999 Format (1X,A,I3,A,I3,A,A)
99998 Format (4(' (',F8.4,',',F8.4,')',:))
99997 Format (1X,5I7)
      End Program f07crfe

```

10.2 Program Data

F07CRF Example Program Data

```

      5                                     :Value of N
      ( 2.0,-1.0) ( 2.0, 1.0) (-1.0, 1.0) ( 1.0,-1.0) :End of DU
      (-1.3, 1.3) (-1.3, 1.3) (-1.3, 3.3) (-0.3, 4.3)
      (-3.3, 1.3)                                     :End of D
      ( 1.0,-2.0) ( 1.0, 1.0) ( 2.0,-3.0) ( 1.0, 1.0) :End of DL

```

10.3 Program Results

F07CRF Example Program Results

Details of factorization

```

      Second superdiagonal of U
      ( 2.0000, 1.0000) ( -1.0000, 1.0000) ( 1.0000, -1.0000)

      First superdiagonal of U
      ( -1.3000, 1.3000) ( -1.3000, 3.3000) ( -0.3000, 4.3000) ( -3.3000, 1.3000)

      Main diagonal of U
      ( 1.0000, -2.0000) ( 1.0000, 1.0000) ( 2.0000, -3.0000) ( 1.0000, 1.0000)
      ( -1.3399, 0.2875)

      Multipliers
      ( -0.7800, -0.2600) ( 0.1620, -0.4860) ( -0.0452, -0.0010) ( -0.3979, -0.0562)

      Vector of interchanges
      2      3      4      5      5

```
