
NAG Library Routine Document

F07FEF (DPOTRS)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

F07FEF (DPOTRS) solves a real symmetric positive definite system of linear equations with multiple
right-hand sides,

AX ¼ B;

where A has been factorized by F07FDF (DPOTRF).

2 Specification

SUBROUTINE F07FEF (UPLO, N, NRHS, A, LDA, B, LDB, INFO)

INTEGER N, NRHS, LDA, LDB, INFO
REAL (KIND=nag_wp) A(LDA,*), B(LDB,*)
CHARACTER(1) UPLO

The routine may be called by its LAPACK name dpotrs.

3 Description

F07FEF (DPOTRS) is used to solve a real symmetric positive definite system of linear equations
AX ¼ B, this routine must be preceded by a call to F07FDF (DPOTRF) which computes the Cholesky
factorization of A. The solution X is computed by forward and backward substitution.

If UPLO ¼ U , A ¼ UTU , where U is upper triangular; the solution X is computed by solving
UTY ¼ B and then UX ¼ Y .

If UPLO ¼ L , A ¼ LLT, where L is lower triangular; the solution X is computed by solving LY ¼ B
and then LTX ¼ Y .

4 References

Golub G H and Van Loan C F (1996) Matrix Computations (3rd Edition) Johns Hopkins University
Press, Baltimore

5 Arguments

1: UPLO – CHARACTER(1) Input

On entry: specifies how A has been factorized.

UPLO ¼ U
A ¼ UTU , where U is upper triangular.

UPLO ¼ L
A ¼ LLT, where L is lower triangular.

Constraint: UPLO ¼ U or L .

2: N – INTEGER Input

On entry: n, the order of the matrix A.

Constraint: N � 0.

3: NRHS – INTEGER Input

On entry: r, the number of right-hand sides.

Constraint: NRHS � 0.

4: AðLDA; �Þ – REAL (KIND=nag_wp) array Input

Note: the second dimension of the array A must be at least max 1;Nð Þ.
On entry: the Cholesky factor of A, as returned by F07FDF (DPOTRF).

5: LDA – INTEGER Input

On entry: the first dimension of the array A as declared in the (sub)program from which F07FEF
(DPOTRS) is called.

Constraint: LDA � max 1;Nð Þ.

6: BðLDB; �Þ – REAL (KIND=nag_wp) array Input/Output

Note: the second dimension of the array B must be at least max 1;NRHSð Þ.
On entry: the n by r right-hand side matrix B.

On exit: the n by r solution matrix X.

7: LDB – INTEGER Input

On entry: the first dimension of the array B as declared in the (sub)program from which F07FEF
(DPOTRS) is called.

Constraint: LDB � max 1;Nð Þ.

8: INFO – INTEGER Output

On exit: INFO ¼ 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO ¼ �i, argument i had an illegal value. An explanatory message is output, and execution
of the program is terminated.

7 Accuracy

For each right-hand side vector b, the computed solution x is the exact solution of a perturbed system of
equations Aþ Eð Þx ¼ b, where

if UPLO ¼ U , Ej j � c nð Þ� UTj j Uj j;
if UPLO ¼ L , Ej j � c nð Þ� Lj j LTj j,

c nð Þ is a modest linear function of n, and � is the machine precision.

If x̂ is the true solution, then the computed solution x satisfies a forward error bound of the form

x� x̂k k1
xk k1

� c nð Þ cond A; xð Þ�

where cond A; xð Þ ¼ A�1
�� �� Aj j xj j�� ��

1= xk k1 � cond Að Þ ¼ A�1
�� �� Aj j�� ��

1 � �1 Að Þ.
Note that cond A; xð Þ can be much smaller than cond Að Þ.
Forward and backward error bounds can be computed by calling F07FHF (DPORFS), and an estimate
for �1 Að Þ ( ¼ �1 Að Þ) can be obtained by calling F07FGF (DPOCON).

8 Parallelism and Performance

F07FEF (DPOTRS) is threaded by NAG for parallel execution in multithreaded implementations of the
NAG Library.

F07FEF (DPOTRS) makes calls to BLAS and/or LAPACK routines, which may be threaded within the
vendor library used by this implementation. Consult the documentation for the vendor library for further
information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the
OpenMP environment used within this routine. Please also consult the Users' Note for your
implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations is approximately 2n2r.

This routine may be followed by a call to F07FHF (DPORFS) to refine the solution and return an error
estimate.

The complex analogue of this routine is F07FSF (ZPOTRS).

10 Example

This example solves the system of equations AX ¼ B, where

A ¼
4:16 �3:12 0:56 �0:10

�3:12 5:03 �0:83 1:18
0:56 �0:83 0:76 0:34

�0:10 1:18 0:34 1:18

0
B@

1
CA and B ¼

8:70 8:30
�13:35 2:13

1:89 1:61
�4:14 5:00

0
B@

1
CA:

Here A is symmetric positive definite and must first be factorized by F07FDF (DPOTRF).

10.1 Program Text

Program f07fefe

! F07FEF Example Program Text

! Mark 26 Release. NAG Copyright 2016.

! .. Use Statements ..
Use nag_library, Only: dpotrf, dpotrs, nag_wp, x04caf

! .. Implicit None Statement ..
Implicit None

! .. Parameters ..
Integer, Parameter :: nin = 5, nout = 6

! .. Local Scalars ..
Integer :: i, ifail, info, lda, ldb, n, nrhs
Character (1) :: uplo

! .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: a(:,:), b(:,:)

! .. Executable Statements ..
Write (nout,*) ’F07FEF Example Program Results’

! Skip heading in data file
Read (nin,*)
Read (nin,*) n, nrhs
lda = n
ldb = n
Allocate (a(lda,n),b(ldb,nrhs))

! Read A and B from data file

Read (nin,*) uplo
If (uplo==’U’) Then

Read (nin,*)(a(i,i:n),i=1,n)
Else If (uplo==’L’) Then

Read (nin,*)(a(i,1:i),i=1,n)
End If
Read (nin,*)(b(i,1:nrhs),i=1,n)

! Factorize A
! The NAG name equivalent of dpotrf is f07fdf

Call dpotrf(uplo,n,a,lda,info)

Write (nout,*)
Flush (nout)
If (info==0) Then

! Compute solution
! The NAG name equivalent of dpotrs is f07fef

Call dpotrs(uplo,n,nrhs,a,lda,b,ldb,info)

! Print solution

! ifail: behaviour on error exit
! =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft

ifail = 0
Call x04caf(’General’,’ ’,n,nrhs,b,ldb,’Solution(s)’,ifail)

Else
Write (nout,*) ’A is not positive definite’

End If

End Program f07fefe

10.2 Program Data

F07FEF Example Program Data
4 2 :Values of N and NRHS
’L’ :Value of UPLO
4.16

-3.12 5.03
0.56 -0.83 0.76

-0.10 1.18 0.34 1.18 :End of matrix A
8.70 8.30

-13.35 2.13
1.89 1.61

-4.14 5.00 :End of matrix B

10.3 Program Results

F07FEF Example Program Results

Solution(s)
1 2

1 1.0000 4.0000
2 -1.0000 3.0000
3 2.0000 2.0000
4 -3.0000 1.0000

