

NAG Library Routine Document

F07FWF (ZPOTRI)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F07FWF (ZPOTRI) computes the inverse of a complex Hermitian positive definite matrix A , where A has been factorized by F07FRF (ZPOTRF).

2 Specification

```
SUBROUTINE F07FWF (UPLO, N, A, LDA, INFO)
  INTEGER          N, LDA, INFO
  COMPLEX (KIND=nag_wp) A(LDA,*)
  CHARACTER(1)     UPLO
```

The routine may be called by its LAPACK name *zpotri*.

3 Description

F07FWF (ZPOTRI) is used to compute the inverse of a complex Hermitian positive definite matrix A , the routine must be preceded by a call to F07FRF (ZPOTRF), which computes the Cholesky factorization of A .

If $UPLO = 'U'$, $A = U^H U$ and A^{-1} is computed by first inverting U and then forming $(U^{-1})U^{-H}$.

If $UPLO = 'L'$, $A = LL^H$ and A^{-1} is computed by first inverting L and then forming $L^{-H}(L^{-1})$.

4 References

Du Croz J J and Higham N J (1992) Stability of methods for matrix inversion *IMA J. Numer. Anal.* **12** 1–19

5 Arguments

- 1: UPLO – CHARACTER(1) *Input*
On entry: specifies how A has been factorized.
 UPLO = 'U'
 $A = U^H U$, where U is upper triangular.
 UPLO = 'L'
 $A = LL^H$, where L is lower triangular.
Constraint: UPLO = 'U' or 'L'.
- 2: N – INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.

3: A(LDA,*) – COMPLEX (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, N)$.

On entry: the upper triangular matrix U if UPLO = 'U' or the lower triangular matrix L if UPLO = 'L', as returned by F07FRF (ZPOTRF).

On exit: U is overwritten by the upper triangle of A^{-1} if UPLO = 'U'; L is overwritten by the lower triangle of A^{-1} if UPLO = 'L'.

4: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F07FWF (ZPOTRI) is called.

Constraint: $LDA \geq \max(1, N)$.

5: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

Diagonal element $\langle value \rangle$ of the Cholesky factor is zero; the Cholesky factor is singular and the inverse of A cannot be computed.

7 Accuracy

The computed inverse X satisfies

$$\|XA - I\|_2 \leq c(n)\epsilon\kappa_2(A) \quad \text{and} \quad \|AX - I\|_2 \leq c(n)\epsilon\kappa_2(A),$$

where $c(n)$ is a modest function of n , ϵ is the **machine precision** and $\kappa_2(A)$ is the condition number of A defined by

$$\kappa_2(A) = \|A\|_2 \|A^{-1}\|_2.$$

8 Parallelism and Performance

F07FWF (ZPOTRI) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of real floating-point operations is approximately $\frac{8}{3}n^3$.

The real analogue of this routine is F07FJF (DPOTRI).

10 Example

This example computes the inverse of the matrix A , where

$$A = \begin{pmatrix} 3.23 + 0.00i & 1.51 - 1.92i & 1.90 + 0.84i & 0.42 + 2.50i \\ 1.51 + 1.92i & 3.58 + 0.00i & -0.23 + 1.11i & -1.18 + 1.37i \\ 1.90 - 0.84i & -0.23 - 1.11i & 4.09 + 0.00i & 2.33 - 0.14i \\ 0.42 - 2.50i & -1.18 - 1.37i & 2.33 + 0.14i & 4.29 + 0.00i \end{pmatrix}.$$

Here A is Hermitian positive definite and must first be factorized by F07FRF (ZPOTRF).

10.1 Program Text

```

Program f07fwfe

!      F07FWF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: nag_wp, x04dbf, zpotrf, zpotri
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Integer                     :: i, ifail, info, lda, n
      Character (1)               :: uplo
!      .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: a(:, :)
      Character (1)               :: clabs(1), rlabs(1)
!      .. Executable Statements ..
      Write (nout,*) 'F07FWF Example Program Results'
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n
      lda = n
      Allocate (a(lda,n))

!      Read A from data file

      Read (nin,*) uplo
      If (uplo=='U') Then
        Read (nin,*)(a(i,i:n),i=1,n)
      Else If (uplo=='L') Then
        Read (nin,*)(a(i,1:i),i=1,n)
      End If

!      Factorize A
!      The NAG name equivalent of zpotrf is f07frf
      Call zpotrf(uplo,n,a,lda,info)

      Write (nout,*)
      Flush (nout)
      If (info==0) Then

!      Compute inverse of A
!      The NAG name equivalent of zpotri is f07fwf
      Call zpotri(uplo,n,a,lda,info)

!      Print inverse

!      ifail: behaviour on error exit
!      =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
      Call x04dbf(uplo,'Nonunit',n,n,a,lda,'Bracketed','F7.4','Inverse',
        &
        'Integer',rlabs,'Integer',clabs,80,0,ifail)

```

```

      Else
        Write (nout,*) 'A is not positive definite'
      End If

      End Program f07fwfe

```

10.2 Program Data

F07FWF Example Program Data

```

      4                                     :Value of N
      'L'                                 :Value of UPLO
      (3.23, 0.00)
      (1.51, 1.92) ( 3.58, 0.00)
      (1.90,-0.84) (-0.23,-1.11) ( 4.09, 0.00)
      (0.42,-2.50) (-1.18,-1.37) ( 2.33, 0.14) ( 4.29, 0.00) :End of matrix A

```

10.3 Program Results

F07FWF Example Program Results

```

Inverse
      1          2          3          4
1  ( 5.4691, 0.0000)
2  (-1.2624,-1.5491) ( 1.1024, 0.0000)
3  (-2.9746,-0.9616) ( 0.8989,-0.5672) ( 2.1589, 0.0000)
4  ( 1.1962, 2.9772) (-0.9826,-0.2566) (-1.3756,-1.4550) ( 2.2934,-0.0000)

```
