

# NAG Library Routine Document

## F07HDF (DPBTRF)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F07HDF (DPBTRF) computes the Cholesky factorization of a real symmetric positive definite band matrix.

### 2 Specification

```
SUBROUTINE F07HDF (UPLO, N, KD, AB, LDAB, INFO)
  INTEGER          N, KD, LDAB, INFO
  REAL (KIND=nag_wp) AB(LDAB,*)
  CHARACTER(1)     UPLO
```

The routine may be called by its LAPACK name *dpbtrf*.

### 3 Description

F07HDF (DPBTRF) forms the Cholesky factorization of a real symmetric positive definite band matrix  $A$  either as  $A = U^T U$  if  $UPLO = 'U'$  or  $A = LL^T$  if  $UPLO = 'L'$ , where  $U$  (or  $L$ ) is an upper (or lower) triangular band matrix with the same number of superdiagonals (or subdiagonals) as  $A$ .

### 4 References

Demmel J W (1989) On floating-point errors in Cholesky *LAPACK Working Note No. 14* University of Tennessee, Knoxville <http://www.netlib.org/lapack/lawnspdf/lawn14.pdf>

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

### 5 Arguments

- 1: UPLO – CHARACTER(1) *Input*  
*On entry:* specifies whether the upper or lower triangular part of  $A$  is stored and how  $A$  is to be factorized.  
 UPLO = 'U'  
     The upper triangular part of  $A$  is stored and  $A$  is factorized as  $U^T U$ , where  $U$  is upper triangular.  
 UPLO = 'L'  
     The lower triangular part of  $A$  is stored and  $A$  is factorized as  $LL^T$ , where  $L$  is lower triangular.  
*Constraint:* UPLO = 'U' or 'L'.
- 2: N – INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .

- 3: KD – INTEGER *Input*  
*On entry:*  $k_d$ , the number of superdiagonals or subdiagonals of the matrix  $A$ .  
*Constraint:*  $KD \geq 0$ .
- 4: AB(LDAB,\*) – REAL (KIND=nag\_wp) array *Input/Output*  
**Note:** the second dimension of the array AB must be at least  $\max(1, N)$ .  
*On entry:* the  $n$  by  $n$  symmetric positive definite band matrix  $A$ .  
The matrix is stored in rows 1 to  $k_d + 1$ , more precisely,  
if UPLO = 'U', the elements of the upper triangle of  $A$  within the band must be stored with element  $A_{ij}$  in  $AB(k_d + 1 + i - j, j)$  for  $\max(1, j - k_d) \leq i \leq j$ ;  
if UPLO = 'L', the elements of the lower triangle of  $A$  within the band must be stored with element  $A_{ij}$  in  $AB(1 + i - j, j)$  for  $j \leq i \leq \min(n, j + k_d)$ .  
*On exit:* the upper or lower triangle of  $A$  is overwritten by the Cholesky factor  $U$  or  $L$  as specified by UPLO, using the same storage format as described above.
- 5: LDAB – INTEGER *Input*  
*On entry:* the first dimension of the array AB as declared in the (sub)program from which F07HDF (DPBTRF) is called.  
*Constraint:*  $LDAB \geq KD + 1$ .
- 6: INFO – INTEGER *Output*  
*On exit:*  $INFO = 0$  unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If  $INFO = -i$ , argument  $i$  had an illegal value.

If  $INFO = -999$ , dynamic memory allocation failed. See Section 3.7 in How to Use the NAG Library and its Documentation for further information. An explanatory message is output, and execution of the program is terminated.

INFO > 0

The leading minor of order  $\langle value \rangle$  is not positive definite and the factorization could not be completed. Hence  $A$  itself is not positive definite. This may indicate an error in forming the matrix  $A$ . There is no routine specifically designed to factorize a symmetric band matrix which is not positive definite; the matrix must be treated either as a nonsymmetric band matrix, by calling F07BDF (DGBTRF) or as a full symmetric matrix, by calling F07MDF (DSYTRF).

## 7 Accuracy

If UPLO = 'U', the computed factor  $U$  is the exact factor of a perturbed matrix  $A + E$ , where

$$|E| \leq c(k+1)\epsilon |U^T| |U|,$$

$c(k+1)$  is a modest linear function of  $k+1$ , and  $\epsilon$  is the *machine precision*.

If UPLO = 'L', a similar statement holds for the computed factor  $L$ . It follows that  $|e_{ij}| \leq c(k+1)\epsilon \sqrt{a_{ii}a_{jj}}$ .

## 8 Parallelism and Performance

F07HDF (DPBTRF) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The total number of floating-point operations is approximately  $n(k+1)^2$ , assuming  $n \gg k$ .

A call to F07HDF (DPBTRF) may be followed by calls to the routines:

F07HEF (DPBTRS) to solve  $AX = B$ ;

F07HGF (DPBCON) to estimate the condition number of  $A$ .

The complex analogue of this routine is F07HRF (ZPBTRF).

## 10 Example

This example computes the Cholesky factorization of the matrix  $A$ , where

$$A = \begin{pmatrix} 5.49 & 2.68 & 0.00 & 0.00 \\ 2.68 & 5.63 & -2.39 & 0.00 \\ 0.00 & -2.39 & 2.60 & -2.22 \\ 0.00 & 0.00 & -2.22 & 5.17 \end{pmatrix}.$$

### 10.1 Program Text

```

Program f07hdfc

!      F07HDF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: dpbtrf, nag_wp, x04cef
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Integer                     :: i, ifail, info, j, kd, ldab, n
      Character (1)               :: uplo
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: ab(:, :)
!      .. Intrinsic Procedures ..
      Intrinsic                   :: max, min
!      .. Executable Statements ..
      Write (nout,*) 'F07HDF Example Program Results'
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, kd
      ldab = kd + 1
      Allocate (ab(ldab,n))

!      Read A from data file

      Read (nin,*) uplo
      If (uplo=='U') Then
        Do i = 1, n
          Read (nin,*)(ab(kd+1+i-j,j),j=i,min(n,i+kd))
        End Do
      End If

```

```

      Else If (uplo=='L') Then
        Do i = 1, n
          Read (nin,*)(ab(1+i-j,j),j=max(1,i-kd),i)
        End Do
      End If

!      Factorize A
!      The NAG name equivalent of dpbtrf is f07hdf
      Call dpbtrf(uplo,n,kd,ab,ldab,info)

      Write (nout,*)
      Flush (nout)
      If (info==0) Then

!        Print factor

!        ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0

      If (uplo=='U') Then

        Call x04cef(n,n,0,kd,ab,ldab,'Factor',ifail)

      Else If (uplo=='L') Then

        Call x04cef(n,n,kd,0,ab,ldab,'Factor',ifail)

      End If

      Else
        Write (nout,*) 'A is not positive definite'
      End If

      End Program f07hdfe

```

## 10.2 Program Data

F07HDF Example Program Data

```

4  1      :Values of N and KD
'L'      :Value of UPLO
5.49
2.68      5.63
          -2.39      2.60
          -2.22      5.17      :End of matrix A

```

## 10.3 Program Results

F07HDF Example Program Results

Factor	1	2	3	4
1	2.3431			
2	1.1438	2.0789		
3		-1.1497	1.1306	
4			-1.9635	1.1465

---