

NAG Library Routine Document

F07JEF (DPTTRS)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F07JEF (DPTTRS) computes the solution to a real system of linear equations $AX = B$, where A is an n by n symmetric positive definite tridiagonal matrix and X and B are n by r matrices, using the LDL^T factorization returned by F07JDF (DPTTRF).

2 Specification

```
SUBROUTINE F07JEF (N, NRHS, D, E, B, LDB, INFO)
  INTEGER          N, NRHS, LDB, INFO
  REAL (KIND=nag_wp) D(*), E(*), B(LDB,*)
```

The routine may be called by its LAPACK name *dpttrs*.

3 Description

F07JEF (DPTTRS) should be preceded by a call to F07JDF (DPTTRF), which computes a modified Cholesky factorization of the matrix A as

$$A = LDL^T,$$

where L is a unit lower bidiagonal matrix and D is a diagonal matrix, with positive diagonal elements. F07JEF (DPTTRS) then utilizes the factorization to solve the required equations. Note that the factorization may also be regarded as having the form $U^T D U$, where U is a unit upper bidiagonal matrix.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

5 Arguments

- 1: N – INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.
- 2: NRHS – INTEGER *Input*
On entry: r , the number of right-hand sides, i.e., the number of columns of the matrix B .
Constraint: $NRHS \geq 0$.
- 3: D(*) – REAL (KIND=nag_wp) array *Input*
Note: the dimension of the array D must be at least $\max(1, N)$.
On entry: must contain the n diagonal elements of the diagonal matrix D from the LDL^T factorization of A .

- 4: $E(*)$ – REAL (KIND=nag_wp) array *Input*
Note: the dimension of the array E must be at least $\max(1, N - 1)$.
On entry: must contain the $(n - 1)$ subdiagonal elements of the unit lower bidiagonal matrix L . (E can also be regarded as the superdiagonal of the unit upper bidiagonal matrix U from the $U^T D U$ factorization of A .)
- 5: $B(\text{LDB}, *)$ – REAL (KIND=nag_wp) array *Input/Output*
Note: the second dimension of the array B must be at least $\max(1, \text{NRHS})$.
On entry: the n by r matrix of right-hand sides B .
On exit: the n by r solution matrix X .
- 6: LDB – INTEGER *Input*
On entry: the first dimension of the array B as declared in the (sub)program from which F07JEF (DPTTRS) is called.
Constraint: $\text{LDB} \geq \max(1, N)$.
- 7: INFO – INTEGER *Output*
On exit: $\text{INFO} = 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

$\text{INFO} < 0$

If $\text{INFO} = -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed solution for a single right-hand side, \hat{x} , satisfies an equation of the form

$$(A + E)\hat{x} = b,$$

where

$$\|E\|_1 = O(\epsilon)\|A\|_1$$

and ϵ is the *machine precision*. An approximate error bound for the computed solution is given by

$$\frac{\|\hat{x} - x\|_1}{\|x\|_1} \leq \kappa(A) \frac{\|E\|_1}{\|A\|_1},$$

where $\kappa(A) = \|A^{-1}\|_1 \|A\|_1$, the condition number of A with respect to the solution of the linear equations. See Section 4.4 of Anderson *et al.* (1999) for further details.

Following the use of this routine F07JGF (DPTCON) can be used to estimate the condition number of A and F07JHF (DPTRFS) can be used to obtain approximate error bounds.

8 Parallelism and Performance

F07JEF (DPTTRS) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations required to solve the equations $AX = B$ is proportional to nr .

The complex analogue of this routine is F07JSF (ZPTTRS).

10 Example

This example solves the equations

$$AX = B,$$

where A is the symmetric positive definite tridiagonal matrix

$$A = \begin{pmatrix} 4.0 & -2.0 & 0 & 0 & 0 \\ -2.0 & 10.0 & -6.0 & 0 & 0 \\ 0 & -6.0 & 29.0 & 15.0 & 0 \\ 0 & 0 & 15.0 & 25.0 & 8.0 \\ 0 & 0 & 0 & 8.0 & 5.0 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} 6.0 & 10.0 \\ 9.0 & 4.0 \\ 2.0 & 9.0 \\ 14.0 & 65.0 \\ 7.0 & 23.0 \end{pmatrix}.$$

10.1 Program Text

```

Program f07jefe

!      F07JEF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: dpttrf, dpttrs, nag_wp, x04caf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Integer                     :: i, ifail, info, ldb, n, nrhs
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: b(:,:), d(:), e(:)
!      .. Executable Statements ..
      Write (nout,*) 'F07JEF Example Program Results'
      Write (nout,*)
      Flush (nout)
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, nrhs
      ldb = n
      Allocate (b(ldb,nrhs),d(n),e(n-1))

!      Read the upper bidiagonal part of the tridiagonal matrix A from
!      data file

      Read (nin,*) e(1:n-1)
      Read (nin,*) d(1:n)

!      Read the right hand matrix B

      Read (nin,*)(b(i,1:nrhs),i=1,n)

!      Factorize the tridiagonal matrix A
!      The NAG name equivalent of dpttrf is f07jdf
      Call dpttrf(n,d,e,info)

      If (info==0) Then

!          Solve the equations AX = B
!          The NAG name equivalent of dpttrs is f07jef
          Call dpttrs(n,nrhs,d,e,b,ldb,info)

```

```

!      Print the solution

!      ifail: behaviour on error exit
!      =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
!      ifail = 0
!      Call x04caf('General',' ',n,nrhs,b,ldb,'Solution(s)',ifail)

      Else
        Write (nout,99999) 'The leading minor of order ', info,
          ' is not positive definite'
      End If

99999 Format (1X,A,I3,A)
      End Program f07jefe

```

10.2 Program Data

F07JEF Example Program Data

```

5      2      :Values of N and NRHS
      -2.0 -6.0 15.0 8.0 :End of super-diagonal E
4.0 10.0 29.0 25.0 5.0 :End of diagonal D
6.0 10.0
9.0 4.0
2.0 9.0
14.0 65.0
7.0 23.0      :End of matrix B

```

10.3 Program Results

F07JEF Example Program Results

Solution(s)

| | 1 | 2 |
|---|---------|---------|
| 1 | 2.5000 | 2.0000 |
| 2 | 2.0000 | -1.0000 |
| 3 | 1.0000 | -3.0000 |
| 4 | -1.0000 | 6.0000 |
| 5 | 3.0000 | -5.0000 |
