

NAG Library Routine Document

F07JGF (DPTCON)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F07JGF (DPTCON) computes the reciprocal condition number of a real n by n symmetric positive definite tridiagonal matrix A , using the LDL^T factorization returned by F07JDF (DPTTRF).

2 Specification

```
SUBROUTINE F07JGF (N, D, E, ANORM, RCOND, WORK, INFO)
  INTEGER          N, INFO
  REAL (KIND=nag_wp) D(*), E(*), ANORM, RCOND, WORK(N)
```

The routine may be called by its LAPACK name *dptcon*.

3 Description

F07JGF (DPTCON) should be preceded by a call to F07JDF (DPTTRF), which computes a modified Cholesky factorization of the matrix A as

$$A = LDL^T,$$

where L is a unit lower bidiagonal matrix and D is a diagonal matrix, with positive diagonal elements. F07JGF (DPTCON) then utilizes the factorization to compute $\|A^{-1}\|_1$ by a direct method, from which the reciprocal of the condition number of A , $1/\kappa(A)$ is computed as

$$1/\kappa_1(A) = 1/(\|A\|_1\|A^{-1}\|_1).$$

$1/\kappa(A)$ is returned, rather than $\kappa(A)$, since when A is singular $\kappa(A)$ is infinite.

4 References

Higham N J (2002) *Accuracy and Stability of Numerical Algorithms* (2nd Edition) SIAM, Philadelphia

5 Arguments

- 1: N – INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.
- 2: D(*) – REAL (KIND=nag_wp) array *Input*
Note: the dimension of the array D must be at least $\max(1, N)$.
On entry: must contain the n diagonal elements of the diagonal matrix D from the LDL^T factorization of A .

- 3: $E(*)$ – REAL (KIND=nag_wp) array *Input*
Note: the dimension of the array E must be at least $\max(1, N - 1)$.
On entry: must contain the $(n - 1)$ subdiagonal elements of the unit lower bidiagonal matrix L . (E can also be regarded as the superdiagonal of the unit upper bidiagonal matrix U from the $U^T D U$ factorization of A .)
- 4: ANORM – REAL (KIND=nag_wp) *Input*
On entry: the 1-norm of the **original** matrix A , which may be computed by calling F06RPF with its argument NORM = '1'. ANORM must be computed either **before** calling F07JDF (DPTTRF) or else from a **copy** of the original matrix A .
Constraint: ANORM ≥ 0.0 .
- 5: RCOND – REAL (KIND=nag_wp) *Output*
On exit: the reciprocal condition number, $1/\kappa_1(A) = 1/(\|A\|_1 \|A^{-1}\|_1)$.
- 6: WORK(N) – REAL (KIND=nag_wp) array *Workspace*
- 7: INFO – INTEGER *Output*
On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed condition number will be the exact condition number for a closely neighbouring matrix.

8 Parallelism and Performance

F07JGF (DPTCON) is not threaded in any implementation.

9 Further Comments

The condition number estimation requires $O(n)$ floating-point operations.

See Section 15.6 of Higham (2002) for further details on computing the condition number of tridiagonal matrices.

The complex analogue of this routine is F07JUF (ZPTCON).

10 Example

This example computes the condition number of the symmetric positive definite tridiagonal matrix A given by

$$A = \begin{pmatrix} 4.0 & -2.0 & 0 & 0 & 0 \\ -2.0 & 10.0 & -6.0 & 0 & 0 \\ 0 & -6.0 & 29.0 & 15.0 & 0 \\ 0 & 0 & 15.0 & 25.0 & 8.0 \\ 0 & 0 & 0 & 8.0 & 5.0 \end{pmatrix}.$$

10.1 Program Text

```

Program f07jgfe

!      F07JGF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: dptcon, dpttrf, f06rpf, nag_wp, x02ajf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: anorm, rcond
      Integer                      :: info, n
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: d(:), e(:), work(:)
!      .. Executable Statements ..
      Write (nout,*) 'F07JGF Example Program Results'
      Write (nout,*)
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n

      Allocate (d(n),e(n-1),work(n))

!      Read the lower bidiagonal part of the tridiagonal matrix A from
!      data file

      Read (nin,*) d(1:n)
      Read (nin,*) e(1:n-1)

!      Compute the 1-norm of A
      anorm = f06rpf('1-norm',n,d,e)

!      Factorize the tridiagonal matrix A
!      The NAG name equivalent of dpttrf is f07jdf
      Call dpttrf(n,d,e,info)

      If (info==0) Then

!          Estimate the condition number of A
!          The NAG name equivalent of dptcon is f07jgf
          Call dptcon(n,d,e,anorm,rcond,work,info)

!          Print the estimated condition number

          If (rcond>=x02ajf()) Then
              Write (nout,99999) 'Estimate of condition number = ',      &
                  1.0_nag_wp/rcond
          Else
              Write (nout,99999) 'A is singular to working precision. RCOND = ',      &
                  rcond
              End If

          Else
              Write (nout,99998) 'The leading minor of order ', info,      &
                  ' is not positive definite'
              End If

99999 Format (1X,A,1P,E10.2)
99998 Format (1X,A,I3,A)
      End Program f07jgfe

```

10.2 Program Data

F07JGF Example Program Data

```
5           :Value of N
4.0  10.0  29.0  25.0   5.0 :End of diagonal D
-2.0  -6.0  15.0   8.0      :End of sub-diagonal E
```

10.3 Program Results

F07JGF Example Program Results

Estimate of condition number = 1.05E+02
