

NAG Library Routine Document

F07JRF (ZPTTRF)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F07JRF (ZPTTRF) computes the modified Cholesky factorization of a complex n by n Hermitian positive definite tridiagonal matrix A .

2 Specification

SUBROUTINE F07JRF (N, D, E, INFO)

INTEGER N, INFO
REAL (KIND=nag_wp) D(*)
COMPLEX (KIND=nag_wp) E(*)

The routine may be called by its LAPACK name *zpttrf*.

3 Description

F07JRF (ZPTTRF) factorizes the matrix A as

$$A = LDL^H,$$

where L is a unit lower bidiagonal matrix and D is a diagonal matrix with positive diagonal elements. The factorization may also be regarded as having the form $U^H DU$, where U is a unit upper bidiagonal matrix.

4 References

None.

5 Arguments

- 1: N – INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.
- 2: D(*) – REAL (KIND=nag_wp) array *Input/Output*
Note: the dimension of the array D must be at least $\max(1, N)$.
On entry: must contain the n diagonal elements of the matrix A .
On exit: is overwritten by the n diagonal elements of the diagonal matrix D from the LDL^H factorization of A .
- 3: E(*) – COMPLEX (KIND=nag_wp) array *Input/Output*
Note: the dimension of the array E must be at least $\max(1, N - 1)$.
On entry: must contain the $(n - 1)$ subdiagonal elements of the matrix A .
On exit: is overwritten by the $(n - 1)$ subdiagonal elements of the lower bidiagonal matrix L . (E can also be regarded as containing the $(n - 1)$ superdiagonal elements of the upper bidiagonal matrix U .)

4: INFO – INTEGER

Output

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0 and INFO < N

The leading minor of order $\langle value \rangle$ is not positive definite, the factorization could not be completed.

INFO > 0 and INFO = N

The leading minor of order n is not positive definite, the factorization was completed, but $D(N) \leq 0$.

7 Accuracy

The computed factorization satisfies an equation of the form

$$A + E = LDL^H,$$

where

$$\|E\|_{\infty} = O(\epsilon)\|A\|_{\infty}$$

and ϵ is the *machine precision*.

Following the use of this routine, F07JSF (ZPTTRS) can be used to solve systems of equations $AX = B$, and F07JUF (ZPTCON) can be used to estimate the condition number of A .

8 Parallelism and Performance

F07JRF (ZPTTRF) is not threaded in any implementation.

9 Further Comments

The total number of floating-point operations required to factorize the matrix A is proportional to n .

The real analogue of this routine is F07JDF (DPTTRF).

10 Example

This example factorizes the Hermitian positive definite tridiagonal matrix A given by

$$A = \begin{pmatrix} 16.0 & 16.0 - 16.0i & 0 & 0 \\ 16.0 + 16.0i & 41.0 & 18.0 + 9.0i & 0 \\ 0 & 18.0 - 9.0i & 46.0 & 1.0 + 4.0i \\ 0 & 0 & 1.0 - 4.0i & 21.0 \end{pmatrix}.$$

10.1 Program Text

```

Program f07jrfe

!      F07JRF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: nag_wp, zpttrf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Integer                     :: info, n
!      .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: e(:)
      Real (Kind=nag_wp), Allocatable   :: d(:)
!      .. Executable Statements ..
      Write (nout,*) 'F07JRF Example Program Results'
      Write (nout,*)
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n

      Allocate (e(n-1),d(n))

!      Read the lower bidiagonal part of the tridiagonal matrix A from
!      data file

      Read (nin,*) d(1:n)
      Read (nin,*) e(1:n-1)

!      Factorize the tridiagonal matrix A
!      The NAG name equivalent of zpttrf is f07jrf
      Call zpttrf(n,d,e,info)

      If (info>0) Then
         Write (nout,99999) 'The leading minor of order ', info,           &
           ' is not positive definite'
      End If

!      Print details of the factorization

      Write (nout,*) 'Details of factorization'
      Write (nout,*)
      Write (nout,*) ' The diagonal elements of D'
      Write (nout,99998) d(1:n)
      Write (nout,*)
      Write (nout,*) ' Subdiagonal elements of the Cholesky factor L'
      Write (nout,99998) e(1:n-1)

99999 Format (1X,A,I3,A)
99998 Format (1X,8F9.4)
      End Program f07jrfe

```

10.2 Program Data

```

F07JRF Example Program Data
  4                                     :Value of N
 16.0          41.0          46.0          21.0       :End of diagonal D
( 16.0, 16.0) ( 18.0, -9.0) (  1.0, -4.0)       :End of sub-diagonal E

```

10.3 Program Results

F07JRF Example Program Results

Details of factorization

The diagonal elements of D

16.0000 9.0000 1.0000 4.0000

Subdiagonal elements of the Cholesky factor L

1.0000 1.0000 2.0000 -1.0000 1.0000 -4.0000
