

NAG Library Routine Document

F07KDF (DPSTRF)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F07KDF (DPSTRF) computes the Cholesky factorization with complete pivoting of a real symmetric positive semidefinite matrix.

2 Specification

```
SUBROUTINE F07KDF (UPLO, N, A, LDA, PIV, RANK, TOL, WORK, INFO)
  INTEGER          N, LDA, PIV(N), RANK, INFO
  REAL (KIND=nag_wp) A(LDA,*), TOL, WORK(2*N)
  CHARACTER(1)     UPLO
```

The routine may be called by its LAPACK name *dpstrf*.

3 Description

F07KDF (DPSTRF) forms the Cholesky factorization of a real symmetric positive semidefinite matrix A either as $P^T A P = U^T U$ if $UPLO = 'U'$ or $P^T A P = L L^T$ if $UPLO = 'L'$, where P is a permutation matrix, U is an upper triangular matrix and L is lower triangular.

This algorithm does not attempt to check that A is positive semidefinite.

4 References

Higham N J (2002) *Accuracy and Stability of Numerical Algorithms* (2nd Edition) SIAM, Philadelphia
 Lucas C (2004) LAPACK-style codes for Level 2 and 3 pivoted Cholesky factorizations *LAPACK Working Note No. 161. Technical Report CS-04-522* Department of Computer Science, University of Tennessee, 107 Ayres Hall, Knoxville, TN 37996-1301, USA <http://www.netlib.org/lapack/lawnspdf/lawn161.pdf>

5 Arguments

- 1: UPLO – CHARACTER(1) *Input*
On entry: specifies whether the upper or lower triangular part of A is stored and how A is to be factorized.
 UPLO = 'U'
 The upper triangular part of A is stored and A is factorized as $U^T U$, where U is upper triangular.
 UPLO = 'L'
 The lower triangular part of A is stored and A is factorized as $L L^T$, where L is lower triangular.
Constraint: UPLO = 'U' or 'L'.
- 2: N – INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.

3: A(LDA,*) – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, N)$.

On entry: the n by n symmetric positive semidefinite matrix A .

If UPLO = 'U', the upper triangular part of A must be stored and the elements of the array below the diagonal are not referenced.

If UPLO = 'L', the lower triangular part of A must be stored and the elements of the array above the diagonal are not referenced.

On exit: if UPLO = 'U', the first RANK rows of the upper triangle of A are overwritten with the nonzero elements of the Cholesky factor U , and the remaining rows of the triangle are destroyed.

If UPLO = 'L', the first RANK columns of the lower triangle of A are overwritten with the nonzero elements of the Cholesky factor L , and the remaining columns of the triangle are destroyed.

4: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F07KDF (DPSTRF) is called.

Constraint: $LDA \geq \max(1, N)$.

5: PIV(N) – INTEGER array *Output*

On exit: PIV is such that the nonzero entries of P are $P(\text{PIV}(k), k) = 1$, for $k = 1, 2, \dots, n$.

6: RANK – INTEGER *Output*

On exit: the computed rank of A given by the number of steps the algorithm completed.

7: TOL – REAL (KIND=nag_wp) *Input*

On entry: user defined tolerance. If $TOL < 0$, then $n \times \max_{k=1,n} |A_{kk}| \times \text{machine precision}$ will be used. The algorithm terminates at the r th step if the $(r + 1)$ th step pivot $< TOL$.

8: WORK(2 * N) – REAL (KIND=nag_wp) array *Workspace*

9: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO = 1

The matrix A is not positive definite. It is either positive semidefinite with computed rank as returned in RANK and less than n , or it may be indefinite, see Section 9.

7 Accuracy

If UPLO = 'L' and RANK = r , the computed Cholesky factor L and permutation matrix P satisfy the following upper bound

$$\frac{\|A - PLL^T P^T\|_2}{\|A\|_2} \leq 2rc(r)\epsilon(\|W\|_2 + 1)^2 + O(\epsilon^2),$$

where

$$W = L_{11}^{-1} L_{12}, \quad L = \begin{pmatrix} L_{11} & 0 \\ L_{12} & 0 \end{pmatrix}, \quad L_{11} \in \mathbb{R}^{r \times r},$$

$c(r)$ is a modest linear function of r , ϵ is **machine precision**, and

$$\|W\|_2 \leq \sqrt{\frac{1}{3}(n-r)(4^r - 1)}.$$

So there is no guarantee of stability of the algorithm for large n and r , although $\|W\|_2$ is generally small in practice.

8 Parallelism and Performance

F07KDF (DPSTRF) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations is approximately $nr^2 - 2/3r^3$, where r is the computed rank of A .

This algorithm does not attempt to check that A is positive semidefinite, and in particular the rank detection criterion in the algorithm is based on A being positive semidefinite. If there is doubt over semidefiniteness then you should use the indefinite factorization F07MDF (DSYTRF). See Lucas (2004) for further information.

The complex analogue of this routine is F07KRF (ZPSTRF).

10 Example

This example computes the Cholesky factorization of the matrix A , where

$$A = \begin{pmatrix} 2.51 & 4.04 & 3.34 & 1.34 & 1.29 \\ 4.04 & 8.22 & 7.38 & 2.68 & 2.44 \\ 3.34 & 7.38 & 7.06 & 2.24 & 2.14 \\ 1.34 & 2.68 & 2.24 & 0.96 & 0.80 \\ 1.29 & 2.44 & 2.14 & 0.80 & 0.74 \end{pmatrix}.$$

10.1 Program Text

```

Program f07kdfe

!      F07KDF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: dpstrf, nag_wp, x04caf, x04ebf
!      .. Implicit None Statement ..
```

```

      Implicit None
!      .. Parameters ..
      Real (Kind=nag_wp), Parameter      :: zero = 0.0E0_nag_wp
      Integer, Parameter                  :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)                  :: tol
      Integer                             :: i, ifail, info, j, lda, n, rank
      Character (1)                       :: upto
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable     :: a(:,,:), work(:)
      Integer, Allocatable                 :: piv(:)
      Character (1)                        :: clabs(1), rlabs(1)
!      .. Executable Statements ..
      Write (nout,*) 'F07KDF Example Program Results'
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, upto
      lda = n
      Allocate (a(lda,n),piv(n),work(2*n))

!      Read A from data file
      If (upto=='U') Then
        Read (nin,*)(a(i,i:n),i=1,n)
      Else If (upto=='L') Then
        Read (nin,*)(a(i,1:i),i=1,n)
      End If
      tol = -1.0_nag_wp

!      Factorize A
!      The NAG name equivalent of dpstrf is f07kdf
      Call dpstrf(upto,n,a,lda,piv,rank,tol,work,info)

!      Zero out columns rank+1 to n
      If (upto=='U') Then
        Do j = rank + 1, n
          a(rank+1:j,j) = zero
        End Do
      Else If (upto=='L') Then
        Do j = rank + 1, n
          a(j:n,j) = zero
        End Do
      End If

!      Print rank
      Write (nout,*)
      Write (nout,'(1X,A15,I3)') 'Computed rank: ', rank

!      Print factor
      Write (nout,*)
      Flush (nout)
      ifail = 0
      Call x04caf(upto,'Nonunit',n,n,a,lda,'Factor',ifail)

!      Print pivot indices
      Write (nout,*)
      Write (nout,*) 'PIV'
      Flush (nout)

      ifail = 0
      Call x04ebf('General','Non-unit',1,n,piv,1,'I11',' ','No',rlabs,'No', &
        clabs,80,1,ifail)

      End Program f07kdfe

```

10.2 Program Data

F07KDF Example Program Data

```

5 'L'                                : n, uplo
2.51
4.04  8.22
3.34  7.38  7.06
1.34  2.68  2.24  0.96
1.29  2.44  2.14  0.80  0.74      : End of matrix A

```

10.3 Program Results

F07KDF Example Program Results

Computed rank: 3

Factor

	1	2	3	4	5
1	2.8671				
2	1.4091	0.7242			
3	2.5741	-0.3965	0.5262		
4	0.9348	0.0315	-0.2920	0.0000	
5	0.8510	0.1254	-0.0018	0.0000	0.0000

PIV

	2	1	3	4	5
