
NAG Library Routine Document

F07MPF (ZHESVX)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

F07MPF (ZHESVX) uses the diagonal pivoting factorization to compute the solution to a complex
system of linear equations

AX ¼ B;

where A is an n by n Hermitian matrix and X and B are n by r matrices. Error bounds on the solution
and a condition estimate are also provided.

2 Specification

SUBROUTINE F07MPF (FACT, UPLO, N, NRHS, A, LDA, AF, LDAF, IPIV, B, LDB,
X, LDX, RCOND, FERR, BERR, WORK, LWORK, RWORK, INFO)

&

INTEGER N, NRHS, LDA, LDAF, IPIV(*), LDB, LDX, LWORK,
INFO

&

REAL (KIND=nag_wp) RCOND, FERR(*), BERR(*), RWORK(*)
COMPLEX (KIND=nag_wp) A(LDA,*), AF(LDAF,*), B(LDB,*), X(LDX,*),

WORK(max(1,LWORK))
&

CHARACTER(1) FACT, UPLO

The routine may be called by its LAPACK name zhesvx.

3 Description

F07MPF (ZHESVX) performs the following steps:

1. If FACT ¼ N , the diagonal pivoting method is used to factor A. The form of the factorization is
A ¼ UDUH if UPLO ¼ U or A ¼ LDLH if UPLO ¼ L , where U (or L) is a product of
permutation and unit upper (lower) triangular matrices, and D is Hermitian and block diagonal
with 1 by 1 and 2 by 2 diagonal blocks.

2. If some dii ¼ 0, so that D is exactly singular, then the routine returns with INFO ¼ i. Otherwise,
the factored form of A is used to estimate the condition number of the matrix A. If the reciprocal
of the condition number is less than machine precision, INFO ¼ Nþ 1 is returned as a warning,
but the routine still goes on to solve for X and compute error bounds as described below.

3. The system of equations is solved for X using the factored form of A.

4. Iterative refinement is applied to improve the computed solution matrix and to calculate error
bounds and backward error estimates for it.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A,
Hammarling S, McKenney A and Sorensen D (1999) LAPACK Users' Guide (3rd Edition) SIAM,
Philadelphia http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) Matrix Computations (3rd Edition) Johns Hopkins University
Press, Baltimore

Higham N J (2002) Accuracy and Stability of Numerical Algorithms (2nd Edition) SIAM, Philadelphia

5 Arguments

1: FACT – CHARACTER(1) Input

On entry: specifies whether or not the factorized form of the matrix A has been supplied.

FACT ¼ F
AF and IPIV contain the factorized form of the matrix A. AF and IPIV will not be
modified.

FACT ¼ N
The matrix A will be copied to AF and factorized.

Constraint: FACT ¼ F or N .

2: UPLO – CHARACTER(1) Input

On entry: if UPLO ¼ U , the upper triangle of A is stored.

If UPLO ¼ L , the lower triangle of A is stored.

Constraint: UPLO ¼ U or L .

3: N – INTEGER Input

On entry: n, the number of linear equations, i.e., the order of the matrix A.

Constraint: N � 0.

4: NRHS – INTEGER Input

On entry: r, the number of right-hand sides, i.e., the number of columns of the matrix B.

Constraint: NRHS � 0.

5: AðLDA; �Þ – COMPLEX (KIND=nag_wp) array Input

Note: the second dimension of the array A must be at least max 1;Nð Þ.
On entry: the n by n Hermitian matrix A.

If UPLO ¼ U , the upper triangular part of A must be stored and the elements of the array
below the diagonal are not referenced.

If UPLO ¼ L , the lower triangular part of A must be stored and the elements of the array
above the diagonal are not referenced.

6: LDA – INTEGER Input

On entry: the first dimension of the array A as declared in the (sub)program from which F07MPF
(ZHESVX) is called.

Constraint: LDA � max 1;Nð Þ.

7: AFðLDAF; �Þ – COMPLEX (KIND=nag_wp) array Input/Output

Note: the second dimension of the array AF must be at least max 1;Nð Þ.
On entry: if FACT ¼ F , AF contains the block diagonal matrix D and the multipliers used to
obtain the factor U or L from the factorization A ¼ UDUH or A ¼ LDLH as computed by
F07MRF (ZHETRF).

On exit: if FACT ¼ N , AF returns the block diagonal matrix D and the multipliers used to
obtain the factor U or L from the factorization A ¼ UDUH or A ¼ LDLH.

8: LDAF – INTEGER Input

On entry: the first dimension of the array AF as declared in the (sub)program from which
F07MPF (ZHESVX) is called.

Constraint: LDAF � max 1;Nð Þ.

9: IPIVð�Þ – INTEGER array Input/Output

Note: the dimension of the array IPIV must be at least max 1;Nð Þ.
On entry: if FACT ¼ F , IPIV contains details of the interchanges and the block structure of D,
as determined by F07MRF (ZHETRF).

if IPIVðiÞ ¼ k > 0, dii is a 1 by 1 pivot block and the ith row and column of A were
interchanged with the kth row and column;

if UPLO ¼ U and IPIVði� 1Þ ¼ IPIVðiÞ ¼ �l < 0, di�1;i�1
�di;i�1

�di;i�1 dii

� �
is a 2 by 2 pivot

block and the i� 1ð Þth row and column of A were interchanged with the lth row and
column;

if UPLO ¼ L and IPIVðiÞ ¼ IPIVðiþ 1Þ ¼ �m < 0, dii diþ1;i
diþ1;i diþ1;iþ1

� �
is a 2 by 2 pivot

block and the iþ 1ð Þth row and column of A were interchanged with the mth row and
column.

On exit: if FACT ¼ N , IPIV contains details of the interchanges and the block structure of D, as
determined by F07MRF (ZHETRF), as described above.

10: BðLDB; �Þ – COMPLEX (KIND=nag_wp) array Input

Note: the second dimension of the array B must be at least max 1;NRHSð Þ.
On entry: the n by r right-hand side matrix B.

11: LDB – INTEGER Input

On entry: the first dimension of the array B as declared in the (sub)program from which F07MPF
(ZHESVX) is called.

Constraint: LDB � max 1;Nð Þ.

12: XðLDX; �Þ – COMPLEX (KIND=nag_wp) array Output

Note: the second dimension of the array X must be at least max 1;NRHSð Þ.
On exit: if INFO ¼ 0 or Nþ 1, the n by r solution matrix X.

13: LDX – INTEGER Input

On entry: the first dimension of the array X as declared in the (sub)program from which F07MPF
(ZHESVX) is called.

Constraint: LDX � max 1;Nð Þ.

14: RCOND – REAL (KIND=nag_wp) Output

On exit: the estimate of the reciprocal condition number of the matrix A. If RCOND ¼ 0:0, the
matrix may be exactly singular. This condition is indicated by INFO > 0 and INFO � N.
Otherwise, if RCOND is less than the machine precision, the matrix is singular to working
precision. This condition is indicated by INFO ¼ Nþ 1.

15: FERRð�Þ – REAL (KIND=nag_wp) array Output

Note: the dimension of the array FERR must be at least max 1;NRHSð Þ.
On exit: if INFO ¼ 0 or Nþ 1, an estimate of the forward error bound for each computed
solution vector, such that x̂j � xj

�� ��
1= xj

�� ��
1 � FERRðjÞ where x̂j is the jth column of the

computed solution returned in the array X and xj is the corresponding column of the exact
solution X. The estimate is as reliable as the estimate for RCOND, and is almost always a slight
overestimate of the true error.

16: BERRð�Þ – REAL (KIND=nag_wp) array Output

Note: the dimension of the array BERR must be at least max 1;NRHSð Þ.
On exit: if INFO ¼ 0 or Nþ 1, an estimate of the component-wise relative backward error of
each computed solution vector x̂j (i.e., the smallest relative change in any element of A or B that
makes x̂j an exact solution).

17: WORKðmax 1;LWORKð ÞÞ – COMPLEX (KIND=nag_wp) array Workspace

On exit: if INFO ¼ 0, WORKð1Þ returns the optimal LWORK.

18: LWORK – INTEGER Input

On entry: the dimension of the array WORK as declared in the (sub)program from which
F07MPF (ZHESVX) is called.

LWORK � max 1; 2� Nð Þ, a n d f o r b e s t p e r f o r m a n c e , w h e n FACT ¼ N ,
LWORK � max 1; 2� N;N� nbð Þ, where nb is the optimal block size for F07MRF (ZHETRF).

If LWORK ¼ �1, a workspace query is assumed; the routine only calculates the optimal size of
the WORK array, returns this value as the first entry of the WORK array, and no error message
related to LWORK is issued.

19: RWORKð�Þ – REAL (KIND=nag_wp) array Workspace

Note: the dimension of the array RWORK must be at least max 1;Nð Þ.

20: INFO – INTEGER Output

On exit: INFO ¼ 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO ¼ �i, argument i had an illegal value. An explanatory message is output, and execution
of the program is terminated.

INFO > 0 and INFO � N

Element valueh i of the diagonal is exactly zero. The factorization has been completed, but the
factor D is exactly singular, so the solution and error bounds could not be computed.
RCOND ¼ 0:0 is returned.

INFO ¼ Nþ 1

D is nonsingular, but RCOND is less than machine precision, meaning that the matrix is singular
to working precision. Nevertheless, the solution and error bounds are computed because there are
a number of situations where the computed solution can be more accurate than the value of
RCOND would suggest.

7 Accuracy

For each right-hand side vector b, the computed solution x̂ is the exact solution of a perturbed system of
equations Aþ Eð Þx̂ ¼ b, where

Ek k1 ¼ O �ð Þ Ak k1;
where � is the machine precision. See Chapter 11 of Higham (2002) for further details.

If x̂ is the true solution, then the computed solution x satisfies a forward error bound of the form

x� x̂k k1
x̂k k1

� wc cond A; x̂; bð Þ

where cond A; x̂; bð Þ ¼ A�1
�� �� Aj j x̂j j þ bj jð Þ�� ��

1= x̂k k1 � cond Að Þ ¼ A�1
�� �� Aj j�� ��

1 � �1 Að Þ. If x̂ is the
jth column of X, then wc is returned in BERRðjÞ and a bound on x� x̂k k1= x̂k k1 is returned in
FERRðjÞ. See Section 4.4 of Anderson et al. (1999) for further details.

8 Parallelism and Performance

F07MPF (ZHESVX) is threaded by NAG for parallel execution in multithreaded implementations of the
NAG Library.

F07MPF (ZHESVX) makes calls to BLAS and/or LAPACK routines, which may be threaded within the
vendor library used by this implementation. Consult the documentation for the vendor library for further
information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the
OpenMP environment used within this routine. Please also consult the Users' Note for your
implementation for any additional implementation-specific information.

9 Further Comments

The factorization of A requires approximately 4
3n

3 floating-point operations.

For each right-hand side, computation of the backward error involves a minimum of 16n2 floating-point
operations. Each step of iterative refinement involves an additional 24n2 operations. At most five steps
of iterative refinement are performed, but usually only one or two steps are required. Estimating the
forward error involves solving a number of systems of equations of the form Ax ¼ b; the number is
usually 4 or 5 and never more than 11. Each solution involves approximately 8n2 operations.

The real analogue of this routine is F07MBF (DSYSVX). The complex symmetric analogue of this
routine is F07NPF (ZSYSVX).

10 Example

This example solves the equations

AX ¼ B;

where A is the Hermitian matrix

A ¼
�1:84 0:11� 0:11i �1:78� 1:18i 3:91� 1:50i
0:11þ 0:11i �4:63 �1:84þ 0:03i 2:21þ 0:21i

�1:78þ 1:18i �1:84� 0:03i �8:87 1:58� 0:90i
3:91þ 1:50i 2:21� 0:21i 1:58þ 0:90i �1:36

0
B@

1
CA

and

B ¼
2:98� 10:18i 28:68� 39:89i

�9:58þ 3:88i �24:79� 8:40i
�0:77� 16:05i 4:23� 70:02i
7:79þ 5:48i �35:39þ 18:01i

0
B@

1
CA:

Error estimates for the solutions, and an estimate of the reciprocal of the condition number of the
matrix A are also output.

10.1 Program Text

Program f07mpfe

! F07MPF Example Program Text

! Mark 26 Release. NAG Copyright 2016.

! .. Use Statements ..
Use nag_library, Only: nag_wp, x04dbf, zhesvx

! .. Implicit None Statement ..
Implicit None

! .. Parameters ..
Integer, Parameter :: nb = 64, nin = 5, nout = 6

! .. Local Scalars ..
Real (Kind=nag_wp) :: rcond
Integer :: i, ifail, info, lda, ldaf, ldb, ldx, &

lwork, n, nrhs
! .. Local Arrays ..

Complex (Kind=nag_wp), Allocatable :: a(:,:), af(:,:), b(:,:), work(:), &
x(:,:)

Real (Kind=nag_wp), Allocatable :: berr(:), ferr(:), rwork(:)
Integer, Allocatable :: ipiv(:)
Character (1) :: clabs(1), rlabs(1)

! .. Executable Statements ..
Write (nout,*) ’F07MPF Example Program Results’
Write (nout,*)
Flush (nout)

! Skip heading in data file
Read (nin,*)
Read (nin,*) n, nrhs
lda = n
ldaf = n
ldb = n
ldx = n
lwork = nb*n
Allocate (a(lda,n),af(ldaf,n),b(ldb,nrhs),work(lwork),x(ldx,nrhs), &

berr(nrhs),ferr(nrhs),rwork(n),ipiv(n))

! Read the upper triangular part of A from data file

Read (nin,*)(a(i,i:n),i=1,n)

! Read B from data file

Read (nin,*)(b(i,1:nrhs),i=1,n)

! Solve the equations AX = B for X
! The NAG name equivalent of zhesvx is f07mpf

Call zhesvx(’Not factored’,’Upper’,n,nrhs,a,lda,af,ldaf,ipiv,b,ldb,x, &
ldx,rcond,ferr,berr,work,lwork,rwork,info)

If ((info==0) .Or. (info==n+1)) Then

! Print solution, error bounds and condition number

! ifail: behaviour on error exit
! =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft

ifail = 0
Call x04dbf(’General’,’ ’,n,nrhs,x,ldx,’Bracketed’,’F7.4’, &

’Solution(s)’,’Integer’,rlabs,’Integer’,clabs,80,0,ifail)

Write (nout,*)
Write (nout,*) ’Backward errors (machine-dependent)’
Write (nout,99999) berr(1:nrhs)
Write (nout,*)
Write (nout,*) ’Estimated forward error bounds (machine-dependent)’

Write (nout,99999) ferr(1:nrhs)
Write (nout,*)
Write (nout,*) ’Estimate of reciprocal condition number’
Write (nout,99999) rcond
Write (nout,*)

If (info==n+1) Then
Write (nout,*)
Write (nout,*) ’The matrix A is singular to working precision’

End If
Else

Write (nout,99998) ’The diagonal block ’, info, ’ of D is zero’
End If

99999 Format ((3X,1P,7E11.1))
99998 Format (1X,A,I3,A)

End Program f07mpfe

10.2 Program Data

F07MPF Example Program Data
4 2 :N and NRHS

( -1.84, 0.00) ( 0.11, -0.11) ( -1.78, -1.18) ( 3.91, -1.50)
( -4.63 , 0.00) ( -1.84, 0.03) ( 2.21, 0.21)

( -8.87, 0.00) ( 1.58, -0.90)
( -1.36 , 0.00) :End matrix A

( 2.98,-10.18) ( 28.68,-39.89)
( -9.58, 3.88) (-24.79, -8.40)
( -0.77,-16.05) ( 4.23,-70.02)
( 7.79, 5.48) (-35.39, 18.01) :End matrix B

10.3 Program Results

F07MPF Example Program Results

Solution(s)
1 2

1 ( 2.0000, 1.0000) (-8.0000, 6.0000)
2 ( 3.0000,-2.0000) ( 7.0000,-2.0000)
3 (-1.0000, 2.0000) (-1.0000, 5.0000)
4 ( 1.0000,-1.0000) ( 3.0000,-4.0000)

Backward errors (machine-dependent)
5.1E-17 5.9E-17

Estimated forward error bounds (machine-dependent)
2.5E-15 3.0E-15

Estimate of reciprocal condition number
1.5E-01

