

NAG Library Routine Document

F07PDF (DSPTRF)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F07PDF (DSPTRF) computes the Bunch–Kaufman factorization of a real symmetric indefinite matrix, using packed storage.

2 Specification

```
SUBROUTINE F07PDF (UPLO, N, AP, IPIV, INFO)
  INTEGER          N, IPIV(N), INFO
  REAL (KIND=nag_wp) AP(*)
  CHARACTER(1)     UPLO
```

The routine may be called by its LAPACK name *dsptrf*.

3 Description

F07PDF (DSPTRF) factorizes a real symmetric matrix A , using the Bunch–Kaufman diagonal pivoting method and packed storage. A is factorized as either $A = PUDU^T P^T$ if $UPLO = 'U'$ or $A = PLDL^T P^T$ if $UPLO = 'L'$, where P is a permutation matrix, U (or L) is a unit upper (or lower) triangular matrix and D is a symmetric block diagonal matrix with 1 by 1 and 2 by 2 diagonal blocks; U (or L) has 2 by 2 unit diagonal blocks corresponding to the 2 by 2 blocks of D . Row and column interchanges are performed to ensure numerical stability while preserving symmetry.

This method is suitable for symmetric matrices which are not known to be positive definite. If A is in fact positive definite, no interchanges are performed and no 2 by 2 blocks occur in D .

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Arguments

- 1: UPLO – CHARACTER(1) *Input*
On entry: specifies whether the upper or lower triangular part of A is stored and how A is to be factorized.
 UPLO = 'U'
 The upper triangular part of A is stored and A is factorized as $PUDU^T P^T$, where U is upper triangular.
 UPLO = 'L'
 The lower triangular part of A is stored and A is factorized as $PLDL^T P^T$, where L is lower triangular.
Constraint: UPLO = 'U' or 'L'.
- 2: N – INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.

3: AP(*) – REAL (KIND=nag_wp) array

Input/Output

Note: the dimension of the array AP must be at least $\max(1, N \times (N + 1)/2)$.

On entry: the n by n symmetric matrix A , packed by columns.

More precisely,

if UPLO = 'U', the upper triangle of A must be stored with element A_{ij} in $AP(i + j(j - 1)/2)$ for $i \leq j$;

if UPLO = 'L', the lower triangle of A must be stored with element A_{ij} in $AP(i + (2n - j)(j - 1)/2)$ for $i \geq j$.

On exit: A is overwritten by details of the block diagonal matrix D and the multipliers used to obtain the factor U or L as specified by UPLO.

4: IPIV(N) – INTEGER array

Output

On exit: details of the interchanges and the block structure of D . More precisely,

if $IPIV(i) = k > 0$, d_{ii} is a 1 by 1 pivot block and the i th row and column of A were interchanged with the k th row and column;

if UPLO = 'U' and $IPIV(i - 1) = IPIV(i) = -l < 0$, $\begin{pmatrix} d_{i-1,i-1} & \bar{d}_{i,i-1} \\ \bar{d}_{i,i-1} & d_{ii} \end{pmatrix}$ is a 2 by 2 pivot block and the $(i - 1)$ th row and column of A were interchanged with the l th row and column;

if UPLO = 'L' and $IPIV(i) = IPIV(i + 1) = -m < 0$, $\begin{pmatrix} d_{ii} & d_{i+1,i} \\ d_{i+1,i} & d_{i+1,i+1} \end{pmatrix}$ is a 2 by 2 pivot block and the $(i + 1)$ th row and column of A were interchanged with the m th row and column.

5: INFO – INTEGER

Output

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

Element $\langle value \rangle$ of the diagonal is exactly zero. The factorization has been completed, but the block diagonal matrix D is exactly singular, and division by zero will occur if it is used to solve a system of equations.

7 Accuracy

If UPLO = 'U', the computed factors U and D are the exact factors of a perturbed matrix $A + E$, where

$$|E| \leq c(n)\epsilon P|U||D||U^T|P^T,$$

$c(n)$ is a modest linear function of n , and ϵ is the **machine precision**.

If UPLO = 'L', a similar statement holds for the computed factors L and D .

8 Parallelism and Performance

F07PDF (DSPTRF) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The elements of D overwrite the corresponding elements of A ; if D has 2 by 2 blocks, only the upper or lower triangle is stored, as specified by UPLO.

The unit diagonal elements of U or L and the 2 by 2 unit diagonal blocks are not stored. The remaining elements of U or L overwrite elements in the corresponding columns of A , but additional row interchanges must be applied to recover U or L explicitly (this is seldom necessary). If $\text{IPIV}(i) = i$, for $i = 1, 2, \dots, n$ (as is the case when A is positive definite), then U or L are stored explicitly in packed form (except for their unit diagonal elements which are equal to 1).

The total number of floating-point operations is approximately $\frac{1}{3}n^3$.

A call to F07PDF (DSPTRF) may be followed by calls to the routines:

F07PEF (DSPTRS) to solve $AX = B$;

F07PGF (DSPCON) to estimate the condition number of A ;

F07PJF (DSPTRI) to compute the inverse of A .

The complex analogues of this routine are F07PRF (ZHPTRF) for Hermitian matrices and F07QRF (ZSPTRF) for symmetric matrices.

10 Example

This example computes the Bunch–Kaufman factorization of the matrix A , where

$$A = \begin{pmatrix} 2.07 & 3.87 & 4.20 & -1.15 \\ 3.87 & -0.21 & 1.87 & 0.63 \\ 4.20 & 1.87 & 1.15 & 2.06 \\ -1.15 & 0.63 & 2.06 & -1.81 \end{pmatrix},$$

using packed storage.

10.1 Program Text

```

Program f07pdfc

!      F07PDF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: dsptrf, nag_wp, x04ccf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Integer                     :: i, ifail, info, j, n
      Character (1)               :: uplo
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: ap(:)
      Integer, Allocatable         :: ipiv(:)
!      .. Executable Statements ..

```

```

      Write (nout,*) 'F07PDF Example Program Results'
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n

      Allocate (ap(n*(n+1)/2),ipiv(n))

!      Read A from data file

      Read (nin,*) uplo
      If (uplo=='U') Then
        Read (nin,*)((ap(i+j*(j-1)/2),j=i,n),i=1,n)
      Else If (uplo=='L') Then
        Read (nin,*)((ap(i+(2*n-j)*(j-1)/2),j=1,i),i=1,n)
      End If

!      Factorize A
!      The NAG name equivalent of dsptf is f07pdf
      Call dsptf(uplo,n,ap,ipiv,info)

      Write (nout,*)
      Flush (nout)

!      Print details of factorization

!      ifail: behaviour on error exit
!      =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
      Call x04ccf(uplo,'Nonunit',n,ap,'Details of factorization',ifail)

!      Print pivot indices

      Write (nout,*)
      Write (nout,*) 'IPIV'
      Write (nout,99999) ipiv(1:n)

      If (info/=0) Then
        Write (nout,*) 'The factor D is singular'
      End If

99999 Format ((3X,7I11))
      End Program f07pdfe

```

10.2 Program Data

```

F07PDF Example Program Data
  4                               :Value of N
  'L'                           :Value of UPLO
  2.07
  3.87 -0.21
  4.20  1.87  1.15
  -1.15  0.63  2.06 -1.81 :End of matrix A

```

10.3 Program Results

F07PDF Example Program Results

```

Details of factorization
      1          2          3          4
1      2.0700
2      4.2000      1.1500
3      0.2230      0.8115      -2.5907
4      0.6537      -0.5960      0.3031      0.4074

IPIV
      -3          -3          3          4

```
