

NAG Library Routine Document

F07PEF (DSPTRS)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F07PEF (DSPTRS) solves a real symmetric indefinite system of linear equations with multiple right-hand sides,

$$AX = B,$$

where A has been factorized by F07PDF (DSPTRF), using packed storage.

2 Specification

```
SUBROUTINE F07PEF (UPLO, N, NRHS, AP, IPIV, B, LDB, INFO)
  INTEGER          N, NRHS, IPIV(*), LDB, INFO
  REAL (KIND=nag_wp) AP(*), B(LDB,*)
  CHARACTER(1)     UPLO
```

The routine may be called by its LAPACK name ***dsptrs***.

3 Description

F07PEF (DSPTRS) is used to solve a real symmetric indefinite system of linear equations $AX = B$, the routine must be preceded by a call to F07PDF (DSPTRF) which computes the Bunch–Kaufman factorization of A , using packed storage.

If $UPLO = 'U'$, $A = PUDU^T P^T$, where P is a permutation matrix, U is an upper triangular matrix and D is a symmetric block diagonal matrix with 1 by 1 and 2 by 2 blocks; the solution X is computed by solving $PUDY = B$ and then $U^T P^T X = Y$.

If $UPLO = 'L'$, $A = PLDL^T P^T$, where L is a lower triangular matrix; the solution X is computed by solving $PLDY = B$ and then $L^T P^T X = Y$.

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Arguments

- 1: UPLO – CHARACTER(1) *Input*
On entry: specifies how A has been factorized.
 UPLO = 'U'
 $A = PUDU^T P^T$, where U is upper triangular.
 UPLO = 'L'
 $A = PLDL^T P^T$, where L is lower triangular.
Constraint: UPLO = 'U' or 'L'.

- 2: N – INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.
- 3: NRHS – INTEGER *Input*
On entry: r , the number of right-hand sides.
Constraint: $NRHS \geq 0$.
- 4: AP(*) – REAL (KIND=nag_wp) array *Input*
Note: the dimension of the array AP must be at least $\max(1, N \times (N + 1)/2)$.
On entry: the factorization of A stored in packed form, as returned by F07PDF (DSPTRF).
- 5: IPIV(*) – INTEGER array *Input*
Note: the dimension of the array IPIV must be at least $\max(1, N)$.
On entry: details of the interchanges and the block structure of D , as returned by F07PDF (DSPTRF).
- 6: B(LDB, *) – REAL (KIND=nag_wp) array *Input/Output*
Note: the second dimension of the array B must be at least $\max(1, NRHS)$.
On entry: the n by r right-hand side matrix B .
On exit: the n by r solution matrix X .
- 7: LDB – INTEGER *Input*
On entry: the first dimension of the array B as declared in the (sub)program from which F07PEF (DSPTRS) is called.
Constraint: $LDB \geq \max(1, N)$.
- 8: INFO – INTEGER *Output*
On exit: $INFO = 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

$INFO < 0$

If $INFO = -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

For each right-hand side vector b , the computed solution x is the exact solution of a perturbed system of equations $(A + E)x = b$, where

$$\text{if UPLO} = 'U', |E| \leq c(n)\epsilon P|U||D||U^T|P^T;$$

$$\text{if UPLO} = 'L', |E| \leq c(n)\epsilon P|L||D||L^T|P^T,$$

$c(n)$ is a modest linear function of n , and ϵ is the *machine precision*.

If \hat{x} is the true solution, then the computed solution x satisfies a forward error bound of the form

$$\frac{\|x - \hat{x}\|_{\infty}}{\|x\|_{\infty}} \leq c(n) \text{cond}(A, x) \epsilon$$

where $\text{cond}(A, x) = \| |A^{-1}| |A| |x| \|_{\infty} / \|x\|_{\infty} \leq \text{cond}(A) = \| |A^{-1}| |A| \|_{\infty} \leq \kappa_{\infty}(A)$.

Note that $\text{cond}(A, x)$ can be much smaller than $\text{cond}(A)$.

Forward and backward error bounds can be computed by calling F07PHF (DSPRFS), and an estimate for $\kappa_{\infty}(A)$ ($= \kappa_1(A)$) can be obtained by calling F07PGF (DSPCON).

8 Parallelism and Performance

F07PEF (DSPTRS) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations is approximately $2n^2r$.

This routine may be followed by a call to F07PHF (DSPRFS) to refine the solution and return an error estimate.

The complex analogues of this routine are F07PSF (ZHPTRS) for Hermitian matrices and F07QSF (ZSPTRS) for symmetric matrices.

10 Example

This example solves the system of equations $AX = B$, where

$$A = \begin{pmatrix} 2.07 & 3.87 & 4.20 & -1.15 \\ 3.87 & -0.21 & 1.87 & 0.63 \\ 4.20 & 1.87 & 1.15 & 2.06 \\ -1.15 & 0.63 & 2.06 & -1.81 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} -9.50 & 27.85 \\ -8.38 & 9.90 \\ -6.07 & 19.25 \\ -0.96 & 3.93 \end{pmatrix}.$$

Here A is symmetric indefinite, stored in packed form, and must first be factorized by F07PDF (DSPTRF).

10.1 Program Text

```

Program f07pefe

!      F07PEF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: dsptf, dsptf, nag_wp, x04caf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Integer                     :: i, ifail, info, j, ldb, n, nrhs
      Character (1)               :: uplo
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: ap(:), b(:, :)
      Integer, Allocatable        :: ipiv(:)
!      .. Executable Statements ..

```

```

      Write (nout,*) 'F07PEF Example Program Results'
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, nrhs
      ldb = n
      Allocate (ap(n*(n+1)/2),b(ldb,nrhs),ipiv(n))

!      Read A and B from data file

      Read (nin,*) uplo
      If (uplo=='U') Then
        Read (nin,*)((ap(i+j*(j-1)/2),j=i,n),i=1,n)
      Else If (uplo=='L') Then
        Read (nin,*)((ap(i+(2*n-j)*(j-1)/2),j=1,i),i=1,n)
      End If
      Read (nin,*)(b(i,1:nrhs),i=1,n)

!      Factorize A
!      The NAG name equivalent of dsptf is f07pdf
      Call dsptf(uplo,n,ap,ipiv,info)

      Write (nout,*)
      Flush (nout)
      If (info==0) Then

!      Compute solution
!      The NAG name equivalent of dsptf is f07pdf
      Call dsptf(uplo,n,nrhs,ap,ipiv,b,ldb,info)

!      Print solution

!      ifail: behaviour on error exit
!      =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
      Call x04caf('General',' ',n,nrhs,b,ldb,'Solution(s)',ifail)

      Else
        Write (nout,*) 'The factor D is singular'
      End If

      End Program f07pefe

```

10.2 Program Data

F07PEF Example Program Data

```

  4  2                      :Values of N and NRHS
  'L'                      :Value of UPLO
  2.07
  3.87 -0.21
  4.20  1.87  1.15
-1.15  0.63  2.06 -1.81    :End of matrix A
-9.50 27.85
-8.38  9.90
-6.07 19.25
-0.96  3.93              :End of matrix B

```

10.3 Program Results

F07PEF Example Program Results

```

Solution(s)
      1      2
  1    -4.0000  1.0000
  2    -1.0000  4.0000
  3     2.0000  3.0000
  4     5.0000  2.0000

```
