

NAG Library Routine Document

F07TWF (ZTRTRI)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F07TWF (ZTRTRI) computes the inverse of a complex triangular matrix.

2 Specification

```
SUBROUTINE F07TWF (UPLO, DIAG, N, A, LDA, INFO)
  INTEGER          N, LDA, INFO
  COMPLEX (KIND=nag_wp) A(LDA,*)
  CHARACTER(1)     UPLO, DIAG
```

The routine may be called by its LAPACK name *ztrtri*.

3 Description

F07TWF (ZTRTRI) forms the inverse of a complex triangular matrix A . Note that the inverse of an upper (lower) triangular matrix is also upper (lower) triangular.

4 References

Du Croz J J and Higham N J (1992) Stability of methods for matrix inversion *IMA J. Numer. Anal.* **12** 1–19

5 Arguments

- 1: UPLO – CHARACTER(1) *Input*
On entry: specifies whether A is upper or lower triangular.
 UPLO = 'U'
 A is upper triangular.
 UPLO = 'L'
 A is lower triangular.
Constraint: UPLO = 'U' or 'L'.
- 2: DIAG – CHARACTER(1) *Input*
On entry: indicates whether A is a nonunit or unit triangular matrix.
 DIAG = 'N'
 A is a nonunit triangular matrix.
 DIAG = 'U'
 A is a unit triangular matrix; the diagonal elements are not referenced and are assumed to be 1.
Constraint: DIAG = 'N' or 'U'.
- 3: N – INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.

4: A(LDA,*) – COMPLEX (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, N)$.

On entry: the n by n triangular matrix A .

If UPLO = 'U', A is upper triangular and the elements of the array below the diagonal are not referenced.

If UPLO = 'L', A is lower triangular and the elements of the array above the diagonal are not referenced.

If DIAG = 'U', the diagonal elements of A are assumed to be 1, and are not referenced.

On exit: A is overwritten by A^{-1} , using the same storage format as described above.

5: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F07TWF (ZTRTRI) is called.

Constraint: $LDA \geq \max(1, N)$.

6: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

Element $\langle value \rangle$ of the diagonal is exactly zero. A is singular its inverse cannot be computed.

7 Accuracy

The computed inverse X satisfies

$$|XA - I| \leq c(n)\epsilon|X||A|,$$

where $c(n)$ is a modest linear function of n , and ϵ is the **machine precision**.

Note that a similar bound for $|AX - I|$ cannot be guaranteed, although it is almost always satisfied.

The computed inverse satisfies the forward error bound

$$|X - A^{-1}| \leq c(n)\epsilon|A^{-1}||A||X|.$$

See Du Croz and Higham (1992).

8 Parallelism and Performance

F07TWF (ZTRTRI) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of real floating-point operations is approximately $\frac{4}{3}n^3$.

The real analogue of this routine is F07TJF (DTRTRI).

10 Example

This example computes the inverse of the matrix A , where

$$A = \begin{pmatrix} 4.78 + 4.56i & 0.00 + 0.00i & 0.00 + 0.00i & 0.00 + 0.00i \\ 2.00 - 0.30i & -4.11 + 1.25i & 0.00 + 0.00i & 0.00 + 0.00i \\ 2.89 - 1.34i & 2.36 - 4.25i & 4.15 + 0.80i & 0.00 + 0.00i \\ -1.89 + 1.15i & 0.04 - 3.69i & -0.02 + 0.46i & 0.33 - 0.26i \end{pmatrix}.$$

10.1 Program Text

```

Program f07twfe

!      F07TWF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: nag_wp, x04dbf, ztrtri
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
      Character (1), Parameter    :: diag = 'N'
!      .. Local Scalars ..
      Integer                     :: i, ifail, info, lda, n
      Character (1)               :: uplo
!      .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: a(:, :)
      Character (1)               :: clabs(1), rlabs(1)
!      .. Executable Statements ..
      Write (nout,*) 'F07TWF Example Program Results'
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n
      lda = n
      Allocate (a(lda,n))

!      Read A from data file

      Read (nin,*) uplo
      If (uplo=='U') Then
         Read (nin,*)(a(i,i:n),i=1,n)
      Else If (uplo=='L') Then
         Read (nin,*)(a(i,1:i),i=1,n)
      End If

!      Compute inverse of A
!      The NAG name equivalent of ztrtri is f07twf
      Call ztrtri(uplo,diag,n,a,lda,info)

!      Print inverse

      Write (nout,*)
      Flush (nout)

!      ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft

```

```

      ifail = 0
      Call x04dbf(uplo,diag,n,n,a,lda,'Bracketed','F7.4','Inverse','Integer', &
        rlabs,'Integer',clabs,80,0,ifail)

      End Program f07twfe

```

10.2 Program Data

F07TWF Example Program Data

```

      4                                     :Value of N
      'L'                                 :Value of UPLO
      ( 4.78, 4.56)
      ( 2.00,-0.30) (-4.11, 1.25)
      ( 2.89,-1.34) ( 2.36,-4.25) ( 4.15, 0.80)
      (-1.89, 1.15) ( 0.04,-3.69) (-0.02, 0.46) ( 0.33,-0.26) :End of matrix A

```

10.3 Program Results

F07TWF Example Program Results

```

Inverse
      1           2           3           4
1  ( 0.1095,-0.1045)
2  ( 0.0582,-0.0411) (-0.2227,-0.0677)
3  ( 0.0032, 0.1905) ( 0.1538,-0.2192) ( 0.2323,-0.0448)
4  ( 0.7602, 0.2814) ( 1.6184,-1.4346) ( 0.1289,-0.2250) ( 1.8697, 1.4731)

```
