

# NAG Library Routine Document

## F07USF (ZTPTRS)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F07USF (ZTPTRS) solves a complex triangular system of linear equations with multiple right-hand sides,  $AX = B$ ,  $A^T X = B$  or  $A^H X = B$ , using packed storage.

### 2 Specification

SUBROUTINE F07USF (UPLO, TRANS, DIAG, N, NRHS, AP, B, LDB, INFO)

INTEGER N, NRHS, LDB, INFO  
 COMPLEX (KIND=nag\_wp) AP(\*), B(LDB,\*)  
 CHARACTER(1) UPLO, TRANS, DIAG

The routine may be called by its LAPACK name *ztptrs*.

### 3 Description

F07USF (ZTPTRS) solves a complex triangular system of linear equations  $AX = B$ ,  $A^T X = B$  or  $A^H X = B$ , using packed storage.

### 4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

Higham N J (1989) The accuracy of solutions to triangular systems *SIAM J. Numer. Anal.* **26** 1252–1265

### 5 Arguments

- 1: UPLO – CHARACTER(1) *Input*  
*On entry:* specifies whether  $A$  is upper or lower triangular.  
 UPLO = 'U'  
      $A$  is upper triangular.  
 UPLO = 'L'  
      $A$  is lower triangular.  
*Constraint:* UPLO = 'U' or 'L'.
- 2: TRANS – CHARACTER(1) *Input*  
*On entry:* indicates the form of the equations.  
 TRANS = 'N'  
     The equations are of the form  $AX = B$ .  
 TRANS = 'T'  
     The equations are of the form  $A^T X = B$ .

TRANS = 'C'

The equations are of the form  $A^H X = B$ .

*Constraint:* TRANS = 'N', 'T' or 'C'.

3: DIAG – CHARACTER(1) *Input*

*On entry:* indicates whether  $A$  is a nonunit or unit triangular matrix.

DIAG = 'N'

$A$  is a nonunit triangular matrix.

DIAG = 'U'

$A$  is a unit triangular matrix; the diagonal elements are not referenced and are assumed to be 1.

*Constraint:* DIAG = 'N' or 'U'.

4: N – INTEGER *Input*

*On entry:*  $n$ , the order of the matrix  $A$ .

*Constraint:*  $N \geq 0$ .

5: NRHS – INTEGER *Input*

*On entry:*  $r$ , the number of right-hand sides.

*Constraint:* NRHS  $\geq 0$ .

6: AP(\*) – COMPLEX (KIND=nag\_wp) array *Input*

**Note:** the dimension of the array AP must be at least  $\max(1, N \times (N + 1)/2)$ .

*On entry:* the  $n$  by  $n$  triangular matrix  $A$ , packed by columns.

More precisely,

if UPLO = 'U', the upper triangle of  $A$  must be stored with element  $A_{ij}$  in  $AP(i + j(j - 1)/2)$  for  $i \leq j$ ;

if UPLO = 'L', the lower triangle of  $A$  must be stored with element  $A_{ij}$  in  $AP(i + (2n - j)(j - 1)/2)$  for  $i \geq j$ .

If DIAG = 'U', the diagonal elements of  $A$  are assumed to be 1, and are not referenced; the same storage scheme is used whether DIAG = 'N' or 'U'.

7: B(LDB,\*) – COMPLEX (KIND=nag\_wp) array *Input/Output*

**Note:** the second dimension of the array B must be at least  $\max(1, \text{NRHS})$ .

*On entry:* the  $n$  by  $r$  right-hand side matrix  $B$ .

*On exit:* the  $n$  by  $r$  solution matrix  $X$ .

8: LDB – INTEGER *Input*

*On entry:* the first dimension of the array B as declared in the (sub)program from which F07USF (ZTPTRS) is called.

*Constraint:* LDB  $\geq \max(1, N)$ .

9: INFO – INTEGER *Output*

*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If INFO =  $-i$ , argument  $i$  had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

Element  $\langle value \rangle$  of the diagonal is exactly zero.  $A$  is singular and the solution has not been computed.

## 7 Accuracy

The solutions of triangular systems of equations are usually computed to high accuracy. See Higham (1989).

For each right-hand side vector  $b$ , the computed solution  $x$  is the exact solution of a perturbed system of equations  $(A + E)x = b$ , where

$$|E| \leq c(n)\epsilon|A|,$$

$c(n)$  is a modest linear function of  $n$ , and  $\epsilon$  is the *machine precision*.

If  $\hat{x}$  is the true solution, then the computed solution  $x$  satisfies a forward error bound of the form

$$\frac{\|x - \hat{x}\|_{\infty}}{\|x\|_{\infty}} \leq c(n) \text{cond}(A, x)\epsilon, \quad \text{provided} \quad c(n) \text{cond}(A, x)\epsilon < 1,$$

where  $\text{cond}(A, x) = \| |A^{-1}| |A| |x| \|_{\infty} / \|x\|_{\infty}$ .

Note that  $\text{cond}(A, x) \leq \text{cond}(A) = \| |A^{-1}| |A| \|_{\infty} \leq \kappa_{\infty}(A)$ ;  $\text{cond}(A, x)$  can be much smaller than  $\text{cond}(A)$  and it is also possible for  $\text{cond}(A^H)$ , which is the same as  $\text{cond}(A^T)$ , to be much larger (or smaller) than  $\text{cond}(A)$ .

Forward and backward error bounds can be computed by calling F07UVF (ZTPRFS), and an estimate for  $\kappa_{\infty}(A)$  can be obtained by calling F07UUF (ZTPCON) with NORM = 'I'.

## 8 Parallelism and Performance

F07USF (ZTPTRS) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F07USF (ZTPTRS) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The total number of real floating-point operations is approximately  $4n^2r$ .

The real analogue of this routine is F07UEF (DTPTRS).

## 10 Example

This example solves the system of equations  $AX = B$ , where

$$A = \begin{pmatrix} 4.78 + 4.56i & 0.00 + 0.00i & 0.00 + 0.00i & 0.00 + 0.00i \\ 2.00 - 0.30i & -4.11 + 1.25i & 0.00 + 0.00i & 0.00 + 0.00i \\ 2.89 - 1.34i & 2.36 - 4.25i & 4.15 + 0.80i & 0.00 + 0.00i \\ -1.89 + 1.15i & 0.04 - 3.69i & -0.02 + 0.46i & 0.33 - 0.26i \end{pmatrix}$$

and

$$B = \begin{pmatrix} -14.78 - 32.36i & -18.02 + 28.46i \\ 2.98 - 2.14i & 14.22 + 15.42i \\ -20.96 + 17.06i & 5.62 + 35.89i \\ 9.54 + 9.91i & -16.46 - 1.73i \end{pmatrix},$$

using packed storage for  $A$ .

### 10.1 Program Text

Program f07usfe

```
!      F07USF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: nag_wp, x04dbf, ztptrs
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
      Character (1), Parameter    :: diag = 'N', trans = 'N'
!      .. Local Scalars ..
      Integer                     :: i, ifail, info, j, ldb, n, nrhs
      Character (1)               :: uplo
!      .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: ap(:), b(:, :)
      Character (1)               :: clabs(1), rlabs(1)
!      .. Executable Statements ..
      Write (nout,*) 'F07USF Example Program Results'
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, nrhs
      ldb = n
      Allocate (ap(n*(n+1)/2), b(ldb, nrhs))

!      Read A and B from data file

      Read (nin,*) uplo
      If (uplo=='U') Then
         Read (nin,*) ((ap(i+j*(j-1)/2), j=i, n), i=1, n)
      Else If (uplo=='L') Then
         Read (nin,*) ((ap(i+(2*n-j)*(j-1)/2), j=1, i), i=1, n)
      End If
      Read (nin,*) (b(i, 1:nrhs), i=1, n)

!      Compute solution
!      The NAG name equivalent of ztptrs is f07usf
      Call ztptrs(uplo, trans, diag, n, nrhs, ap, b, ldb, info)

!      Print solution

      Write (nout,*)
      Flush (nout)
      If (info==0) Then

!          ifail: behaviour on error exit
!          =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
```

```

        ifail = 0
        Call x04dbf('General',' ',n,nrhs,b,ldb,'Bracketed','F7.4',
            'Solution(s)','Integer',rlabs,'Integer',clabs,80,0,ifail)
        &
    Else
        Write (nout,*) 'A is singular'
    End If

End Program f07usfe

```

## 10.2 Program Data

F07USF Example Program Data

```

    4  2                                     :Values of N and NRHS
    'L'                                     :Value of UPLO
    ( 4.78, 4.56)
    ( 2.00,-0.30) (-4.11, 1.25)
    ( 2.89,-1.34) ( 2.36,-4.25) ( 4.15, 0.80)
    (-1.89, 1.15) ( 0.04,-3.69) (-0.02, 0.46) ( 0.33,-0.26) :End of matrix A
    (-14.78,-32.36) (-18.02, 28.46)
    ( 2.98, -2.14) ( 14.22, 15.42)
    (-20.96, 17.06) ( 5.62, 35.89)
    ( 9.54, 9.91) (-16.46, -1.73)           :End of matrix B

```

## 10.3 Program Results

F07USF Example Program Results

```

Solution(s)
           1           2
1  (-5.0000,-2.0000) ( 1.0000, 5.0000)
2  (-3.0000,-1.0000) (-2.0000,-2.0000)
3  ( 2.0000, 1.0000) ( 3.0000, 4.0000)
4  ( 4.0000, 3.0000) ( 4.0000,-3.0000)

```

---