

NAG Library Routine Document

F07VSF (ZTBTRS)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F07VSF (ZTBTRS) solves a complex triangular band system of linear equations with multiple right-hand sides, $AX = B$, $A^T X = B$ or $A^H X = B$.

2 Specification

```
SUBROUTINE F07VSF (UPLO, TRANS, DIAG, N, KD, NRHS, AB, LDAB, B, LDB,      &
                  INFO)
```

```
INTEGER          N, KD, NRHS, LDAB, LDB, INFO
COMPLEX (KIND=nag_wp) AB(LDAB,*), B(LDB,*)
CHARACTER(1)     UPLO, TRANS, DIAG
```

The routine may be called by its LAPACK name *ztbtrs*.

3 Description

F07VSF (ZTBTRS) solves a complex triangular band system of linear equations $AX = B$, $A^T X = B$ or $A^H X = B$.

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

Higham N J (1989) The accuracy of solutions to triangular systems *SIAM J. Numer. Anal.* **26** 1252–1265

5 Arguments

- 1: UPLO – CHARACTER(1) *Input*
On entry: specifies whether A is upper or lower triangular.
UPLO = 'U'
 A is upper triangular.
UPLO = 'L'
 A is lower triangular.
Constraint: UPLO = 'U' or 'L'.
- 2: TRANS – CHARACTER(1) *Input*
On entry: indicates the form of the equations.
TRANS = 'N'
The equations are of the form $AX = B$.
TRANS = 'T'
The equations are of the form $A^T X = B$.

TRANS = 'C'

The equations are of the form $A^H X = B$.

Constraint: TRANS = 'N', 'T' or 'C'.

3: DIAG – CHARACTER(1)

Input

On entry: indicates whether A is a nonunit or unit triangular matrix.

DIAG = 'N'

A is a nonunit triangular matrix.

DIAG = 'U'

A is a unit triangular matrix; the diagonal elements are not referenced and are assumed to be 1.

Constraint: DIAG = 'N' or 'U'.

4: N – INTEGER

Input

On entry: n , the order of the matrix A .

Constraint: $N \geq 0$.

5: KD – INTEGER

Input

On entry: k_d , the number of superdiagonals of the matrix A if UPLO = 'U', or the number of subdiagonals if UPLO = 'L'.

Constraint: $KD \geq 0$.

6: NRHS – INTEGER

Input

On entry: r , the number of right-hand sides.

Constraint: NRHS ≥ 0 .

7: AB(LDAB,*) – COMPLEX (KIND=nag_wp) array

Input

Note: the second dimension of the array AB must be at least $\max(1, N)$.

On entry: the n by n triangular band matrix A .

The matrix is stored in rows 1 to $k_d + 1$, more precisely,

if UPLO = 'U', the elements of the upper triangle of A within the band must be stored with element A_{ij} in $AB(k_d + 1 + i - j, j)$ for $\max(1, j - k_d) \leq i \leq j$;

if UPLO = 'L', the elements of the lower triangle of A within the band must be stored with element A_{ij} in $AB(1 + i - j, j)$ for $j \leq i \leq \min(n, j + k_d)$.

If DIAG = 'U', the diagonal elements of A are assumed to be 1, and are not referenced.

8: LDAB – INTEGER

Input

On entry: the first dimension of the array AB as declared in the (sub)program from which F07VSF (ZTBTRS) is called.

Constraint: LDAB $\geq KD + 1$.

9: B(LDB,*) – COMPLEX (KIND=nag_wp) array

Input/Output

Note: the second dimension of the array B must be at least $\max(1, NRHS)$.

On entry: the n by r right-hand side matrix B .

On exit: the n by r solution matrix X .

10: LDB – INTEGER

Input

On entry: the first dimension of the array B as declared in the (sub)program from which F07VSF (ZTBTRS) is called.

Constraint: $LDB \geq \max(1, N)$.

11: INFO – INTEGER

Output

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

Element $\langle value \rangle$ of the diagonal is exactly zero. A is singular and the solution has not been computed.

7 Accuracy

The solutions of triangular systems of equations are usually computed to high accuracy. See Higham (1989).

For each right-hand side vector b , the computed solution x is the exact solution of a perturbed system of equations $(A + E)x = b$, where

$$|E| \leq c(k)\epsilon|A|,$$

$c(k)$ is a modest linear function of k , and ϵ is the *machine precision*.

If \hat{x} is the true solution, then the computed solution x satisfies a forward error bound of the form

$$\frac{\|x - \hat{x}\|_{\infty}}{\|x\|_{\infty}} \leq c(k) \text{cond}(A, x)\epsilon, \quad \text{provided} \quad c(k) \text{cond}(A, x)\epsilon < 1,$$

where $\text{cond}(A, x) = \|A^{-1}\| \|A\| \|x\|_{\infty} / \|x\|_{\infty}$.

Note that $\text{cond}(A, x) \leq \text{cond}(A) = \|A^{-1}\| \|A\|_{\infty} \leq \kappa_{\infty}(A)$; $\text{cond}(A, x)$ can be much smaller than $\text{cond}(A)$ and it is also possible for $\text{cond}(A^H)$, which is the same as $\text{cond}(A^T)$, to be much larger (or smaller) than $\text{cond}(A)$.

Forward and backward error bounds can be computed by calling F07VVF (ZTBRFS), and an estimate for $\kappa_{\infty}(A)$ can be obtained by calling F07VUF (ZTBCON) with NORM = 'I'.

8 Parallelism and Performance

F07VSF (ZTBTRS) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F07VSF (ZTBTRS) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of real floating-point operations is approximately $8nkr$ if $k \ll n$.

The real analogue of this routine is F07VEF (DTBTRS).

10 Example

This example solves the system of equations $AX = B$, where

$$A = \begin{pmatrix} -1.94 + 4.43i & 0.00 + 0.00i & 0.00 + 0.00i & 0.00 + 0.00i \\ -3.39 + 3.44i & 4.12 - 4.27i & 0.00 + 0.00i & 0.00 + 0.00i \\ 1.62 + 3.68i & -1.84 + 5.53i & 0.43 - 2.66i & 0.00 + 0.00i \\ 0.00 + 0.00i & -2.77 - 1.93i & 1.74 - 0.04i & 0.44 + 0.10i \end{pmatrix}$$

and

$$B = \begin{pmatrix} -8.86 - 3.88i & -24.09 - 5.27i \\ -15.57 - 23.41i & -57.97 + 8.14i \\ -7.63 + 22.78i & 19.09 - 29.51i \\ -14.74 - 2.40i & 19.17 + 21.33i \end{pmatrix}.$$

Here A is treated as a lower triangular band matrix with two subdiagonals.

10.1 Program Text

Program f07vsfe

```
!      F07VSF Example Program Text
!
!      Mark 26 Release. NAG Copyright 2016.
!
!      .. Use Statements ..
      Use nag_library, Only: nag_wp, x04dbf, ztbtrs
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
      Character (1), Parameter    :: diag = 'N', trans = 'N'
!      .. Local Scalars ..
      Integer                     :: i, ifail, info, j, kd, ldab, ldb, n, &
                                   nrhs
      Character (1)               :: uplo
!      .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: ab(:,,:), b(:,:)
      Character (1)               :: clabs(1), rlabs(1)
!      .. Intrinsic Procedures ..
      Intrinsic                   :: max, min
!      .. Executable Statements ..
      Write (nout,*) 'F07VSF Example Program Results'
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, kd, nrhs
      ldab = kd + 1
      ldb = n
      Allocate (ab(ldab,n),b(ldb,nrhs))
!
!      Read A and B from data file
!
      Read (nin,*) uplo
      If (uplo=='U') Then
        Do i = 1, n
          Read (nin,*)(ab(kd+1+i-j,j),j=i,min(n,i+kd))
        End Do
      Else If (uplo=='L') Then
        Do i = 1, n
          Read (nin,*)(ab(1+i-j,j),j=max(1,i-kd),i)
        End Do
      End If
```

```

      End If
      Read (nin,*)(b(i,1:nrhs),i=1,n)

!      Compute solution
!      The NAG name equivalent of ztbtrs is f07vsf
      Call ztbtrs(uplo,trans,diag,n,kd,nrhs,ab,ldab,b,ldb,info)

!      Print solution

      Write (nout,*)
      Flush (nout)
      If (info==0) Then

!          ifail: behaviour on error exit
!          =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
          ifail = 0
          Call x04dbf('General',' ',n,nrhs,b,ldb,'Bracketed','F7.4',           &
                    'Solution(s)','Integer',rlabs,'Integer',clabs,80,0,ifail)

      Else
          Write (nout,*) 'A is singular'
      End If

      End Program f07vsfe

```

10.2 Program Data

F07VSF Example Program Data

```

  4  2  2                                :Values of N, KD and NRHS
  'L'                                    :Value of UPLO
(-1.94, 4.43)
(-3.39, 3.44) ( 4.12,-4.27)
( 1.62, 3.68) (-1.84, 5.53) ( 0.43,-2.66)
              (-2.77,-1.93) ( 1.74,-0.04) ( 0.44, 0.10) :End of matrix A
( -8.86, -3.88) (-24.09, -5.27)
(-15.57,-23.41) (-57.97,  8.14)
( -7.63, 22.78) ( 19.09,-29.51)
(-14.74, -2.40) ( 19.17, 21.33)           :End of matrix B

```

10.3 Program Results

F07VSF Example Program Results

```

Solution(s)
              1              2
1  ( 0.0000, 2.0000) ( 1.0000, 5.0000)
2  ( 1.0000,-3.0000) (-7.0000,-2.0000)
3  (-4.0000,-5.0000) ( 3.0000, 4.0000)
4  ( 2.0000,-1.0000) (-6.0000,-9.0000)

```
