NAG Library Routine Document

F08BBF (DTPQRT)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08BBF (DTPQRT) computes the QR factorization of a real $(m+n)$ by n triangular-pentagonal matrix.

2 Specification

```
SUBROUTINE F08BBF (M, N, L, NB, A, LDA, B, LDB, T, LDT, WORK, INFO)
INTEGER             M, N, L, NB, LDA, LDB, LDT, INFO
REAL (KIND=nag_wp) A(LDA,*), B(LDB,*), T(LDT,*), WORK(*)
```

The routine may be called by its LAPACK name ***dtpqrt***.

3 Description

F08BBF (DTPQRT) forms the QR factorization of a real $(m+n)$ by n triangular-pentagonal matrix C,

$$C = \begin{pmatrix} A \\ B \end{pmatrix}$$

where A is an upper triangular n by n matrix and B is an m by n pentagonal matrix consisting of an $(m-l)$ by n rectangular matrix B_1 on top of an l by n upper trapezoidal matrix B_2:

$$B = \begin{pmatrix} B_1 \\ B_2 \end{pmatrix}.$$

The upper trapezoidal matrix B_2 consists of the first l rows of an n by n upper triangular matrix, where $0 \leq l \leq \min(m,n)$. If $l = 0$, B is m by n rectangular; if $l = n$ and $m = n$, B is upper triangular.

A recursive, explicitly blocked, QR factorization (see F08ABF (DGEQRT)) is performed on the matrix C. The upper triangular matrix R, details of the orthogonal matrix Q, and further details (the block reflector factors) of Q are returned.

Typically the matrix A or B_2 contains the matrix R from the QR factorization of a subproblem and F08BBF (DTPQRT) performs the QR update operation from the inclusion of matrix B_1.

For example, consider the QR factorization of an l by n matrix $\hat{B}$ with $l < n$: $\hat{B} = \hat{Q}\hat{R}$, $\hat{R} = \begin{pmatrix} \hat{R}_1 & \hat{R}_2 \end{pmatrix}$, where $\hat{R}_1$ is l by l upper triangular and $\hat{R}_2$ is $(n-l)$ by n rectangular (this can be performed by F08ABF (DGEQRT)). Given an initial least-squares problem $\hat{B}\hat{X} = \hat{Y}$ where X and Y are l by $nrhs$ matrices, we have $\hat{R}\hat{X} = \hat{Q}^{\mathrm{T}}\hat{Y}$.

Now, adding an additional $m-l$ rows to the original system gives the augmented least squares problem

$$BX = Y$$

where B is an m by n matrix formed by adding $m-l$ rows on top of $\hat{R}$ and Y is an m by $nrhs$ matrix formed by adding $m-l$ rows on top of $\hat{Q}^{\mathrm{T}}\hat{Y}$.

F08BBF (DTPQRT) can then be used to perform the QR factorization of the pentagonal matrix B; the n by n matrix A will be zero on input and contain R on output.

In the case where $\hat{B}$ is r by n, $r \geq n$, $\hat{R}$ is n by n upper triangular (forming A) on top of $r-n$ rows of zeros (forming first $r-n$ rows of B). Augmentation is then performed by adding rows to the bottom of B with $l = 0$.

4 References

Elmroth E and Gustavson F (2000) Applying Recursion to Serial and Parallel QR Factorization Leads to Better Performance *IBM Journal of Research and Development. (Volume 44)* **4** 605–624

Golub G H and Van Loan C F (2012) *Matrix Computations* (4th Edition) Johns Hopkins University Press, Baltimore

5 Arguments

1: M – INTEGER *Input*

On entry: m, the number of rows of the matrix B.

Constraint: $\mathrm{M} \geq 0$.

2: N – INTEGER *Input*

On entry: n, the number of columns of the matrix B and the order of the upper triangular matrix A.

Constraint: $\mathrm{N} \geq 0$.

3: L – INTEGER *Input*

On entry: l, the number of rows of the trapezoidal part of B (i.e., B_2).

Constraint: $0 \leq \mathrm{L} \leq \min(\mathrm{M}, \mathrm{N})$.

4: NB – INTEGER *Input*

On entry: the explicitly chosen block-size to be used in the algorithm for computing the QR factorization. See Section 9 for details.

Constraints:

$\mathrm{NB} \geq 1$;
if $\mathrm{N} > 0$, $\mathrm{NB} \leq \mathrm{N}$.

5: A(LDA, $*$) – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: the n by n upper triangular matrix A.

On exit: the upper triangle is overwritten by the corresponding elements of the n by n upper triangular matrix R.

6: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F08BBF (DTPQRT) is called.

Constraint: $\mathrm{LDA} \geq \max(1, \mathrm{N})$.

7: B(LDB, $*$) – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array B must be at least $\max(1, \mathrm{N})$.

On entry: the m by n pentagonal matrix B composed of an $(m - l)$ by n rectangular matrix B_1 above an l by n upper trapezoidal matrix B_2.

On exit: details of the orthogonal matrix Q.

8: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F08BBF (DTPQRT) is called.

Constraint: LDB $\geq \max(1, \mathrm{M})$.

9: T(LDT, $*$) – REAL (KIND=nag_wp) array *Output*

Note: the second dimension of the array T must be at least N.

On exit: further details of the orthogonal matrix Q. The number of blocks is $b = \left\lceil \frac{k}{\mathrm{NB}} \right\rceil$, where $k = \min(m, n)$ and each block is of order NB except for the last block, which is of order $k - (b-1) \times \mathrm{NB}$. For each of the blocks, an upper triangular block reflector factor is computed: $\boldsymbol{T}_1, \boldsymbol{T}_2, \ldots, \boldsymbol{T}_b$. These are stored in the NB by n matrix T as $\boldsymbol{T} = [\boldsymbol{T}_1 | \boldsymbol{T}_2 | \ldots | \boldsymbol{T}_b]$.

10: LDT – INTEGER *Input*

On entry: the first dimension of the array T as declared in the (sub)program from which F08BBF (DTPQRT) is called.

Constraint: LDT $\geq$ NB.

11: WORK($*$) – REAL (KIND=nag_wp) array *Workspace*

Note: the dimension of the array WORK must be at least NB $\times$ N.

12: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO $= -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed factorization is the exact factorization of a nearby matrix $(A + E)$, where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and ϵ is the ***machine precision***.

8 Parallelism and Performance

F08BBF (DTPQRT) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations is approximately $\frac{2}{3}n^2(3m - n)$ if $m \geq n$ or $\frac{2}{3}m^2(3n - m)$ if $m < n$.

The block size, NB, used by F08BBF (DTPQRT) is supplied explicitly through the interface. For moderate and large sizes of matrix, the block size can have a marked effect on the efficiency of the

algorithm with the optimal value being dependent on problem size and platform. A value of $\mathrm{NB} = 64 \ll \min(m, n)$ is likely to achieve good efficiency and it is unlikely that an optimal value would exceed 340.

To apply Q to an arbitrary real rectangular matrix C, F08BBF (DTPQRT) may be followed by a call to F08BCF (DTPMQRT). For example,

```
CALL DTPMQRT('Left','Transpose',M,P,N,L,NB,B,LDB, &
              T,LDT,C,LDC,C(n+1,1),LDC,WORK,INFO)
```

forms $C = Q^{\mathrm{T}}C$, where C is $(m + n)$ by p.

To form the orthogonal matrix Q explicitly set $p = m + n$, initialize C to the identity matrix and make a call to F08BCF (DTPMQRT) as above.

10 Example

This example finds the basic solutions for the linear least squares problems

$$\text{minimize } \|Ax_i - b_i\|_2, \quad i = 1, 2$$

where b_1 and b_2 are the columns of the matrix B,

$$A = \begin{pmatrix} -0.57 & -1.28 & -0.39 & 0.25 \\ -1.93 & 1.08 & -0.31 & -2.14 \\ 2.30 & 0.24 & 0.40 & -0.35 \\ -1.93 & 0.64 & -0.66 & 0.08 \\ 0.15 & 0.30 & 0.15 & -2.13 \\ -0.02 & 1.03 & -1.43 & 0.50 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} -2.67 & 0.41 \\ -0.55 & -3.10 \\ 3.34 & -4.01 \\ -0.77 & 2.76 \\ 0.48 & -6.17 \\ 4.10 & 0.21 \end{pmatrix}.$$

A QR factorization is performed on the first 4 rows of A using F08ABF (DGEQRT) after which the first 4 rows of B are updated by applying Q^T using F08ACF (DGEMQRT). The remaining row is added by performing a QR update using F08BBF (DTPQRT); B is updated by applying the new Q^T using F08BCF (DTPMQRT); the solution is finally obtained by triangular solve using R from the updated QR.

10.1 Program Text

```
    Program f08bbfe

!     F08BBF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: dgemqrt, dgeqrt, dnrm2, dtpmqrt, dtpqrt, dtrtrs,  &
                             nag_wp, x04caf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter                :: nbmax = 64, nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                           :: i, ifail, info, j, lda, ldb, ldt,    &
                                           lwork, m, n, nb, nrhs
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable   :: a(:,:), b(:,:), c(:,:), rnorm(:),    &
                                           t(:,:), work(:)
!     .. Intrinsic Procedures ..
      Intrinsic                         :: max, min
!     .. Executable Statements ..
      Write (nout,*) 'F08BBF Example Program Results'
      Write (nout,*)
      Flush (nout)
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) m, n, nrhs
      lda = m
```

```
      ldb = m
      nb = min(m,n,nbmax)
      ldt = nb
      lwork = nb*max(n,m)
      Allocate (a(lda,n),b(ldb,nrhs),c(ldb,nrhs),rnorm(nrhs),t(ldt,min(m,      &
        n)),work(lwork))

!     Read A and B from data file

      Read (nin,*)(a(i,1:n),i=1,m)
      Read (nin,*)(b(i,1:nrhs),i=1,m)

      c(1:m,1:nrhs) = b(1:m,1:nrhs)
!     Compute the QR factorization of first n rows of A
!     The NAG name equivalent of dgeqrt is f08abf
      Call dgeqrt(n,n,nb,a,lda,t,ldt,work,info)

!     Compute C = (C1) = (Q**T)*B, storing the result in C
!                 (C2)
!     The NAG name equivalent of dgemqrt is f08acf
      Call dgemqrt('Left','Transpose',n,nrhs,n,nb,a,lda,t,ldt,c,ldb,work,info)

      b(1:n,1:nrhs) = c(1:n,1:nrhs)
!     Compute least squares solutions for first n rows by back-substitution in
!     R*X = C1
!     The NAG name equivalent of dtrtrs is f07tef
      Call dtrtrs('Upper','No transpose','Non-Unit',n,nrhs,a,lda,c,ldb,info)

      If (info>0) Then
        Write (nout,*) 'The upper triangular factor, R, of A is singular, '
        Write (nout,*) 'the least squares solution could not be computed'
      Else

!       Print solution using first n rows

!       ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
        ifail = 0
        Call x04caf('General',' ',n,nrhs,c,ldb,'solution(s) for n rows',ifail)

      End If

!     Now add the remaining rows and perform QR update
!     The NAG name equivalent of dtpqrt is f08bbf
      Call dtpqrt(m-n,n,0,nb,a,lda,a(n+1,1),lda,t,ldt,work,info)

!     Apply orthogonal transformations to C
!     The NAG name equivalent of dtpmqrt is f08bcf
      Call dtpmqrt('Left','Transpose',m-n,nrhs,n,0,nb,a(n+1,1),lda,t,ldt,b,    &
        ldb,b(5,1),ldb,work,info)
!     Compute least squares solutions for first n rows by bac-substitution in
!     R*X = C1
!     The NAG name equivalent of dtrtrs is f07tef
      Call dtrtrs('Upper','No transpose','Non-Unit',n,nrhs,a,lda,b,ldb,info)

      If (info>0) Then
        Write (nout,*) 'The upper triangular factor, R, of A is singular, '
        Write (nout,*) 'the least squares solution could not be computed'
      Else

!       Print least squares solutions
        Write (nout,*)
        ifail = 0
        Call x04caf('G',' ',n,nrhs,b,ldb,                                      &
          'Least squares solution(s) for all rows',ifail)

!       Compute and print estimates of the square roots of the residual
!       sums of squares

!       The NAG name equivalent of dnrm2 is f06ejf
        Do j = 1, nrhs
```

```
          rnorm(j) = dnrm2(m-n,b(n+1,j),1)
        End Do

        Write (nout,*)
        Write (nout,*) 'Square root(s) of the residual sum(s) of squares'
        Write (nout,99999) rnorm(1:nrhs)
      End If

99999 Format (5X,1P,7E11.2)
    End Program f08bbfe
```

10.2 Program Data

```
F08BBF Example Program Data

  6      4      2           : m, n and nrhs

 -0.57  -1.28  -0.39   0.25
 -1.93   1.08  -0.31  -2.14
  2.30   0.24   0.40  -0.35
 -1.93   0.64  -0.66   0.08
  0.15   0.30   0.15  -2.13
 -0.02   1.03  -1.43   0.50 : matrix A

 -2.67   0.41
 -0.55  -3.10
  3.34  -4.01
 -0.77   2.76
  0.48  -6.17
  4.10   0.21               : matrix B
```

10.3 Program Results

```
 F08BBF Example Program Results

 solution(s) for n rows
             1          2
 1      1.5179    -1.5850
 2      1.8629     0.5531
 3     -1.4608     1.3485
 4      0.0398     2.9619

 Least squares solution(s) for all rows
             1          2
 1      1.5339    -1.5753
 2      1.8707     0.5559
 3     -1.5241     1.3119
 4      0.0392     2.9585

 Square root(s) of the residual sum(s) of squares
        2.22E-02   1.38E-02
```