

# NAG Library Routine Document

## F08BHF (DTZRZF)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F08BHF (DTZRZF) reduces the  $m$  by  $n$  ( $m \leq n$ ) real upper trapezoidal matrix  $A$  to upper triangular form by means of orthogonal transformations.

### 2 Specification

```
SUBROUTINE F08BHF (M, N, A, LDA, TAU, WORK, LWORK, INFO)
  INTEGER          M, N, LDA, LWORK, INFO
  REAL (KIND=nag_wp) A(LDA,*), TAU(*), WORK(max(1,LWORK))
```

The routine may be called by its LAPACK name *dtzrzf*.

### 3 Description

The  $m$  by  $n$  ( $m \leq n$ ) real upper trapezoidal matrix  $A$  given by

$$A = \begin{pmatrix} R_1 & R_2 \end{pmatrix},$$

where  $R_1$  is an  $m$  by  $m$  upper triangular matrix and  $R_2$  is an  $m$  by  $(n - m)$  matrix, is factorized as

$$A = \begin{pmatrix} R & 0 \end{pmatrix} Z,$$

where  $R$  is also an  $m$  by  $m$  upper triangular matrix and  $Z$  is an  $n$  by  $n$  orthogonal matrix.

### 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

### 5 Arguments

- 1: M – INTEGER *Input*  
*On entry:*  $m$ , the number of rows of the matrix  $A$ .  
*Constraint:*  $M \geq 0$ .
- 2: N – INTEGER *Input*  
*On entry:*  $n$ , the number of columns of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 3: A(LDA,\*) – REAL (KIND=nag\_wp) array *Input/Output*  
**Note:** the second dimension of the array  $A$  must be at least  $\max(1, N)$ .  
*On entry:* the leading  $m$  by  $n$  upper trapezoidal part of the array  $A$  must contain the matrix to be factorized.  
*On exit:* the leading  $m$  by  $m$  upper triangular part of  $A$  contains the upper triangular matrix  $R$ , and elements  $M + 1$  to  $N$  of the first  $m$  rows of  $A$ , with the array  $TAU$ , represent the orthogonal

matrix  $Z$  as a product of  $m$  elementary reflectors (see Section 3.3.6 in the F08 Chapter Introduction).

- 4: LDA – INTEGER *Input*  
*On entry:* the first dimension of the array A as declared in the (sub)program from which F08BHF (DTZRZF) is called.  
*Constraint:*  $LDA \geq \max(1, M)$ .
- 5: TAU(\*) – REAL (KIND=nag\_wp) array *Output*  
**Note:** the dimension of the array TAU must be at least  $\max(1, M)$ .  
*On exit:* the scalar factors of the elementary reflectors.
- 6: WORK(max(1,LWORK)) – REAL (KIND=nag\_wp) array *Workspace*  
*On exit:* if INFO = 0, WORK(1) contains the minimum value of LWORK required for optimal performance.
- 7: LWORK – INTEGER *Input*  
*On entry:* the dimension of the array WORK as declared in the (sub)program from which F08BHF (DTZRZF) is called.  
 If LWORK = -1, a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.  
*Suggested value:* for optimal performance,  $LWORK \geq M \times nb$ , where  $nb$  is the optimal **block size**.  
*Constraint:*  $LWORK \geq \max(1, M)$  or  $LWORK = -1$ .
- 8: INFO – INTEGER *Output*  
*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If INFO = - $i$ , argument  $i$  had an illegal value. An explanatory message is output, and execution of the program is terminated.

## 7 Accuracy

The computed factorization is the exact factorization of a nearby matrix  $A + E$ , where

$$\|E\|_2 = O\epsilon \|A\|_2$$

and  $\epsilon$  is the *machine precision*.

## 8 Parallelism and Performance

F08BHF (DTZRZF) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The total number of floating-point operations is approximately  $4m^2(n - m)$ .

The complex analogue of this routine is F08BVF (ZTZRZF).

## 10 Example

This example solves the linear least squares problems

$$\min_x \|b_j - Ax_j\|_2, \quad j = 1, 2$$

for the minimum norm solutions  $x_1$  and  $x_2$ , where  $b_j$  is the  $j$ th column of the matrix  $B$ ,

$$A = \begin{pmatrix} -0.09 & 0.14 & -0.46 & 0.68 & 1.29 \\ -1.56 & 0.20 & 0.29 & 1.09 & 0.51 \\ -1.48 & -0.43 & 0.89 & -0.71 & -0.96 \\ -1.09 & 0.84 & 0.77 & 2.11 & -1.27 \\ 0.08 & 0.55 & -1.13 & 0.14 & 1.74 \\ -1.59 & -0.72 & 1.06 & 1.24 & 0.34 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} 7.4 & 2.7 \\ 4.2 & -3.0 \\ -8.3 & -9.6 \\ 1.8 & 1.1 \\ 8.6 & 4.0 \\ 2.1 & -5.7 \end{pmatrix}.$$

The solution is obtained by first obtaining a  $QR$  factorization with column pivoting of the matrix  $A$ , and then the  $RZ$  factorization of the leading  $k$  by  $k$  part of  $R$  is computed, where  $k$  is the estimated rank of  $A$ . A tolerance of 0.01 is used to estimate the rank of  $A$  from the upper triangular factor,  $R$ .

Note that the block size (NB) of 64 assumed in this example is not realistic for such a small problem, but should be suitable for large problems.

### 10.1 Program Text

```

Program f08bhfe

!      F08BHF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: dgeqp3, dnrn2, dormqr, dormrz, dtrsm, dtzrzf,      &
                               nag_wp, x04caf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Real (Kind=nag_wp), Parameter      :: one = 1.0E0_nag_wp
      Real (Kind=nag_wp), Parameter      :: zero = 0.0E0_nag_wp
      Integer, Parameter                  :: incl = 1, nb = 64, nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)                  :: tol
      Integer                              :: i, ifail, info, j, k, lda, ldb,      &
                                          lwork, m, n, nrhs
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable     :: a(:, :), b(:, :), rnorm(:), tau(:),      &
                                          work(:)
      Integer, Allocatable                 :: jpvt(:)
!      .. Intrinsic Procedures ..
      Intrinsic                            :: abs
!      .. Executable Statements ..
      Write (nout,*) 'F08BHF Example Program Results'
      Write (nout,*)
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) m, n, nrhs
      lda = m
      ldb = m
      lwork = 2*n + (n+1)*nb
      Allocate (a(lda,n),b(ldb,nrhs),rnorm(n),tau(n),work(lwork),jpvt(n))

!      Read A and B from data file

```

```

      Read (nin,*)(a(i,1:n),i=1,m)
      Read (nin,*)(b(i,1:nrhs),i=1,m)

!      Initialize JPVT to be zero so that all columns are free
      jpvt(1:n) = 0

!      Compute the QR factorization of A with column pivoting as
!      A = Q*(R11 R12)*(P**T)
!      ( 0 R22)

!      The NAG name equivalent of dgeqp3 is f08bff
      Call dgeqp3(m,n,a,lda,jpvt,tau,work,lwork,info)

!      Compute C = (C1) = (Q**T)*B, storing the result in B
!      (C2)
!      The NAG name equivalent of dormqr is f08agf
      Call dormqr('Left','Transpose',m,nrhs,n,a,lda,tau,b,ldb,work,lwork,info)

!      Choose TOL to reflect the relative accuracy of the input data
      tol = 0.01_nag_wp

!      Determine and print the rank, K, of R relative to TOL
loop: Do k = 1, n
      If (abs(a(k,k))<=tol*abs(a(1,1))) Then
        Exit loop
      End If
    End Do loop
    k = k - 1

    Write (nout,*) 'Tolerance used to estimate the rank of A'
    Write (nout,99999) tol
    Write (nout,*) 'Estimated rank of A'
    Write (nout,99998) k
    Write (nout,*)
    Flush (nout)

!      Compute the RZ factorization of the K by K part of R as
!      (R11 R12) = (T 0)*Z
!      The NAG name equivalent of dtzrzf is f08bhf
      Call dtzrzf(k,n,a,lda,tau,work,lwork,info)

!      Compute least squares solutions of triangular problems by
!      back-substitution in T*Y1 = C1, storing the result in B
!      The NAG name equivalent of dtrsm is f06yjf
      Call dtrsm('Left','Upper','No transpose','Non-Unit',k,nrhs,one,a,lda,b, &
        ldb)

!      Compute estimates of the square roots of the residual sums of
!      squares (2-norm of each of the columns of C2)
!      The NAG name equivalent of dnorm2 is f06ejf
      Do j = 1, nrhs
        rnorm(j) = dnorm2(m-k,b(k+1,j),incl)
      End Do

!      Set the remaining elements of the solutions to zero (to give
!      the minimum-norm solutions), Y2 = 0

      b(k+1:n,1:nrhs) = zero

!      Form W = (Z**T)*Y

!      The NAG name equivalent of dormrz is f08bkf
      Call dormrz('Left','Transpose',n,nrhs,k,n-k,a,lda,tau,b,ldb,work,lwork, &
        info)

!      Permute the least squares solutions stored in B to give X = P*W

```

```

      Do j = 1, nrhs
        work(jpvt(1:n)) = b(1:n,j)
        b(1:n,j) = work(1:n)
      End Do

!      Print least squares solutions

!      ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
      Call x04caf('General',' ',n,nrhs,b,ldb,'Least squares solution(s)',      &
        ifail)

!      Print the square roots of the residual sums of squares

      Write (nout,*)
      Write (nout,*) 'Square root(s) of the residual sum(s) of squares'
      Write (nout,99999) rnorm(1:nrhs)

99999 Format (5X,1P,6E11.2)
99998 Format (1X,I8)
      End Program f08bhfe

```

## 10.2 Program Data

F08BHF Example Program Data

```

  6  5  2                                :Values of M, N and NRHS

-0.09  0.14 -0.46  0.68  1.29
-1.56  0.20  0.29  1.09  0.51
-1.48 -0.43  0.89 -0.71 -0.96
-1.09  0.84  0.77  2.11 -1.27
  0.08  0.55 -1.13  0.14  1.74
-1.59 -0.72  1.06  1.24  0.34 :End of matrix A

  7.4   2.7
  4.2  -3.0
 -8.3  -9.6
  1.8   1.1
  8.6   4.0
  2.1  -5.7                                :End of matrix B

```

## 10.3 Program Results

F08BHF Example Program Results

Tolerance used to estimate the rank of A  
1.00E-02

Estimated rank of A  
4

Least squares solution(s)

	1	2
1	0.6344	3.6258
2	0.9699	1.8284
3	-1.4402	-1.6416
4	3.3678	2.4307
5	3.3992	0.2818

Square root(s) of the residual sum(s) of squares  
2.54E-02 3.65E-02

---