NAG Library Routine Document

F08GCF (DSPEVD)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

Warning. The specification of the arguments LWORK and LIWORK changed at Mark 20 in the case where JOB = 'V' and N > 1: the minimum dimension of the array WORK has been reduced whereas the minimum dimension of the array IWORK has been increased.

1 Purpose

F08GCF (DSPEVD) computes all the eigenvalues and, optionally, all the eigenvectors of a real symmetric matrix held in packed storage. If the eigenvectors are requested, then it uses a divide-and-conquer algorithm to compute eigenvalues and eigenvectors. However, if only eigenvalues are required, then it uses the Pal–Walker–Kahan variant of the QL or QR algorithm.

2 Specification

```
SUBROUTINE F08GCF (JOB, UPLO, N, AP, W, Z, LDZ, WORK, LWORK, IWORK,        &
                   LIWORK, INFO)
INTEGER             N, LDZ, LWORK, IWORK(max(1,LIWORK)), LIWORK, INFO
REAL (KIND=nag_wp) AP(*), W(*), Z(LDZ,*), WORK(max(1,LWORK))
CHARACTER(1)        JOB, UPLO
```

The routine may be called by its LAPACK name ***dspevd***.

3 Description

F08GCF (DSPEVD) computes all the eigenvalues and, optionally, all the eigenvectors of a real symmetric matrix A (held in packed storage). In other words, it can compute the spectral factorization of A as

$$A = Z\Lambda Z^{\mathrm{T}},$$

where Λ is a diagonal matrix whose diagonal elements are the eigenvalues λ_i, and Z is the orthogonal matrix whose columns are the eigenvectors z_i. Thus

$$Az_i = \lambda_i z_i, \quad i = 1, 2, \ldots, n.$$

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Arguments

1: JOB – CHARACTER(1) *Input*

On entry: indicates whether eigenvectors are computed.

JOB = 'N'
Only eigenvalues are computed.

JOB = 'V'
Eigenvalues and eigenvectors are computed.

Constraint: JOB = 'N' or 'V'.

2: UPLO – CHARACTER(1) *Input*

On entry: indicates whether the upper or lower triangular part of A is stored.

UPLO = 'U'
The upper triangular part of A is stored.

UPLO = 'L'
The lower triangular part of A is stored.

Constraint: UPLO = 'U' or 'L'.

3: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

4: AP$(*)$ – REAL (KIND=nag_wp) array *Input/Output*

Note: the dimension of the array AP must be at least $\max(1, \mathrm{N} \times (\mathrm{N}+1)/2)$.

On entry: the upper or lower triangle of the n by n symmetric matrix A, packed by columns.

More precisely,

if UPLO = 'U', the upper triangle of A must be stored with element A_{ij} in AP$(i + j(j-1)/2)$ for $i \leq j$;

if UPLO = 'L', the lower triangle of A must be stored with element A_{ij} in AP$(i + (2n-j)(j-1)/2)$ for $i \geq j$.

On exit: AP is overwritten by the values generated during the reduction to tridiagonal form. The elements of the diagonal and the off-diagonal of the tridiagonal matrix overwrite the corresponding elements of A.

5: W$(*)$ – REAL (KIND=nag_wp) array *Output*

Note: the dimension of the array W must be at least $\max(1, \mathrm{N})$.

On exit: the eigenvalues of the matrix A in ascending order.

6: Z(LDZ, $*$) – REAL (KIND=nag_wp) array *Output*

Note: the second dimension of the array Z must be at least $\max(1, \mathrm{N})$ if JOB = 'V' and at least 1 if JOB = 'N'.

On exit: if JOB = 'V', Z is overwritten by the orthogonal matrix Z which contains the eigenvectors of A.

If JOB = 'N', Z is not referenced.

7: LDZ – INTEGER *Input*

On entry: the first dimension of the array Z as declared in the (sub)program from which F08GCF (DSPEVD) is called.

Constraints:

if JOB = 'V', $\mathrm{LDZ} \geq \max(1, \mathrm{N})$;
if JOB = 'N', $\mathrm{LDZ} \geq 1$.

8: WORK(max(1, LWORK)) – REAL (KIND=nag_wp) array *Workspace*

On exit: if INFO = 0, WORK(1) contains the required minimal size of LWORK.

9: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08GCF (DSPEVD) is called.

If LWORK = −1, a workspace query is assumed; the routine only calculates the minimum dimension of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.

Constraints:

if N ≤ 1, LWORK ≥ 1 or LWORK = −1;
if JOB = 'N' and N > 1, LWORK ≥ 2 × N or LWORK = −1;
if JOB = 'V' and N > 1, LWORK ≥ $N^2 + 6 \times N + 1$ or LWORK = −1.

10: IWORK(max(1, LIWORK)) – INTEGER array *Workspace*

On exit: if INFO = 0, IWORK(1) contains the required minimal size of LIWORK.

11: LIWORK – INTEGER *Input*

On entry: the dimension of the array IWORK as declared in the (sub)program from which F08GCF (DSPEVD) is called.

If LIWORK = −1, a workspace query is assumed; the routine only calculates the minimum dimension of the IWORK array, returns this value as the first entry of the IWORK array, and no error message related to LIWORK is issued.

Constraints:

if JOB = 'N' or N ≤ 1, LIWORK ≥ 1 or LIWORK = −1;
if JOB = 'V' and N > 1, LIWORK ≥ 5 × N + 3 or LIWORK = −1.

12: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

if INFO = i and JOB = 'N', the algorithm failed to converge; i elements of an intermediate tridiagonal form did not converge to zero; if INFO = i and JOB = 'V', then the algorithm failed to compute an eigenvalue while working on the submatrix lying in rows and column $i/(N + 1)$ through i mod $(N + 1)$.

7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix $(A + E)$, where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and ϵ is the ***machine precision***. See Section 4.7 of Anderson *et al.* (1999) for further details.

8 Parallelism and Performance

F08GCF (DSPEVD) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F08GCF (DSPEVD) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The complex analogue of this routine is F08GQF (ZHPEVD).

10 Example

This example computes all the eigenvalues and eigenvectors of the symmetric matrix A, where

$$A = \begin{pmatrix} 1.0 & 2.0 & 3.0 & 4.0 \\ 2.0 & 2.0 & 3.0 & 4.0 \\ 3.0 & 3.0 & 3.0 & 4.0 \\ 4.0 & 4.0 & 4.0 & 4.0 \end{pmatrix}.$$

10.1 Program Text

```
    Program f08gcfe

!     F08GCF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: blas_damax_val, dspevd, nag_wp, x04caf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Real (Kind=nag_wp), Parameter    :: zero = 0.0E0_nag_wp
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: r
      Integer                          :: i, ifail, info, j, k, ldz, liwork,   &
                                          lwork, n
      Character (1)                    :: job, uplo
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: ap(:), w(:), work(:), z(:,:)
      Integer, Allocatable             :: iwork(:)
!     .. Executable Statements ..
      Write (nout,*) 'F08GCF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n
      ldz = n
      liwork = 5*n + 3
      lwork = n*n + 6*n + 1
      Allocate (ap(n*(n+1)/2),w(n),work(lwork),z(ldz,n),iwork(liwork))

!     Read A from data file

      Read (nin,*) uplo
      If (uplo=='U') Then
        Read (nin,*)((ap(i+j*(j-1)/2),j=i,n),i=1,n)
      Else If (uplo=='L') Then
        Read (nin,*)((ap(i+(2*n-j)*(j-1)/2),j=1,i),i=1,n)
```

```
      End If

      Read (nin,*) job

!     Calculate all the eigenvalues and eigenvectors of A
!     The NAG name equivalent of dspevd is f08gcf
      Call dspevd(job,uplo,n,ap,w,z,ldz,work,lwork,iwork,liwork,info)

      Write (nout,*)
      If (info>0) Then
        Write (nout,*) 'Failure to converge.'
      Else

!       Print eigenvalues and eigenvectors

        Write (nout,*) 'Eigenvalues'
        Write (nout,99999) w(1:n)
        Write (nout,*)
        Flush (nout)

!       Normalize the eigenvectors: largest element positive
        Do i = 1, n
          Call blas_damax_val(n,z(1,i),1,k,r)
          If (z(k,i)<zero) Then
            z(1:n,i) = -z(1:n,i)
          End If
        End Do

!       ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
        ifail = 0
        Call x04caf('General',' ',n,n,z,ldz,'Eigenvectors',ifail)

      End If

99999 Format (3X,(8F8.4))
    End Program f08gcfe
```

10.2 Program Data

```
F08GCF Example Program Data
  4                          :Value of N
  'L'                        :Value of UPLO
  1.0
  2.0   2.0
  3.0   3.0   3.0
  4.0   4.0   4.0   4.0      :End of matrix A
  'V'                        :Value of JOB
```

10.3 Program Results

```
 F08GCF Example Program Results

 Eigenvalues
    -2.0531 -0.5146 -0.2943 12.8621

 Eigenvectors
          1       2       3       4
 1   0.7003 -0.5144 -0.2767  0.4103
 2   0.3592  0.4851  0.6634  0.4422
 3  -0.1569  0.5420 -0.6504  0.5085
 4  -0.5965 -0.4543  0.2457  0.6144
```