

NAG Library Routine Document

F08GEF (DSPTRD)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F08GEF (DSPTRD) reduces a real symmetric matrix to tridiagonal form, using packed storage.

2 Specification

```
SUBROUTINE F08GEF (UPLO, N, AP, D, E, TAU, INFO)
  INTEGER          N, INFO
  REAL (KIND=nag_wp) AP(*), D(N), E(N-1), TAU(N-1)
  CHARACTER(1)     UPLO
```

The routine may be called by its LAPACK name *dsptrd*.

3 Description

F08GEF (DSPTRD) reduces a real symmetric matrix A , held in packed storage, to symmetric tridiagonal form T by an orthogonal similarity transformation: $A = QTQ^T$.

The matrix Q is not formed explicitly but is represented as a product of $n - 1$ elementary reflectors (see the F08 Chapter Introduction for details). Routines are provided to work with Q in this representation (see Section 9).

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Arguments

- 1: UPLO – CHARACTER(1) *Input*
On entry: indicates whether the upper or lower triangular part of A is stored.
 UPLO = 'U'
 The upper triangular part of A is stored.
 UPLO = 'L'
 The lower triangular part of A is stored.
Constraint: UPLO = 'U' or 'L'.
- 2: N – INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.
- 3: AP(*) – REAL (KIND=nag_wp) array *Input/Output*
Note: the dimension of the array AP must be at least $\max(1, N \times (N + 1)/2)$.
On entry: the upper or lower triangle of the n by n symmetric matrix A , packed by columns.
 More precisely,

if $UPLO = 'U'$, the upper triangle of A must be stored with element A_{ij} in $AP(i + j(j - 1)/2)$ for $i \leq j$;

if $UPLO = 'L'$, the lower triangle of A must be stored with element A_{ij} in $AP(i + (2n - j)(j - 1)/2)$ for $i \geq j$.

On exit: AP is overwritten by the tridiagonal matrix T and details of the orthogonal matrix Q .

4: $D(N)$ – REAL (KIND=nag_wp) array *Output*

On exit: the diagonal elements of the tridiagonal matrix T .

5: $E(N - 1)$ – REAL (KIND=nag_wp) array *Output*

On exit: the off-diagonal elements of the tridiagonal matrix T .

6: $TAU(N - 1)$ – REAL (KIND=nag_wp) array *Output*

On exit: further details of the orthogonal matrix Q .

7: $INFO$ – INTEGER *Output*

On exit: $INFO = 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

$INFO < 0$

If $INFO = -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed tridiagonal matrix T is exactly similar to a nearby matrix $(A + E)$, where

$$\|E\|_2 \leq c(n)\epsilon\|A\|_2,$$

$c(n)$ is a modestly increasing function of n , and ϵ is the *machine precision*.

The elements of T themselves may be sensitive to small perturbations in A or to rounding errors in the computation, but this does not affect the stability of the eigenvalues and eigenvectors.

8 Parallelism and Performance

F08GEF (DSPTRD) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations is approximately $\frac{4}{3}n^3$.

To form the orthogonal matrix Q F08GEF (DSPTRD) may be followed by a call to F08GFF (DOPGTR):

```
CALL DOPGTR(UPLO,N,AP,TAU,Q,LDQ,WORK,INFO)
```

To apply Q to an n by p real matrix C F08GEF (DSPTRD) may be followed by a call to F08GGF (DOPMTR). For example,

```
CALL DOPMTR('Left',UPLO,'No Transpose',N,P,AP,TAU,C,LDC,WORK, &
           INFO)
```

forms the matrix product QC .

The complex analogue of this routine is F08GSF (ZHPTRD).

10 Example

This example reduces the matrix A to tridiagonal form, where

$$A = \begin{pmatrix} 2.07 & 3.87 & 4.20 & -1.15 \\ 3.87 & -0.21 & 1.87 & 0.63 \\ 4.20 & 1.87 & 1.15 & 2.06 \\ -1.15 & 0.63 & 2.06 & -1.81 \end{pmatrix},$$

using packed storage.

10.1 Program Text

```
Program f08gefe

!      F08GEF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: dsptd, nag_wp
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Integer                     :: i, info, j, n
      Character (1)               :: uplo
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: ap(:), d(:), e(:), tau(:)
!      .. Intrinsic Procedures ..
      Intrinsic                   :: abs
!      .. Executable Statements ..
      Write (nout,*) 'F08GEF Example Program Results'
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n

      Allocate (ap(n*(n+1)/2),d(n),e(n-1),tau(n-1))

!      Read A from data file and copy A into AW

      Read (nin,*) uplo
      If (uplo=='U') Then
         Read (nin,*)((ap(i+j*(j-1)/2),j=i,n),i=1,n)
      Else If (uplo=='L') Then
         Read (nin,*)((ap(i+(2*n-j)*(j-1)/2),j=1,i),i=1,n)
      End If

!      Reduce A to tridiagonal form
!      The NAG name equivalent of dsptd is f08gef
      Call dsptd(uplo,n,ap,d,e,tau,info)

      If (info==0) Then
!      Print the diagonal and off-diagonal of tridiagonal T.
!      The absolute value of E is printed since this can vary by a change of
!      sign (corresponding to multiplying through a column of Q by -1).

         Write (nout,*)
         Write (nout,*)
         'Diagonal and off-diagonal elements of tridiagonal form'
         Write (nout,*)

```

```

      Write (nout,99999) 'i', 'D', 'E'
      Do i = 1, n - 1
        Write (nout,99998) i, d(i), abs(e(i))
      End Do
      Write (nout,99998) n, d(n)

      Else
        Write (nout,99997) info
      End If

99999 Format (5X,A,9X,A,12X,A)
99998 Format (1X,I5,2(1X,F12.5))
99997 Format (1X,'** DSPTRD/F08GEF retuned with INFO = ',I10)

      End Program f08gefe

```

10.2 Program Data

F08GEF Example Program Data

```

4                               :Value of N
'L'                             :Value of UPLO
2.07
3.87 -0.21
4.20  1.87  1.15
-1.15  0.63  2.06 -1.81      :End of matrix A

```

10.3 Program Results

F08GEF Example Program Results

Diagonal and off-diagonal elements of tridiagonal form

i	D	E
1	2.07000	5.82575
2	1.47409	2.62405
3	-0.64916	0.91627
4	-1.69493	
