# NAG Library Routine Document

# F08GSF (ZHPTRD)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

## 1 Purpose

F08GSF (ZHPTRD) reduces a complex Hermitian matrix to tridiagonal form, using packed storage.

## 2 Specification

```
SUBROUTINE F08GSF (UPLO, N, AP, D, E, TAU, INFO)
INTEGER                 N, INFO
REAL (KIND=nag_wp)      D(N), E(N-1)
COMPLEX (KIND=nag_wp) AP(*), TAU(N-1)
CHARACTER(1)            UPLO
```

The routine may be called by its LAPACK name ***zhptrd***.

## 3 Description

F08GSF (ZHPTRD) reduces a complex Hermitian matrix $A$, held in packed storage, to real symmetric tridiagonal form $T$ by a unitary similarity transformation: $A = QTQ^{\mathrm{H}}$.

The matrix $Q$ is not formed explicitly but is represented as a product of $n-1$ elementary reflectors (see the F08 Chapter Introduction for details). Routines are provided to work with $Q$ in this representation (see Section 9).

## 4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

## 5 Arguments

1: UPLO – CHARACTER(1) *Input*

*On entry*: indicates whether the upper or lower triangular part of $A$ is stored.

UPLO = 'U'
The upper triangular part of $A$ is stored.

UPLO = 'L'
The lower triangular part of $A$ is stored.

*Constraint*: UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

*On entry*: $n$, the order of the matrix $A$.

*Constraint*: $\mathrm{N} \geq 0$.

3: AP($*$) – COMPLEX (KIND=nag_wp) array *Input/Output*

**Note**: the dimension of the array AP must be at least $\max(1, \mathrm{N} \times (\mathrm{N}+1)/2)$.

*On entry*: the upper or lower triangle of the $n$ by $n$ Hermitian matrix $A$, packed by columns.

More precisely,

if UPLO = 'U', the upper triangle of $A$ must be stored with element $A_{ij}$ in AP$(i+j(j-1)/2)$ for $i \leq j$;

if UPLO = 'L', the lower triangle of $A$ must be stored with element $A_{ij}$ in AP$(i+(2n-j)(j-1)/2)$ for $i \geq j$.

*On exit*: AP is overwritten by the tridiagonal matrix $T$ and details of the unitary matrix $Q$.

4: D(N) – REAL (KIND=nag_wp) array *Output*

*On exit*: the diagonal elements of the tridiagonal matrix $T$.

5: E(N − 1) – REAL (KIND=nag_wp) array *Output*

*On exit*: the off-diagonal elements of the tridiagonal matrix $T$.

6: TAU(N − 1) – COMPLEX (KIND=nag_wp) array *Output*

*On exit*: further details of the unitary matrix $Q$.

7: INFO – INTEGER *Output*

*On exit*: INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, argument $i$ had an illegal value. An explanatory message is output, and execution of the program is terminated.

## 7 Accuracy

The computed tridiagonal matrix $T$ is exactly similar to a nearby matrix $(A+E)$, where

$$\|E\|_2 \leq c(n)\epsilon\|A\|_2,$$

$c(n)$ is a modestly increasing function of $n$, and $\epsilon$ is the ***machine precision***.

The elements of $T$ themselves may be sensitive to small perturbations in $A$ or to rounding errors in the computation, but this does not affect the stability of the eigenvalues and eigenvectors.

## 8 Parallelism and Performance

F08GSF (ZHPTRD) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The total number of real floating-point operations is approximately $\frac{16}{3}n^3$.

To form the unitary matrix $Q$ F08GSF (ZHPTRD) may be followed by a call to F08GTF (ZUPGTR):

```
CALL ZUPGTR(UPLO,N,AP,TAU,Q,LDQ,WORK,INFO)
```

To apply $Q$ to an $n$ by $p$ complex matrix $C$ F08GSF (ZHPTRD) may be followed by a call to F08GUF (ZUPMTR). For example,

```
      CALL ZUPMTR('Left',UPLO,'No Transpose',N,P,AP,TAU,C,LDC,WORK, &
                  INFO)
```

forms the matrix product $QC$.

The real analogue of this routine is F08GEF (DSPTRD).

## 10 Example

This example reduces the matrix $A$ to tridiagonal form, where

$$A = \begin{pmatrix} -2.28+0.00i & 1.78-2.03i & 2.26+0.10i & -0.12+2.53i \\ 1.78+2.03i & -1.12+0.00i & 0.01+0.43i & -1.07+0.86i \\ 2.26-0.10i & 0.01-0.43i & -0.37+0.00i & 2.31-0.92i \\ -0.12-2.53i & -1.07-0.86i & 2.31+0.92i & -0.73+0.00i \end{pmatrix},$$

using packed storage.

### 10.1 Program Text

```
    Program f08gsfe

!     F08GSF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: nag_wp, zhptrd
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: i, info, j, n
      Character (1)                    :: uplo
!     .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: ap(:), tau(:)
      Real (Kind=nag_wp), Allocatable  :: d(:), e(:)
!     .. Intrinsic Procedures ..
      Intrinsic                        :: abs
!     .. Executable Statements ..
      Write (nout,*) 'F08GSF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n

      Allocate (ap(n*(n+1)/2),tau(n-1),d(n),e(n-1))

!     Read A from data file and copy a into AW

      Read (nin,*) uplo
      If (uplo=='U') Then
        Read (nin,*)((ap(i+j*(j-1)/2),j=i,n),i=1,n)
      Else If (uplo=='L') Then
        Read (nin,*)((ap(i+(2*n-j)*(j-1)/2),j=1,i),i=1,n)
      End If

!     Reduce A to tridiagonal form
!     The NAG name equivalent of zhptrd is f08gsf
      Call zhptrd(uplo,n,ap,d,e,tau,info)

      If (info==0) Then
!       Print the diagonal and off-diagonal of tridiagonal T.
!       The absolute value of E is printed since this can vary by a change of
!       sign (corresponding to multiplying through a column of Q by -1).

        Write (nout,*)
        Write (nout,*)                                                    &
          'Diagonal and off-diagonal elements of tridiagonal form'
```

```
        Write (nout,*)
        Write (nout,99999) 'i', 'D', 'E'
        Do i = 1, n - 1
          Write (nout,99998) i, d(i), abs(e(i))
        End Do
        Write (nout,99998) n, d(n)

      Else
        Write (nout,99997) info
      End If

99999 Format (5X,A,9X,A,12X,A)
99998 Format (1X,I5,2(1X,F12.5))
99997 Format (1X,'** ZHPTRD/F08GSF retuned with INFO = ',I10)

    End Program f08gsfe
```

## 10.2 Program Data

```
F08GSF Example Program Data
  4                                                              :Value of N
  'L'                                                            :Value of UPLO
 (-2.28, 0.00)
 ( 1.78, 2.03) (-1.12, 0.00)
 ( 2.26,-0.10) ( 0.01,-0.43) (-0.37, 0.00)
 (-0.12,-2.53) (-1.07,-0.86) ( 2.31, 0.92) (-0.73, 0.00)   :End of matrix A
```

## 10.3 Program Results

```
 F08GSF Example Program Results

 Diagonal and off-diagonal elements of tridiagonal form

     i         D            E
     1     -2.28000      4.33846
     2     -0.12846      2.02259
     3     -0.16659      1.80232
     4     -1.92495
```