

NAG Library Routine Document

F08HSF (ZHBTRD)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F08HSF (ZHBTRD) reduces a complex Hermitian band matrix to tridiagonal form.

2 Specification

```
SUBROUTINE F08HSF (VECT, UPLO, N, KD, AB, LDAB, D, E, Q, LDQ, WORK,      &
                  INFO)
INTEGER                N, KD, LDAB, LDQ, INFO
REAL (KIND=nag_wp)    D(N), E(N-1)
COMPLEX (KIND=nag_wp) AB(LDAB,*), Q(LDQ,*), WORK(N)
CHARACTER(1)          VECT, UPLO
```

The routine may be called by its LAPACK name *zhbtrd*.

3 Description

F08HSF (ZHBTRD) reduces a Hermitian band matrix A to real symmetric tridiagonal form T by a unitary similarity transformation:

$$T = Q^H A Q.$$

The unitary matrix Q is determined as a product of Givens rotation matrices, and may be formed explicitly by the routine if required.

The routine uses a vectorizable form of the reduction, due to Kaufman (1984).

4 References

Kaufman L (1984) Banded eigenvalue solvers on vector machines *ACM Trans. Math. Software* **10** 73–86

Parlett B N (1998) *The Symmetric Eigenvalue Problem* SIAM, Philadelphia

5 Arguments

- 1: VECT – CHARACTER(1) *Input*
- On entry:* indicates whether Q is to be returned.
- VECT = 'V'
 Q is returned.
- VECT = 'U'
 Q is updated (and the array Q must contain a matrix on entry).
- VECT = 'N'
 Q is not required.
- Constraint:* VECT = 'V', 'U' or 'N'.

- 2: UPLO – CHARACTER(1) *Input*
On entry: indicates whether the upper or lower triangular part of A is stored.
UPLO = 'U'
The upper triangular part of A is stored.
UPLO = 'L'
The lower triangular part of A is stored.
Constraint: UPLO = 'U' or 'L'.
- 3: N – INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.
- 4: KD – INTEGER *Input*
On entry: if UPLO = 'U', the number of superdiagonals, k_d , of the matrix A .
If UPLO = 'L', the number of subdiagonals, k_d , of the matrix A .
Constraint: $KD \geq 0$.
- 5: AB(LDAB,*) – COMPLEX (KIND=nag_wp) array *Input/Output*
Note: the second dimension of the array AB must be at least $\max(1, N)$.
On entry: the upper or lower triangle of the n by n Hermitian band matrix A .
The matrix is stored in rows 1 to $k_d + 1$, more precisely,
if UPLO = 'U', the elements of the upper triangle of A within the band must be stored with element A_{ij} in $AB(k_d + 1 + i - j, j)$ for $\max(1, j - k_d) \leq i \leq j$;
if UPLO = 'L', the elements of the lower triangle of A within the band must be stored with element A_{ij} in $AB(1 + i - j, j)$ for $j \leq i \leq \min(n, j + k_d)$.
On exit: AB is overwritten by values generated during the reduction to tridiagonal form.
The first superdiagonal or subdiagonal and the diagonal of the tridiagonal matrix T are returned in AB using the same storage format as described above.
- 6: LDAB – INTEGER *Input*
On entry: the first dimension of the array AB as declared in the (sub)program from which F08HSF (ZHBTRD) is called.
Constraint: $LDAB \geq \max(1, KD + 1)$.
- 7: D(N) – REAL (KIND=nag_wp) array *Output*
On exit: the diagonal elements of the tridiagonal matrix T .
- 8: E(N - 1) – REAL (KIND=nag_wp) array *Output*
On exit: the off-diagonal elements of the tridiagonal matrix T .
- 9: Q(LDQ,*) – COMPLEX (KIND=nag_wp) array *Input/Output*
Note: the second dimension of the array Q must be at least $\max(1, N)$ if VECT = 'V' or 'U' and at least 1 if VECT = 'N'.
On entry: if VECT = 'U', Q must contain the matrix formed in a previous stage of the reduction (for example, the reduction of a banded Hermitian-definite generalized eigenproblem); otherwise Q need not be set.

On exit: if VECT = 'V' or 'U', the n by n matrix Q .

If VECT = 'N', Q is not referenced.

10: LDQ – INTEGER

Input

On entry: the first dimension of the array Q as declared in the (sub)program from which F08HSF (ZHBTRD) is called.

Constraints:

if VECT = 'V' or 'U', $LDQ \geq \max(1, N)$;
if VECT = 'N', $LDQ \geq 1$.

11: WORK(N) – COMPLEX (KIND=nag_wp) array

Workspace

12: INFO – INTEGER

Output

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed tridiagonal matrix T is exactly similar to a nearby matrix $(A + E)$, where

$$\|E\|_2 \leq c(n)\epsilon\|A\|_2,$$

$c(n)$ is a modestly increasing function of n , and ϵ is the *machine precision*.

The elements of T themselves may be sensitive to small perturbations in A or to rounding errors in the computation, but this does not affect the stability of the eigenvalues and eigenvectors.

The computed matrix Q differs from an exactly unitary matrix by a matrix E such that

$$\|E\|_2 = O(\epsilon),$$

where ϵ is the *machine precision*.

8 Parallelism and Performance

F08HSF (ZHBTRD) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F08HSF (ZHBTRD) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of real floating-point operations is approximately $20n^2k$ if VECT = 'N' with $10n^3(k-1)/k$ additional operations if VECT = 'V'.

The real analogue of this routine is F08HEF (DSBTRD).

10 Example

This example computes all the eigenvalues and eigenvectors of the matrix A , where

$$A = \begin{pmatrix} -3.13 + 0.00i & 1.94 - 2.10i & -3.40 + 0.25i & 0.00 + 0.00i \\ 1.94 + 2.10i & -1.91 + 0.00i & -0.82 - 0.89i & -0.67 + 0.34i \\ -3.40 - 0.25i & -0.82 + 0.89i & -2.87 + 0.00i & -2.10 - 0.16i \\ 0.00 + 0.00i & -0.67 - 0.34i & -2.10 + 0.16i & 0.50 + 0.00i \end{pmatrix}.$$

Here A is Hermitian and is treated as a band matrix. The program first calls F08HSF (ZHBTRD) to reduce A to tridiagonal form T , and to form the unitary matrix Q ; the results are then passed to F08JSF (ZSTEQR) which computes the eigenvalues and eigenvectors of A .

10.1 Program Text

Program f08hsfe

```
!      F08HSF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: dznrm2, nag_wp, x04dbf, zhbtrd, zsteqr
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Complex (Kind=nag_wp)      :: scal
      Integer                    :: i, ifail, info, j, k, kd, ldab, ldq, &
                                   n
      Character (1)              :: uplo
!      .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: ab(:,,:), q(:,,:), work(:)
      Real (Kind=nag_wp), Allocatable   :: d(:), e(:), rwork(:)
      Character (1)                    :: clabs(1), rlabs(1)
!      .. Intrinsic Procedures ..
      Intrinsic                      :: abs, conjg, max, maxloc, min
!      .. Executable Statements ..
      Write (nout,*) 'F08HSF Example Program Results'
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, kd
      ldab = kd + 1
      ldq = n
      Allocate (ab(ldab,n),q(ldq,n),work(n),d(n),e(n-1),rwork(2*n-2))

!      Read A from data file

      Read (nin,*) uplo
      If (uplo=='U') Then
        Do i = 1, n
          Read (nin,*)(ab(kd+1+i-j,j),j=i,min(n,i+kd))
        End Do
      Else If (uplo=='L') Then
        Do i = 1, n
          Read (nin,*)(ab(1+i-j,j),j=max(1,i-kd),i)
        End Do
      End If

!      Reduce A to tridiagonal form T = (Q**H)*A*Q (and form Q)
!      The NAG name equivalent of zhbtrd is f08hsf
      Call zhbtrd('V',uplo,n,kd,ab,ldab,d,e,q,ldq,work,info)

!      Calculate all the eigenvalues and eigenvectors of A
!      The NAG name equivalent of zsteqr is f08jsf
      Call zsteqr('V',n,d,e,q,ldq,rwork,info)

      Write (nout,*)
```

```

      If (info>0) Then
        Write (nout,*) 'Failure to converge.'
      Else

!       Print eigenvalues and eigenvectors

        Write (nout,*) 'Eigenvalues'
        Write (nout,99999) d(1:n)
        Write (nout,*)
        Flush (nout)

!       Normalize the eigenvectors, largest element real
        Do i = 1, n
          rwork(1:n) = abs(q(1:n,i))
          k = maxloc(rwork(1:n),1)
          scal = conjg(q(k,i))/abs(q(k,i))/dznrm2(n,q(1,i),1)
          q(1:n,i) = q(1:n,i)*scal
        End Do

!       ifail: behaviour on error exit
!       =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
        ifail = 0
        Call x04dbf('General',' ',n,n,q,ldq,'Bracketed','F7.4','Eigenvectors', &
          'Integer',rlabs,'Integer',clabs,80,0,ifail)

      End If

99999 Format (8X,4(F7.4,11X,:))
      End Program f08hsfe

```

10.2 Program Data

F08HSF Example Program Data

```

  4  2                                     :Values of N and KD
  'L'                                     :Value of UPLO
(-3.13, 0.00)
( 1.94, 2.10) (-1.91, 0.00)
(-3.40,-0.25) (-0.82, 0.89) (-2.87, 0.00)
              (-0.67,-0.34) (-2.10, 0.16) ( 0.50, 0.00) :End of matrix A

```

10.3 Program Results

F08HSF Example Program Results

```

Eigenvalues
      -7.0042           -4.0038           0.5968           3.0012

```

```

Eigenvectors
              1              2              3              4
1 ( 0.7293, 0.0000) (-0.2128, 0.1511) (-0.3354,-0.1604) (-0.5114,-0.0163)
2 (-0.1654,-0.2046) ( 0.7316, 0.0000) (-0.2804,-0.3413) (-0.2374,-0.3796)
3 ( 0.6081, 0.0301) ( 0.3910,-0.3843) (-0.0144, 0.1532) ( 0.5523,-0.0000)
4 ( 0.1653,-0.0303) ( 0.2775,-0.1378) ( 0.8019, 0.0000) (-0.4517, 0.1693)

```
