

NAG Library Routine Document

F08KZF (ZGESVDX)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08KZF (ZGESVDX) computes the singular value decomposition (SVD) of a complex m by n matrix A , optionally computing the left and/or right singular vectors. All singular values or a selected set of singular values may be computed.

2 Specification

```
SUBROUTINE F08KZF (JOBV, JOBV, RANGE, M, N, A, LDA, VL, VU, IL, IU, NS,      &
                  S, U, LDU, VT, LDVT, WORK, LWORK, RWORK, IWORK, INFO)
INTEGER          M, N, LDA, IL, IU, NS, LDU, LDVT, LWORK,      &
                  IWORK(12*min(N,M)), INFO
REAL (KIND=nag_wp) VL, VU, S(min(M,N)), RWORK(*)
COMPLEX (KIND=nag_wp) A(LDA,*), U(LDU,*), VT(LDVT,*),      &
                  WORK(max(1,LWORK))
CHARACTER(1)     JOBV, JOBV, RANGE
```

The routine may be called by its LAPACK name ***zgesvdx***.

3 Description

The SVD is written as

$$A = U\Sigma V^H,$$

where Σ is an m by n matrix which is zero except for its $\min(m, n)$ diagonal elements, U is an m by m unitary matrix, and V is an n by n unitary matrix. The diagonal elements of Σ are the singular values of A ; they are complex and non-negative, and are returned in descending order. The first $\min(m, n)$ columns of U and V are the left and right singular vectors of A .

Note that the routine returns V^H , not V .

Alternative to computing all singular values of A , a selected set can be computed. The set is either those singular values lying in a given interval, $\sigma \in (v_l, v_u]$, or those whose index (counting from largest to smallest in magnitude) lies in a given range $1 \leq i_l, \dots, i_u \leq n$. In these cases, the corresponding left and right singular vectors can optionally be computed.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Arguments

- 1: JOBV – CHARACTER(1) *Input*
On entry: specifies options for computing all or part of the matrix U .
 JOBV = 'V'
 The NS columns of U , as specified by RANGE, are returned in array U.

- JOBV = 'N'
No columns of U (no left singular vectors) are computed.
Constraint: JOBV = 'V' or 'N'.
- 2: JOBVT – CHARACTER(1) *Input*
On entry: specifies options for computing all or part of the matrix V^T .
JOBVT = 'V'
 The NS rows of V^T , as specified by RANGE, are returned in the array VT.
JOBVT = 'N'
 No rows of V^T (no right singular vectors) are computed.
Constraint: JOBVT = 'V' or 'N'.
- 3: RANGE – CHARACTER(1) *Input*
On entry: indicates which singular values should be returned.
RANGE = 'A'
 All singular values will be found.
RANGE = 'V'
 All singular values in the half-open interval $(VL, VU]$ will be found.
RANGE = 'I'
 The ILth through IUth singular values will be found.
Constraint: RANGE = 'A', 'V' or 'I'.
- 4: M – INTEGER *Input*
On entry: m , the number of rows of the matrix A .
Constraint: $M \geq 0$.
- 5: N – INTEGER *Input*
On entry: n , the number of columns of the matrix A .
Constraint: $N \geq 0$.
- 6: A(LDA,*) – COMPLEX (KIND=nag_wp) array *Input/Output*
Note: the second dimension of the array A must be at least $\max(1, N)$.
On entry: the m by n matrix A .
On exit: if $JOBV \neq 'N'$ and $JOBVT \neq 'N'$, the contents of A are destroyed.
- 7: LDA – INTEGER *Input*
On entry: the first dimension of the array A as declared in the (sub)program from which F08KZF (ZGESV DX) is called.
Constraint: $LDA \geq \max(1, M)$.
- 8: VL – REAL (KIND=nag_wp) *Input*
On entry: if RANGE = 'V', the lower bounds of the interval to be searched for singular values.
If RANGE = 'A' or 'I', VL is not referenced.
Constraint: if RANGE = 'V', $0.0 \leq VL$.

- 9: VU – REAL (KIND=nag_wp) *Input*
On entry: if RANGE = 'V', the upper bounds of the interval to be searched for singular values.
 If RANGE = 'A' or 'I', VU is not referenced.
Constraint: if RANGE = 'V', VL < VU.
- 10: IL – INTEGER *Input*
 11: IU – INTEGER *Input*
On entry: if RANGE = 'I', the indices (in ascending order) of the smallest and largest singular values to be returned.
 If RANGE = 'A' or 'V', IL and IU are not referenced.
Constraints:
 if RANGE = 'I' and $\min(M, N) = 0$, IL = 1 and IU = 0;
 if RANGE = 'I' and $\min(M, N) > 0$, $1 \leq IL \leq IU \leq \min(M, N)$.
- 12: NS – INTEGER *Output*
On exit: the total number of singular values found. $0 \leq NS \leq \min(M, N)$.
 If RANGE = 'A', NS = $\min(M, N)$.
 If RANGE = 'I', NS = IU – IL + 1.
- 13: S(min(M,N)) – REAL (KIND=nag_wp) array *Output*
On exit: the singular values of A, sorted so that $S(i) \geq S(i + 1)$.
- 14: U(LDU,*) – COMPLEX (KIND=nag_wp) array *Output*
Note: the second dimension of the array U must be at least $\max(1, NS)$ if JOBU = 'V', and at least 1 otherwise.
On exit: m by NS.
 If JOBU = 'V', U contains the first $\min(m, n)$ NS columns of U (the left singular vectors, stored column-wise).
 If JOBU = 'N', U is not referenced.
 If RANGE = 'V' then the value of NS is not known in advance and so an upper limit should be used, e.g., $\min(M, N)$.
- 15: LDU – INTEGER *Input*
On entry: the first dimension of the array U as declared in the (sub)program from which F08KZF (ZGESVDX) is called.
Constraints:
 if JOBU = 'V', $LDU \geq \max(1, M)$;
 otherwise $LDU \geq 1$.
- 16: VT(LDVT,*) – COMPLEX (KIND=nag_wp) array *Output*
Note: the second dimension of the array VT must be at least $\max(1, N)$ if JOBVT = 'V', and at least 1 otherwise.
On exit: if JOBVT = 'V', VT contains the first NS rows of V^H (the right singular vectors, stored row-wise).
 If JOBVT = 'N', VT is not referenced.

17: LDVT – INTEGER *Input*

Note: If JOBVT = 'V' and RANGE = 'V' then the value of NS is not known in advance and so an upper limit should be used, e.g., $\min(M, N)$.

On entry: the first dimension of the array VT as declared in the (sub)program from which F08KZF (ZGESV DX) is called.

Constraints:

if JOBVT = 'V', $LDVT \geq \max(1, NS)$;
otherwise $LDVT \geq 1$.

18: WORK(max(1, LWORK)) – COMPLEX (KIND=nag_wp) array *Workspace*

On exit: if INFO = 0, WORK(1) returns the optimal LWORK.

If INFO > 0, WORK(2 : min(M, N)) contains the unconverged superdiagonal elements of an upper bidiagonal matrix B whose diagonal is in S (not necessarily sorted). B satisfies $A = UBV^H$, so it has the same singular values as A , and singular vectors related by U and V^H .

19: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08KZF (ZGESV DX) is called.

If LWORK = -1, a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.

Suggested value: for optimal performance, LWORK should generally be larger. Consider increasing LWORK by at least $nb \times \min(M, N)$, where nb is the optimal **block size**.

Constraint: $LWORK \geq \max(1, \min(M, N) \times (\min(M, N) + 5))$.

20: RWORK(*) – REAL (KIND=nag_wp) array *Workspace*

Note: the dimension of the array RWORK must be at least $\min(M, N) \times (2 \times \min(M, N) + 17)$.

21: IWORK(12 × min(N, M)) – INTEGER array *Workspace*

On exit:

if INFO = 0, the first NS elements of IWORK are zero;

if INFO > 0, IWORK contains the indices of the eigenvectors that failed to converge in F08JBF (DSTEVX) and F08MBF (DBDSV DX).

22: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = - i , argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If F08KZF (ZGESV DX) did not converge, INFO specifies how many superdiagonals of an intermediate bidiagonal form did not converge to zero.

7 Accuracy

The computed singular value decomposition is nearly the exact singular value decomposition for a nearby matrix $(A + E)$, where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and ϵ is the *machine precision*. In addition, the computed singular vectors are nearly unitary to working precision. See Section 4.9 of Anderson *et al.* (1999) for further details.

8 Parallelism and Performance

F08KZF (ZGESVDX) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F08KZF (ZGESVDX) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations is approximately proportional to mn^2 when $m > n$ and m^2n otherwise.

The singular values are returned in descending order.

The real analogue of this routine is F08KMF (DGESVDX).

10 Example

This example finds the singular values and left and right singular vectors of the 6 by 4 matrix

$$A = \begin{pmatrix} 0.96 - 0.81i & -0.03 + 0.96i & -0.91 + 2.06i & -0.05 + 0.04i \\ -0.98 + 1.98i & -1.20 + 0.19i & -0.66 + 0.42i & -0.81 + 0.56i \\ 0.62 - 0.46i & 1.01 + 0.02i & 0.63 - 0.17i & -1.11 + 0.60i \\ -0.37 + 0.38i & 0.19 - 0.54i & -0.98 - 0.36i & 0.22 - 0.20i \\ 0.83 + 0.51i & 0.20 + 0.01i & -0.17 - 0.46i & 1.47 + 1.59i \\ 1.08 + 0.28i & 0.20 - 0.12i & -0.07 + 1.23i & 0.26 + 0.26i \end{pmatrix},$$

together with approximate error bounds for the computed singular values and vectors.

The example program for F08KRF (ZGESDD) illustrates finding a singular value decomposition for the case $m \leq n$.

10.1 Program Text

Program f08kzfe

```
!      F08KZF Example Program Text
!
!      Mark 26 Release. NAG Copyright 2016.
!
!      .. Use Statements ..
!      Use nag_library, Only: nag_wp, zgesvdx
!      .. Implicit None Statement ..
!      Implicit None
!      .. Parameters ..
!      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
!      Real (Kind=nag_wp)         :: vl, vu
!      Integer                     :: i, il, info, iu, lda, ldu, ldvt,      &
```

```

                                liwork, lrwork, lwork, m, n, ns
Character (1)                  :: range
! .. Local Arrays ..
Complex (Kind=nag_wp), Allocatable :: a(:,,:), u(:,,:), vt(:,,:), work(:)
Real (Kind=nag_wp), Allocatable  :: rwork(:), s(:)
Integer, Allocatable              :: iwork(:)
! .. Intrinsic Procedures ..
Intrinsic                      :: nint
! .. Executable Statements ..
Write (nout,*) 'F08KZF Example Program Results'
Write (nout,*)
! Skip heading in data file
Read (nin,*)
Read (nin,*) m, n
lda = m
ldu = m
ldvt = n
lwork = 2*n**2 + 4*n
lrwork = 2*n**2 + 34*n
liwork = 12*n
Allocate (a(lda,n),s(n),vt(ldvt,n),u(ldu,m),iwork(liwork),work(lwork), &
         rwork(lrwork))

! Read the m by n matrix A from data file
Read (nin,*)(a(i,1:n),i=1,m)

! Read range for selected singular values
Read (nin,*) range

If (range=='I' .Or. range=='i') Then
    Read (nin,*) il, iu
Else If (range=='V' .Or. range=='v') Then
    Read (nin,*) vl, vu
End If

! Compute the singular values and left and right singular vectors
! of A.

! The NAG name equivalent of zgesvd is f08kzf
Call zgesvdx('V','V',range,m,n,a,lda,vl,vu,il,iu,ns,s,u,ldu,vt,ldvt, &
            work,lwork,rwork,iwork,info)

If (info/=0) Then
    Write (nout,99999) 'Failure in ZGESVDX. INFO =', info
99999 Format (1X,A,I4)
    Go To 100
End If

! Print the selected singular values of A

Write (nout,*) 'Singular values of A:'
Write (nout,99998) s(1:ns)
99998 Format (1X,4(3X,F11.4))

Call compute_error_bounds(m,ns,s)

100 Continue

Contains
Subroutine compute_error_bounds(m,n,s)

! Error estimates for singular values and vectors is computed
! and printed here.

! .. Use Statements ..
Use nag_library, Only: ddisna, nag_wp, x02ajf
! .. Implicit None Statement ..
Implicit None
! .. Scalar Arguments ..
Integer, Intent (In)      :: m, n
! .. Array Arguments ..

```

```

      Real (Kind=nag_wp), Intent (In) :: s(n)
!      .. Local Scalars ..
      Real (Kind=nag_wp)                :: eps, serrbd
      Integer                          :: i, info
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: rcondu(:), rcondv(:), uerrbd(:), &
                                         verrbd(:)
!      .. Executable Statements ..
      Allocate (rcondu(n),rcondv(n),uerrbd(n),verrbd(n))

!      Get the machine precision, EPS and compute the approximate
!      error bound for the computed singular values. Note that for
!      the 2-norm, S(1) = norm(A)

      eps = x02ajf()
      serrbd = eps*s(1)

!      Call DDISNA (F08FLF) to estimate reciprocal condition
!      numbers for the singular vectors

      Call ddisna('Left',m,n,s,rcondu,info)
      Call ddisna('Right',m,n,s,rcondv,info)

!      Compute the error estimates for the singular vectors

      Do i = 1, n
         uerrbd(i) = serrbd/rcondu(i)
         verrbd(i) = serrbd/rcondv(i)
      End Do

!      Print the approximate error bounds for the singular values
!      and vectors

      Write (nout,*)
      Write (nout,*) 'Error estimates (as multiples of machine precision):'
      Write (nout,*) '  for the singular values'
      Write (nout,99999) nint(serrbd/x02ajf())
      Write (nout,*)
      Write (nout,*) '  for the left singular vectors'
      Write (nout,99999) nint(uerrbd(1:n)/x02ajf())
      Write (nout,*)
      Write (nout,*) '  for the right singular vectors'
      Write (nout,99999) nint(verrbd(1:n)/x02ajf())

99999   Format (4X,6I11)

      End Subroutine compute_error_bounds

End Program f08kzfe

```

10.2 Program Data

F08KZF Example Program Data

```

      6              4                      : m and n

( 0.96,-0.81) (-0.03, 0.96) (-0.91, 2.06) (-0.05, 0.41)
(-0.98, 1.98) (-1.20, 0.19) (-0.66, 0.42) (-0.81, 0.56)
( 0.62,-0.46) ( 1.01, 0.02) ( 0.63,-0.17) (-1.11, 0.60)
(-0.37, 0.38) ( 0.19,-0.54) (-0.98,-0.36) ( 0.22,-0.20)
( 0.83, 0.51) ( 0.20, 0.01) (-0.17,-0.46) ( 1.47, 1.59)
( 1.08,-0.28) ( 0.20,-0.12) (-0.07, 1.23) ( 0.26, 0.26) : matrix A

'v'
1.00   4.00                      : range

```

10.3 Program Results

F08KZF Example Program Results

Singular values of A:

3.9994	3.0003	1.9944
--------	--------	--------

Error estimates (as multiples of machine precision):

for the singular values

4

for the left singular vectors

4

4

4

for the right singular vectors

4

4

4
