

NAG Library Routine Document

F08RAF (DORCSD)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08RAF (DORCSD) computes the CS decomposition of a real m by m orthogonal matrix X , partitioned into a 2 by 2 array of submatrices.

2 Specification

```
SUBROUTINE F08RAF (JOBU1, JOBU2, JOBV1T, JOBV2T, TRANS, SIGNS, M, P, Q,      &
  X11, LDX11, X12, LDX12, X21, LDX21, X22, LDX22,                          &
  THETA, U1, LDU1, U2, LDU2, V1T, LDV1T, V2T, LDV2T,                      &
  WORK, LWORK, IWORK, INFO)
INTEGER      M, P, Q, LDX11, LDX12, LDX21, LDX22, LDU1, LDU2,            &
  LDV1T, LDV2T, LWORK, IWORK(M-min(P,M-P,Q,M-Q)),                      &
  INFO
REAL (KIND=nag_wp) X11(LDX11,*), X12(LDX12,*), X21(LDX21,*),            &
  X22(LDX22,*), THETA(min(P,M-P,Q,M-Q)), U1(LDU1,*),                  &
  U2(LDU2,*), V1T(LDV1T,*), V2T(LDV2T,*),                               &
  WORK(max(1,LWORK))
CHARACTER(1)  JOBU1, JOBU2, JOBV1T, JOBV2T, TRANS, SIGNS
```

The routine may be called by its LAPACK name ***dorcscd***.

3 Description

The m by m orthogonal matrix X is partitioned as

$$X = \begin{pmatrix} X_{11} & X_{12} \\ X_{21} & X_{22} \end{pmatrix}$$

where X_{11} is a p by q submatrix and the dimensions of the other submatrices X_{12} , X_{21} and X_{22} are such that X remains m by m .

The CS decomposition of X is $X = U \Sigma_p V^T$ where U , V and Σ_p are m by m matrices, such that

$$U = \begin{pmatrix} U_1 & \mathbf{0} \\ \mathbf{0} & U_2 \end{pmatrix}$$

is an orthogonal matrix containing the p by p orthogonal matrix U_1 and the $(m-p)$ by $(m-p)$ orthogonal matrix U_2 ;

$$V = \begin{pmatrix} V_1 & \mathbf{0} \\ \mathbf{0} & V_2 \end{pmatrix}$$

is an orthogonal matrix containing the q by q orthogonal matrix V_1 and the $(m-q)$ by $(m-q)$ orthogonal matrix V_2 ; and

$$\Sigma_p = \left(\begin{array}{ccc|ccc} I_{11} & & \mathbf{0} & & \mathbf{0} & \mathbf{0} \\ & C & \mathbf{0} & & \mathbf{0} & -S \\ \mathbf{0} & \mathbf{0} & & & \mathbf{0} & -I_{12} \\ \hline & \mathbf{0} & \mathbf{0} & I_{22} & & \mathbf{0} \\ \mathbf{0} & S & & & C & \mathbf{0} \\ \mathbf{0} & & I_{21} & & \mathbf{0} & \mathbf{0} \end{array} \right)$$

contains the r by r non-negative diagonal submatrices C and S satisfying $C^2 + S^2 = I$, where $r = \min(p, m - p, q, m - q)$ and the top left partition is p by q .

The identity matrix I_{11} is of order $\min(p, q) - r$ and vanishes if $\min(p, q) = r$.

The identity matrix I_{12} is of order $\min(p, m - q) - r$ and vanishes if $\min(p, m - q) = r$.

The identity matrix I_{21} is of order $\min(m - p, q) - r$ and vanishes if $\min(m - p, q) = r$.

The identity matrix I_{22} is of order $\min(m - p, m - q) - r$ and vanishes if $\min(m - p, m - q) = r$.

In each of the four cases $r = p, q, m - p, m - q$ at least two of the identity matrices vanish.

The indicated zeros represent augmentations by additional rows or columns (but not both) to the square diagonal matrices formed by I_{ij} and C or S .

Σ_p does not need to be stored in full; it is sufficient to return only the values θ_i for $i = 1, 2, \dots, r$ where $C_{ii} = \cos(\theta_i)$ and $S_{ii} = \sin(\theta_i)$.

The algorithm used to perform the complete CS decomposition is described fully in Sutton (2009) including discussions of the stability and accuracy of the algorithm.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Golub G H and Van Loan C F (2012) *Matrix Computations* (4th Edition) Johns Hopkins University Press, Baltimore

Sutton B D (2009) Computing the complete CS decomposition *Numerical Algorithms (Volume 50)* 1017–1398 Springer US 33–65 <http://dx.doi.org/10.1007/s11075-008-9215-6>

5 Arguments

1: JOBU1 – CHARACTER(1) Input

On entry:

if JOBU1 = 'Y', U_1 is computed;

otherwise, U_1 is not computed.

2: JOBU2 – CHARACTER(1) Input

On entry:

if JOBU2 = 'Y', U_2 is computed;

otherwise, U_2 is not computed.

3: JOBV1T – CHARACTER(1) Input

On entry:

if JOBV1T = 'Y', V_1^T is computed;

otherwise, V_1^T is not computed.

4: JOBV2T – CHARACTER(1) Input

On entry:

if JOBV2T = 'Y', V_2^T is computed;

otherwise, V_2^T is not computed.

- 5: TRANS – CHARACTER(1) *Input*
On entry:
 if TRANS = 'T', X , U_1 , U_2 , V_1^T and V_2^T are stored in row-major order;
 otherwise, X , U_1 , U_2 , V_1^T and V_2^T are stored in column-major order.
- 6: SIGNS – CHARACTER(1) *Input*
On entry:
 if SIGNS = 'O', the lower-left block is made nonpositive (the other convention);
 otherwise, the upper-right block is made nonpositive (the default convention).
- 7: M – INTEGER *Input*
On entry: m , the number of rows and columns in the orthogonal matrix X .
Constraint: $M \geq 0$.
- 8: P – INTEGER *Input*
On entry: p , the number of rows in X_{11} and X_{12} .
Constraint: $0 \leq P \leq M$.
- 9: Q – INTEGER *Input*
On entry: q , the number of columns in X_{11} and X_{21} .
Constraint: $0 \leq Q \leq M$.
- 10: X11(LDX11,*) – REAL (KIND=nag_wp) array *Input/Output*
Note: the second dimension of the array X11 must be at least $\max(1, P)$ if TRANS = 'T', and at least Q otherwise.
On entry: the upper left partition of the orthogonal matrix X whose CSD is desired.
On exit: contains details of the orthogonal matrix used in a simultaneous bidiagonalization process.
- 11: LDX11 – INTEGER *Input*
On entry: the first dimension of the array X11 as declared in the (sub)program from which F08RAF (DORCSD) is called.
Constraints:
 if TRANS = 'T', $LDX11 \geq \max(1, Q)$;
 otherwise $LDX11 \geq \max(1, P)$.
- 12: X12(LDX12,*) – REAL (KIND=nag_wp) array *Input/Output*
Note: the second dimension of the array X12 must be at least $\max(1, P)$ if TRANS = 'T', and at least $M - Q$ otherwise.
On entry: the upper right partition of the orthogonal matrix X whose CSD is desired.
On exit: contains details of the orthogonal matrix used in a simultaneous bidiagonalization process.
- 13: LDX12 – INTEGER *Input*
On entry: the first dimension of the array X12 as declared in the (sub)program from which F08RAF (DORCSD) is called.

Constraints:

if TRANS = 'T', $\text{LDX12} \geq \max(1, M - Q)$;
 otherwise $\text{LDX12} \geq \max(1, P)$.

- 14: X21(LDX21,*) – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array X21 must be at least $\max(1, M - P)$ if TRANS = 'T', and at least Q otherwise.

On entry: the lower left partition of the orthogonal matrix X whose CSD is desired.

On exit: contains details of the orthogonal matrix used in a simultaneous bidiagonalization process.

- 15: LDX21 – INTEGER *Input*

On entry: the first dimension of the array X21 as declared in the (sub)program from which F08RAF (DORCSD) is called.

Constraints:

if TRANS = 'T', $\text{LDX21} \geq \max(1, Q)$;
 otherwise $\text{LDX21} \geq \max(1, M - P)$.

- 16: X22(LDX22,*) – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array X22 must be at least $\max(1, M - P)$ if TRANS = 'T', and at least $M - Q$ otherwise.

On entry: the lower right partition of the orthogonal matrix X CSD is desired.

On exit: contains details of the orthogonal matrix used in a simultaneous bidiagonalization process.

- 17: LDX22 – INTEGER *Input*

On entry: the first dimension of the array X22 as declared in the (sub)program from which F08RAF (DORCSD) is called.

Constraints:

if TRANS = 'T', $\text{LDX22} \geq \max(1, M - Q)$;
 otherwise $\text{LDX22} \geq \max(1, M - P)$.

- 18: THETA(min(P, M - P, Q, M - Q)) – REAL (KIND=nag_wp) array *Output*

On exit: the values θ_i for $i = 1, 2, \dots, r$ where $r = \min(p, m - p, q, m - q)$. The diagonal submatrices C and S of Σ_p are constructed from these values as

$$C = \text{diag}(\cos(\text{THETA}(1)), \dots, \cos(\text{THETA}(r))) \text{ and} \\ S = \text{diag}(\sin(\text{THETA}(1)), \dots, \sin(\text{THETA}(r))).$$

- 19: U1(LDU1,*) – REAL (KIND=nag_wp) array *Output*

Note: the second dimension of the array U1 must be at least $\max(1, P)$ if JOBU1 = 'Y', and at least 1 otherwise.

On exit: if JOBU1 = 'Y', U1 contains the p by p orthogonal matrix U_1 .

- 20: LDU1 – INTEGER *Input*

On entry: the first dimension of the array U1 as declared in the (sub)program from which F08RAF (DORCSD) is called.

Constraint: if JOBU1 = 'Y', $\text{LDU1} \geq \max(1, P)$.

- 21: U2(LDU2,*) – REAL (KIND=nag_wp) array Output
Note: the second dimension of the array U2 must be at least $\max(1, M - P)$ if JOBU2 = 'Y', and at least 1 otherwise.
On exit: if JOBU2 = 'Y', U2 contains the $m - p$ by $m - p$ orthogonal matrix U_2 .
- 22: LDU2 – INTEGER Input
On entry: the first dimension of the array U2 as declared in the (sub)program from which F08RAF (DORCSD) is called.
Constraint: if JOBU2 = 'Y', $LDU2 \geq \max(1, M - P)$.
- 23: V1T(LDV1T,*) – REAL (KIND=nag_wp) array Output
Note: the second dimension of the array V1T must be at least $\max(1, Q)$ if JOBV1T = 'Y', and at least 1 otherwise.
On exit: if JOBV1T = 'Y', V1T contains the q by q orthogonal matrix V_1^T .
- 24: LDV1T – INTEGER Input
On entry: the first dimension of the array V1T as declared in the (sub)program from which F08RAF (DORCSD) is called.
Constraint: if JOBV1T = 'Y', $LDV1T \geq \max(1, Q)$.
- 25: V2T(LDV2T,*) – REAL (KIND=nag_wp) array Output
Note: the second dimension of the array V2T must be at least $\max(1, M - Q)$ if JOBV2T = 'Y', and at least 1 otherwise.
On exit: if JOBV2T = 'Y', V2T contains the $m - q$ by $m - q$ orthogonal matrix V_2^T .
- 26: LDV2T – INTEGER Input
On entry: the first dimension of the array V2T as declared in the (sub)program from which F08RAF (DORCSD) is called.
Constraint: if JOBV2T = 'Y', $LDV2T \geq \max(1, M - Q)$.
- 27: WORK(max(1,LWORK)) – REAL (KIND=nag_wp) array Workspace
On exit: if INFO = 0, WORK(1) returns the optimal LWORK.
If INFO > 0 on exit, WORK(2 : r) contains the values PHI(1), ..., PHI(r - 1) that, together with THETA(1), ..., THETA(r), define the matrix in intermediate bidiagonal-block form remaining after nonconvergence. INFO specifies the number of nonzero PHI's.
- 28: LWORK – INTEGER Input
On entry:
If LWORK = -1, a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.
The minimum workspace required is $5 \times \max(1, r - 1) + 4 \times \max(1, r) + \max(1, 8r) + \max(1, p) + \max(1, m - p) + \max(1, q) + \max(1, m - q) + 1$ where $r = \min(p, m - p, q, m - q)$. The optimal workspace depends on internal block sizes and the relative dimensions of the problem.
Constraint: LWORK = -1 or
 $LWORK \geq 5 \times \max(1, r - 1) + 4 \times \max(1, r) + \max(1, 8r) + \max(1, P) + \max(1, M - P) + \max(1, Q) + \max(1, M - Q) + 1$.

29: IWORK(M – min(P, M – P, Q, M – Q)) – INTEGER array Workspace

30: INFO – INTEGER Output

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

The Jacobi-type procedure failed to converge during an internal reduction to bidiagonal-block form. The process requires convergence to min(P, M – P, Q, M – Q) values, the value of INFO gives the number of converged values.

7 Accuracy

The computed CS decomposition is nearly the exact CS decomposition for the nearby matrix $(X + E)$, where

$$\|E\|_2 = O(\epsilon),$$

and ϵ is the *machine precision*.

8 Parallelism and Performance

F08RAF (DORCSD) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F08RAF (DORCSD) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations required to perform the full CS decomposition is approximately $2m^3$.

The complex analogue of this routine is F08RNF (ZUNCSD).

10 Example

This example finds the full CS decomposition of

$$X = \begin{pmatrix} -0.7576 & 0.3697 & 0.3838 & 0.2126 & -0.3112 \\ -0.4077 & -0.1552 & -0.1129 & 0.2676 & 0.8517 \\ -0.0488 & 0.7240 & -0.6730 & -0.1301 & 0.0602 \\ -0.2287 & 0.0088 & 0.2235 & -0.9235 & 0.2120 \\ 0.4530 & 0.5612 & 0.5806 & 0.1162 & 0.3595 \end{pmatrix}$$

partitioned so that the top left block is 3 by 2.

The decomposition is performed both on submatrices of the orthogonal matrix X and on separated partition matrices. Code is also provided to perform a recombining check if required.

10.1 Program Text

Program f08rafe

```
!      F08RAF Example Program Text
!
!      Mark 26 Release. NAG Copyright 2016.
!
!      .. Use Statements ..
!      Use nag_library, Only: dgemm, dorcsd, nag_wp, x04caf
!      .. Implicit None Statement ..
!      Implicit None
!      .. Parameters ..
!      Real (Kind=nag_wp), Parameter      :: one = 1.0_nag_wp
!      Real (Kind=nag_wp), Parameter      :: zero = 0.0_nag_wp
!      Integer, Parameter                  :: nin = 5, nout = 6, recombine = 1, &
!                                          reprint = 0
!
!      .. Local Scalars ..
!      Integer                             :: i, ifail, info, info2, j, ldu, ldu1, &
!                                          ldu2, ldv, ldv1t, ldv2t, ldx, ldx11, &
!                                          ldx12, ldx21, ldx22, lwork, m, n11, &
!                                          n12, n21, n22, p, q, r
!
!      .. Local Arrays ..
!      Real (Kind=nag_wp), Allocatable     :: theta(:), u(:,,:), u1(:,,:), u2(:,,:), &
!                                          v(:,,:), v1t(:,,:), v2t(:,,:), w(:,,:), &
!                                          work(:), x(:,,:), x11(:,,:), x12(:,,:), &
!                                          x21(:,,:), x22(:,,:)
!
!      Real (Kind=nag_wp)                  :: wdum(1)
!      Integer, Allocatable                 :: iwork(:)
!
!      .. Intrinsic Procedures ..
!      Intrinsic                           :: cos, min, nint, sin
!
!      .. Executable Statements ..
!      Write (nout,*) 'F08RAF Example Program Results'
!      Write (nout,*)
!      Flush (nout)
!      Skip heading in data file
!      Read (nin,*)
!      Read (nin,*) m, p, q
!
!      r = min(min(p,q),min(m-p,m-q))
!
!      ldx = m
!      ldx11 = p
!      ldx12 = p
!      ldx21 = m - p
!      ldx22 = m - p
!      ldu = m
!      ldu1 = p
!      ldu2 = m - p
!      ldv = m
!      ldv1t = q
!      ldv2t = m - q
!      Allocate (x(ldx,m),u(ldu,m),v(ldv,m),theta(r),iwork(m),w(ldx,m))
!      Allocate (x11(ldx11,q),x12(ldx12,m-q),x21(ldx21,q),x22(ldx22,m-q))
!      Allocate (u1(ldu1,p),u2(ldu2,m-p),v1t(ldv1t,q),v2t(ldv2t,m-q))
!
!      Read and print orthogonal X from data file
!      (as, say, generated by a generalized singular value decomposition).
!
!      Read (nin,*)(x(i,1:m),i=1,m)
!
!      Print general matrix using simple matrix printing routine x04caf.
!      ifail: behaviour on error exit
!      =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
!      ifail = 0
!      Call x04caf('G','N',m,m,x,ldx,'      Orthogonal matrix X',ifail)
!      Write (nout,*)
!
!      Compute the complete CS factorization of X:
!      X11 is stored in X(1:p, 1:q), X12 is stored in X(1:p, q+1:m)
!      X21 is stored in X(p+1:m, 1:q), X22 is stored in X(p+1:m, q+1:m)
```

```

!      U1 is stored in U(1:p, 1:p), U2 is stored in U(p+1:m, p+1:m)
!      V1 is stored in V(1:q, 1:q), V2 is stored in V(q+1:m, q+1:m)
x11(1:p,1:q) = x(1:p,1:q)
x12(1:p,1:m-q) = x(1:p,q+1:m)
x21(1:m-p,1:q) = x(p+1:m,1:q)
x22(1:m-p,1:m-q) = x(p+1:m,q+1:m)

!      Workspace query first to get length of work array for complete CS
!      factorization routine dorcsd/f08raf.
lwork = -1
Call dorcsd('Yes U1','Yes U2','Yes V1T','Yes V2T','Column','Default',m, &
  p,q,x,ldx,x(1,q+1),ldx,x(p+1,1),ldx,x(p+1,q+1),ldx,theta,u,ldu, &
  u(p+1,p+1),ldu,v,ldv,v(q+1,q+1),ldv,wdum,lwork,iwork,info)
lwork = nint(wdum(1))
Allocate (work(lwork))
!      Initialize all of u, v to zero.
u(1:m,1:m) = zero
v(1:m,1:m) = zero

!      This is how you might pass partitions as sub-matrices
Call dorcsd('Yes U1','Yes U2','Yes V1T','Yes V2T','Column','Default',m, &
  p,q,x,ldx,x(1,q+1),ldx,x(p+1,1),ldx,x(p+1,q+1),ldx,theta,u,ldu, &
  u(p+1,p+1),ldu,v,ldv,v(q+1,q+1),ldv,work,lwork,iwork,info)
If (info/=0) Then
  Write (nout,99999) 'Failure in DORCSD/F08RAF. info =', info
  Go To 100
End If
!      Print Theta, U1, U2, V1T, V2T using matrix printing routine x04caf.
Write (nout,99998) 'Components of CS factorization of X:'
ifail = 0
Call x04caf('G','N',r,1,theta,r,'      Theta',ifail)
Write (nout,*)
!      By changes of sign the first r elements of the first row of U1 can be
!      made positive.
Do i = 1, r
  If (u(1,i)<0.0_nag_wp) Then
    u(1:p,i) = -u(1:p,i)
    u(p+1:m,p+i) = -u(p+1:m,p+i)
    v(i,1:q) = -v(i,1:q)
    v(q+i,q+1:m) = -v(q+i,q+1:m)
  End If
End Do
ifail = 0
Call x04caf('G','N',p,p,u,ldu,'      U1',ifail)
Write (nout,*)
ifail = 0
Call x04caf('G','N',m-p,m-p,u(p+1,p+1),ldu,'      U2',ifail)
Write (nout,*)
ifail = 0
Call x04caf('G','N',q,q,v,ldv,'      V1T',ifail)
Write (nout,*)
ifail = 0
Call x04caf('G','N',m-q,m-q,v(q+1,q+1),ldv,'      V2T',ifail)
Write (nout,*)

!      And this is how you might pass partitions as separate matrices.
Call dorcsd('Yes U1','Yes U2','Yes V1T','Yes V2T','Column','Default',m, &
  p,q,x11,ldx11,x12,ldx12,x21,ldx21,x22,ldx22,theta,u1,ldu1,u2,ldu2,v1t, &
  ldv1t,v2t,ldv2t,work,lwork,iwork,info2)
If (info2/=0) Then
  Write (nout,99999) 'Failure in DORCSD/F08RAF. info =', info
  Go To 100
End If

!      Print Theta, U1, U2, V1T, V2T using matrix printing routine x04caf.
If (reprint/=0) Then
!      By changes of sign the first r elements of the first row of U1 can be
!      made positive.
  Do i = 1, r
    If (u1(1,i)<0.0_nag_wp) Then
      u1(1:p,i) = -u1(1:p,i)

```



```

        u2(1:m-p,i) = -u2(1:m-p,i)
        v1t(i,1:q) = -v1t(i,1:q)
        v2t(i,1:m-q) = -v2t(i,1:m-q)
    End If
End Do
Write (nout,99998) 'Components of CS factorization of X:'
ifail = 0
Call x04caf('G','N',r,1,theta,r,'    Theta',ifail)
Write (nout,*)
ifail = 0
Call x04caf('G','N',p,p,u1,ldu1,'    U1',ifail)
Write (nout,*)
ifail = 0
Call x04caf('G','N',m-p,m-p,u2,ldu2,'    U2',ifail)
Write (nout,*)
ifail = 0
Call x04caf('G','N',q,q,v1t,ldv1t,'    V1T',ifail)
Write (nout,*)
ifail = 0
Call x04caf('G','N',m-q,m-q,v2t,ldv2t,'    V2T',ifail)
Write (nout,*)
End If
If (recombine/=0) Then
!   Recombining should return the original matrix
!   Assemble Sigma_p into X
    x(1:m,1:m) = zero
    n11 = min(p,q) - r
    n12 = min(p,m-q) - r
    n21 = min(m-p,q) - r
    n22 = min(m-p,m-1) - r
!   Top Half
    Do j = 1, n11
        x(j,j) = one
    End Do
    Do j = 1, r
        x(j+n11,j+n11) = cos(theta(j))
        x(j+n11,j+n11+r+n21+n22) = -sin(theta(j))
    End Do
    Do j = 1, n12
        x(j+n11+r,j+n11+r+n21+n22+r) = -one
    End Do
!   Bottom half
    Do j = 1, n22
        x(p+j,q+j) = one
    End Do
    Do j = 1, r
        x(p+n22+j,j+n11) = sin(theta(j))
        x(p+n22+j,j+r+n21+n22) = cos(theta(j))
    End Do
    Do j = 1, n21
        x(p+n22+r+j,n11+r+j) = one
    End Do
!   multiply U * Sigma_p into w
    Call dgemm('n','n',m,m,m,one,u,ldu,x,ldx,zero,w,ldx)
!   form U * Sigma_p * V^T into u
    Call dgemm('n','n',m,m,m,one,w,ldx,v,ldv,zero,u,ldu)

    ifail = 0
    Call x04caf('G','N',m,m,u,ldu,
        '    Recombined matrix X = U * Sigma_p * V^T',ifail)
End If

100    Continue

99999 Format (1X,A,I4)
99998 Format (1X,A,/)
End Program f08rafe

```

10.2 Program Data

F08RAF Example Program Data

```

5      3      2      : m, p, q
-0.7576 0.3697 0.3838 0.2126 -0.3112
-0.4077 -0.1552 -0.1129 0.2676 0.8517
-0.0488 0.7240 -0.6730 -0.1301 0.0602
-0.2287 0.0088 0.2235 -0.9235 0.2120
0.4530 0.5612 0.5806 0.1162 0.3595 : orthogonal matrix X

```

10.3 Program Results

F08RAF Example Program Results

```

Orthogonal matrix X
1      2      3      4      5
1 -0.7576 0.3697 0.3838 0.2126 -0.3112
2 -0.4077 -0.1552 -0.1129 0.2676 0.8517
3 -0.0488 0.7240 -0.6730 -0.1301 0.0602
4 -0.2287 0.0088 0.2235 -0.9235 0.2120
5 0.4530 0.5612 0.5806 0.1162 0.3595

```

Components of CS factorization of X:

```

Theta
1
1 0.1811
2 0.8255

```

```

U1
1      2      3
1 0.8249 0.3370 -0.4538
2 0.2042 0.5710 0.7952
3 0.5271 -0.7486 0.4022

```

```

U2
1      2
1 0.9802 0.1982
2 0.1982 -0.9802

```

```

V1T
1      2
1 -0.7461 0.6658
2 -0.6658 -0.7461

```

```

V2T
1      2      3
1 0.3397 -0.8967 0.2837
2 -0.7738 -0.4379 -0.4576
3 0.5346 -0.0640 -0.8427

```

```

Recombined matrix X = U * Sigma_p * V^T
1      2      3      4      5
1 -0.7576 0.3697 0.3838 0.2126 -0.3112
2 -0.4077 -0.1551 -0.1129 0.2677 0.8517
3 -0.0488 0.7240 -0.6730 -0.1300 0.0602
4 -0.2287 0.0088 0.2235 -0.9234 0.2120
5 0.4530 0.5612 0.5806 0.1162 0.3595

```
