

NAG Library Routine Document

F08SNF (ZHEGV)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08SNF (ZHEGV) computes all the eigenvalues and, optionally, the eigenvectors of a complex generalized Hermitian-definite eigenproblem, of the form

$$Az = \lambda Bz, \quad ABz = \lambda z \quad \text{or} \quad BAz = \lambda z,$$

where A and B are Hermitian and B is also positive definite.

2 Specification

```
SUBROUTINE F08SNF ( ITYPE, JOBZ, UPLO, N, A, LDA, B, LDB, W, WORK, LWORK, &
                    RWORK, INFO )
```

```
INTEGER                ITYPE, N, LDA, LDB, LWORK, INFO
REAL (KIND=nag_wp)    W(N), RWORK(3*N-2)
COMPLEX (KIND=nag_wp) A(LDA,*), B(LDB,*), WORK(max(1,LWORK))
CHARACTER(1)          JOBZ, UPLO
```

The routine may be called by its LAPACK name ***zhegv***.

3 Description

F08SNF (ZHEGV) first performs a Cholesky factorization of the matrix B as $B = U^H U$, when $UPLO = 'U'$ or $B = LL^H$, when $UPLO = 'L'$. The generalized problem is then reduced to a standard symmetric eigenvalue problem

$$Cx = \lambda x,$$

which is solved for the eigenvalues and, optionally, the eigenvectors; the eigenvectors are then backtransformed to give the eigenvectors of the original problem.

For the problem $Az = \lambda Bz$, the eigenvectors are normalized so that the matrix of eigenvectors, z , satisfies

$$Z^H A Z = \Lambda \quad \text{and} \quad Z^H B Z = I,$$

where Λ is the diagonal matrix whose diagonal elements are the eigenvalues. For the problem $ABz = \lambda z$ we correspondingly have

$$Z^{-1} A Z^{-H} = \Lambda \quad \text{and} \quad Z^H B Z = I,$$

and for $BAz = \lambda z$ we have

$$Z^H A Z = \Lambda \quad \text{and} \quad Z^H B^{-1} Z = I.$$

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Arguments

- 1: ITYPE – INTEGER *Input*
- On entry:* specifies the problem type to be solved.
- ITYPE = 1
 $Az = \lambda Bz.$
- ITYPE = 2
 $ABz = \lambda z.$
- ITYPE = 3
 $BAz = \lambda z.$
- Constraint:* ITYPE = 1, 2 or 3.
- 2: JOBZ – CHARACTER(1) *Input*
- On entry:* indicates whether eigenvectors are computed.
- JOBZ = 'N'
 Only eigenvalues are computed.
- JOBZ = 'V'
 Eigenvalues and eigenvectors are computed.
- Constraint:* JOBZ = 'N' or 'V'.
- 3: UPLO – CHARACTER(1) *Input*
- On entry:* if UPLO = 'U', the upper triangles of A and B are stored.
 If UPLO = 'L', the lower triangles of A and B are stored.
- Constraint:* UPLO = 'U' or 'L'.
- 4: N – INTEGER *Input*
- On entry:* n , the order of the matrices A and B .
- Constraint:* $N \geq 0$.
- 5: A(LDA,*) – COMPLEX (KIND=nag_wp) array *Input/Output*
- Note:** the second dimension of the array A must be at least $\max(1, N)$.
- On entry:* the n by n Hermitian matrix A .
- If UPLO = 'U', the upper triangular part of A must be stored and the elements of the array below the diagonal are not referenced.
- If UPLO = 'L', the lower triangular part of A must be stored and the elements of the array above the diagonal are not referenced.
- On exit:* if JOBZ = 'V', A contains the matrix Z of eigenvectors. The eigenvectors are normalized as follows:
- if ITYPE = 1 or 2, $Z^H B Z = I$;
 if ITYPE = 3, $Z^H B^{-1} Z = I$.
- If JOBZ = 'N', the upper triangle (if UPLO = 'U') or the lower triangle (if UPLO = 'L') of A , including the diagonal, is overwritten.

- 6: LDA – INTEGER *Input*
On entry: the first dimension of the array A as declared in the (sub)program from which F08SNF (ZHEGV) is called.
Constraint: $LDA \geq \max(1, N)$.
- 7: B(LDB, *) – COMPLEX (KIND=nag_wp) array *Input/Output*
Note: the second dimension of the array B must be at least $\max(1, N)$.
On entry: the n by n Hermitian positive definite matrix B .
 If UPLO = 'U', the upper triangular part of B must be stored and the elements of the array below the diagonal are not referenced.
 If UPLO = 'L', the lower triangular part of B must be stored and the elements of the array above the diagonal are not referenced.
On exit: if $0 \leq \text{INFO} \leq N$, the part of B containing the matrix is overwritten by the triangular factor U or L from the Cholesky factorization $B = U^H U$ or $B = L L^H$.
- 8: LDB – INTEGER *Input*
On entry: the first dimension of the array B as declared in the (sub)program from which F08SNF (ZHEGV) is called.
Constraint: $LDB \geq \max(1, N)$.
- 9: W(N) – REAL (KIND=nag_wp) array *Output*
On exit: the eigenvalues in ascending order.
- 10: WORK(max(1, LWORK)) – COMPLEX (KIND=nag_wp) array *Workspace*
On exit: if $\text{INFO} = 0$, the real part of WORK(1) contains the minimum value of LWORK required for optimal performance.
- 11: LWORK – INTEGER *Input*
On entry: the dimension of the array WORK as declared in the (sub)program from which F08SNF (ZHEGV) is called.
 If LWORK = -1, a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.
 Suggested value: for optimal performance, $LWORK \geq (nb + 1) \times N$, where nb is the optimal **block size** for F08FSF (ZHETRD).
Constraint: $LWORK \geq \max(1, 2 \times N)$.
- 12: RWORK($3 \times N - 2$) – REAL (KIND=nag_wp) array *Workspace*
- 13: INFO – INTEGER *Output*
On exit: $\text{INFO} = 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If $\text{INFO} = -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO = 1 to N

If INFO = i , F08FNF (ZHEEV) failed to converge; i off-diagonal elements of an intermediate tridiagonal form did not converge to zero.

INFO > N

F07FRF (ZPOTRF) returned an error code; i.e., if INFO = $N + i$, for $1 \leq i \leq N$, then the leading minor of order i of B is not positive definite. The factorization of B could not be completed and no eigenvalues or eigenvectors were computed.

7 Accuracy

If B is ill-conditioned with respect to inversion, then the error bounds for the computed eigenvalues and vectors may be large, although when the diagonal elements of B differ widely in magnitude the eigenvalues and eigenvectors may be less sensitive than the condition of B would suggest. See Section 4.10 of Anderson *et al.* (1999) for details of the error bounds.

The example program below illustrates the computation of approximate error bounds.

8 Parallelism and Performance

F08SNF (ZHEGV) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F08SNF (ZHEGV) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations is proportional to n^3 .

The real analogue of this routine is F08SAF (DSYGV).

10 Example

This example finds all the eigenvalues and eigenvectors of the generalized Hermitian eigenproblem $Az = \lambda Bz$, where

$$A = \begin{pmatrix} -7.36 & 0.77 - 0.43i & -0.64 - 0.92i & 3.01 - 6.97i \\ 0.77 + 0.43i & 3.49 & 2.19 + 4.45i & 1.90 + 3.73i \\ -0.64 + 0.92i & 2.19 - 4.45i & 0.12 & 2.88 - 3.17i \\ 3.01 + 6.97i & 1.90 - 3.73i & 2.88 + 3.17i & -2.54 \end{pmatrix}$$

and

$$B = \begin{pmatrix} 3.23 & 1.51 - 1.92i & 1.90 + 0.84i & 0.42 + 2.50i \\ 1.51 + 1.92i & 3.58 & -0.23 + 1.11i & -1.18 + 1.37i \\ 1.90 - 0.84i & -0.23 - 1.11i & 4.09 & 2.33 - 0.14i \\ 0.42 - 2.50i & -1.18 - 1.37i & 2.33 + 0.14i & 4.29 \end{pmatrix},$$

together with an estimate of the condition number of B , and approximate error bounds for the computed eigenvalues and eigenvectors.

The example program for F08SQF (ZHEGVD) illustrates solving a generalized Hermitian eigenproblem of the form $ABz = \lambda z$.

10.1 Program Text

```

Program f08snfe

!      F08SNF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: ddisna, f06ucf, nag_wp, x02ajf, x04daf, zhegv,      &
                               ztrcon
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nb = 64, nin = 5, nout = 6
!      .. Local Scalars ..
      Complex (Kind=nag_wp)       :: scal
      Real (Kind=nag_wp)          :: anorm, bnorm, eps, rcond, rcondb,      &
                                     t1, t2, t3
      Integer                     :: i, ifail, info, k, lda, ldb, lwork, &
                                     n
!      .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: a(:,,:), b(:,,:), work(:)
      Complex (Kind=nag_wp)              :: dummy(1)
      Real (Kind=nag_wp), Allocatable    :: eerbnd(:), rcondz(:), rwork(:),      &
                                     w(:), zerbnd(:)
!      .. Intrinsic Procedures ..
      Intrinsic                      :: abs, conjg, max, maxloc, nint, real
!      .. Executable Statements ..
      Write (nout,*) 'F08SNF Example Program Results'
      Write (nout,*)
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n
      lda = n
      ldb = n
      Allocate (a(lda,n),b(ldb,n),eerbnd(n),rcondz(n),rwork(3*n-2),w(n),      &
               zerbnd(n))

!      Use routine workspace query to get optimal workspace.
      lwork = -1
!      The NAG name equivalent of zhegv is f08snf
      Call zhegv(1,'Vectors','Upper',n,a,lda,b,ldb,w,dummy,lwork,rwork,info)

!      Make sure that there is enough workspace for block size nb.
      lwork = max((nb+1)*n,nint(real(dummy(1))))
      Allocate (work(lwork))

!      Read the upper triangular parts of the matrices A and B

      Read (nin,*)(a(i,i:n),i=1,n)
      Read (nin,*)(b(i,i:n),i=1,n)

!      Compute the one-norms of the symmetric matrices A and B

      anorm = f06ucf('One norm','Upper',n,a,lda,rwork)
      bnorm = f06ucf('One norm','Upper',n,b,ldb,rwork)

!      Solve the generalized Hermitian eigenvalue problem
!      A*x = lambda*B*x (itype = 1)

!      The NAG name equivalent of zhegv is f08snf
      Call zhegv(1,'Vectors','Upper',n,a,lda,b,ldb,w,work,lwork,rwork,info)

      If (info==0) Then

!          Print solution

          Write (nout,*) 'Eigenvalues'
          Write (nout,99999) w(1:n)
          Flush (nout)

```

```

!      Normalize the eigenvectors, largest element real
!      (normalization w.r.t B unaffected:  $Z^H B Z = I$ ).
      Do i = 1, n
        rwork(1:n) = abs(a(1:n,i))
        k = maxloc(rwork(1:n),1)
        scal = conjg(a(k,i))/abs(a(k,i))
        a(1:n,i) = a(1:n,i)*scal
      End Do

!      ifail: behaviour on error exit
!      =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
      Call x04daf('General',' ',n,n,a,lda,'Eigenvectors',ifail)

!      Call ZTRCON (F07TUF) to estimate the reciprocal condition
!      number of the Cholesky factor of B. Note that:
!      cond(B) = 1/rcond**2
      Call ztrcon('One norm','Upper','Non-unit',n,b,ldb,rcond,work,rwork,      &
        info)

!      Print the reciprocal condition number of B

      rcondb = rcond**2
      Write (nout,*)
      Write (nout,*) 'Estimate of reciprocal condition number for B'
      Write (nout,99998) rcondb
      Flush (nout)

!      Get the machine precision, eps, and if rcondb is not less
!      than eps**2, compute error estimates for the eigenvalues and
!      eigenvectors

      eps = x02ajf()
      If (rcond>=eps) Then

!      Call DDISNA (F08FLF) to estimate reciprocal condition
!      numbers for the eigenvectors of (A - lambda*B)

        Call ddisna('Eigenvectors',n,n,w,rcondz,info)

!      Compute the error estimates for the eigenvalues and
!      eigenvectors

        t1 = eps/rcondb
        t2 = anorm/bnorm
        t3 = t2/rcond
        Do i = 1, n
          eerbnd(i) = t1*(t2+abs(w(i)))
          zerbnd(i) = t1*(t3+abs(w(i)))/rcondz(i)
        End Do

!      Print the approximate error bounds for the eigenvalues
!      and vectors

        Write (nout,*)
        Write (nout,*) 'Error estimates for the eigenvalues'
        Write (nout,99998) eerbnd(1:n)
        Write (nout,*)
        Write (nout,*) 'Error estimates for the eigenvectors'
        Write (nout,99998) zerbnd(1:n)
      Else
        Write (nout,*)
        Write (nout,*) 'B is very ill-conditioned, error ',      &
          'estimates have not been computed'
      End If
    Else If (info>n) Then
      i = info - n
      Write (nout,99997) 'The leading minor of order ', i,      &
        ' of B is not positive definite'
    Else

```

```

      Write (nout,99996) 'Failure in ZHEGV. INFO =', info
      End If

99999 Format (3X,(6F11.4))
99998 Format (4X,1P,6E11.1)
99997 Format (1X,A,I4,A)
99996 Format (1X,A,I4)
      End Program f08snfe

```

10.2 Program Data

F08SNF Example Program Data

```

4                                     :Value of N

(-7.36, 0.00) ( 0.77, -0.43) (-0.64, -0.92) ( 3.01, -6.97)
              ( 3.49, 0.00) ( 2.19, 4.45) ( 1.90, 3.73)
              ( 0.12, 0.00) ( 2.88, -3.17)
              (-2.54, 0.00) :End of matrix A

( 3.23, 0.00) ( 1.51, -1.92) ( 1.90, 0.84) ( 0.42, 2.50)
              ( 3.58, 0.00) (-0.23, 1.11) (-1.18, 1.37)
              ( 4.09, 0.00) ( 2.33, -0.14)
              ( 4.29, 0.00) :End of matrix B

```

10.3 Program Results

F08SNF Example Program Results

```

Eigenvalues
-5.9990   -2.9936    0.5047    3.9990
Eigenvectors
      1      2      3      4
1      1.7405  -0.6626   0.2835   1.2378
      0.0000   0.2258  -0.5806   0.0000

2      -0.4136  -0.1164  -0.3769  -0.5608
      -0.4689  -0.0178  -0.3194  -0.3729

3      -0.8404   0.9098  -0.3338  -0.6643
      -0.2483  -0.0000  -0.0134  -0.1021

4       0.3021  -0.6120   0.6663   0.1589
       0.6103  -0.5348   0.0000   0.8366

Estimate of reciprocal condition number for B
2.5E-03

Error estimates for the eigenvalues
3.4E-13  2.0E-13  9.6E-14  2.5E-13

Error estimates for the eigenvectors
5.8E-13  5.3E-13  4.3E-13  4.7E-13

```
