
NAG Library Routine Document

F08UPF (ZHBGVX)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

F08UPF (ZHBGVX) computes selected the eigenvalues and, optionally, the eigenvectors of a complex
generalized Hermitian-definite banded eigenproblem, of the form

Az ¼ �Bz;

where A and B are Hermitian and banded, and B is also positive definite. Eigenvalues and eigenvectors
can be selected by specifying either all eigenvalues, a range of values or a range of indices for the
desired eigenvalues.

2 Specification

SUBROUTINE F08UPF (JOBZ, RANGE, UPLO, N, KA, KB, AB, LDAB, BB, LDBB, Q,
LDQ, VL, VU, IL, IU, ABSTOL, M, W, Z, LDZ, WORK,
RWORK, IWORK, JFAIL, INFO)

&
&

INTEGER N, KA, KB, LDAB, LDBB, LDQ, IL, IU, M, LDZ,
IWORK(5*N), JFAIL(*), INFO

&

REAL (KIND=nag_wp) VL, VU, ABSTOL, W(N), RWORK(7*N)
COMPLEX (KIND=nag_wp) AB(LDAB,*), BB(LDBB,*), Q(LDQ,*), Z(LDZ,*),

WORK(N)
&

CHARACTER(1) JOBZ, RANGE, UPLO

The routine may be called by its LAPACK name zhbgvx.

3 Description

The generalized Hermitian-definite band problem

Az ¼ �Bz

is first reduced to a standard band Hermitian problem

Cx ¼ �x;

where C is a Hermitian band matrix, using Wilkinson's modification to Crawford's algorithm (see
Crawford (1973) and Wilkinson (1977)). The Hermitian eigenvalue problem is then solved for the
required eigenvalues and eigenvectors, and the eigenvectors are then backtransformed to the
eigenvectors of the original problem.

The eigenvectors are normalized so that

zHAz ¼ � and zHBz ¼ 1:

5 Arguments

1: JOBZ – CHARACTER(1) Input

On entry: indicates whether eigenvectors are computed.

JOBZ ¼ N
Only eigenvalues are computed.

JOBZ ¼ V
Eigenvalues and eigenvectors are computed.

Constraint: JOBZ ¼ N or V .

2: RANGE – CHARACTER(1) Input

On entry: if RANGE ¼ A , all eigenvalues will be found.

If RANGE ¼ V , all eigenvalues in the half-open interval VL;VUð � will be found.

If RANGE ¼ I , the ILth to IUth eigenvalues will be found.

Constraint: RANGE ¼ A , V or I .

3: UPLO – CHARACTER(1) Input

On entry: if UPLO ¼ U , the upper triangles of A and B are stored.

If UPLO ¼ L , the lower triangles of A and B are stored.

Constraint: UPLO ¼ U or L .

4: N – INTEGER Input

On entry: n, the order of the matrices A and B.

Constraint: N � 0.

5: KA – INTEGER Input

On entry: if UPLO ¼ U , the number of superdiagonals, ka, of the matrix A.

If UPLO ¼ L , the number of subdiagonals, ka, of the matrix A.

Constraint: KA � 0.

6: KB – INTEGER Input

On entry: if UPLO ¼ U , the number of superdiagonals, kb, of the matrix B.

If UPLO ¼ L , the number of subdiagonals, kb, of the matrix B.

Constraint: KA � KB � 0.

7: ABðLDAB; �Þ – COMPLEX (KIND=nag_wp) array Input/Output

Note: the second dimension of the array AB must be at least max 1;Nð Þ.
On entry: the upper or lower triangle of the n by n Hermitian band matrix A.

The matrix is stored in rows 1 to ka þ 1, more precisely,

if UPLO ¼ U , the elements of the upper triangle of A within the band must be stored with
element Aij in ABðka þ 1þ i� j; jÞ for max 1; j� kað Þ � i � j;

if UPLO ¼ L , the elements of the lower triangle of A within the band must be stored with
element Aij in ABð1þ i� j; jÞ for j � i � min n; jþ kað Þ:

On exit: the contents of AB are overwritten.

8: LDAB – INTEGER Input

On entry: the first dimension of the array AB as declared in the (sub)program from which
F08UPF (ZHBGVX) is called.

Constraint: LDAB � KAþ 1.

9: BBðLDBB; �Þ – COMPLEX (KIND=nag_wp) array Input/Output

Note: the second dimension of the array BB must be at least max 1;Nð Þ.
On entry: the upper or lower triangle of the n by n Hermitian positive definite band matrix B.

The matrix is stored in rows 1 to kb þ 1, more precisely,

if UPLO ¼ U , the elements of the upper triangle of B within the band must be stored with
element Bij in BBðkb þ 1þ i� j; jÞ for max 1; j� kbð Þ � i � j;

if UPLO ¼ L , the elements of the lower triangle of B within the band must be stored with
element Bij in BBð1þ i� j; jÞ for j � i � min n; jþ kbð Þ:

On exit: the factor S from the split Cholesky factorization B ¼ SHS, as returned by F08UTF
(ZPBSTF).

10: LDBB – INTEGER Input

On entry: the first dimension of the array BB as declared in the (sub)program from which
F08UPF (ZHBGVX) is called.

Constraint: LDBB � KBþ 1.

11: QðLDQ; �Þ – COMPLEX (KIND=nag_wp) array Output

Note: the second dimension of the array Q must be at least max 1;Nð Þ if JOBZ ¼ V , and at least
1 otherwise.

On exit: if JOBZ ¼ V , the n by n matrix, Q used in the reduction of the standard form, i.e.,
Cx ¼ �x, from symmetric banded to tridiagonal form.

If JOBZ ¼ N , Q is not referenced.

12: LDQ – INTEGER Input

On entry: the first dimension of the array Q as declared in the (sub)program from which F08UPF
(ZHBGVX) is called.

Constraints:

if JOBZ ¼ V , LDQ � max 1;Nð Þ;
otherwise LDQ � 1.

13: VL – REAL (KIND=nag_wp) Input
14: VU – REAL (KIND=nag_wp) Input

On entry: if RANGE ¼ V , the lower and upper bounds of the interval to be searched for
eigenvalues.

If RANGE ¼ A or I , VL and VU are not referenced.

Constraint: if RANGE ¼ V , VL < VU.

15: IL – INTEGER Input
16: IU – INTEGER Input

On entry: if RANGE ¼ I , the indices (in ascending order) of the smallest and largest eigenvalues
to be returned.

If RANGE ¼ A or V , IL and IU are not referenced.

Constraints:

if RANGE ¼ I and N ¼ 0, IL ¼ 1 and IU ¼ 0;
if RANGE ¼ I and N > 0, 1 � IL � IU � N.

17: ABSTOL – REAL (KIND=nag_wp) Input

On entry: the absolute error tolerance for the eigenvalues. An approximate eigenvalue is accepted
as converged when it is determined to lie in an interval a; b½ � of width less than or equal to

ABSTOLþ �max aj j; bj jð Þ;
where � is the machine precision. If ABSTOL is less than or equal to zero, then � Tk k1 will be
used in its place, where T is the tridiagonal matrix obtained by reducing C to tridiagonal form.
Eigenvalues will be computed most accurately when ABSTOL is set to twice the underflow
threshold 2� X02AMFð Þ, not zero. If this routine returns with INFO ¼ 1 to N, indicating that
some eigenvectors did not converge, try setting ABSTOL to 2� X02AMFð Þ. See Demmel and
Kahan (1990).

18: M – INTEGER Output

On exit: the total number of eigenvalues found. 0 � M � N.

If RANGE ¼ A , M ¼ N.

If RANGE ¼ I , M ¼ IU� ILþ 1.

19: WðNÞ – REAL (KIND=nag_wp) array Output

On exit: the eigenvalues in ascending order.

20: ZðLDZ; �Þ – COMPLEX (KIND=nag_wp) array Output

Note: the second dimension of the array Z must be at least max 1;Nð Þ if JOBZ ¼ V , and at least
1 otherwise.

On exit: if JOBZ ¼ V , Z contains the matrix Z of eigenvectors, with the ith column of Z
holding the eigenvector associated with WðiÞ. The eigenvectors are normalized so that
ZHBZ ¼ I.

If JOBZ ¼ N , Z is not referenced.

21: LDZ – INTEGER Input

On entry: the first dimension of the array Z as declared in the (sub)program from which F08UPF
(ZHBGVX) is called.

Constraints:

if JOBZ ¼ V , LDZ � max 1;Nð Þ;
otherwise LDZ � 1.

22: WORKðNÞ – COMPLEX (KIND=nag_wp) array Workspace

23: RWORKð7� NÞ – REAL (KIND=nag_wp) array Workspace

24: IWORKð5� NÞ – INTEGER array Workspace

25: JFAILð�Þ – INTEGER array Output

Note: the dimension of the array JFAIL must be at least max 1;Nð Þ.
On exit: if JOBZ ¼ V , then

if INFO ¼ 0, the first M elements of JFAIL are zero;

if INFO ¼ 1 to N, JFAIL contains the indices of the eigenvectors that failed to converge.

If JOBZ ¼ N , JFAIL is not referenced.

26: INFO – INTEGER Output

On exit: INFO ¼ 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO ¼ �i, argument i had an illegal value. An explanatory message is output, and execution
of the program is terminated.

INFO ¼ 1 to N

If INFO ¼ i, then i eigenvectors failed to converge. Their indices are stored in array JFAIL.
Please see ABSTOL.

INFO > N

F08UFF (DPBSTF) returned an error code; i.e., if INFO ¼ Nþ i, for 1 � i � N, then the leading
minor of order i of B is not positive definite. The factorization of B could not be completed and
no eigenvalues or eigenvectors were computed.

7 Accuracy

If B is ill-conditioned with respect to inversion, then the error bounds for the computed eigenvalues and
vectors may be large, although when the diagonal elements of B differ widely in magnitude the
eigenvalues and eigenvectors may be less sensitive than the condition of B would suggest. See Section
4.10 of Anderson et al. (1999) for details of the error bounds.

8 Parallelism and Performance

F08UPF (ZHBGVX) is threaded by NAG for parallel execution in multithreaded implementations of the
NAG Library.

F08UPF (ZHBGVX) makes calls to BLAS and/or LAPACK routines, which may be threaded within the
vendor library used by this implementation. Consult the documentation for the vendor library for further
information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the
OpenMP environment used within this routine. Please also consult the Users' Note for your
implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations is proportional to n3 if JOBZ ¼ V and RANGE ¼ A ,
and assuming that n � ka, is approximately proportional to n2ka if JOBZ ¼ N . Otherwise the number
of floating-point operations depends upon the number of eigenvectors computed.

The real analogue of this routine is F08UBF (DSBGVX).

10 Example

This example finds the eigenvalues in the half-open interval 0:0; 2:0ð �, and corresponding eigenvectors,
of the generalized band Hermitian eigenproblem Az ¼ �Bz, where

A ¼
�1:13 1:94� 2:10i �1:40þ 0:25i 0
1:94þ 2:10i �1:91 �0:82� 0:89i �0:67þ 0:34i

�1:40� 0:25i �0:82þ 0:89i �1:87 �1:10� 0:16i
0 �0:67� 0:34i �1:10þ 0:16i 0:50

0
B@

1
CA

and

B ¼
9:89 1:08� 1:73i 0 0
1:08þ 1:73i 1:69 �0:04þ 0:29i 0
0 �0:04� 0:29i 2:65 �0:33þ 2:24i
0 0 �0:33� 2:24i 2:17

0
B@

1
CA:

10.1 Program Text

Program f08upfe

! F08UPF Example Program Text

! Mark 26 Release. NAG Copyright 2016.

! .. Use Statements ..
Use nag_library, Only: nag_wp, x04daf, zhbgvx

! .. Implicit None Statement ..
Implicit None

! .. Parameters ..
Real (Kind=nag_wp), Parameter :: zero = 0.0E+0_nag_wp
Integer, Parameter :: nin = 5, nout = 6
Character (1), Parameter :: uplo = ’U’

! .. Local Scalars ..
Real (Kind=nag_wp) :: abstol, vl, vu
Integer :: i, ifail, il, info, iu, j, ka, kb, &

ldab, ldbb, ldq, ldz, m, n
! .. Local Arrays ..

Complex (Kind=nag_wp), Allocatable :: ab(:,:), bb(:,:), q(:,:), work(:), &
z(:,:)

Real (Kind=nag_wp), Allocatable :: rwork(:), w(:)
Integer, Allocatable :: iwork(:), jfail(:)

! .. Intrinsic Procedures ..
Intrinsic :: max, min

! .. Executable Statements ..
Write (nout,*) ’F08UPF Example Program Results’
Write (nout,*)

! Skip heading in data file
Read (nin,*)
Read (nin,*) n, ka, kb
ldab = ka + 1
ldbb = kb + 1
ldq = n
ldz = n
m = n
Allocate (ab(ldab,n),bb(ldbb,n),q(ldq,n),work(n),z(ldz,m),rwork(7*n), &

w(n),iwork(5*n),jfail(n))

! Read the lower and upper bounds of the interval to be searched,
! and read the upper or lower triangular parts of the matrices A
! and B from data file

Read (nin,*) vl, vu
If (uplo==’U’) Then

Read (nin,*)((ab(ka+1+i-j,j),j=i,min(n,i+ka)),i=1,n)
Read (nin,*)((bb(kb+1+i-j,j),j=i,min(n,i+kb)),i=1,n)

Else If (uplo==’L’) Then
Read (nin,*)((ab(1+i-j,j),j=max(1,i-ka),i),i=1,n)
Read (nin,*)((bb(1+i-j,j),j=max(1,i-kb),i),i=1,n)

End If

! Set the absolute error tolerance for eigenvalues. With abstol
! set to zero, the default value is used instead

abstol = zero

! Solve the generalized symmetric eigenvalue problem
! A*x = lambda*B*x

! The NAG name equivalent of zhbgvx is f08upf
Call zhbgvx(’Vectors’,’Values in range’,uplo,n,ka,kb,ab,ldab,bb,ldbb,q, &

ldq,vl,vu,il,iu,abstol,m,w,z,ldz,work,rwork,iwork,jfail,info)

If (info>=0 .And. info<=n) Then

! Print solution

Write (nout,99999) ’Number of eigenvalues found =’, m
Write (nout,*)
Write (nout,*) ’Eigenvalues’
Write (nout,99998) w(1:m)
Flush (nout)

! ifail: behaviour on error exit
! =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft

ifail = 0
Call x04daf(’General’,’ ’,n,m,z,ldz,’Selected eigenvectors’,ifail)

If (info>0) Then
Write (nout,99999) ’INFO eigenvectors failed to converge, INFO =’, &

info
Write (nout,*) ’Indices of eigenvectors that did not converge’
Write (nout,99997) jfail(1:m)

End If
Else If (info>n .And. info<=2*n) Then

i = info - n
Write (nout,99996) ’The leading minor of order ’, i, &

’ of B is not positive definite’
Else

Write (nout,99999) ’Failure in ZHBGVX. INFO =’, info
End If

99999 Format (1X,A,I5)
99998 Format (3X,(8F8.4))
99997 Format (3X,(8I8))
99996 Format (1X,A,I4,A)

End Program f08upfe

10.2 Program Data

F08UPF Example Program Data

4 2 1 :Values of N, KA and KB

0.0 2.0 :Values of VL and VU

(-1.13, 0.00) ( 1.94,-2.10) (-1.40, 0.25)
(-1.91, 0.00) (-0.82,-0.89) (-0.67, 0.34)

(-1.87, 0.00) (-1.10,-0.16)
( 0.50, 0.00) :End of matrix A

( 9.89, 0.00) ( 1.08,-1.73)
( 1.69, 0.00) (-0.04, 0.29)

( 2.65, 0.00) (-0.33, 2.24)
( 2.17, 0.00) :End of matrix B

10.3 Program Results

F08UPF Example Program Results

Number of eigenvalues found = 2

Eigenvalues
0.1603 1.7712

Selected eigenvectors
1 2

1 0.1908 0.0494
0.0137 -0.0045

2 0.1413 0.2505
0.1012 0.4427

3 -0.0437 -0.9705
-0.0905 0.0679

4 -0.2135 0.0606
0.2880 -1.3227

